

DONALD ALCOCK

ISMERD MEG A

BASIC

NYELVET !

MŰSZAKI KÖNYVKIADÓ

DONALD ALCOCK

ISMERD MEG A BASIC NYELVET!

DONALD ALCOCK
ISMERD MEG A
BASIC
NYELVET!

4. KIADÁS

MŰSZAKI KÖNYVKIADÓ,
BUDAPEST, 1984

AZ EREDETI MŰ:

DONALD ALCOCK: ILLUSTRATING BASIC

SYNDICATE OF THE CAMBRIDGE UNIVERSITY PRESS, CAMBRIDGE

© CAMBRIDGE UNIVERSITY PRESS, 1977

A MAGYAR FORDÍTÁS A 6. JAVÍTOTT KIADÁS ALAPJÁN KÉSZÜLT

FORDÍTOTTA:

RÁNKY MIKLÓS, OKL. GÉPÉSZMÉRNÖK, OKL. MÉRNÖKTANÁR

LEKTORÁLTA:

BAKOS TAMÁS, OKL. MATEMATIKUS

FÜGGELÉKET KÉSZÍTETTE:

TRINGER ÉVA, OKL. MATEMATIKUS

HUNGARIAN TRANSLATION

© MŰSZAKI KÖNYVKIADÓ, BUDAPEST, 1983

ETO : 800. 92 BASIC

ISBN : 963 10 5117 X (AZ ELSŐ KIADÁSÉ)

963 10 5939 1 (A NEGYEDIK KIADÁSÉ)

FELELŐS SZERKESZTŐ: DR. KEMÉNY TAMÁSNE, OKL. VILLAMOSMÉRNÖK

TARTALOMJEGYZÉK

ELŐSZÓ

vii

ELŐSZÓ A MAGYAR KIADÁSHOZ

ix

1. A NYELV ELEMEI

FELADAT FELVETÉSE ∞ BEVEZETŐ PÉLDA	1
ELŐSZÖR TÁROLD A PROGRAMOT, MAJD GÉPELD BE A "RUN"-T	2
AZ ÍRÓGÉPHEZ HASONLÓ BILLENTYŰZET	4
GÉPELÉS ∞ NÉHÁNY SZÓKÖZ SZABÁLY	5
A SOROK SZÁMOZÁSA ∞ A SZÁMÍTÓGÉP SORRENDBE RAKJA ŐKET	6
UTASÍTÁSOK, PARANCSONK ÉS MEGJEGYZÉSEK	7
NAGY ÉS KIS SZÁMOK	8
VÁLTOZÓK ∞ EGYSZERŰ NUMERIKUSOK	9
SZÁM HELYETTESÍTÉSE EGY MÁSIKKAL	10
SZÖVEGEK ∞ KINYOMTATÁSUK MÓDJA	11
SZÖVEGES VÁLTOZÓK ∞ EGYSZERŰEK	12
	13

2. BEVITEL ÉS KIVITEL, KIFEJEZÉSEK ÉS FÜGGVÉNYEK

DATA UTASÍTÁSOK ÉS AZ ABBÓL VALÓ OLVASÁS	15
RESTORE ∞ AZ ADATOK ÚJRAOLVASHATÓSÁGA	16
INPUT ∞ ADATBEVITEL KÉRÉSE A SZÁMÍTÓGÉPPAL	17
KIFEJEZÉSEK ∞ MINT AZ ALGEBRÁBAN	18
FÜGGVÉNYEK ∞ NÉGYZETGYÖK, LOGARITMUS STB.	20
TRIGONOMETRIKUS FÜGGVÉNYEK ∞ SINUS, COSINUS STB.	22
VELETLEN SZÁMOK ∞ A SZÁMÍTÓGÉP KOCKÁT VET	24
SAJÁT FÜGGVÉNYEK DEFINIÁLÁSA	25
PRINT ∞ A LEGSOKOLDALÚBB UTASÍTÁS	26
TAB () ∞ HASZNOS NYERS ÁBRÁK KIRAJZOLÁSÁRA	28
PRINT USING ∞ MÉG TÖBB OLDALÚ	32
	34

3. VEZÉRLÉS

GO TO ∞ A SORREND MEGSZAKAD	39
IF... THEN ∞ A SORRENDVÁLTOZTATÁS FELTÉTELES	40
STOP ∞ A FUTÁS MEGÁLLÍTÁSA ∞ MINTAPROGRAMMAL	41
MÁS PÉLDA A VEZÉRLÉS BEMUTATÁSÁRA	42
ON... GO TO ∞ A TÖBBÁLLÁSÚ KAPCSOLÓ	44
FOR... NEXT ∞ A CIKLUSOK VEZÉRLÉSE	46
GO SUB... RETURN ∞ A SZUBRUTIN FOGALMA	48
MOO ∞ TALÁLD KI A JÁTÉKA A GO SUB BEMUTATÁSÁRA	52
	56

4. TÖMBÖK	59
TÖMBÖK \rightarrow INDEXELT VÁLTOZÓK BEVEZETÉSE	60
DIM \rightarrow A TÖMBÖK DIMENZIÓINAK DEKLARÁLÁSA	62
INDEXEK \rightarrow INDEXEK KIFEJEZÉSEKKÉNT VALÓ KEZELÉSE	64
FELCSERÉLÉSES RENDEZÉS \rightarrow EGY PÉLDA AZ INDEXEK BEMUTATÁSÁRA	66
LÁNCOK \rightarrow INDEXEK HASZNÁLATA LISTA FELDOLGOZÁSBAN	68
SZÉTVÁLOGATÁS \rightarrow EGY PÉLDA: NÉPESSÉG CSOPORTOSÍTÁS	72
5. MÁTRIXOK	75
MÁTRIXOK \rightarrow NE IJEDJ MEG: EZEK CSUPÁN TÖMBÖK	76
HELYETTESÍTÉS \rightarrow EGYSZERRE EGY EGÉSZ MÁTRIX	78
ÚJRADIMENZIONÁLÁS \rightarrow MÁTRIXOK MEGNAGYOBBODNAK ÉS ÖSSZEMENNEK	79
MÁTRIXÖSSZEADÁS ÉS -KIVONÁS	80
MÁTRIXOK SKÁLÁRIS SZORZÁSA	82
TRANSZPONÁLÁS \rightarrow A SOROK OSZLOPOKKÁ VÁLNAK	84
ZER, CON ÉS IDN \rightarrow MÁTRIXOK TELI 1-EKKEL ÉS 0-ÁKKAL	86
MÁTRIXSZORZÁS \rightarrow "BELSŐ SZORZATOK"	88
INVERZIÓ \rightarrow A FOGALOM BEVEZETÉSE	90
MÁTRIXINVERZIÓ \rightarrow AZ "INV" UTASÍTÁS	92
TELJES MÁTRIXOK BEOLVASÁSA DATA UTASÍTÁSBÓL	94
MÁTRIXOK BEVITELE \rightarrow NE TEDD EZT A BILLENTYŰZETRŐL!	96
TELJES MÁTRIXOK KINYOMTATÁSA \rightarrow HASZNOS A PROGRAM- FEJLESZTÉS FOLYAMÁN	98
6. TELJES PÉLDAPROGRAMOK	101
MCDXCII \rightarrow SZIMBÓLUM - ÁLLAPOTTÁBLÁZAT BEMUTATÁSA	102
A LEGRÖVIDEBB HAZAÚT \rightarrow EGY ÚTKERESŐ MÓDSZER BEMUTATÁSA	106
7. PARANCSONK ÉS BEJELENTKEZÉS	111
BEJELENTKEZÉS \rightarrow A PROBLÉMA ÁLTALÁNOS MEGKÖZELÍTÉSE	112
PARANCSONK \rightarrow MINDEGYIK AZONNALI VÉGREHAJTÁST VÁLT KI	114
8. ADATÁLLOMÁNYOK (FILE-OK)	119
ADATÁLLOMÁNYOK \rightarrow A SZÁMÍTÓGÉP HÁTTÉRTÁRA	120
ADATÁLLOMÁNY FAJTÁK \rightarrow VÁZLATOS ÁTTEKINTÉS	124
9. SZINTAXIS	127
SZINTAXIS \rightarrow A BASIC FORMAI SZABÁLYAINAK DEFINÍCIÓJA	128
FÜGGELÉK	133
TÁRGYMUTATÓ	161

ELŐSZÓ

BÁRMILYEN EGYSZERŰ IS LEGYEN EGY SZÁMÍTÁS, HA SZÁMÍTÓGÉPPAL AKARJUK VÉGEZTETNI, MINDEN LÉPÉSÉT A GÉP SZÁMÁRA ÉRTHETŐ NYELVEN KELL LEÍRNUK. EZT A LEÍRÁST NEVEZIK PROGRAMNAK. EZ A KÖNYV A BASIC NEVŰ NÉPSZERŰ, IGEN SZÉLES KÖRBE ELTERJEDT NYELVET MUTATJA BE, ÉS ELMAGYARAZZA, HOGYAN IS KELL EGYSZERŰ PROGRAMOKAT ÍRNI EZEN A NYELVEN. BÁR A HIVATÁSOS PROGRAMOZÓK AZ ENNÉL ELEGÁNSABB ÉS HATÉKONYABB NYELVEKET RÉSZESÍTIK ELŐNYBEN, KEZDŐK ÉS AMATŐRÖK SZÁMÁRA A BASIC KIVÁLÓAN ALKALMAZHATÓ.

A BASIC AMERIKÁBAN, A DARTMOUTH COLLEGE-BEN, NEW HAMPSHIRE-BAN "SZÜLETETT" 1964-BEN EGYSZERŰ, KEZDŐKNEK SZÁNT SZÁMÍTÓGÉPNYELVKÉNT. NÉPSZERŰNEK BIZONYULT, ÉS SOK SZÁMÍTÓGÉPGYÁRTÓ, FŐISKOLA, EGYETEM ÉS IDŐOSZTÁSOS SZÁMÍTÓGÉPES SZOLGÁLTATÓIRODA VETTE ÁT ÉS BŐVÍTETTE KI. MANAPSÁG, AZ ANGOL NYELVHEZ HASONLÓAN A BASIC-NEK IS DIALEKTUSAI VANNAK: AZ EGYIK SZÁMÍTÓGÉPRE ÍRT BASIC PROGRAM NEM VALÓSZÍNŰ, HOGY MŰKÖDIK EGY MÁSIK SZÁMÍTÓGÉPEN.

AZ EGYESÜLT ÁLLAMOK SZABVÁNYÜGYI HIVATALA (AMERICAN NATIONAL STANDARDS INSTITUTE) RÖVIDESEN KIAD EGY "MINIMÁLIS BASIC" SZABVÁNYT† (A.N.S.I. X3J2). TOVÁBBÁ BULL, FREEMAN ÉS GARLAND "SZABVÁNY BASIC SPECIFIKÁCIÓJÁT" A NATIONAL COMPUTING CENTRE, U.K.* (1973) ADTA KI. EZEK A SZABVÁNYOK MÉG TÚL FRISSEK AHHÓZ, HOGY MINDENKI IGAZODNA HOZZÁJUK, ÍGY AZ A BASIC, AMELLYEL TALÁLKOZOL, NEM VALÓSZÍNŰ, HOGY SZABVÁNYOS LESZ. EZT AZ ÉLET TÉNYEKÉNT FOGADTAM EL, ÉS E KÖNYV ÍRÁSA KÖZBEN 11 KÉZIKÖNYVET TARTOTTAM KÉSZENLÉTBEN, MINDEGYIKŰK MÁS ÉS MÁS BASIC-KEL ÍRT. EZEK KÖZÜL NÉGY OLYAN NAGYSZÁMÍTÓGÉPEKEN ÉRTHETŐ EL, AMELYEKET NEMZETKÖZI IDŐOSZTÁSOS SZÁMÍTÓGÉPES SZOLGÁLTATÓIRODÁK ÜZEMELTETNEK, A MÁSIK HÉT VISZONT NAGY- ÉS ASZTALI KISSZÁMÍTÓGÉPEKEN EGYARÁNT HASZNÁLATOS. E TIZENEGY KÉZIKÖNYVBŐL PRÓBÁLTAM FELDERÍTENI ÉS ÖSSZEGYŰJTENI, HOGY ÁLTALÁBAN MIBEN KÜLÖNBÖZIK EGYIK BASIC A MÁSIKTÓL, ÉS MEGKÍSÉRELTEM UTAT MUTATNI, HOGYAN KERÜLHETŐ EL AZ EGYIK VAGY A MÁSIK VÁLTOZATTÓL VALÓ FÜGGŐSÉG. A "HORDOZHATÓ" SZÓT OLYAN BASIC PROGRAMOK JELLEMZÉSÉRE HASZNÁLTAM, AMELYEK EZEN FÜGGETLENSÉG JEGYÉBEN IRÓDTAK, A "HORDOZHATÓSÁGOT" ALAPVETŐ IGÉNYKÉNT KEZELTEM.

† A.N.S.I. X3J2 /76 -01 SZABVÁNYT 1976 JANUÁRBAN ADTÁK KI KÖZVÉLEMÉNYEZÉSRE.

* A SZÖVEGBEN MINT "N.C.C. SZABVÁNYOS BASIC"-RE HIVATKOZOM.

NEM KELL SZÁMÍTÓGÉPES SZAKEMBERNEK LENNED AHHOZ, HOGY EL-
OLVASHASD EZT A KÖNYVET: OLYANOKNAK ÍRTAM, AKIK ELŐSZÖR TALÁL-
KOZNAK SZÁMÍTÓGÉPPEL. HOGY KIKNEK? PÉLDAUL AZOKNAK, AKIK AZ IPAR-
BAN DOLGOZNAK (FŐLEG OLYAN MÉRNÖKÖK, AKIK HIVATALOSAN SOHASEM
TANULTAK SZÁMÍTÁSTECHNIKÁT, DE SZERETNÉNEK EGYSZERŰ PROGRAMOKAT
ÍRNI), MENEDZSEREKNEK (AKIK NEM AKARNAK PROGRAMOZNI, DE SZERET-
NÉNEK TÖBBET TUDNI ERRŐL A TERÜLETRŐL, AMELYEN GYAKRAN KELL
DÖNTÉSEKET HOZNIUK) ÉS AZOK RÉSZÉRE, AKIK MÁR ISMERIK A BASIC-ET,
DE ÉRDEKLŐDNEK A "HORDOZHATÓ" PROGRAMOZÁS SZÉLESEBB LEHETŐSÉGEI
ÉS OLYAN PROGRAMOZÁSI TECHNIKÁK IRÁNT, MINT AZ "ÁLLAPOTTÁBLÁZATOK"
ÉS A "LISTAFELDOLGOZÁS".

A KÖNYV SZÖVEGÉNEK NAGY RÉSZÉT OLDALPÁROKRA CSOPORTOSÍTOTTAM.
MINDEGYIKÜK A BASIC EGY-EGY "UTASÍTÁSÁVAL" FOGLALKOZIK. MIVEL SOK
UTASÍTÁS KÖLCSÖNÖS FÜGGŐSÉGBEN ÁLL EGYMÁSSAL, EZ AZ ELRENDEZÉS NÉ-
HA TOVÁBBI HIVATKOZÁSOKAT IGÉNYEL, DE A KÖNYVBEN SZEREPLŐ ELŐRE
UTALÁSOKAT A SZÁMÍTÁSTECHNIKÁBAN MÉG KEZDŐK NYUGODTAN FIGYEL-
MEN KIVÜL HAGYHATJÁK (NEM KELL ATTÓL TARTANIUK, HOGY A TÉMAKÖR
MEGÉRTÉSÉHEZ ELENGEDHETETLEN RÉSZT HAGYNAK KI.



MÉLY HÁLÁMAT SZERETNÉM KIFEJEZNI FELESÉGEMNEK, FAY-NEK,
AKI A KÖNYV ÍRÁSA ALATT KÉNYTELEN VOLT SZALMAÖZVEGYKÉNT ÉLNI (ENNEK
ELLENÉRE MINDVÉGIG MELEGEN BÍZTATOTT; TÁRSAMNAK, BRIAN
SHEARING-NEK, AKITŐL OLY SOKAT TANULTAM "SZÁMÍTÁSTECHNIKÁBÓL", ÉS
AKI LEHETŐVÉ TETTE, HOGY MUNKAIDŐM EGY RÉSZÉT E KÖNYV MEGÍRÁSÁRA
FORDÍTSAM, VALAMINT CHARLES LANG-NEK, AKI HITT ELKÉPZELESEM BEN,
ÉS MEGGYŐZÖTT ARRÓL, HOGY EZEKET FORMÁBA KELL ÖNTENI.

KÖNYVEM NYELVEZETE KÖZNAPI, FORMÁJA SZOKATLAN, HISZEN KÉZZEL
ÍRTAM, MELYEN A FIGYELMES OLVASÓ KOMOLY SZEMTENGÉLYFERDÜLÉSEM
ÉS FOLYTONOS KÉZRESZKETÉSEM NYOMAIT FEDEZHETI FEL.

REIGATE,
SURREY, U. K.

DONALD ALCOCK
1977 JANUÁR

ELOSZÓ A MAGYAR KIADÁSHOZ

MAPJAINKBAN GYAKRAN HANGOZTATOTT ÉS SOKAK ÁLTAL MÁR KÖZHELYNEK TEKINTETT MEGÁLLAPÍTÁS, HOGY KORSZAKUNKAT A TUDOMÁNY ÉS TECHNIKA EGYRE ROHAMOSABB FEJLŐDÉSE JELLEMZI. JÓSZERÉVEL A KÖZVÉLEMÉNY EHHEZ MÁR HOZZÁ IS SZOKOTT. ÍGY VAN EZ A SZÁMÍTÁSTECHNIKÁVAL KAPCSOLATOSAN IS. AZT, HOGY EGYRE NAGYOBB GÉPEK  AMELYEK EGYRE KOMPLEXEBB FELADATOKAT TUDNAK CSÖKKENŐ IDŐ ALATT MEGOLDANI  JELENNEK MEG, JÓFORMÁN CSAK A SZAKEMBEREK ÉRZÉKELIK.

AZ ELMÚLT IDŐSZAKNAK AZ ÁLTALÁNOS FEJLŐDÉSŐL KIEMELKEDŐ JELENSÉGE UGYANAKKOR AZONBAN AZ, HOGY A SZÁMÍTÁSTECHNIKAI BERENDEZÉSEK, ILL. ALKALMAZÁSAIK ALSÓ TELJESÍTMÉNY- ÉS ÁRKATEGÓRIÁIBAN IS MEGINDULT NÉHÁNY ÉVE A ROHAMOS FEJLŐDÉS. ERRE A FOLYAMATRA A TÁRSADALOM ÉREZHETŐEN FELFIGYELT. AZ UGYANIS, HOGY A SZÁMÍTÁSTECHNIKA ÉS AZ ALKALMAZÓK KÖZÖTT "SZEMÉLYES" KAPCSOLAT ALAKULHAT KI, MINŐSÉGI VÁLTOZÁST JELENTETT A SZÁMÍTÁSTECHNIKA ELTERJEDÉSE SZEMPONTJÁBÓL. KÜLÖNÖSEN, HA FIGYELEMBE VESSZÜK, HOGY ENNEK A "SZEMÉLYI SZÁMÍTÁSTECHNIKÁNAK" AZ ESZKÖZEI OLYAN ÁRÓN KERÜLNEK FORGALOMBA, HOGY SZÉLESEBB KÖRŰ (PL. KÖZÉPISKOLAI) ELTERJEDÉSE IS VÁRHATÓ.

TERMÉSZETESEN E SZÁMÍTÁSTECHNIKAI BERENDEZÉSEK (KÜLFÖLDÖN PERSONAL COMPUTERNEK, RÖVIDÍTVE PC-NEK NEVEZIK) ELTERJEDÉSE "KÖZÖS NYELVET" IGÉNYEL, AMELYET A GÉP ÉS A FELHASZNÁLÓ EGYARÁNT MEGÉRT.

MA A VILÁGON ILYEN NYELVNEK TEKINTIK A BASIC-ET, AMELYNEK DONALD ALCOCK KÖNYVE KIEMELKEDŐ SZÍNVONALÚ BEVEZETŐ LEÍRÁSÁT ADJA. A KÖNYV ELŐADÁSMÓDJA TALÁN SZO-

KATLAN. A KÖNNYED, GRAFIKUS MEGJELENÉSI FORMA AZONBAN
NE TÉVESSZEN MEG SENKIT! A KÖNYV A BASIC NYELV IGEN
ALAPOS ∞ A SZERZŐ SOK ÉVES BASIC-KEL KAPCSOLATOS
GYAKORLATÁRA ALAPOZOTT ∞ LEÍRÁSA. TERMÉSZETESEN
SEMILYEN SZÁMÍTASTECHNIKAI PROGRAMNYELVET LEÍRÓ KÖNYV,
ÍGY E KÖNYV ∞ AMELYET A KEDVES OLVASÓ A KEZÉBEN
TART ∞ SEM HELYETTESÍTI A BASIC NYELV KONKRÉT GÉP-
TÍPUSRA VAGY GÉPTÍPUSCSALÁDRA ALAPOZOTT KÉZIKÖNYVÉT.
EBBÓL TALAÁL AZ OLVASÓ (A KÖNYV STÍLUSÁHOZ ALKALMAZKO-
DÓ KIVITELBEN) EGY VÁLTOZATOT A JELEN KÖNYV FÜGGELEKÉ-
BEN.

Náray Anu.
1983. FEBRUÁR




A KÖNYVBEN A "POLOSKA" (BUG) JELET A PROGRAMHIBÁK JEJÖLÉSÉRE HASZNÁLTAM. (A SZÁMÍTÁSTECHNIKAI SZAKZSARGONBAN A HIBA KERESÉSÉT BUGGING-NAK : BOGARÁSZÁS-NAK NEVEZIK.)



A NYELV ELEMEI

FELADAT



HÁNY DOBOZ FESTÉKRE VAN SZÜKSÉGED AHHOZ, HOGY E VÍZTARTÁLY TETEJÉT ÉS FALÁT LEFESTHESD ?

ÍGY IS NEKIKEZDHETÜNK:

a tető területe, $T = \pi \times 6.5^2 \div 4 = 33.2$
a fal területe, $S = \pi \times 6.5 \times 27 = 551$
összterület, $A = T + S = 584.2$
dobozok száma, $G = A \div 236 = 2.48$
Kerekítve, $R = 3$

\therefore 3 DOBOZ FESTÉKRE VAN SZÜKSÉGED

AVAGY ÍRHA TNÁNK EGY PROGRAMOT A FELADAT MEGOLDÁSÁRA.

1. REMARK (megjegyzés) PROGRAM TERMÉSZETES NYELVEN
2. AZ ADATOK (DATA) 6.5, 27, 236
3. OLVASD BE AZ ADATOKAT
D, H ÉS C-NEK NEVEZVE ŐKET
(úgy képzeld el, mintha az adatokat D, H és C címkéjű kis dobozokba raknád) L. A MÁSÍK OLDALON!
4. SZÁMOLD KI $3.14 \times D^2 \div 4$ -ET, ÉS NEVEZD AZ EREDMÉNYT T-NEK
(azaz rakd az eredményt a T címkéjű kis dobozba)!
5. SZÁMOLD KI $3.14 \times D \times H$ -T ÉS LEGYEN S AZ EREDMÉNY!
6. ADD HOZZÁ A T-T AZ S-HEZ! LEGYEN A AZ EREDMÉNY!
7. SZÁMOLD KI $A \div C$ -T ÉS LEGYEN G AZ EREDMÉNY!
8. KEREKÍTSD G-T A LEGKÖZELEBBI EGÉSZ SZÁMRA, ÉS LEGYEN R AZ EREDMÉNY
(adj 1-et G-hez és vedd az eredmény egész részét)!
9. NYOMTASD (PRINT) KI, HOGY R DOBOZRA VAN SZÜKSÉG
(azaz nyomtasd ki R értékét)!
10. VÉGE (END)

E PROGRAM ELŐNYE AZ HOGY BÁRMILYEN MÉRETŰ TARTÁLYNÁL ÉS DOBOZ FESTÉKNÉL ALKALMAZHATÓ A MÁSODIK SORBAN LEVŐ ADATOKAT KELL CSAK KICSERÉLNED.

MOST

PRÓBÁLD MEG VÉGREHAJTANI A BAL OLDALI PROGRAMOT α PRÓBÁLD KI, MILYEN IS LENNE, HA TE LENNÉL A SZÁMÍTÓGÉP α TÖLTSD KI EZT AZ OLDALT SZÁMOKAT ÍRVA AZ ALÁBBI DOBOZOKRA!

adat 6.5



adat 27

adat 236

D	
H	
C	
T	
S	
A	
G	
R	

$$\pi \times D^2 \div 4$$

$$\pi \times D \times H$$

$$T + S$$

$$A \div C$$

$$G + 1 \text{ EGÉSZ RÉSZE}$$

ÍME MÉG EGYSZER UGYANEZ A PROGRAM **BASIC** BEN.

BASIC

FIGYELMESEN HASONLÍTSD ÖSSZE A BAL OLDALON LEVŐ PROGRAMMAL!

DOBOZRA VAN SZÜKSÉGED

```
1 REM BASIC PROGRAM
2 DATA 6.5, 27, 236
3 READ D, H, C
4 LET T = 3.14 * D ↑ 2 / 4
5 LET S = 3.14 * D * H
6 LET A = T + S
7 LET G = A / C
8 LET R = INT (G + 1)
9 PRINT R ; " DOBOZRA VAN SZÜKSÉGED "
10 END
```

jegyezd meg

* = azt jelenti, szorozd össze

↑ = azt jelenti, hogy emeld hatványra

/ = azt jelenti, hogy oszd el

ÉS EZT VÉGREHAJTVA A KÖVETKEZŐT ADJA :

3 DOBOZRA VAN SZÜKSÉGED

ELOSZÖR

KÉSZÍTSD EL A PROGRAMODAT
A BILLENTYŰZETEN α EKKOR A
SZÁMÍTÓGÉP EGYSZERŰEN EL-
RAKTÁROZZA A PROGRAMOT :

**EGYETLEN UTASÍTÁST
SEM NAJT VÉGRE**



```
1 REM BASIC PROGRAM
2 DATA 6.5 , 27, 236
3 READ D, H, C
4 LET T=3.14 * D ^ 2 / 4
5 LET S=3.14 * D * H
6 LET A=T+S
7 LET G=A/C
8 LET R=INT(G+1)
9 PRINT R; "DOBOZRA VAN SZÜKSÉGED"
10 END
```

AZTÁN

IRJ RUN-T

AMI A SZÁMÍTÓGÉPPAL VÉGREHAJTATJA α SORSZÁMOZÁS SZERINT α A
TÁROLT UTASÍTÁSOKAT, MIALATT TE PIHENSZ.



VÉGŰL A SZÁMÍTÓGÉP AZ "END"
UTASÍTÁST VÉGREHAJTVA MEGÁLL.

RUN
3 DOBOZRA VAN SZÜKSÉGED

DE


MIELÖTT MÉG MEGTENNÉD AZ ELSŐ LÉPÉST ÉS KEZDENÉD A
PROGRAM BEVITELÉT, TŰL KELL ESZNED A RITUÁLIS BEJELENTKE-
ZÉSEN, ÉS JELEZNED KELL A SZÁMÍTÓGÉPNEK, HOGY BASIC-ET
AKARSZ HASZNALNI.

KÜLÖNBÖZŐ SZÁMÍTÓGÉPEK α MÉG AZONOS TÍPUSU SZÁMÍTÓGÉPEK ESETÉBEN
IS, MELYEKET KÜLÖNBÖZŐ SZERVEZETEK ÜZEMELTETNEK β SOKSZOR ELTÉRŐ
MÓDON VÉGIK EZEKET A DOLGOKAT, ÍGY HA MOST KI AKAROD PRÓBÁLNI
A PROGRAMOT, HÍVJ ODA VALAKIT, AKI "ISMERI A RENDSZERT", HOGY
BEJELENTKEZZEN HELYETTED ÉS HÍVJA MEG A BASIC-ET!

BILLENTYŰZET

MINDEN BASIC
PROGRAMOT
BILLENTYŰZETTEL
KELL BEÍRNI.

☞ EZ FELTEHETŐLEG VALAMI ILYESMI ☞





BÁR A BETŰK ÉS SZÁMOK ELHELYEZÉSE A LEGTÖBB BILLENTYŰZETEN EGYFORMA, A FENTI KÉPEN LÁTHATÓ  ÉS  BILLENTYŰK NÉVBEN, ELHELYEZÉSBN, ÉS FUNKCIÓBAN GÉPENKÉNT VÁLTOZHATNAK.

ÉRDEMES MEGJEGYEZNI, HOGY MINDEN BETŰ NAGYBETŰ, TOVÁBBÁ AZI IS, HOGY AZ 1-NEK ÉS A 0-NAK KÜLÖN BILLENTYŰJE VAN (MINDKETTŐ A FELSŐ SORBAN TALÁLHATÓ). SOHASE HASZNÁLD HELYETTÜK AZ I ÉS AZ O BETŰT.

MINT MINDEN SZOKÁSOS ÍRÓGÉPEN, A  ((BETŰVÁLTÓ)) BILLENTYŰT EGYIDŐBEN EGY MÁSIK KÉT FELÍRÁSÚ BILLENTYŰVEL LENYOMVA AZ UTÓBBI FELSŐ RÉSZÉN FELTŰNTETETT KARAKTER ÍRÓDIK KI, ÍGY  EGYÜTT A  -TEL #T AD, MÍG  EGYEDÜL 3-AT AD.

A "VISSZANYÍL" ← A BAL OLDALÁN LEVŐ KARAKTERT TÖRLI A SZÁMÍTÓGÉP TÁRÁBAN KETTŐT HASZNÁLVA, AZ ELŐZŐ KÉT KARAKTERT TÖRLI ÉS ÍGY TOVÁBB. ÍGY, HA AZT GÉPELED PRIMP ← ← NT, AKKOR A BASIC A PRINT SZÓT ÉRTI MEG. NÉHÁNY BASIC AZONBAN ALÁHÚZÓKARAKTERT HASZNÁL E CÉLRA: PRIMP _ _ NT.

A LEGTÖBB BASIC EGY OLYAN BILLENTYŰT ((GYAKRAN "RUBOUT"-OT)) HASZNÁL, AMELYET HA LEÜTÜNK, A SZÁMÍTÓGÉP TÁRÁBAN TÖRLI AZ EGÉSZ BEGÉPELT SORT; EGY MÁSIK ((GYAKRAN "BREAK")) MEGÁLLÍTTJA A PROGRAMOT.

A BASIC-BEN AZ ÚJ SOR KEZDÉSÉHEZ ÜSD LE A  ((KÖCSI VISSZA))-ÁT, MAJD A  ((SOREMELEÉS)) BILLENTYŰT KELL LEÜTÖD.

GÉPELÉS

HA SOKSZOR ÍROD A PROGRAMOD BASIC-BEN, TANULJ MEG VAKON GÉPELNI! TÍZ UJ SOKKAL GYORSABB ÉS FREDMÉNYESEBB

MINT KETTŐ. A GÉPELT SOR HOSSZA KORLÁTOZOTT A LEGTÖBB BASIC LEGFELJEBB 72 KARAKTERBŐL ÁLLÓ SORT ENGED MEG. NÉMELYIK HOSSZABBAT IS MEGENGED, DE LEGJOBB A 72-ES KORLÁTOT ELFOGADNI.

NÉMELYIK BASIC NAGY SZABADSÁGOT ENGED MEG A SZÓKÖZ BILLENTYŰ VONATKOZÁSÁBAN; VAN, AMELYIK MINDEN SZÓKÖZT FIGYELMEN KÍVÜL HAGY, KIVEVE AZ IDEZŐJEL KÖZÖTTIEKET. ÍGY MEGENGEDHETŐ A KÖVETKEZŐ SZÖVEG:

```
8 FORD = STOP
```

A

```
8 FOR D= S TO P
```

HÉLYETT,

NO PERSZE BUTASÁG LENNE ELHOMÁLYOSÍTANI A PROGRAM JELENTÉSÉT CSAK AZÉRT, HOGY NÉHANY SZÓKÖZT MEGTAKARÍTSUNK.

NÉMELYIK BASIC A NYELV KULCSSZAVAIN BELÜL NEM ENGED MEG BETŰKÖZÖKET. EZEKNÉL A KÖVETKEZŐ SOR HIBAS LENNE:

```
23 L E T A = B + C
```

NÉMELY BASIC LEGALÁBB 1 SZÓKÖZT ÍR ELŐ MINDEN KULCSSZÓ ELŐTT VAGY UTÁN VAGY MINDKÉT ESETBEN.

```
20 DATA 6.5, 27, 236
80 PRINT R;"DOBOZRA VAN SZÜKSÉG"
```

NÉMELYIK BASIC NEM FOGAD EL SORSZÁMOKON BELÜL SZÓKÖZÖKET, DE ELFOGADJA ŐKET MÁS SZÁMOKON BELÜL:

```
1.000 LET A = 1.000.0
1000 LET A = 1000.0
```

NÉMELYIK BASIC EZEKET SEM FOGADJA EL.

NÉMELYIK BASIC NEM FOGAD EL SZÓKÖZÖKET SORSZÁMOK ELŐTT:

```
95 LET A = B
100 LET C = D * F + G
```

A SZÓKÖZ ITT TETSZÉS SZERINT ALKALMAZHATÓ VAGY ELHAGYHATÓ

ÁLTALÁBAN HA EGY SZÓKÖZ MEGENGEDETT (VAGY ELŐÍRT), AKKOR TÖBB HASZNÁLATA IS ENGEDÉLYEZETT. TOVÁBBÁ ÁLTALÁBAN A SZÓKÖZ TETSZÉS SZERINT HASZNÁLHATÓ A KÖVETKEZŐ JELEK BÁRMELYIK OLDALÁN: (, ; * + / - = † > <), DE NEM 1.5E2 (L.9. OLD.) ESETÉN, SEM > ÉS = KÖZÖTT (L.41. OLD.).

AZ A PROGRAM, AMELY MEGFELEL MINDEN KORLÁTOZÁSNAK, BÁRMELY VÁLTOZATNÁL ELFOGADHATÓ LESZ.

SORSZÁMOK

NE
FOLYAMATOS
SORSZÁMOZÁST
HASZNÁLJ:

HASZNÁLJ
10-ES
VAGY 5-ÖS
LÉPÉSKÖZT

```
10 REM BASIC PROGRAM
20 DATA 6.5, 27, 286 ← ← 36
30 READ D, H, C
40 LET T = 3.14 * D ↑ 2/4
50 LET S = 3.14 * D * H
60 LET A = T + S
70 LET G = A / C
80 PRINT R; "DOBOZRA VAN SZÜKSÉGED"
90 END
```

EZ 236
2.5. OLDALT

A PROGRAMBAN HIBA VAN: AZ UTOLSÓ LET ELMARADT. A BESZÚRÁS EGYSZERŰ, CSAK AZT KELL LEÍRNI, HOGY:

```
75 LET R = INT (G + 1)
```

ÉS A SZÁMÍTÓGÉP A 75 SORT A 70 ÉS A 80 KÖZÉ RAKJA, ATTÓL FÜGGETLENŰL, HOGY A GÉPELÉSKOR NEM EZ VOLT A SORREND.

B A TÖBB SORT ÍRUNK LE AZONOS SORSZÁMMAL, A SZÁMÍTÓGÉP MINDEN EGYES ELŐZŐ VÁLTOZATOT TÖRÖL, ÍGY CSAK AZ UTOLSÓKÉNT BEVITT SORT FOGADJA EL. HA AZ UTOLSÓKÉNT BEVITT SOR CSUPÁN EGY SORSZÁMBÓL ÁLL, AKKOR AZ EGÉSZ SOR TÖRLŐDIK A SZÁMÍTÓGÉP TÁRÁBÓL ~~Ö~~ BELEÉRTVE MEG A SORSZÁMOT IS. ÍGY LEHET TÖRÖLNI A NEM KÍVÁNT SOROKAT:

```
120 LET A = B + C
125 LET E = F
120 LET A = B + G
125
120 LET A = B
```

EZ A SZÁMÍTÓGÉP TÁRÁBAN CSUPÁN ÍGY FOG MEGJELENNI:

```
120 LET A = B
```

A AZ ELSŐ SORSZÁMNAK NAGYOBBNAK KELL LENNIE 0-NÁL. A SORSZÁM, NAGYSÁGA KORLÁTOZOTT: MIVEL NÉMELYIK BASIC MEGÁLL A 9999-NEL, A LEGJOBB, HA EZT FOGADJUK EL LEGNAGYOBB SORSZÁMKÉNT.

MINDEN PROGRAM UTOLSÓ UTASÍTÁSÁNAK AZ **END**-NEK KELL LENNIE.

UTASÍTÁSOK

A BASIC PROGRAM UTASÍTÁSOKNAK NEVEZETT, SORSZÁMOZOTT SOROK SOROZATÁBÓL ÁLL.

AZ UTASÍTÁS EGYSZERŰEN ÁLLÍTHAT VALAMIT

110 DATA 1, 2, 4
120 END

30 READ A, B, C
40 LET G = A * B ↑ 2 + C
50 PRINT "A VÁLASZ"; G

VAGY VALAMINEK AZ ELVÉGZÉSÉRE UTASÍTHATJA A SZÁMÍTÓGÉPET. AZ UTASÍTÁS SZÓRA JÓL ISMERT SZINONIMA A PARANCS;

AZOK AZ UTASÍTÁSOK, AMELYEK VALAMILYEN TEVÉKENYSÉGET VÉGEZNEK, VÉGREHAJTHATÓ UTASÍTÁSOK.

A SZÁMÍTÓGÉP ÁLTALÁBAN AZ ELSŐ SZÓ ALAPJÁN ELDÖNTI, HOGY MIT ÁLLÍTUNK VAGY MIT KELL TENNIE. ILYENEK: DATA, END, READ, LET stb.

NÉHÁNY ESETBEN A DÖNTÉSHEZ EZEK ELÉ A MAT SZÓ KELL: MAT READ, MAT PRINT, stb. (A MAT-TAL A 76. OLDALON TALÁLKOZUNK MAJD).

VAN AZONBAN EGY FONTOS KIVÉTEL:

LET

A **LET** SZÓ, AMELY A BASIC LEGTÖBB VÁLTOZTÁTNÁL ELHAGYHATÓ.

40 G = A * B ↑ 2 + C

REM

REM AZT JELENTI, HOGY MEGJEGYZÉS, REMARK. A REM UTASÍTÁSOK, NEM EREDMÉNYEZNEK MŰVELET-VÉGREHAJTÁST; AZÉRT ALKALMAZOD, HOGY VILÁGOSÁ TEGYE A PROGRAMODAT.

```
10 REM          * * * VÍZTARTÁLYOK * * *
20 REM
30 REM PROGRAM A BASIC BEMUTATÁSÁRA
40 REM - * - * - * - * - * - * - * - * - * -
50 DATA 6.5, 27, 236
60 REM A'TMÉRO' MAGASSÁG BEFEDÉS
```

REM AZ ÜRES SOR HELYÉN

REM A DÍSZÍTÉSRE

REM AZ EGYERTELMŰSÉG MIATT

E KÖNYV PÉLDÁIBAN NINCS SOK REM UTASÍTÁS, MERT KIS NYILAKKAL ÉS FELHŐKKEL LÁTTAM EL A PROGRAMOKAT, HOGY HELYET TAKARÍTSAK MEG.

SZÁMOK

HÁROMFÉLE SZÁM HASZNÁLHATÓ ∞
∞ EGÉSZ (INTEGER), VALÓS (REAL) VAGY E-FORMÁJÚ.

EGÉSZ FORMA
(EGÉSZ SZÁMOK)

160 DATA 0, 2, +4, 1000, -30

VALÓS FORMA
(TIZEDES SZÁMOK)

170 DATA + 0.70, 4., .6, -1.3, 987.65

A 4. ITT 4.0-T JELENT

E-FORMA
(EXPONENCIÁLIS ALAK)

190 DATA 1E3, 13.6 E-4, -13.6E6, -.0136E9

AZ E AZT JELENTI HOGY
"SZOR TÍZ A VALAHÁNYADIKON"

1.0×10^3
= 1,000

13.6×10^{-4}
= 0.00136

-13.6×10^6
= -13,600,000

$-.0136 \times 10^9$
= -13,600,000

AZ E UTÁN ÁLLÓ EGÉSZ AZT MUTATJA, HOGY HÁNY HELLYEL KERÜL A TIZEDESPONT ARRÉBB. JOBBRA KELL ELTOLNI, HA AZ EGÉSZ SZÁM POZITÍV; KÜLÖNBEN BALRA.

190 DATA E3, 13.6E1.2, 13.6 E 2

AZ E-FORMÁBAN MINDÍG KELL LENNIE EGY SZÁMNAK, AZ E-ELŐTT ÉS EGY EGÉSZNEK UTÁNA. NÉMELYIK BASIC MEGENGEDI A SZÓKÖZT AZ E-FORMÁN BELÜL, DE JOBB, HA NEM HAGYUNK KI HELYET.

$\pm 10^{38}$

NÉMELYIK BASIC RENDSZERBEN A LEGNAGYOBB ÁBRÁZOLHATÓ SZÁM MEGKÖZELÍTŐLEG $\pm 10^{38}$ (NAGYSZÁMON AZT ÉRTJÜK, HOGY MESSZE A 0-TÓL BÁRMELYIK OLDALON; KICSIN PEDIG AZT, HOGY KÖZEL A 0-HOZ BÁRMELYIK OLDALON.) MÁS BASIC RENDSZEREK A $\pm 10^{38}$ -NÁL SOKKAL NAGYOBB SZÁMOKKAL IS TUDNAK SZÁMOLNI; EZ A SZÁMÍTÓGÉP "SZÓHOSSZÚSÁGÁTÓL" FÜGG, ATTÓL, HOGY A "SZAVAKAT" EGYENKÉNT, PÁRSÁVAL VAGY NAGYOBB CSOPORTOKBAN HASZNÁLJA-E. DE EGYETLEN BASIC SEM UTASÍTHATJA VISSZA EGY OLYAN SZÁM KEZELÉSÉT, MINT: $\pm 10000, 000,000, 000,000,000, 000,000,000,000,000,000,000,000,000,000$.

6-7
szignifikáns jegy

NÉMELYIK BASIC RENDSZERBEN A SZÁMOK TÁROLÁSA ÉS AZ ARITMETIKAI MŰVELETEK PONTOSSÁGA 6-7 ÉRTEKES TÍZESES RENDSZERŰ SZÁMJEGY KÖRÜL VAN. A 987,654,321 KB. 987,654,000-KÉNT KERÜLNE TÁROLÁSRA. MÁS BASIC ENNÉL SOKKAL NAGYOBB PONTOSSÁGOT AD, ÁLTALÁNOS A 15 ÉRTEKES SZÁMJEGYIG TERJEDŐ PONTOSSÁG. EZ IS A SZÁMÍTÓGÉP "SZÓHOSSZÁTÓL" FÜGG, ÉS ATTÓL, HOGYAN HASZNÁLJA FEL A SZAVAKAT. EGYETLEN BASIC SEM DOLGOZIK 6-7 SZÁMJEGYNÉL KISEBB PONTOSSÁGGAL. (A "6 VAGY 7" BIZONYTALANSÁG AZÉRT VAN, MERT A SZÁMÍTÓGÉPEK LEGTÖBBJE BINÁRIS ARITMETIKÁT HASZNÁL ÉS NEM DECIMÁLISAT. ENNEK MÉG PONTOSABB MEGFOGALMAZÁSA AZ ALÁBBI LENNE "24 BINÁRIS SZÁMJEGY POZITÍV SZÁMOK ESETÉBEN, 23 NEGATÍV SZÁMOKNÁL; VAGY FORDÍTVÁ" AZONBAN EZEKEL A DOLGOKKAL A BASIC-BEN A KEZDŐKNEK NEM KELL TÖRÖDNIÜK.)

VÁLTOZÓK

286 EGYSZERŰ
NUMERIKUS VÁLTOZÓ
VAN A BASIC-BEN.

- D átmérő
- H magasság
- C befedés
- T felső terület
- S oldal
- A össz-terület
- C mennyiség
- R kerület mennyiség



EZEK KÖZÜL NÉHÁNYAT A 3. OLDALON LÁTTUNK. EZEK AZOK A KIS DOBOZOK, AMELYEKBE SZÁMOK TÁROLÓDNAK. A TELJES 286 ITT LÁTHATÓ. HASZNOS, HA EHEZ HASONLÓ NAGYMÉRETŰ TÁBLAZATOT KÉSZÍTESZ, ÉS MINDEN ÚJ PROGRAMNÁL ERŐL FÉNYMÁSOLATOT CSINÁLSZ. AHOGY FELHASZNALOD AZ EGYES VÁLTOZÓKAT, JEGYEZD FEL DOBOZÁBAN, HOGY MIRE HASZNÁLTAD. ÍGY ELKERÜLSZ, HOGY A MÁR EGYSZER FELHASZNÁLT VÁLTOZÓKAT ÚJRA HASZNÁLD MÁS CÉLBŐL. EZ SOK HIBA FORRÁSA.

A	A ⁰	A ¹	A ²	A ³	A ⁴	A ⁵	A ⁶	A ⁷	A ⁸	A ⁹
B	B ⁰	B ¹	B ²	B ³	B ⁴	B ⁵	B ⁶	B ⁷	B ⁸	B ⁹
C	C ⁰	C ¹	C ²	C ³	C ⁴	C ⁵	C ⁶	C ⁷	C ⁸	C ⁹
D	D ⁰	D ¹	D ²	D ³	D ⁴	D ⁵	D ⁶	D ⁷	D ⁸	D ⁹
E	E ⁰	E ¹	E ²	E ³	E ⁴	E ⁵	E ⁶	E ⁷	E ⁸	E ⁹
F	F ⁰	F ¹	F ²	F ³	F ⁴	F ⁵	F ⁶	F ⁷	F ⁸	F ⁹
G	G ⁰	G ¹	G ²	G ³	G ⁴	G ⁵	G ⁶	G ⁷	G ⁸	G ⁹
H	H ⁰	H ¹	H ²	H ³	H ⁴	H ⁵	H ⁶	H ⁷	H ⁸	H ⁹
I	I ⁰	I ¹	I ²	I ³	I ⁴	I ⁵	I ⁶	I ⁷	I ⁸	I ⁹
J	J ⁰	J ¹	J ²	J ³	J ⁴	J ⁵	J ⁶	J ⁷	J ⁸	J ⁹
K	K ⁰	K ¹	K ²	K ³	K ⁴	K ⁵	K ⁶	K ⁷	K ⁸	K ⁹
L	L ⁰	L ¹	L ²	L ³	L ⁴	L ⁵	L ⁶	L ⁷	L ⁸	L ⁹
M	M ⁰	M ¹	M ²	M ³	M ⁴	M ⁵	M ⁶	M ⁷	M ⁸	M ⁹
N	N ⁰	N ¹	N ²	N ³	N ⁴	N ⁵	N ⁶	N ⁷	N ⁸	N ⁹
O	O ⁰	O ¹	O ²	O ³	O ⁴	O ⁵	O ⁶	O ⁷	O ⁸	O ⁹
P	P ⁰	P ¹	P ²	P ³	P ⁴	P ⁵	P ⁶	P ⁷	P ⁸	P ⁹
Q	Q ⁰	Q ¹	Q ²	Q ³	Q ⁴	Q ⁵	Q ⁶	Q ⁷	Q ⁸	Q ⁹
R	R ⁰	R ¹	R ²	R ³	R ⁴	R ⁵	R ⁶	R ⁷	R ⁸	R ⁹
S	S ⁰	S ¹	S ²	S ³	S ⁴	S ⁵	S ⁶	S ⁷	S ⁸	S ⁹
T	T ⁰	T ¹	T ²	T ³	T ⁴	T ⁵	T ⁶	T ⁷	T ⁸	T ⁹
U	U ⁰	U ¹	U ²	U ³	U ⁴	U ⁵	U ⁶	U ⁷	U ⁸	U ⁹
V	V ⁰	V ¹	V ²	V ³	V ⁴	V ⁵	V ⁶	V ⁷	V ⁸	V ⁹
W	W ⁰	W ¹	W ²	W ³	W ⁴	W ⁵	W ⁶	W ⁷	W ⁸	W ⁹
X	X ⁰	X ¹	X ²	X ³	X ⁴	X ⁵	X ⁶	X ⁷	X ⁸	X ⁹
Y	Y ⁰	Y ¹	Y ²	Y ³	Y ⁴	Y ⁵	Y ⁶	Y ⁷	Y ⁸	Y ⁹
Z	Z ⁰	Z ¹	Z ²	Z ³	Z ⁴	Z ⁵	Z ⁶	Z ⁷	Z ⁸	Z ⁹

HELYETTESÍTÉS


MIN-
DEN
VÁLTO-
ZÓT
TÖBB-

SZÖR FELHASZNALHATSZ.


A VÁLTOZÓKBA RAKOTT SZÁM EGYSZERŰEN FELVÁLTJA A MÁR BENNE LÉVŐT, HASONLÍTSD ÖSSZE A KÖVETKEZŐ KÉT PROGRAMOT; AZ EGYIK A 3. OLDALON LÉVŐ PROGRAM ÚJ SORSZÁMOKKAL:

```

10 REM BASIC PROGRAM
20 DATA 6.5, 27, 236
30 READ D, H, C
40 LET T = 3.14 * D ^ 2 / 4
50 LET S = 3.14 * D * H
60 LET A = T + S
70 LET G = A / C
80 LET R = INT(G + 1)
90 PRINT R; "DOBOZRA VAN SZÜKSÉGED"
100 END
    
```



A MÁSIK UGYANEZT A FELADATOT OLDJA MEG, DE EZ A T VÁLTOZÓT ÚJRA MEG ÚJRA FELHASZNALJA:



```

10 REM HELYETTESÍTÉS BEMUTATÁSA
20 DATA 6.5, 27, 236
30 READ D, H, C
40 LET T = 3.14 * D ^ 2 / 4
50 LET T = T + 3.14 * D * H
60 LET T = T / C
70 LET T = INT(T + 1)
80 PRINT T; "DOBOZRA VAN SZÜKSÉGED"
90 END
    
```

AMIKOR A PROGRAM FUTNI KEZD, MINDEN VÁLTOZÓNAK VAN VALAMILYEN KEZDETI ÉRTÉKE. A BASIC RENDSZERTŐL FÜGG, HOGY MI EZ AZ ÉRTÉK. NÉMELY BASICBEN EZ **MULLA**; MÁSBAN AZ A SZÁM, AMIT AZ ELŐZŐ PROGRAM HÁTRAHAGYOTT; VAN OLYAN BASIC, AMELYIKBEN EZ AZ ÉRTÉK AZ ALÁBBI ÜZENET **A VÁLTOZÓNAK NEM ADTAL ÉRTÉKET** ÍGY A KÖVETKEZŐ TELJES PROGRAMBAN:

```

10 PRINT V
20 END
    
```

KAPHATSZ 0-T VAGY OLYAN "SZEMETET", MINT -123, 456 VAGY EGY ÜZENETET A SZÁMÍTÓGÉPTŐL, AMI AZT MONDJA, HOGY NEM TUD TOVABB LÉPNI, MERT A V VÁLTOZÓNAK ELSŐ HASZNÁLATAKOR MEG NINC S ÉRTÉKE. NE BÍZZ ABBAN, HOGY BARMELYIK VÁLTOZÓ ÉRTÉKE 0 A PROGRAM KEZDETEKOR; ADD MEG A KEZDŐ ÉRTÉKET:

```

10 LET V = 0
    
```

SZÖVEGEK

A BASIC EGYARÁNT TUD SZAVAKAT ÉS SZÁMOKAT KEZELNI. EZT MÁR LÁTUK A 4. OLDALON.

```
9 PRINT R;"DOBOZRA VAN SZÜKSÉGED"  
10 END  
RUN  
3 DOBOZRA VAN SZÜKSÉGED
```

E KÖNYVBEN A "DOBOZRA VAN SZÜKSÉGED"-ET SZÖVEGNEK NEVEZZÜK. A SZAKSZARON EZT KARAKTERLÁNCNAK, IDÉZETT KARAKTERLÁNCNAK; ALFAMETRIKUS SZÖVEGNEK ÉS MÉG MÁSNAK IS NEVEZI.

A SZÖVEGEK IDÉZŐJELBE TETT SZAVAK VAGY KARAKTERSOROZATOK. E DEFINÍCIÓBÓL KÖVETKEZIK, HOGY NEM HASZNÁLHATSZ OLYAN SZÖVEGET, AMELY IDÉZŐJELEKET TARTALMAZ, HISZ A SZÁMÍTÓGÉP, AZT HINNÉ, HOGY A SZÖVEG VÉGÉT JELÖLIK; AZONBAN LEHET APOSZTRÓFOT HASZNÁLNI A SZÖVEGEKBEN:

```
20 PRINT "ROCK'N'ROLL"
```

VANNAK AZONBAN OLYAN BASIC VÁLTOZATOK IS, AMELYEK MEGENGEDIK, HOGY A SZÖVEGET APOSZTRÓFOK KÖZÉ RAKJUK; ILYEN SZÖVEGEKBEN ELŐFORDULHATNAK IDÉZŐJELEK, APOSZTRÓFOK AZONBAN NEM. MÁS BASIC NYELVEK ÚGY KERÜLNEK MEG A PROBLÉMÁT, HOGY A SZÖVEGBEN SZEREPLŐ IDÉZŐJELPÁRT EGY IDÉZŐJELNEK VESZIK:

```
30 PRINT "AZT MONDTA""OOH!""
```

```
AZT MONDTA "OOH!"
```

LESZ.

A LEGJOBB AZ, HA ELKERÜLJÜK A SZÖVEGKÖZI IDÉZŐJELEKET.

A "PRINT" UTASÍTÁSBAN SZEREPLŐ SZÖVEG BÁRMILYEN HOSSZÚSÁGÚ LEHET, CSAK KIFÉRJEN EGY SORBA. HA A KIÍRÁS SORÁN A LAP TELJES SORÁT KI AKAROD HASZNÁLNI, AKKOR KÉT VAGY TÖBB SZÖVEGET KELL KIÍRATNI. MINDEGYIK UTÁN PONTOSVESSZŐVEL, KIVÉVE AZ UTOLSÓT.

```
120 PRINT "-----";  
130 PRINT "-----";  
140 PRINT "-----"
```

A "PRINT" UTASÍTÁSBAN SZEREPLŐ PONTOSVESSZŐKNÉL A SZÁMÍTÓGÉP EGYMÁSHOZ ILLESZTVE NYOMTATJA KI AZ EGYES TÉTELEKET; A SZÁMÍTÓGÉP A VESSZŐKNÉL A TÉTELEKET SZÉTSZTJA AZ OLDALON. MINDERRE A 28. OLDALON ADOK RÉSZLETES MAGYARÁZATOT, AHOL A "PRINT" UTASÍTÁST TAGLALOM.

SZÖVEGES VÁLTOZÓK

A 286 KIS SZÁMTAROLÓ-DOBOZON KIVÜL LEGALÁBB 2.6-BAN SZÖVEGEKET LEHET TÁROLNI & EZEKET SZÖVEGES VÁLTOZÓKNAK NEVEZIK. ÚGY KOVÁCSOLTUK EZT AZ ÚJ KIFEJEZÉST, HOGY ALKALMAZKODJON A NUMERIKUS VÁLTOZÓ KIFEJEZÉSHEZ, DE A SZAKZSARGONBAN HASZNÁLJÁK MÉG A SZÖVEGVÁLTOZÓ, A KARAKTERLÁNC-VÁLTOZÓ ÉS AZ ALFAMETRIKUS VÁLTOZÓ SZÓT IS.

MINDEN BASIC-BEN KORLÁTOZOTT A SZÖVEGES VÁLTOZÓBAN TÁROLHATÓ SZÖVEG HOSSZÚSÁGA & EGYESEK 4095, MÍG MÁSOK VISZONT CSAK 18 KARAKTERT ENGEDNEK MEG & EZ EGYBEN A "HORDOZHATÓSÁG" FENNTARTÁSA MELLETT A MAXIMUM.

1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18	1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18
A\$	N\$
B\$	O\$
C\$	P\$
D\$	Q\$
E\$	R\$
F\$	S\$
G\$	T\$
H\$	U\$
I\$	V\$
J\$	W\$ * , * , *
K\$	X\$
L\$	Y\$
M\$	Z\$ C S A K 1 8 K A R A K T E R

NÉMELYIK BASIC RENDSZERNEK 286 SZÖVEGES VÁLTOZÓJA VAN. A\$, A0\$ A1\$, A2\$ stb. Z8\$, Z9\$

AMIKOR A PROGRAM FUTNI KEZD, A SZÖVEGES VÁLTOZÓKBAN LEHETNEK ÜRES HELYEK, AZ ELŐZŐ PROGRAM MARADÉK SZÖVEGEI, VAGY TARTALMAZHATNAK SPECIÁLIS ÜZENETET. PL. "A VÁLTOZÓNAK NEM ADTÁL ÉRTÉKET", MINT A NUMERIKUS VÁLTOZÓK ESETÉBEN (L. 10. OLD.). EZ AZ ALKALMAZOTT BASIC-RENDSZERTŐL FÜGG. HA IGAZÁN "HORDOZHATÓ" PROGRAMOT AKARSZ ÍRNI, AKKOR ADJ A VÁLTOZÓKNAK ÉRTÉKET, MIELŐTT HASZNÁLNÁD ŐKET. A SZÖVEGES VÁLTOZÓKNAK A LET UTASÍTÁSSAL ADHATUNK ÉRTÉKET &

Kiírásukat pedig a PRINT utasítással végezhetjük.

```
10 LET W$ = "***"
20 LET Z$ = "CSAK 18 TÁROLT KARAKTER"
```

```
30 PRINT W$; Z$; W$
40 END
RUN
*** CSAK 18 TÁROLT KAR ***
```

ennél több karakter nem fér a szöveges változóba

A SZÖVEGES VÁLTOZÓKNAK MÉG A READ-DEL (L. 16. OLD.) ES AZ INPUT-TAL ADHATUNK ÉRTÉKET (L. 18. OLD.) ÖSSZE IS HASONLÍTHATJUK ŐKET AZ IF UTASÍTÁS (L. 41. OLD.) SEGÍTSÉGÉVEL.

```
50 IF I$ = "IGEN" THEN 80
60 IF I$ = "W$" THEN 9000
```




BEVITEL ÉS KIVITEL,
KIFEJEZÉSEK ÉS FÜGGVÉNYEK

DATA + READ

A DATA UTASÍTÁS KIINDULÁSI ADATOKKAL ÉS SZÖVEGEKKEL LÁTJA EL AZ EZEKRE ÉPÜLŐ READ UTASÍTÁSOKAT.

AKÁRHÁNY DATA UTASÍTÁST HASZNÁLHATSZ BÁRHOL A PROGRAMBAN. MIELŐTT A PROGRAM NEKILÁT A MUNKÁNAK, EGY HOSSZÚ "SOR" FORMÁLVA BELŐLÜK ÖSSZEKAPCSOLJA ÖKET. A DATA UTASÍTÁS NEM VÉGREHAJTHATÓ, DE A SZÁMÍTÓGÉP A READ UTASÍTÁS VÉGREHAJTÁSA SORÁN ANNYI ADATOT SZED ÖSSZE A SOR ELEJÉRŐL, AMENNYIRE SZÜKSÉGE VAN, MEGHAGYVA A MARADÉKOT A RÁ KÖVETKEZŐ READ UTASÍTÁSOKNAK. A KÖVETKEZŐ KÉT PROGRAM MINDEGYIKE A MÁSODIK PROGRAM VÉGÉN LÁTHATÓ EREDMÉNYT HOZZA KI.

```

10 REM FOLYAMATOS SOR
20 DATA "PÁRATLANOK" 7, -23, -17, "PÁROSAK", -12, 36
30 READ Q$, B, C, D, E$, E, F
40 PRINT Q$; B; C; D; E$; E; F
50 END
    
```

```

10 REM TÖRDELT SOR
20 DATA "PÁRATLANOK", 7
30 READ Q$, B, C, D, E$, E, F
40 DATA -23, -17
50 PRINT Q$; B; C; D; E$; E; F
60 DATA "PÁROSAK", -12, 36
70 END
RUN
PÁRATLANOK 7 -23 -17 PÁROSAK -12 36
    
```

AZ MÁR RAJTAD MŰLIK, HOGY A SZÁMOKAT ÉS A SZÖVEGEKET ÚGY HELYEZD EL, HOGY AZOK A MEGFELELŐ TÍPUSU VÁLTOZÓKBA KERÜLJENEK. A KÖVETKEZŐ PROGRAM

HIBAÜZENETET ADNA

N OLVASÁSOKOR.

(Egyik vagy másik párt fel kell cserélni.)

```

10 DATA "ANNO", 1977
20 READ N, C$
30 PRINT N; C$
40 END
    
```

NÉMELYIK BASIC RENDSZER AZONBAN KÜLÖN KEZELI A SZÁMOKAT ÉS A SZÖVEGEKET. OTT E PROGRAM MŰKÖDNI FOG. UGYANAKKOR AZ ILYEN KÜLÖNVÁLASZTOTT ADATSOROK ELÉGGÉ SZOKATLANOK.

HA TÖBB "READ" VAN, MINT ADAT, A SZÁMÍTÓGÉP ÜZENETET GÉPEL KI, MISZERINT KIFOGYOTT AZ ADATBÓL, ÉS LEALL.

```

10 DATA 1.5, 2.5, -3.5, 4
20 READ A, B
30 READ C, D, E
40 PRINT A; B; C; D; E
50 END
RUN
* OUT OF DATA AT LINE 30 *
    
```

adat kifogyott a 30. sornál

RESTORE

A RESTORE UTASÍTÁS VISSZAVISZ A DATA UTASÍTÁSOKBÓL ALAKÍTOTT ADATSOR ELEJÉRE.

AMIKOR A SZÁMÍTÓGÉP READ UTASÍTÁST HAJT VÉGRE, ANNYI ADATOT OLVAS BE (ÚJRA) A SOR ELEJÉRŐL, AMENNYIRE SZÜKSÉGE VAN, ÉS A MARADÉKOT A KÖVETKEZŐ READ UTASÍTÁSOKNAK HAGYJA MEG. A RESTORE NEM TÖRLI A SORT.

```
10 DATA 1, 2, 3, 4, 5, 6, 7, 8, 9
20 READ A, B, C
30 PRINT A; B; C
40 READ A, B, C
50 PRINT A; B; C
60 RESTORE
70 READ A, B, C
80 PRINT A; B; C
90 END
RUN
1 2 3
4 5 6
1 2 3
```

Ezek sohasem kerülnek beolvasásra

Az N.C.C. "Standard Basic" a RESET szót használja

ADATOK

A DATA UTASÍTÁSBAN SZEREPLŐ ADATOK LEHETNEK SZÁMOK (+ ÉS - JELEKKEL VAGY ANÉLKÜL) VAGY SZÖVEGEK, VAGY MINDKETTŐ; DE NEM LEHETNEK KIFEJEZÉSEK (L. A 20. OLDALT).

```
10 REM SZÁM ÉS SZÖVEG MEGENGEDETT
20 DATA 1.5, +16.28, -13.47, "IGEN", "NEM"
30 REM KIFEJEZÉS MEGTILTVA
40 DATA 3.14 * 6.5 + 2/4
```

AZ ADATOKAT EGYMÁSÓL VESSZŐVEL KELL ELVÁLASZTANI. AZONBAN EGYES LISTÁK VÉGÉN NINC S VESSZŐ.

HA EGY ADOTT VÁLTOZÓNAK TÚL HOSSZÚ A SZÖVEG, AKKOR A TÖBBLET KARAKTEREK ELVESZNEK.

```
10 DATA "EZ TÚL HOSSZÚ A VÁLTOZÓNAK"
20 READ V$
```

Ez az V\$-ben tárolt utolsó karakter

l. a 13. old.

AZ EREDETI DARTMOUTH BASIC-BEN A DATA UTASÍTÁS VOLT AZ EGYETLEN ESZKÖZ ARRRA, HOGY ADATOKAT A PROGRAMBA BEVIGYÜNK. A LEGTÖBB BASIC RENDSZER MA MÁR A KÖZISMERT INPUT UTASÍTÁST HASZNÁLJA.

INPUT

A BASIC PROGRAMMAL ADATOT IS KEZELHETSZ ?
A GÉP VÁR, MÍG A SORT LEGÉPELED.

```
10 REM INPUT BEMUTATÁSA
20 PRINT "GÉPELJ EGY SZÁMOT"
30 INPUT X
40 PRINT X; "-T GÉPELTÉL"
50 END
RUN
GÉPELJ EGY SZÁMOT
?
```

A PROGRAM VÁRAKOZIK A 30. UTASÍTÁSNÁL, ADATOT VÁR. A VÉGTELENSÉGIG FOG VÁRNI, EGÉSZEN ADDIG, MÍG BE NEM GÉPELSZ VALAMILVEN ADATOT, ÉS LE NEM NYOMOD A **RETURN** BILLENTYŰT. (Némelyik BASIC-rendszer egy idő múlva kikapcsol magától, ha nem kap választ.)

```
? 6.5
6.5-T GÉPELTÉL
```

AZ INPUT UTASÍTÁS KÉRHET EGYNÉL TÖBB SZÁMOT IS ?

```
30 INPUT X, Y, Z
```

```
? 6.5, 7.5, 8.5
```

BEOLVASHAT SZÖVEGEKET IS ?

```
40 INPUT A$, B$
```

```
? "FELFEDEZÉSEK", "HAMISÍTVÁNYOK"
```

ÉS SZÁMOK ÉS SZÖVEGEK KEVERÉKÉT IS ?

```
30 INPUT T$, C, V
```

```
? "REGÉNYEK", 15, 4.2
```

DE AKÁRHÁNY ADATOT IS AKARUNK LEÍRNI AZ INPUT UTASÍTÁSSAL, A LEKÉRDEZÉS MÓDJA (AMELYET PROMPT-NAK HÍVNAK) MINDEN ADATRA AZONOS ? NÉMELYIK BASIC-RENDSZERBEN CSUPÁN EGY HANG A JEL, MÁSOKBAN EGY KÉRDŐJEL, EGY NYÍL VAGY EGY "NAGYOBB, MINT" JEL.

```
? >
```

SEGÍTSÉGÜL A PROGRAM HASZNÁLÓJÁNAK (AKI NEM BIZTOS HOGY AZONOS A MEGÍRÓJÁVAL) MINDEN INPUT UTASÍTÁS ELÉ TEGYÉL EGY MAGYARÁZÓ

PRINT
utasítást.

```
25 PRINT "GÉPELJ EGY SZÖVEGET ÉS EGY SZÁMOT"
30 INPUT T$, C
```

AMIKOR A VÁLTOZÓKNAK INDEXÜK VAN (L. A 64. OLD.), BALRÓL JOBBRA OLVASHATSZ BE, ÉS HASZNÁLHATOD FEL ÉRTÉKEIKET ? PL. INPUT I, J, A (I, J).



MI TÖRTÉNIK AKKOR, HA A PROMPT -RA ROSSZ VÁLASZT ADSZ?

SAJNOS KÜLÖNBÖZŐ BASIC -EK ERRE ELTÉRŐ MÓDON REAGÁLNAK.

TÚL SOK ADAT : AZ EGÉSZ SOR ÚJRAGÉPELÉSÉT KÉRI.

10 INPUT A, B, C

? 10, 20, 30, 40, 50
TÚL SOK ADAT; GÉPELD ÚJRA A SORT
?

NÉMELYIK BASIC -RENDSZER EGYSZERŰEN NEM VESZI TUDOMÁSUL A TÖBBLET-ADATOKAT ; MÁS RENDSZEREK FÉLRETESZIK EZEKET A KÖVETKEZŐ INPUT UTASÍTÁSRA VAGY CSAK AKKOR TESZNEK ÍGY, HA AZ ADATSOR VESSZŐVEL VÉGZŐDIK.

NEM ELÉG ADAT : AZ EGÉSZ SOR ÚJRAGÉPELÉSÉT KÉRI.

20 INPUT A, B, C, D, E

? 10
NEM ELÉG ADAT; GÉPELD ÚJRA A SORT
?

NÉMELYIK BASIC RENDSZER EGYSZERŰEN TOVÁBB KÉRDEZ EGÉSZEN ADDIG, MÍG NEM KAPOTT ELÉG ADATOT.

HIBÁS PÁROSÍTÁS : AZ EGÉSZ SOR ÚJRAGÉPELÉSÉT KÉRI.

30 INPUT T \$, C, V

? "REGÉNYEK", 15, "HAT"
ROSSZ ADATFAJTA; GÉPELD ÚJRA A SORT
?

DE NÉMELYIK BASIC -RENDSZER ELFOGADJA A FENTI ELSŐ KÉT ADATOT, ÉS CSAK A HARMADIKRA KÉRDEZNE RÁ ÚJRA; ISMÉT MÁS RENDSZER AZ EGÉSZ PROGRAMOT LEÁLLÍTANÁ.

SZÖVEGEK : HA A SZÖVEGBEN NINCS SZÓKÖZ, AZ IDÉZŐJELEK ELHAGYHATÓK :

40 INPUT T \$, C

? REGÉNYEK, 15

DE NÉMELY BASIC -RENDSZER EZT NEM ENGEDI MEG; ESETLEG A PROGRAM FUTÁSÁT IS LEÁLLÍTJA. HA A SZÖVEG TÚL HOSSZÚA VÁLTOZÓNÁK, AZ EGÉSZ SOR ÚJRAGÉPELÉSÉT KÉRI A RENDSZER :

50 INPUT T \$, C

? "SZT. JÁNOS EVANGÉLIUMA", 2
TÚL HOSSZÚ SZÖVEG; GÉPELD ÚJRA A SORT
?

NÉMELYIK BASIC -RENDSZER AZONBAN EGYSZERŰEN MEGCSONKÍTANÁ A TÚL HOSSZÚ SZÖVEGET, MÁSOK CSAK IDÉZŐJELEK KÖZÖTT LÉVŐ SZÖVEGET ENGEDÉLYEZNEK (l. a 12. oldalt).

VESSZŐK ÉS SZÓKÖZÖK : A LISTA ADATAIT VESSZŐKKEL KELL ELVÁLASZTANI, SZÓKÖZ NE LEGYEN AZ EGYES ADATOKON BELÜL. NÉMELY BASIC -RENDSZER ENGEDI A VESSZŐK ELHAGYÁSÁT, DE RAGASZKODIK A SZÓKÖZÖK HELYES HASZNÁLATÁHOZ (l. a 6. oldalt); MÁR RENDSZER FIGYELMEN KÍVÜL HAGYJA A SZÓKÖZÖKET (Kivéve szövegekben), DE RAGASZKODIK A VESSZŐKHOZ.

KIFEJEZÉSEK

BONYOLULT KIFEJEZÉSEKET ÍRHATSZ BASIC-BEN.

PÉLDA EGY KIFEJEZÉSRE ∞ A 2. OLDAL VÍZTARTÁLYÁNAK BEFESTETT TERÜLETE.

$$3.14 * D \uparrow 2/4 + 3.14 * D * H$$

ADOTT $D=6.5$, $H=27$, HA MEGKÉRNÉNEK, HOGY SZÁMÍTSD KI ENNEK A KIFEJEZÉSNEK AZ ÉRTÉKÉT, BIZONYÁRA NEM A $4+3.14=7.14$ ~~*~~, SEM A $2/4=0.5$, MAJD A $0.5+3.14=3.64$ ~~*~~ KISZÁMÍTÁSÁVAL KEZDENÉD.

MIÉRT NEM $?$ AZÉRT, MERT A MEGÁLLAPODÁS SZERINT MINDANNYIAN ELFOGADJUK, HOGY $*$ A HATVÁNYOZÁS TÖRTÉNIK ELŐSZÖR \circ $6.5^2=42.25$

$*$ MAJD A SZORZÁSOK ÉS OSZTÁSOK,

$$\begin{aligned} \circ & 3.14 * 42.25 = 132.665 \\ \text{majd} & 132.665 \div 4 = 33.1663 \\ \text{majd} & 3.14 * 6.5 = 20.4100 \\ \text{majd} & 20.4100 * 27 = 551.07 \end{aligned}$$

$*$ VÉGÜL AZ ÖSSZEADÁSOK ÉS KIVONÁSOK \circ $33.1663 + 551.07 = 584.236$ KÖVETKEZNEK, ÉS PONTOSAN ÍGY CSINÁLJA A BASIC IS.

L ME UGYANEZ A KÉPLET MÁSKÉPPEN KIFEJEZVE \circ $(3.14 * D * (D/4 + H))$.

BÁRMILYEN ÉRTELMESEN MÉLYSÉGIG EGYMÁSBA SKATULYÁZHATSZ ZÁRÓJELEKET. A SZÁMÍTÓGÉP A LEGBELSŐ ZÁRÓJELPÁRBAN FOGLALT KIFEJEZÉS ÉRTÉKÉT SZÁMOLJA KI, ÉS INNEN HALAD KIFELEÉ ∞ ÉS AZT AZ EREDMÉNYT FOGJA ADNI, AMIT AZ ALGEBRA TÖRVÉNYE SZERINT VÁRNÁL.

NEM LEHET JELOLETLEN SZORZÁST ALKALMAZNI \circ NEM ÍRTHATOD EZT \circ

$$3.14 * D * (D/4 + H)$$

NEM TEHETSZ A \Rightarrow \uparrow $/$ $*$ $-$ $+$ MŰVELETI JELEKBŐL KETTŐT KÖZVETLENÜL EGYMÁS MELLÉ ANÉLKÜL, HOGY ZÁRÓJELET NE ALKALMAZNÁL \circ

$$3.14 * \uparrow D * (\uparrow D / \uparrow 4 + H)$$

itt rendben, mert van közbenső zárójel.

DE AKKOR HOGY FEJEZZÜK KI 10^{-2} -ONT $?$ NEM MINT $10 \uparrow 2$ -T, HANEM ÚGY, HOGY $10 \uparrow (-2)$ VAGY $1 / (10 \uparrow 2)$.

KIFEJEZÉSEK GÉPELÉSÉNÉL SZÓKÖZÖKET HAGYHATSZ VAGY TETSZÉS SZERINT TÖMÖRITHETSZ \circ

$$10 \text{ LET } A = 3.14 * D \uparrow 2 / 4 + 3.14 * D * H$$

A KULCSSZAVAKBAN ÉS A SORSZÁMOKBAN ELŐFORDULÓ SZÓKÖZÖKRE VONATKOZÓ TILTÓ SZABÁLYOK A 6. OLDALON TALÁLHATÓK \circ




TEDD VILÁGOSABBÁ A KIFEJEZÉSEKET ZÁRÓJELEK HASZNÁLATÁVAL \odot MÉG AKKOR IS, HA NEM FELTÉTLENÜL SZÜKSÉGESEK!

MIT A/B/C? A LEGTÖBB BASIC-RENDSZER (A/B)/C-KÉNT KEZELI, DE NÉMELYIK A/(B/C)-KÉNT. (PRÓBÁLD MEG MINDKÉT MÓDON KISZÁMÍTANI 8/4/2-T, HOGY LÁSD A KÜLÖNBSÉGET!) MINDIG HASZNÁLJ ZÁRÓJELET, HOGY A SZÁMÍTÓGÉP BIZTOSAN ÚGY KEZELJE A KÉTÉRTELMŰ KIFEJEZÉSEKET, AHOGY TE AKAROD. AZ $A \uparrow B \uparrow C$ EGY MÁSIK KÉTÉRTELMŰ KIFEJEZÉS: $(A \uparrow B) \uparrow C$ VAGY $A \uparrow (B \uparrow C)$? ÉS A^{2^M} -T, $A \uparrow (2 * M)$ -KÉNT KELL ÍRNI NEM PEDIG $A \uparrow 2 * M$ -NEK.

ÍME KÉT PROGRAM A KIFEJEZÉSEK BEMUTATÁSÁRA

AZ USZÓDESZKA BEHAJLÁSÁT A KÖVETKEZŐ KIFEJEZÉS ADJA MEG:

AHOL:
 $I = BD^3 / 12$
 ÉS B, A FA
 YOUNG-FÉLE
 MODULUSA: KB.



```

10 PRINT "AZ ADATOK: S(KP), L(M), B(CM), D(CM)"
20 INPUT S, L, B, D
30 LET E = 13000
40 LET I = B * D ^ 3 / 12
50 LET V = S * (L * 100) ^ 3 / (3 * E * I)
60 PRINT "A DESZKA"; V; "CM-T HAJLIK BE"
70 END
  
```

```

10 PRINT "AZ ADATOK: ÖSSZEG, ÉVEK, KAMAT%"
20 INPUT S, N, P
30 LET R = P / 100
40 LET M = S * R * (1 + R) ^ N / (12 * ((1 + R) ^ N - 1))
50 PRINT "HAVI TÖRLESZTÉS: "; M; "FT"
60 PRINT "VESZTESEG: "; M * (12 * N) - S; "FT"
70 END
  
```

AZ S' ÖSSZEGŰ HÁZ KÖLCSÖN HAVI RÉSZLETE

ÉVI P% KAMAT ÉS N ÉVI TÖRLESZTÉS ESETÉN A KÖVETKEZŐ:

$$\frac{SR(1+R)^N}{12[(1+R)^N - 1]}$$

AHOL

$$R = P \div 100$$



FIGYELEMREMÉLTÓ PONTOK:

HA AZT ÍROD, HOGY " $LET A = B/C$ ", ÉS C NETALÁN 0 LESZ (VAGY VALÁMILYEN NAGYON KIS SZÁM.), A SZÁMÍTÓGÉP AZT AZ ÜZENETET GÉPELI KI, HOGY "DIVISION BY ZERO" (azaz 0-val való osztás) VAGY "NUMBER TOO BIG" (azaz túl nagy szám).

ILYEN ESETEKBEN NÉMELYIK BASIC-RENDSZER ÉRTÉKET AD A-NAK, MIVEL AZ EREDMÉNY $\pm 10^{38}$ (vagy nagyobb, ha a számítógép tárolni tudja), AHOL AZ ELŐJEL A BAJT OKOZÓ KIFEJEZÉS SZÁMLALÓJÁNAK AZ ELŐJELE. MÁS BASIC-RENDSZER ESETLEG LEÁLLÍTJA A PROGRAMOT, VAGY PEDIG VÉGREHAJTJA A FENTI ÉRTÉKADÁST, DE NEM JELEZ HIBÁT.

- $A \uparrow B$ -T A LEGTÖBB BASIC-RENDSZER $-(A \uparrow B)$ -KÉNT KEZELI, AZONBAN A LEGBIZTOSABB, HA ZÁRÓJELET HASZNALUNK.

$A \uparrow B$ -T ÁLTALÁBAN 1.0-NAK VESZIK, HA $B = 0$ ÉS $A > 0$. LEÁLL A PROGRAM, HA $A < 0$, BÁR NÉMELYIK BASIC-RENDSZER MEGENGEDI EZT, HA B EGÉSZ SZÁM. $A \uparrow B$ 10^{38} -AT AD (l. fentebb), HA $A = 0$ ÉS $B < 0$.

FÜGGVÉNYEK

A $*$, $/$, $+$, $-$, \uparrow
OPERÁTOROKON
KIVÜL OLYAN FÜGG-
VÉNYEK IS TARTOZ-
NAK A BASIC-HEZ,
mint pl. a NEGYZETGYÖK.

MÁR TALÁLKOZTUNK EGY FÜGGVÉNNYEL A 3. OLDALON :

```
8 LET R = INT (G + 1)
```

AMELYBEN R A ZÁRÓJELBEN LEVŐ KIFEJEZÉS EGÉSZ RÉSZÉVEL LESZ EGYENLŐ. A G+1 KIFEJEZÉS LEHETNE BONYOLULTABB IS ; TEKINTSÜK ÚJRA A VÍZTARTÁLY PROGRAMOT :

```
10 PRINT "IRD BE: ÁTMÉRŐ, MAGASSÁG, BEFEDÉS"  
20 INPUT D, H, C  
30 LET T = INT ( 1 + (3.14 * D * (D/4 + H)) / C )  
40 PRINT T ; "DOBOZRA VAN SZÜKSÉGED"  
50 END
```

A KIFEJEZÉS MAGA IS TARTALMAZHAT FÜGGVÉNYEKET ; MÉG UGYANAZT A FÜGGVÉNYT IS :

```
10 INPUT X  
20 PRINT SQR (SQR (ABS (X)))  
30 END
```

EZ A PROGRAM EGY SZÁM POZITÍV ÉRTÉKÉNEK NEGYEDIK GYÖKÉT NYOMTATJA KI .

A KIFEJEZÉSBEN SZEREPLŐ FÜGGVÉNYT PONTOSAN ÚGY LEHET KEZELNI, MINT A NUMERIKUS VÁLTOZÓT ∞ AHOVÁ X-ET ÍRHATSZ, ÍRHATSZ INT(X)-ET IS. NEM SZABAD A FÜGGVÉNY NEVÉBEN SZÓKÖZT HAGYNI A NÉV ÉS A NYITÓ ZÁRÓJEL KÖZÖTT SEM. VAN AZONBAN OLYAN BASIC-RENDSZER, AMELY ELFOGADJA AZ ILYEN SZÓKÖZÖKET. A ZÁRÓJELEK KÖZÖTT LEVŐ KIFEJEZÉST LEHET LEVEGŐSEN VAGY ÖSSZETÖMÖRÍTVE ÍRNI.

SOK BASIC-RENDSZER A FÜGGVÉNYEK NAGYOBB VÁLASZTÉKÁT NYÚJTJA, DE MINDEGYIK TARTALMAZZA LEGALÁBB AZ ITT LEÍRT 11 STANDARD FÜGGVÉNYT. BELSŐ FÜGGVÉNYEKNEK HÍVJÁK ÖKET, MERT A BASIC-RENDSZER RÉSZEI. KITALÁLHATSZ MÁS FÜGGVÉNYEKET, AMELYEK NEM BELSŐ FÜGGVÉNYEK, EZEKRE A 26. OLDAL AD MAGYARÁZATOT.

A SGN(X), ABS(X), SQR(X), INT(X), LOG(X), EXP(X) FÜGGVÉNYEK MAGYARÁZATA A KÖVETKEZŐ OLDALON TALÁLHATÓ. A SIN(X), COS(X) TAN(X) TRIGONOMETRIKUS FÜGGVÉNYEK A 24. OLDALON KERÜLTEK LEÍRÁSRA, A 25. OLDALT AZ RND FÜGGVÉNYNEK SZENTELTÜK.

SGN (X) = +1 HA $X > 0$
 = 0 HA $X = 0$
 = -1 HA $X < 0$

"ELŐJEL"

10 PRINT SGN(7.2); SGN(0); SGN(-2)

1 0 -1

ABS (X) = +X HA $X \geq 0$
 = -X HA $X < 0$

"ABSZOLÚT ÉRTÉK"

20 PRINT ABS(7.2); ABS(0); ABS(-.2)

7.2 0 0.2

SQR (X) = \sqrt{X} HA $X \geq 0$
 = HIBA HA $X < 0$ (néhány BASIC \sqrt{x} -et hoz ki)

"NÉGYZETGYÖK"

30 PRINT SQR(16); SQR(0); SQR(-16)

4 0
 HIBA: SQR(NEGATIV)

INT (X) (az a legnagyobb szám, amelyik kisebb vagy egyenlő X-szel) (ahol 0 nagyobb, mint -1)

"EGÉSZ RÉSZE"

40 PRINT INT(3.4); INT(-3.4); INT(3)

3 -4 3

LOG (X) = $\log_e(x)$ HA $X > 0$
 = HIBA HA $X = 0$
 = HIBA HA $X < 0$ (azonban néhány BASIC $\log_e(x)$ -ot)

"TERMÉSZETES LOGARITMUSA"

50 PRINT LOG(1); LOG(10); LOG(2.71828)

0 2.30258 1

EXP (X) = e^x (ahol $e = 2.71828...$)

"A TERMÉSZETES ANTILOGARITMUS"
 "HATVÁNYA"

60 PRINT EXP(0); EXP(2.30258); EXP(1)

1 10 2.71828

ÍME EGY KIS PROGRAM A SGN, ABS ÉS AZ INT BEMUTATÁSÁRA.

```
10 PRINT "ÍRJ BE EGY PÉNZÖSSZEGET; + VAGY -"
20 INPUT L
30 LET S = INT(ABS(L) * 100 + .5) / 100
40 LET G = INT(S) * SGN(L)
50 LET F = INT((S - ABS(G)) * 100)
60 PRINT G; "FORINT ÉS"; F; "FILLÉR"
70 END
```

RUN

ÍRJ BE EGY PÉNZÖSSZEGET; + VAGY-

? - 123.456

- 123 FORINT ÉS 46 FILLÉR


ABS(L) pozitív, bármilyen előjele legyen is L-nek; * 100 fillérre vált; fél fillért hozzáad; INT() a legközelebbi fillérre kerekíti; (a/100 ft.ff alakra hozza az összeget az eredeti ± ft.ffff -ről.)

G a forintokat jelenti, visszaállított előjellel.

F a filléreket jelenti és ezek mindig pozitívek

TRIGONOMETRIKUS FÜGGVÉNYEK

MINDEN SZÖGET RADIÁNBAN MÉRÜNK



AZ ALÁBBI PÉLDÁKBAN P EGYENLŐ π -VEL.

70 LET P = 3.141593

SIN (A) "SINUS"

80 PRINT SIN(-P/6); SIN(0); SIN(P/2)

COS (A) "COSINUS"

90 PRINT COS(-P/3); COS(0); COS(P/2)

TAN (A) "TANGENS"




100 PRINT TAN(-P/4); TAN(0); TAN(5*P/4)

110 PRINT TAN(P/2)


$\tan(\frac{\pi}{2} + n\pi) = \infty$

ATN (T) "AZ A SZÖG, AMELYIKNEK A TANGENSE T," (vagy arcus tangens)

120 PRINT ATN(1E38); ATN(0); ATN(-1)



HIBA; TANGENS TÚL NAGY



1.57080 0 -.78540

A KÖVETKEZŐ PROGRAMPÉLDA TRIGONOMETRIKUS FÜGGVÉNYEKET HASZNÁL

10 PRINT "ÍRD BE P ÉS Q-T ÉS A KÖZBEZÁRT SZÖGET FOKBAN"

20 LET K = 3.141593 / 180

30 INPUT P, Q, D

40 LET C = D * K

FOKRÓL RADIÁNRA

RADIÁN RÓL FOKRA

50 LET S = 0.5 * P * Q * SIN(C)

60 LET A = ATN(Q * SIN(C) / (P - Q * COS(C))) / K


70 LET B = ATN(P * SIN(C) / (Q - P * COS(C))) / K

80 LET G = SQR(P * P + Q * Q - 2 * P * Q * COS(C))

90 PRINT "TERÜLET"; S; "SZEMKÖZTI OLDAL"; G

100 PRINT "ALAPSZÖGEK"; A; "ÉS"; B

110 END



ADOTT A P, Q ÉS A C SZÖG.

★ A TERÜLET $\frac{1}{2} PQ \sin C$

★ AZ A SZÖG $\arctan(Q \sin C / (P - Q \cos C))$

★ A B SZÖG $\arctan(P \sin C / (Q - P \cos C))$

★ A G OLDAL $\sqrt{P^2 + Q^2 - 2PQ \cos C}$

RND

A BASIC TUD "KOCKÁT VETNI", HOGY VÉLETLENSZÁMOKAT ∞ PONTOSABBAN PSZUDOVÉLETLEN-SZÁMOKAT ÁLLÍTSON ELŐ.

10 REM RND BEMUTATÁSA

20 LET X=RND

30 LET Y=RND

40 LET Z=RND

50 PRINT X;Y;Z

60 END

RUN

.240643 .417191 .192204

RND ARGUMENTUM NÉLKÜLI FÜGGVÉNY

MEGJEGYZÉS:

AZ EREDETI DARTMOUTH BASIC ÉS TÖBB MAS RENDSZER MEGKÖVETEL EGY ARGUMENTUMOT:

LET X = RND(A)

NÉMELYIK BASIC RENDSZER EZT NEM IGÉNYLI; MASOK KIFINOMULT MÓDON HASZNÁLJAK A KÜLÖNBÖZŐ VÉLETLEN SOROZATOK MEGJELELÉSÉRE.

AZ RND-T A KIFEJEZÉSEKBE PONTOSAN UGYANÚGY LEHET KEZELNI, MINT A NUMERIKUS VÁLTOZÓT; AHOVÁ X ÍRTHATÓ OTT RND IS SZEREPELHET. A FENTI PÉLDA 20-TÓL 50-IG TERJEDŐ UTASÍTÁSAIT EGY UTASÍTÁSBAN ÖSSZE LEHETNE FOGLALNI:

20 PRINT RND; RND; RND

A PROGRAM FUTÁSA ALATT, MINDEN ALKALOMMAL, AMIKOR A BASIC RENDSZER TALÁLKOZIK EGY RND-VEL EGY n VÉLETLENSZÁMOT AD, AHOL $0 \leq n < 1$ (MEGJEGYZÉS: NÉHA 0, DE SOHASEM 1). IGAZÁBÓL NEM VÉLETLENSZÁM, PSZUDOVÉLETLENNEK HÍVJÁK. HA A SZÁMÍTÓGÉP RND-VEL TALÁLKOZIK, AKKOR A SZÁMOK KÖTÖTT SORRENDJEBEN GENERÁLJA ÉS ADJA A KÖVETKEZŐ SZÁMOT. RENDSZERENKÉNT VÁLTOZIK, DE AZ ELTERJEDTEN HASZNÁLT CIKLUS EGY MILLIÓ DARAB HATJEGYŰ TÖRTET TARTALMAZ 0.000000-TÓL 0.999999-IG. MINDEN CIKLUSBAN MINDEN TÖRT PONTOSAN EGYSZER FORDUL ELŐ. A CIKLUS MINDEN FUTÁSKOR ÚJRAKEZDŐDIK, HA A CIKLUS ÖSSZES ELEMÉT ELHASZNÁLJA A PROGRAM, AKKOR A CIKLUS ÚJRAKEZDŐDIK. A FENTI PROGRAM MINDEN FUTTATÁSBAN UGYANAZT AZ EREDMÉNYT ADNA, DE EGY MÁSIK BASIC-RENDSZER LEHET, HOGY HÁROM TELJESEN KÜLÖNBÖZŐ SZÁMOT ADNA. AZONBAN...

RANDOMIZE

AMIKOR EZZEL AZ UTASÍTÁSSAL TALÁLKOZIK A SZÁMÍTÓGÉP, EZ AZ UTASÍTÁS A CIKLUSBAN ELŐRE KI NEM SZÁMÍTHATÓ HELYRE KÜLDI A PROGRAMOT.

A KÖVETKEZŐ PROGRAM
2 KOCKÁT VET



10 RANDOMIZE

20 LET X=INT(1+6 * RND)

30 LET Y=INT(1+6 * RND)

40 PRINT "DOBJ:";X;" ÉS";Y

50 END

RUN

DOBJ: 3 ÉS 5

"RANDOMIZE" MINDEN ALKALOMMAL ELŐRE KI NEM SZÁMÍTHATÓ DOBÁST BIZTOSÍT


NÉMELYIK BASIC-BEN NINCIS RANDOMIZE UTASÍTÁS, EHELYETT AZ RND(A) ARGUMENTUMOT HASZNÁLJÁK; AZ A ARGUMENTUM HASZNÁLATÁNAK MÓDJA RENDSZERENKÉNT VÁLTOZIK; A MEGFELELŐ KÉZIKÖNYVBEN KELL UTÁNAKÉZNI.

DEF

A BASIC-BEN LEGFÖLJEBB 26 SAJÁTFÜGGVÉNYT DEFINIÁLHATSZ A DEF UTASÍTÁS SEGÍTSÉGÉVEL. E FÜGGVÉNYEKET FNA, FNB, FNC-NEK stb. NEVEZHETED.

HASONLÍTSD ÖSSZE E KÉT PROGRAMOT, MINDEGYIKÜK EGY ADOTT HOSSZÚSÁGÚ CSŐ ANYAGÁNAK TÉRFOGATÁT SZÁMÍJTJA ÉS NYOMTATJA KI.

```
10 PRINT "ÍRD BE : L, D ÉS I"
20 INPUT L, D, I
30 LET A1 = 3.141592 * D ^ 2 / 4
40 LET A2 = 3.141592 * I ^ 2 / 4
50 LET V = L * (A1 - A2)
60 PRINT "TÉRFOGAT" ; V
70 END
```



$$\text{ANYAG TÉRFOGATA} = L \left(\frac{\pi D^2}{4} - \frac{\pi I^2}{4} \right)$$

"DEF" A DEFINIÁL RÖVIDÍTÉSE

```
10 DEF FNC(X) = 3.141592 * X ^ 2 / 4
20 PRINT "GÉPELD BE : L, D ÉS I"
30 INPUT L, D, I
40 LET V = L * (FNC(D) - FNC(I))
50 PRINT "TÉRFOGAT" ; V
60 END
```

HA EGYSZER DEFINIÁLTÁL EGY FÜGGVÉNYT (MINT A FENTI 10. SORBAN), UGYANÓGY HASZNÁLHATOD, MINTHA BELSŐ FÜGGVÉNY LENNE.

MI TÖRTÉNT AZ X VÁLTOZÓVAL? SEMMI : CSAK MOST FORMÁLIS ARGUMENTUMNAK NEVEZIK. EZ A DEF UTASÍTÁS AZT MONDJA, HOGY SZÁMOLD KI A ZÁRÓJELBEN LEVŐ ÉRTÉKET ; VEDD A NÉGYZETÉT, SZOROZD MEG A NÉGYZETÉT 3.141592 -VEL, OSZD EL EZT A SZORZATOT 4-GYEL, ÉS AZ EREDMÉNYT RENDELD FNC() -HEZ! UGYANEZZEL A LOGIKÁVAL ÍRHA TNÁNK AZT IS, HOGY

```
10 DEF FNC(X) = 3.141592 * X ^ 2 / 4
```

AZONBAN NEM LÉTEZIK OLYAN KARAKTER, HOGY A BILLENTYŰZETEN, ÍGY A BASIC ELŐIRJA, HOGY VÁLTOZÓNEVET HASZNÁLJUNK FORMÁLIS ARGUMENTUMKENT. AZ ELŐBB AZ X VÁLTOZÓT HASZNALTUK, DE AZ X ÉRTÉKÉT NEM VÁLTOZTATJA MEG A PROGRAM A 40. UTASÍTÁSNÁL.

SAJNOS VAN OLYAN BASIC RENDSZER, AMELY MEGVÁLTOZTATNÁ AZ X ÉRTÉKÉT A 40. UTASÍTÁSNÁL. ELLENŐRIZHETED SAJÁT VÁLTOZATOD, LEFUTTATVA A KÖVETKEZŐ PROGRAMOT.

```
10 REM FORMALIS ARGUMENTUM TESZTJE
20 DEF FNA(T) = T + 2
30 LET T = 0
40 LET A = FNA(3)
50 PRINT T ; "O KELLENE HOGY MARADJON"
60 END
```

Ha a változatod egy ötös számot ír ki, akkor meg kell különböztetned a változókat az argumentumoktól : lásd a következő oldalt.

FÜGGVÉNYEKET ARGUMENTUMOK NÉLKÜL IS DEFINIÁLHATSZ L. A KÖVETKEZŐ OLDALT.

NÉHÁNY PÉLDA A DEFINIÁLT FÜGGVÉNYEKRE :

10 REM ARGUMENTUM NÉLKÜLI FÜGGVÉNYEK

20 DEF FNI = INT(2 + 11 * RND)

30 DEF FNR = 3.141592 / 180

40 REM

50 REM FÜGGVÉNYEK FORMÁLIS ARGUMENTUMMAL

60 DEF FNC(X) = 3.141592 * X * X / 4

70 DEF FNM(X) = X * SGN(1 + SGN(X))

VÉLETLEN EGÉSZ A(2,12) TARTOMÁNYBAN.

A FOKOKAT SZOROZD MEG FNR-REL, HOGY RADIÁNT KAPJ.

"X" ÁTMÉRŐJŰ KÖR TERÜLETE.

X - "MACAULAY" VAGY "DIRAC" FÜGGVÉNYE

= X HA $X \geq 0$

= 0 HA $X < 0$

LEHETNEK OLYAN ARGUMENTUMOK IS, AMELYEK A FÜGGVÉNY A PROGRAMDBAN SZEREPLŐ VÁLTOZÓK ÉRTÉKEIT HASZNÁLJA FEL.

FORMÁLISAK;

80 REM PROGRAM VÁLTOZÓIT HASZNÁLÓ FÜGGVÉNYEK

90 DEF FNH = SQR(A * A + B * B)

100 DEF FNA(X) = 0.5 * A * B * SIN(X)

110 DEF FNL = P - INT(P/Q) * Q

120 DEF FNT(X) = INT(X * 10 ↑ D + .5) / 10 ↑ D

130 DEF FNG = INT(A + (B - 1) * RND)

A ÉS B OLDALÚ DERÉKSZÖGŰ HÁROMSZÖG ÁTFOGÓJA

A ÉS B OLDALÚ ÉS X SZÖGŰ HÁROMSZÖG TERÜLETE

FNL A P/Q OSZTÁS MARADÉKA

KERÉKÍTSD "X" -ET D TIZEDESRE

FNG VÉLETLEN EGÉSZ AZ A, B TARTOMÁNYBAN

A BASIC RENDSZEREK ELTÉRŐEK. HA "HORDOZHATÓ" PROGRAMOKAT AKARSZ ÍRNI, A KÖVETKEZŐ SZABÁLYOKAT KELL BETARTANOD :

- ★ BIZTOSÍTSD AZT, HOGY A DEF UTASÍTÁS ALACSONYABB SORSZÁM ALATT SZEREPELJEN, MINT A FÜGGVÉNY ELSŐ FELHASZNÁLÁSA!
- ★ BIZTOSÍTSD AZT, HOGY A PROGRAMOD LEGALÁBB EGYSZER ÉRINTI A DEF UTASÍTÁST! (NE KERÜLD MEG "GO TO"-VAL L. A 40. OLDALT!)
- ★ MÉG HA A BASIC VÁLTOZAT MEG IS ENGEDI, NE LEGYEN EGYNÉL TÖBB FORMÁLIS ARGUMENTUMOD! PL. NE LEGYEN `DEF FNA(A,B,C) = A+B+C`!
- ★ NE LEGYEN EGY SORNÁL HOSSZABB A FÜGGVÉNY! NÉMELYIK BASIC RENDSZER MEGENGEDI EZT; A TÖBBSOROS DEFINÍCIÓ VÉGÉT AZ FNEND UTASÍTÁSSAL KELL JELEZNI.
- ★ MINDIG, ALKALMAZZ EGY FORMÁLIS ARGUMENTUMOT, MÉG HA NEM IS HASZNÁLOD FEL! PL. A FENTI 30. SORT ÍRD ÚGY, HOGY `DEF FNR(Q) = 3.141592 / 180`! NÉMELYIK BASIC NEM ENGED MEG ARGUMENTUM NÉLKÜLI FÜGGVÉNYEKET.
- ★ A BIZTONSÁG KEDVÉÉRT KIZÁRÓLAG AZ @ BETŰT (01, 02 stb.) HASZNÁLD FORMÁLIS ARGUMENTUMKÉNT @ DE EZEKET SOHASE HASZNÁLD VÁLTOZÓKÉNT; NÉMELYIK BASIC A FORMÁLIS ARGUMENTUMOKAT VÁLTOZÓKÉNT KEZELI (lásd a bal oldalon).

PRINT

EZ A LEGSOKOLDALÚBB UTASÍTÁS A BASIC-BEN, ÉRTELMEZÉSE BASIC-RÓL BASIC-RE VÁLTOZIK.

A KÖVETKEZŐ PROGRAM EGY SZÁM NÉGYZETÉT NYOMTATJA KI :

```


10 LET E$ = "EGYENLŐ"
20 PRINT " ÍRJ BE EGY SZÁMOT "
30 INPUT A
40 PRINT A; " NÉGYZETE " ; E$ ; A * A
50 END
RUN
ÍRJ BE EGY SZÁMOT
? 4
4 NÉGYZETE EGYENLŐ 16
    
```

A 40. SOR BEMUTAT 4-ET AZON LEHETSÉGES DOLGOK KÖZÜL, AMELYEK A PRINT SZÓT KÖVETŐ LISTÁBAN SZEREPELHETNEK :

- ★ "NÉGYZETE" : SZÖVEG (LEHET BÁRMILYEN HOSSZÚSÁGÚ, AMI KIFÉR A SORBA), AMELY AZ IDÉZŐJELEK KIVÉTELEVEL PONTOSAN ÚGY KERÜL KINYOMTATÁSRA, MINT AHOGY A PROGRAMBAN SZEREPEL.
- ★ A : EGY NUMERIKUS VÁLTOZÓ, AMELYNEK ÉRTÉKE () ÉS NEM PEDIG NEVE () KERÜL KIÍRÁSRA.
- ★ E\$: EGY SZÖVEGES VÁLTOZÓ, AMELYNEK ÉRTÉKE KIÍRÁSRA KERÜL.
- ★ A * A : EGY KIFEJEZÉS, AMELYNEK ÉRTÉKE KIÍRÁSRA KERÜL.

A PRINT UTASÍTÁS BÁRMILYEN SZÖVEG, VÁLTOZÓ ÉS KIFEJEZÉS KEVERÉKÉBŐL ÁLLHAT, CSAK A LISTA EGY SORBAN FÉRJEN KI. PONTOSVESSZÓK () MINT FENTEBB () VAGY VESSZÓK VÁLASZTIJÁK EL ŐKET EGVMÁSTÓL : A VESSZÓ HATÁSÁRA A BASIC AZ ADATOKAT A KÖVETKEZŐKBEN LEÍRT MÓDON ZÓNÁKBAN NYOMTATJA.

AZ ALÁBBI KÉP EGY SORONKÉNT 72 POZÍCIÓS OLDALT MUTAT BE. A BASIC A SOROKAT 4 DB 15 ÉS EGY 12 POZÍCIÓS ZÓNÁRA OSZTJA FEL. (EBBEN A DOLOGBAN A BASIC RENDSZEREK MIND OLDAL-, MIND ZÓNASZÉLESSÉG TEKINTÉJÉBEN IGEN ELTÉRNEK, DE A KONCEPCIÓ ÁLTALÁNOS). A KARAKTERPOZÍCIÓKAT 1-TŐL KEZDVE SZÁMOZZUK, AHOGY EZT MÁS BASIC RENDSZEREK IS TESZIK, BÁR SOK VÁLTOZAT 0-TŐL SZÁMOZ (PL. 0-TŐL 71-IG, 1-72-IG HELYETT).



1 PRINT A ; B ; C

A PONTOSVESSZŐ AZT EREDMÉNYEZI, HOGY A NYOMTATÓFEJ NEM MOZDUL A KÖVETKEZŐ SZÁM VAGY SZÖVEG KINYOMTATÁSA ELŐTT (EBBEN AZ ESETBEN B ÉRTÉKE). HA NINCS ELÉG HELY A SORBAN, AKKOR A NYOMTATÓFEJ VISSZAFORDUL, HOGY ÚJ SORT TUDJON KEZDENI.

2 PRINT A ; B , C

A VESSZŐ AZT EREDMÉNYEZI, HOGY A NYOMTATÓFEJ A KÖVETKEZŐ ZÓNA ELSŐ POZÍCIÓJÁBA ÁLL, MIelőtt KINYOMTATJA A JELZETT SZÁMOT VAGY SZÖVEGET (EZ EBBEN AZ ESETBEN, A C ÉRTÉKE). HA A NYOMTATÓFEJ MÁR AMÜGYIS VALAHOL AZ 5. ZÓNÁBAN TALALHATÓ, AKKOR A KÖVETKEZŐ ZÓNA AZ 1-ES, A RA KÖVETKEZŐ SORBAN.

3 PRINT A : B,
4 PRINT C , D

MI TÖRTÉNIK ILYENKOR A LISTA ELSŐ ADATÁVAL (PL. C -VEL) ? NINCS KIVÉTEL : HA A BASIC AZ ELSŐ LISTA VÉGÉN VESSZŐT VAGY PONTOSVESSZŐT TALÁL, PONTOSAN ÚGY HAJTJA VÉGRE A PARANCSOT, MINTHA AZ A KÖVETKEZŐ FORMÁJÚ VOLNA : **3 PRINT A ; B , C , D**.

5 PRINT E ; F
6 PRINT G , H

MI TÖRTÉNIK AKKOR, HA NINCS VESSZŐ VAGY PONTOSVESSZŐ AZ ELŐZŐ LISTA VÉGÉN VAGY EGYÁLTALÁN NINCS ELŐZŐ LISTA ? VÁLASZ : HA A BASIC RENDSZER OLYAN PRINT UTASÍTÁST HAJT VÉGRE, AMELYNEK VÉGÉN NINCS SEM VESSZŐ, SEM PONTOSVESSZŐ, AKKOR A NYOMTATÓFEJ VISSZAFORDUL ÉS ÚJ SORT KEZD. FELTÉTELEZHETJÜK AZT, HOGY UGYANAZ TÖRTÉNIK A LEGELSŐ PRINT UTASÍTÁS VÉGREHATÁSA ELŐTT IS.

7 PRINT

MI TÖRTÉNIK HA ÜRES A LISTA ? EKKOR A BASIC A SEMMIT NYOMTATJA KI ÉS MIUTÁN NINCS SEM VESSZŐ SEM PONTOSVESSZŐ, EZÉRT A NYOMTATÓFEJ VISSZAFORDUL ÚJ SORT KEZDENI, AHOGY MÁR A FENTIEKBEN LEÍRTAM. RÖVIDEN : ÍGY KELL A BASIC -BEN ÜRES SORT LEÍRNI.

8 PRINT I, "SQD=" I*I

NEMELYIK BASIC RENDSZER MEGENGEDI, HOGY A SZÖVEG BÁRMELY OLDALÁN ELHAGVD AZ ÍRÁSJELEKET ÉS VESSZŐT VAGY PONTOSVESSZŐT TÉTELEZZ FEL VÁLTOZATTÓL FÜGGŐEN. NE TEDD EZT!

ME EGY PÉLDA A VESSZŐK ÉS PONTOSVESSZŐK HASZNALATÁNAK BEMUTATÁSÁRA:

10 DATA	"NAP", 28, 30, 31
20 READ	T\$, A, B, C
30 PRINT	"JAN", "FEBR", "MÁRC", "ÁPR",
40 PRINT	, "(29 SZÖKŐÉVEKBEN)"
50 PRINT	C ; T\$, A ; T\$, C ; T\$, B ; T\$
60 END	
RUN	

JAN	FEBR	MÁRC	ÁPR
	(29 SZÖKŐÉVEKBEN)		
31 NAP	28 NAP	31 NAP	30 NAP

FIGYELJÜK MEG A VESSZŐT A LISTA KEZDETÉN, AMELY A MÁSODIK ZÓNÁBA VALÓ NYOMTATÁST EREDMÉNYEZI

PRINT (FOLYTATÁSA)

MIÉRT ÚGY NYOMTATJA A BASIC
4 NÉGYZETE EGYENLŐ 16

ÉS MIÉRT NEM ÚGY, HOGY "4.000.00 NÉGYZETE EGYENLŐ 16.0000"
(TUDVÁN, HOGY A BASIC LEGALÁBB 6 SZÁMJEGYIG VÉGZI EL AZ ARITMETIKAI MŰVELETEKET)?

AZÉRT, MERT A BASIC FELTÉTELEZI, HOGY NEM ÉRDEKELNEK A TIZEDESPONT UTÁNI, ÉRTÉKTELEN 0-ÁK. A BASIC MÁST IS ELDÖNT A KINYOMTATOTT SZÁMOK SZÉLESSÉGÉVEL ÉS FORMÁJÁVAL KAPCSOLATBAN.

HA EGY SZÁMOT PONTOSAN KI LEHET FEJEZNI EGY LEGFELJEBB 6 SZÁMJEGYŰ EGÉSSZEL, AKKOR A BASIC EGÉSZKÉNT FOGJA KINYOMTATNI.

10 LET A = 654 321
20 PRINT A ; -A
30 PRINT 0.00 ; 20.0 ; 20.00002

A ▲ ÉS ▼ RE VONATKOZÓ MAGYARÁZATOT
LÁSD KÉŐBB
▼ 654 321 ▲ -654 321
▼ 0 ▲▼ 20 ▲▼ 20

HA EGY SZÁM KISEBB (KÖZELEBB VAN A 0-HOZ), MINT 0.1, AKKOR A BASIC E FORMÁBAN FOGJA KINYOMTATNI. LEGTÖBB ESETBEN EGY SZÁMJEGYET NYOMTAT A TIZEDESPONT ELÉ, ÉS ÖTÖT MÖGÉ, DE NÉMELYIK RENDSZER MIND A 6 SZÁMJEGYET A PONT UTÁN NYOMTATJA KI. EZUTÁN A LEGTÖBB BASIC RENDSZER EGY E-T NYOMTAT KI, AMELYET EGY PLUSZ VAGY EGY MÍNUSZJEL KÖVET, MAJD EZUTÁN KÖVETKEZIK A HATVÁNYKITEVŐ KÉT SZÁMJEGYE.

40 PRINT A / 10¹⁷ ; -A / 10¹⁹
50 PRINT 0.0060 ; -0.006

▼ 6.54321E-02 ▲ -6.54321E-04 ▲
▼ 6.00000E-02 ▲ -6.00000E-04 ▲

A BASIC RENDSZER A NAGY SZÁMOKAT IS (10⁶-T VAGY NAGYOBBAT) E-FORMÁBAN NYOMTATJA KI.

60 PRINT 100 * A
70 PRINT -1000000000 * A

▼ 6.54321E+07 ▲ -6.54321E+14 ▲

A BASIC RENDSZER 0.1 ÉS 10⁶ KÖZÖTTI SZÁMOKAT DECIMÁLIS ALAKBAN NYOMTATJA KI.

80 PRINT A / 10 ; -A / 100
90 PRINT -A / 1000 ; A / 1000000

▼ 65432.1 ▲ -6543.21 ▲
▼ -654.321 ▲▼ .654321 ▲

ÖSSZEFOGLALVA, SZÁMOK NYOMTATÁSÁKOR A KÖVETKEZŐ TÖRTÉNIK:

- ▼ MÍNUSZ JELET NYOMTAT HA A SZÁM NEGATÍV, KÜLÖNBEN EGY SZÓKÖZT ÍR KI.
- # A FENTIEKBEN ELMAGYARÁZOTT MÓDON ÉRTÉKTŐL ÉS PONTOSSÁGTÓL FÜGGŐEN EGY E-FORMÁJÚ VAGY DECIMÁLIS ALAKÚ SZÁMOT NYOMTAT KI.
- ▲ VÉGÜL EGY SZÁMVÉGI SZÓKÖZT NYOMTAT.

EGYES BASIC RENDSZEREK A FENTIEKET KÖVETŐEN 2, 1 VAGY 0 TOVÁBBI SZÓKÖZT IS KIÍRNAK, ÉS EZZEL AZ ÖSSZES NYOMTATOTT KARAKTEREK SZÁMÁT 3 TOBBSZÖRÖSÉVÉ TESZIK; MÁR RENDSZEREK VISZONT MINDIG 2 SZÁMVÉGI SZÓKÖZT VISZNEK KI.

▲ LEGTÖBB BASIC RENDSZER A SZÖVEGET ÉS A SZÖVEGES VÁLTOZÓKAT KÖVETŐ SZÓKÖZÖK HOZZÁADÁSA ÉS A SZÖVEGEK SZÓVÉGI SZÓKÖZEINEK ELHAGYÁSA NÉLKÜL NYOMTATJA KI

```

100 DATA "CATCH" , "DALMATIANS" , "STEPS"
110 READ C$ , D$ , S$
120 DATA 22.00 , 1.01 E2 , 39
130 READ C , D , S
140 PRINT C$ ; C , D ; D$ , S ; S$
150 PRINT
160 PRINT C$ ; -C , S ; S$ , -D ; D$
170 END
RUN

```

OUTPUT A 140. SORTÓL KEZDVE

CATCH 22 101 DALMATIANS 39 STEPS

ÜRES SOR

CATCH - 22 39 STEPS - 101 DALMATIANS

← 1.ZÓNA 2.ZÓNA 3.ZÓNA 4.ZÓNA 5.ZÓNA →

VIDD BE ÉS FUTTASD LE A FENTI PROGRAMOT (A 100.SORTÓL A 170.SORIG) ANNAK ELLENŐRZÉSÉRE , HOGY BASIC VÁLTOZATOD ELTÉRŐEN VISELKEDIK-E AZOKNÁL A VÁLTOZATOKNÁL , AMELYEK MINDEN SZÁM HOSSZÁT 3 TÖBBSZÖRÖSÉNEK MEGFELELŐ KARAKTER SZÁMRA ALAKÍJTJÁK , A KIMENET A KÖVETKEZŐ :

CATCH 22 101 DALMATIANS 39 STEPS

CATCH - 22 39 STEPS - 101 DALMATIANS

← 1.ZÓNA 2.ZÓNA 3.ZÓNA 4.ZÓNA 5.ZÓNA →

TAB () ;

A LEGTÖBB BASIC RENDSZER-
BEN HASZNÁLHATÓ A TAB
FÜGGVÉNY A PRINT-EN
BELÜL. PRINT

A KINYOMTATOTT EREDMÉNYEKNEK A PROGRAM ÁLTAL VEZÉRELT FORMÁ-
JÁT, VÁLTOZTATNI TUDJUK A TAB FÜGGVÉNYEKNEK A PRINT-ET KÖVETŐ LISTÁBAN
VALÓ SZEREPELTETÉSÉVEL.

```
10 PRINT "ADJ A-NAK ÉRTÉKET"  
20 INPUT A  
30 PRINT TAB(A); "HAL"; TAB(2*A); "EVŐ"; TAB(3*A+5); "EMBER"  
40 END  
RUN  
ADJ A-NAK ÉRTÉKET  
? 3  
    HAL EVŐ    EMBER  
RUN  
ADJ A-NAK ÉRTÉKET  
? 5
```



A TAB FÜGGVÉNYNEK CSAK A PRINT SZÓT KÖVETŐ LISTÁBAN VAN ÉRTELME.
A ZÁRÓJELBEN EGY VÁLTOZÓ VAGY EGY KIFEJEZÉS ÁLLHAT, A BASIC RENDSZER
KISZÁMÍJA AZ ÉRTÉKÉT, ÉS A NYOMTATÓFEJET AZ EREDMÉNY ÁLTAL MEGADOTT
KARAKTERPOZÍCIÓBA KÜLDI. NÉHÁNY TRÜKKRE VIGYÁZNI KELL A TAB HASZNÁLATÁNÁL:

- ★ NEMELYIK BASIC RENDSZER A KARAKTERPOZÍCIÓKAT 1-TŐL (L. 28. OLD.)
SZÁMOZZA, AZONBAN SOK BASIC 0-NÁL KEZDI A SZÁMOZÁST.
"HORDOZHATÓVÁ" PROGRAMODAT ÚGY TEHETED, HOGY A SZÁMOZÁST
MINDIG 1-TŐL KEZDED, ÉS SOHASEM HASZNÁLSZ 71-NÉL NAGYOBB
POZÍCIÓSZÁMOT.
- ★ NEMELYIK BASIC RENDSZER A KIFEJEZÉS ÉRTÉKÉHEZ LEGKÖZELEBB
ESŐ EGÉSZT HASZNÁLJA, MÍG MÁS RENDSZER ANNAK EGÉSZ RÉSZÉT
VESZI. HA A FÜGGVÉNYED PRODUKÁLHAT, NEM EGÉSZ EREDMÉNYT IS
HASZNÁLD AZ INT FÜGGVÉNYT, HOGY A KÍVÁNT EREDMÉNYT
MEGKAPD. PL. `; TAB(INT(P/3 + .5));`
- ★ NEMELYIK BASIC RENDSZER NEM VESZI FIGYELEMBE A TAB()-T
KÖVETŐ ÍRÁSJELEKET. PL. A TAB(6);-T UGYANÚGY KEZELI, MINT
A TAB(6)-OT. AZONBAN MÁSOK A TAB UTÁNI VESSZÓT HIBÁNAK FOG-
JÁK FEL; ISMÉT MÁSOK A VESSZÓ LÁTTÁN A NYOMTATÓFEJET A KÖ-
VETKEZŐ ZÓNÁBA KÜLDIK. EZÉRT MINDIG HASZNÁLD PONTOSVESSZÓT A TAB(); UTÁN.

★ HA A NYOMTATÓFEJ, MÁR TÚLHALADT A TAB FÜGGVÉNY ÁLTAL MEGHATÁROZOTT POZÍCIÓN ∞ VAGY HA A MEGHATÁROZOTT POZÍCIÓ NINCS A PAPIRON ∞, AZ EGYES BASIC RENDSZEREK KÜLÖNBÖZŐ MÓDON REAGÁLNAK, ÉS AZ EREDMÉNY ÁLTALÁBAN EGY KUSZA KIMENET LESZ. NE TÁMASZKODJ SAJÁTOS ÉRTELMEZÉSEKRE; INKÁBB ÍRD LE HELYESEN A KIFEJEZÉSEKET!

★ A TAB()-NAK A LISTA EGY TAGJAKÉNT, EGYEDÜL KELL ÁLLNIA. NEM LEHET KIFEJEZÉSBE BEVONNI, BÁRMILYEN EGYSZERŰ LEGYEN IS EZ A KIFEJEZÉS :

100 PRINT 4 * TAB(A*6); X

ATAB IGEN HASZNOS NYOMTATOTT NAGYLÉPTÉKŰ GRAFIKONOK KÉSZÍTÉSÉNÉL. ENNEK ILLUSZTRÁLÁSÁRA ELŐRE KELL UGORNUNK A 48. OLDALRA, AHOL MAGYARÁZATOT KAPUNK ARRRA, HOGY AZ ALÁBBI 30-AS SORSZÁMÚ UTASÍTÁS MIÉRT, AD AZ X-NEK 0, 15, 30 STB. EGÉSZEN 180-IG TERJEDŐ, EGYMÁST KÖVETŐ ÉRTEKEKET.

```
10 PRINT "COSINUS (X) GRAFIKONJA X=0-TÓL 180-IG";
```

```
15 PRINT " 15 FOKOS LÉPESEK BEN "
```

```
20 PRINT "X" ; TAB(62); "COS(X)"
```

```
30 FOR X = 0 TO 180 STEP 15
```

```
40 LET C = COS(X * 3.141592 / 180)
```

```
50 LET P = INT(30 * ABS(C) + .5) * SGN(C)
```

```
60 PRINT X ; TAB(P + 36); "X"
```

```
70 NEXT X
```

```
80 END
```

```
RUN
```

```
COSINUS (X) GRAFIKONJA X=0-TÓL 180-IG 15 FOKOS LÉPESEK BEN
```

```
X
```

```
0
```

```
15
```

```
30
```

```
45
```

```
60
```

```
75
```

```
90
```

```
105
```

```
120
```

```
135
```

```
150
```

```
165
```

```
180
```

```
COS(X)
```

```
*
```

```
*
```

```
*
```

```
*
```

```
*
```

```
*
```

```
*
```

```
*
```

```
*
```

```
*
```

```
*
```

```
*
```

```
*
```

```
*
```

A FÜGGŐLEGES TENGELY AZ OLDAL KÖZEPÉRE KERÜL, 36 POZÍCIÓVAL ELTOLVA

SZÁMÍTSD ÁT RADIÁNBA, ÉS LEGYEN C A COSINUS

COS(0) ÉRTÉKÉT A 30 PRINT POZÍCIÓRA VÁLASZTJUK ÉS A LEGKÖZELEBBI POZÍCIÓRA KEREKÍTJÜK.

PRINT USING

NEM MINDEN
BASIC REND-
SZERNEK VAN
ILYEN UTASÍTÁSA,

AMELYEKBEN EZ MEGVAN, AZOKNÁL A RÉSZLETES SZABÁLYOK ELTÉRŐEK. OLYAN ESZKÖZ, AMELLYEL A KIVÁNT KIMENŐ ADATOT EGY MINTA ALAPJÁN LEHET LÉTREHOZNI.

A KETTŐSPONTOT
A "MINTA"
KÖVETI
VAN BASIC, MELY
% JELET HASZNÁL
ÉS NEM :-OT

10 LET G = 1000.00
20 PRINT USING 30, G, G/20
30 : # # # # FORINT! GYERE, VIGYÉL EL ##-ET!
40 END
RUN
1000 FORINT! GYERE, VIGYÉL EL 50-ET!

A HASZNÁLANDÓ "MINTA" SOR
SZÁMA

A LISTA VESSZŐVEL
KEZDŐDIK

G "MASZKJA"

A G/20
"MASZKJA"

A PRINT USING UTASÍTÁS A PROGRAM EGY MÁSIK SORÁRA HIVATKOZIK, AMELYIK A KINYOMTATANDÓ MINTÁT TARTALMAZZA. E KÉP MASZKOKAT TARTALMAZHAT (EZEKET ÁLTALÁBAN FORMÁTUMOKNAK HÍVJÁK). MINDEN MASZK MEGHATÁROZZA A HOZZÁ TARTOZÓ ADAT HELYÉT ÉS ALAKJÁT A KINYOMTATANDÓ PRINT LISTÁN.

A PRINT USING SZAVAK UTÁN AZ A PROGRAMSORSZÁM ÁLL, Ahol a minta található. (EZ BÁRHOL LEHET A PROGRAMBAN, ÉS TÖBB PRINT UTASÍTÁS IS HASZNÁLHATJA.) MAJD EGY VESSZŐ[†] KÖVETKEZIK, AMELY BEVEZETI A VESSZŐVEL ELVÁLÁSZTOTT VÁLTOZÓK ÉS KIFEJEZÉSEK LISTÁJÁT. NE TEGY VESSZŐT A LISTA VÉGÉRE; A NYOMTATÓFEJ AUTOMATIKUSAN ÚJ SORT KEZD, AMIKOR A PRINT USING UTASÍTÁS VÉGREHAJTÁSA BEFEJEZŐDIK. (EZZEL SZEMBEN A NYOMTATÓFEJ NEM KEZD ÚJ SORT AZ UTASÍTÁS ELŐTT; OTT MARAD, Ahol az előző "PRINT" VAGY "PRINT USING" HAGYTA.)

A MINTASOR: A SORSZÁM UTÁN EGY KETTŐSPONT ÁLL. EZUTÁN PONTOSAN AZ KÖVETKEZIK, AMIT A SZÁMÍTÓGÉPPAL KI AKARUNK ÍRATNI. DE MINDEN SZÁMJEGYET EGY # JEL HELYETTESÍT ABBÓL A CÉLBÓL, HOGY A KIÍRANDÓ SZÁM KÖTÖTT FORMÁT KAPJON. HA EGY TIZEDESPONTOT AKARSZ KINYOMTATNI, AKKOR EGY TIZEDESPONTOT KELL ÍRNI A MASZKBA.

10 FL # # # # . # # ! GYERE, VIDD EL (AZ) FL # # # . # #-ET!
20 PRINT USING 10, 1000, 50
30 END
RUN
FL 1000.00! GYERE, VIDD EL (AZ) FL 50.00(-ET)!

LÁSD A
LÁBJEGYZETET

FL A HOLLAND FORINT
PÉNZJELE

[†] N.C.C. "STANDARD BASIC" KETTŐSPONTOT KÍVÁN MEG ITT, EZ SOKKAL SZEBBEN NÉZNE KI, MINT A VESSZŐ, DE MÉG NEM TALALTAH OLYAN BASIC-ET, AMELYIK ITT KETTŐSPONTOT MEGEN-
GED.

HA A PROGRAM NAGYOBB SZÁMOKAT AD, MINT AMI A MASZKBA BELEFÉR, AKKOR A RENDSZER NEM FOGADJA EL NEMELY BASIC A MASZKOT TELEGÉPELI CSILLAGGAL, NEMELYIK PEDIG A MASZKOT MEGVÁLTOZTATJA VAGY KITERJESZTI, MÁSKOR AZ EGÉSZ PROGRAM VÉGREHAJTÁSÁT LEÁLLÍTJÁK.

###.## ← E MASZK 999.99-NÉL NEM NAGYOBB POZITÍV ÉS
 999.99 - 99.99-NÉL NEM KISEBB NEGATÍV SZÁMOKAT KEZEL.
 -99.99

NEM SZÁRMAZIK SEMMI BAJ ABBÓL, HA A MASZKOK A FELTÉTLENÜL SZÜKSÉGESNÉL HOSSZABBAK.

```
5: INTEGRAL ## DECIMAL ###.##
10 LET A = 9.999
15 PRINT USING 5 , A , A
20 END

RUN
INTEGRAL 9 DECIMAL 10.00
```

9.999 EGÉSZ RÉSZE

9.999 KÉT TIZEDESJEGYRE KEREKÍTVE

A BASIC AZ EGÉSZ FORMÁJÚ MASZKOT JOBBRÓL BALRA TÖLTI FEL: ←
 #####

A BASIC A PONTTÓL KIFELE HALADVA TÖLTI FEL A TIZEDES FORMÁJÚ MASZKOKAT: ← →

###.##

A SZÁM EGÉSZ RÉSZÉT EGY EGÉSZ FORMÁJÚ MASZKBA; A TÖRT RÉSZT A JELEZETT JEGYEK SZÁMÁRA KEREKÍTVE HELYEZI EL A MASZK TIZEDESEK SZÁMÁRA SZOLGÁLÓ RÉSZÉN.

HA A PRINT LISTA TÖBB ADATOT TARTALMAZ, MINT AMENNYIT A MINTA FELTUDNA DOLGOZNI, A RENDSZER A MINTÁT ÚJRA MEG ÚJRA FELHASZNÁLJA EGÉSZEN ADDIG, MIG EL NEM FOGY A LISTA.

```
10 : ## ÁRPAKENYÉR ÉS ## KIS HAL
20 PRINT USING 10 , 5, 2, 10, 4, 20
30 END

RUN
5 ÁRPAKENYÉR ÉS 2 KIS HAL
10 ÁRPAKENYÉR ÉS 4 KIS HAL
20 ÁRPAKENYÉR ÉS
```

E PÉLDA AZT IS BEMUTATJA, HOGY MI TÖRTÉNIK, HA A LISTA TÚL KEVÉS ADATOT TARTALMAZ A MINTA SZÁMÁRA. A NYOMTATÁS AZ ELSŐ HIÁNYZÓ ADATNÁL LEÁLL.

EDDIG A SZABÁLYOK AZONOSAK VOLTAK A BASIC MAJD NEM MINDEN OLYAN VÁLTOZATÁBAN, AMELY PRINT USING UTASÍTÁSSAL RENDELKEZIK.

(LEGALÁBB EGY KÖZÜLÜK AZONBAN, EZT A FORMÁT KIVÁNJA MEG SOK BASIC RENDSZERBEN EZEK OPCIONÁLIS SZOLGÁLTATÁSOK).

```
100 LET A$ = "ANS = ##.#"
110 PRINT USING A$ , X
```

```
100 PRINT USING "ANS = ##.#" X
```

A TÚLOLDALON SZEREPLŐ SZOLGÁLTATÁSOK ÉS SZABÁLYOK JOBBAN VÁLTOZNAK BASICRÓL BASICRE, MINT AZ EDDIG TÁRGYALTAK; A "HORDOZHATÓSÁG" KEDVÉÉRT LEGJOBB, HA EZEKET ELKERÜLJÜK.

PRINT USING (FOLYTATÁS)

↑↑↑↑ OLYAN MASZKOKAT KÉSZÍTHETSZ, AMELYEK A SZÁMOKAT E FORMÁBAN NYOMTATJÁK KI, OLY MÓDON, HOGY FELFELÉ MUTATÓ NYILAKAT TESZEL A TIZEDES FORMÁJÚ MASZK UTÁN. A BASIC EGYES VÁLTOZATÁNÁL A NYILAK SZÁMA MEGHATÁROZOTT & EGYIKBEN 4, MÁSIKBAN 5 (NÉMELY BASIC RENDSZER FELKIÁLTÓJELET HASZNÁL A NYIL HELYETT).

-34.5600 → #.### ↑↑↑ → -.346E+02

LEGALÁBB EGY # JELNEK KELL MEGELŐZNI A TIZEDESPONTOT, HOGY AZ ELŐJELNEK HELYET TARTSON FENN.



```

10 : *** A CÍM * * *
20 :
30 PRINT USING 10
40 PRINT USING 20
50 END

RUN

*** A CÍM * * *
    
```

ÜRES SOR
A 20-AT
HASZNÁLVA

NÉMELY BASIC RENDSZERBEN A PRINT LISTA SÓT A KÉPLISTA IS LEHET ÜRES.

(EZEKET A DOLGOKAT KÖNNYEBBEN MEGOLDHATJUK, HA PRINT-ET HASZNÁLUNK USING NÉLKÜL.)

NÉMELY BASIC MEGENGED PÉNZJELEKET IS A MASZKBAN.



1.234 → \$\$\$\$. ## → \$ 1.23
23.456 → \$ 23.46
345.678 → \$ 345.68

A PÉNZJEL (£ AZ EGYESÜLT KIRÁLYSÁGBAN) A BAL OLDALON "LEBEG". MÁS BASIC RENDSZER UGYANEZT A TRÜKKÖT CSINALJA CSILLAGOKKAL.




NÉMELYIK BASIC RENDSZER A MASZK ELŐTT EGY PLUSZ VAGY EGY MINUSZ JELET ENGED MEG, PÉLDAUL
+ ###. # ÉS - #. ##.

A + JEL HATÁSA : "ÍRJ EGY PLUSZ JELET A SZÁM ELÉ, HA AZ POZITÍV & MÍNUSZ JELET, HA AZ NEGATÍV."
A - JEL HATÁSA : "ÍRJ EGY SZÓKÖZT A SZÁM ELÉ, HA AZ POZITÍV & EGY MÍNUSZ JELET HA AZ NEGATÍV."

NEM MINDIG DERÜL KI A KÉZIKÖNYVEKBŐL, HOGY VAJON E SZOLGÁLTATÁST LEHET-E A PÉNZJELLEL EGYÜTT HASZNÁLNI.

† AZ N.C.C. STANDARD BASIC MEGKÖVETELI AZ ELŐJEL HASZNÁLATÁT A FORMÁTUM ELŐTT.



LEGTÖBB OLYAN BASIC-BEN, AMELYNEK VAN PRINT USING UTASÍTÁSA, LEHET SZÖVEGVÁLTOZÓKRA IS MASZKOT KÉSZÍTENI. EZEK IGEN HASZNOSAK, DE SAJNOS RÉSZLETEIKBEN ELTÉRNEK EGYMÁSTÓL.

```
10 REM      KÖZÖS START
20 LET T$ = "TO"
30 LET B$ = "BE"
40 PRINT USING 50 , T$ , B$ , T$ , B$
```

NÉZD AZ 50. SOR ALÁBBI 4 KÜLÖNBÖZŐ VÁLTOZATÁT

```
60 REM      KÖZÖS BEFEJEZÉS
70 END

RUN
```

A KÜLÖNBÖZŐ BASIC RENDSZEREK ESETÉRE AZ 50. SOR LEHETSÉGES VÁLTOZATAIT, AZ ALÁBBIKBAN MUTATJUK BE. AZ EGYES MINTÁK FELHASZNÁLÁSÁBÓL EREDŐ EREDMÉNYT KÖZVETLENÜL A SOR UTÁN LATHATJUK.

```
50: ## ### OR NOT #.##
```

T O BE OR NOT T O B

A KITÖLTÉS BALRÓL JOBBRA TÖRTÉNIK

A TIZEDESPONT (ÉS A # JEL) UGYANÚGY VESELKEDIK, MINT A # JEL

A SZÖVEG CSONKÍTÁSA A JOBB OLDALRÓL JÖRTÉNK, HA A SZÖVEG TUL HOSSZÚ

```
50: 'LL 'RRR OR NOT 'CCC
```

T O BE OR NOT T O B

AZ L-EK AZT JELENTIK, HOGY BALRA IGAZÍT

AZ R-EK AZT JELENTIK, HOGY JOBBRA IGAZÍT

A C-K AZT JELENTIK HOGY KÖZÉPRE IGAZÍT

AZ APOSZTRÓF A SZÖVEGMASZKOT JELZI

```
50: \ \ OR NOT \ \ !
```

T O BE OR NOT T O B

! EGYETLEN KARAKTERNÉL

A SZÓKÖZÖK AZ \ ÉS A \ KÖZÖTT FONTOSAK A MASZK BALRÓL JOBBRA TÖLTŐDIK FEL

```
50: <# <# OR NOT >#### >
```

T O BE OR NOT T O B

< AZT JELENTI, HOGY BALRA IGAZÍT ÉS JOBBRÓL CSONKÍT

> AZT JELENTI, HOGY JOBBRA IGAZÍT ÉS BALRÓL CSONKÍT

EZ N.C.C. "STANDARD BASIC".

EGYETLEN KARAKTERBŐL ÁLLÓ MASZK ILYEN < IS.

EZENKÍVÜL TOVÁBBI VÁLTOZATOK IS VANNAK. MIÉRT NEM SZABVÁNYOSÍTUNK? MÉG A ROSSZ SZABVÁNY IS JOBB LENNE A SEMMINÉL.

3

VEZÉRLÉS

GO TO

EZZEL AZ UTASÍTÁSSAL MEGVÁLTOZTATHATOD AZT A SORRENDET, AMELYBEN A BASIC RENDSZER A PROGRAM SORSZÁMOZOTT UTASÍTÁSAIT VÉGREHAJTJA.

EZ A PROGRAM SOHASEM ÉR VÉGET. D ÉRTÉKÉT MEGADVÁN, KISZÁMÍJTJA ÉS KINYOMTATJA V-T, MAJD ÚJRA VISSZATÉR A 20. SORHOZ D-NEK ÚJABB ÉRTÉKET KÉRVE ÉS ÍGY TOVÁBB ÉS ÍGY TOVÁBB.

A PROGRAMOT A **BREAK** BILLENTYŰVEL LEHET MEGÁLLÍTANI

(VAGY AZZAL A BILLENTYŰVEL AMIT RENDSZERED ERRE A CÉLRA HASZNÁL).

```
10 PRINT "LABDÁK TÉRFOGATA"  
20 PRINT  
30 PRINT "ÍRJ BE EGY ÁTMÉRŐT"  
40 INPUT D  
50 LET V = 3.141592 * D^3 / 6  
60 PRINT "A LABDA TÉRFOGATA"; V  
70 GO TO 20  
80 END
```

```
RUN  
LABDÁK TÉRFOGATA  
ÍRJ BE EGY ÁTMÉRŐT  
? 6.5  
A LABDA TÉRFOGATA 143.793  
ÍRJ BE EGY ÁTMÉRŐT  
?
```

BREAK

EZ A PÉLDA NEM IS TRIVIÁLIS. A BASIC IGEN HASZNOSAN FELHASZNÁLHATÓ KALKULÁTORKÉNT, HOGY A BEVITT VÁLTOZÓK EGYMÁST KÖVETŐ ÉRTÉKEIVEL KÉPLETEKET SZÁMOLJUNK KI, ÉS A LEGEGYSZERŰBB, HA A FOLYAMAT EGYETLEN BILLENTYŰ LEÜTÉSÉVEL LEÁLLÍTHATÓ.

SZINTAXIS :

70 GO TO 20

ÍRHAJD A GOTO-T EGY SZÓBA

IDE CSAK SORSZÁMOK ILLENÉK: NEM ÍRHAJD AZT, HOGY LET L=20, ÉS EZT KÖVETŐEN GO TO L

A PROGRAM BÁRMELYIK SORÁBA ÁTUGORHATSZ (MÉG AKKOR IS, HA AZ EGY OLYAN NEM VÉGREHAJTHATÓ UTASÍTÁST TARTALMAZ, MINT A REM), ÉS A VÉGREHAJTÁS INNEN FOLYTATÓDIK TOVÁBB. HA A GO TO EGY NEM LÉTEZŐ SORRA MUTAT, AKKOR A RUN PARANCSNÁL A LEGTÖBB BASIC RENDSZER ELSEM INDÍTTJA A PROGRAMOT. (UGYANEZ TÖRTÉNIK AZ IF ÉS AZ ON ESETEBEN IS.)

IF-THEN

EZZEL AZ UTASÍTÁSPÁRRAL
A VÉGREHAJTÁS MEG-
SZOKOTT SORRENDJÉT

VÁLTOZTATHATOD MEG DE FELTÉTELHEZ KÖTÖTTEN, A FELTÉTELEK A KÖVETKEZŐK:

```
20 IF A=B*C THEN GO
```

E SORNAK
LÉTEZNI KELL

A HAT LEHETSÉGES
FELTÉTEL EGYIKE

MENJ EHHEZ A
SORHOZ (ÉS
FOLYTASD INNEN),
HA IGAZ A FELTÉTEL;
MÁSKÜLÖN BEN MENJ
TÖVÁBB.



NE LEGVEN
SZÖKÖZ A SZIMBÓ-
LUMOK KÖZÖTT A
FELTÉTEL BEN

CSAK SORSZÁMOK:
NEM ÍRTHATOD AZT,
HOGY "THEN 1"

$=$	EGYENLŐ
$>$	NAGYOBB, MINT
$<$	KISEBB, MINT
$>=$	NAGYOBB, MINT VAGY EGYENLŐ
$<=$	KISEBB, MINT VAGY EGYENLŐ
$<>$	NEM EGYENLŐ

"FELTÉTEL" BÁRMELYIK OLDALÁN
LÉVŐ VALAMI LEHET SZÁM VAGY KIFEJEZÉS :

```
30 IF 1 + SQRT(A^2 + B^2) > 0.2 THEN 10  
40 IF ABS(A - B) <= 0.01 THEN 15
```

AZ A ÉS B "KÖRÜLBELÜLI"
EGYENLŐSÉGÉNEK
ELLENŐRZÉSI MÓDJA

VAGY LEHET SZÖVEG, ILL. SZÖVEGVÁLTOZÓ, AMIKOR A FELTÉTEL NEM BONYO-
LULTABB AZ EGYENLŐ VAGY A NEM EGYENLŐ ESETEKNÉL (L. A LAP ALJÁT) :

```
50 IF Q$ = "IGEN" THEN 150  
60 IF "FEJEZD BE" <> A$ THEN 10  
70 IF R$ = T$ THEN 230
```

MEGJEGYZÉS:
AZ "IGEN"
ÉS AZ "IGEN"
SZÖVEG NEM EGYENLŐ

ÉRTELMETLEN DOLOG NUMERIKUS VÁLTOZÓT SZÖVEGVÁLTOZÓKKAL ÖSSZEHASONLÍTANI.

```
80 IF Q$ = Y THEN 99
```

SOK BASIC RENDSZER MEGENGEDI A "GO TO" (VAGY "GOTO") SZÓ "THEN"
HELYETTI HASZNÁLATÁT, DE A HORDOZHATÓSÁG KEDVÉÉRT JOBB, HA A "THEN"
HEZ RAGASZKODUNK.

LEGTÖBB BASIC RENDSZER BONYOLULTABB SZÖVEG-ÖSSZEHASONLÍTÁSOKAT IS MEGEN-
GED. ÁLTALÁBAN A "Z"-T "NAGYOBBNAK" VESZIK, MINT "A"-T ÉS A "9"-ET
NAGYOBBNAK, MINT A "0"-ÁT. A SZÖKÖZ " " KISEBB, MINT BÁRMELY BETŰ VAGY SZÁM.
ÍGY A NEVEKET ALFABETIKUS SORRENDBE RENDEZHETED:

"A" < "ABLAK" < "ADAG" ÉS "V2" < "VB"

STOP

EZT AZ UTASÍTÁST HASZNÁLVA, PROGRAM BÁRMELYIK SZORÁNÁL LEÁLLÍTHATOD A VÉGREHAJTÁST.

MINDEN PROGRAM UTOLSÓ UTASÍTÁSÁNAK AZ "END" -NEK KELL LENNIE, ÉS CSAK AZ UTOLSÓ UTASÍTÁS LEHET "END".

AZ "END" KETTŐS SZEREPET JÁTSZIK:

- ★ A BASIC RENDSZER SZÁMÁRA MEGJELÖLT PROGRAM VÉGÉTELZI, AMIKOR AZ A "RUN" HATÁSÁRA A BASIC NYELVŰ PROGRAMOT ÉRTELMEZI.
- ★ AMIKOR AZ "END" TÉNYLEGESEN VÉGREHAJTÓDIK, LEÁLLÍJA A PROGRAM FUTASÁT.

LEHET, AZONBAN TÖBB OLYAN HELY IS A PROGRAMBAN, AHOL LE SZERETNÉNK ÁLLÍTANI ANNAK VÉGREHAJTÁST. EZT MEGTEHETED A "GO TO"-VAL AMELY A VEZÉRLÉST ÁTADJA AZ END-NEK, VAGY HASZNÁLHATSZ "STOP" UTASÍTÁST IS. A "STOP" ELTÉRŐEN AZ END-TŐL, AKÁRHÁNYSZOR ÉS BÁRHOL ELŐFORDULHAT A PROGRAMBAN.

```
10 PRINT "SZERETSZ PROGRAMOZNI ? "  
20 INPUT A$  
30 IF A$ = "NEM" THEN 70  
40 IF A$ = "IGEN" THEN 90  
50 PRINT "A VÁLASZ NEM VILÁGOS "  
60 GO TO 100  
70 PRINT "LÉGY KITARTÓ! MEG FOGOD TANULNI "  
80 GO TO 100  
90 PRINT "ÉRDEKFESZÍTŐ, NEMDE?"  
100 END
```

E KÉT BUTA PROGRAM UGYANAZT VÉGZI EL, AZ "END" ÉS A "STOP" HASZNÁLATÁT MUTATJA BE.

```
10 PRINT "SZERETSZ PROGRAMOZNI ? "  
20 INPUT A$  
30 IF A$ = "NEM" THEN 70  
40 IF A$ = "IGEN" THEN 90  
50 PRINT "A VÁLASZ NEM VILÁGOS "  
60 STOP  
70 PRINT "LÉGY KITARTÓ! MEG FOGOD TANULNI "  
80 STOP  
90 PRINT "ÉRDEKFESZÍTŐ, NEMDE ? "  
100 END
```

PÉLDA

A GO TO
IF THEN BEMUTATÁSÁRA
ÉS A STOP

A PROGRAM OLYAN KÉTISMERETLENES EGYENLETRENDSZERT OLD MEG, AMELYNEK TETSZŐLEGES SZÁMÚ JOBB OLDALA LEHET. LEGYEN A KÉT EGYENLET A KÖVETKEZŐ:

$$ax + by = p$$

$$cx + dy = q$$

CRAMER SZABÁLYÁT FELHASZNÁLVA A MEGOLDÁST A KÖVETKEZŐ MÓDON LEHET FELÍRNI:

$$x = \frac{\begin{vmatrix} p & b \\ q & d \end{vmatrix}}{\begin{vmatrix} a & b \\ c & d \end{vmatrix}}, \quad y = \frac{\begin{vmatrix} a & p \\ c & q \end{vmatrix}}{\begin{vmatrix} a & b \\ c & d \end{vmatrix}}$$

AHOL A FÜGGŐLEGES VONALAK A KÖVETKEZŐ MÓDON KISZÁMÍTHATÓ DETERMINÁNSOKAT JEJÖLİK:

$$\begin{vmatrix} a & b \\ c & d \end{vmatrix} = a \cdot d - c \cdot b$$

HA A NEVEZŐBEN LEVŐ DETERMINÁNS NULLA (VAGY NAGYON KÖZEL ÁLL A NULLÁHOZ), AKKOR NINCS MEGOLDÁS.

```
10 PRINT " KÉTISMERETLENES EGYENLET "
20 PRINT " GÉPELD BE X ÉS Y EGYÜTTTHATÓIT ; ELSŐ SOR "
30 INPUT A,B
40 PRINT " MOST A MÁSODIK SORT "
50 INPUT C,D
60 REM SZÁMOLD KI M NEVEZŐT
70 LET M = A * D - C * B
80 IF ABS(M) > 0.00001 THEN 110
90 PRINT " NINCS MEGOLDÁS ; DET = " ; M
100 STOP
110 PRINT " GÉPELD BE KÉT ÉRTÉKET A JOBB OLDALRA "
120 INPUT P,Q
130 LET X = (P * D - Q * B) / M
140 LET Y = (A * Q - C * P) / M
150 PRINT " X = " ; X ; " Y = " ; Y
160 PRINT " VAN MÉG JOBB OLDAL ? IGEN ? "
170 INPUT A$
180 IF A$ = "IGEN" THEN 110
190 REM ITT LEHETSÉGES A "STOP"
200 END
```

PÉLDA



A GO TO ÉS IF THEN BEMUTATÁSÁRA.

ÍME EGY OLYAN PROGRAM, AMELY TÉGLALAPOK, HÁROMSZÖGEK ÉS KÖRÖK TERÜLETÉNEK KISZÁMÍTÁSÁRA KÉSZÜLT. MEGKÉRDEZI AZ ALAKZAT NEVÉT, MAJD KÉRI AZ ERRE AZ IDOMRA JELLEMZŐ MÉRTEKET. HA BÁRMILYEN MÁS NEVET GÉPELSZ BE, MINT "TÉGLALAP", "HÁROMSZÖG" VAGY "KÖR", A PROGRAM LEALL.

A PROGRAM LOGIKAI FELÉPÍTÉSÉT AZ ALÁBBI FOLYAMATÁBRA MUTATJA BE. ILYEN PROGRAMOK SZERVEZÉSÉNEK ENNÉL SOKKAL TETSZETŐSEBB MÓDJAI IS VANNAK, HA MÉG EDDIG BE NEM MUTATOTT UTASÍTÁSOKAT IS FELHASZNÁLUNK. EZ A PROGRAM AZONBAN ARRRA KÉSZÜLT, HOGY A "GO TO" ÉS AZ "IF THEN" UTASÍTÁSOKAT BEMUTASSA.

FOLYAMATÁBRA

EGY OLYAN PROGRAM FOLYAMATÁBRÁJA, AMELY EGYSZERŰ ALAKZATOK TERÜLETÉT SZÁMOLJA KI.



```

10 PRINT " * * * TERÜLET SZÁMOLÓ * * * "
20 PRINT
30 PRINT " ÍRD BE : TÉGLALAP, HÁROMSZÖG VAGY KÖR "
40 INPUT S$
50 REM
60 IF S$ = "TÉGLALAP" THEN 130 .....
70 IF S$ = "HÁROMSZÖG" THEN 180 .....
80 IF S$ = "KÖR" THEN 240 .....
90 PRINT S$, " AZT JELENTI, HOGY STOP "
100 GO TO 320 ← EGYENESEN AZ END-HEZ MEGY
110 REM
120 REM
130 PRINT " ÍRD BE A KÉT OLDAL HOSSZÁT "
140 INPUT B, D
150 LET Z = B * D
160 GO TO 280 -----
170 REM
180 PRINT " ÍRD BE A HÁROM OLDAL HOSSZÁT "
190 INPUT A, B, C
200 LET S = 0.5 * (A + B + C)
210 LET Z = SQR(S * (S - A) * (S - B) * (S - C))
220 GO TO 280 -----
230 REM
240 PRINT " ÍRD BE AZ ÁTMÉRŐT "
250 INPUT D
260 LET Z = 3.141592 * D ↑ 2/4 -----
270 REM
280 REM AZ ÁGAK ITT TALÁLKOZNAK ←
290 PRINT
300 PRINT S$, " TERÜLETE " ; Z
310 GO TO 20
320 END

```

SZÓKÖZ

```

RUN
* * * TERÜLET SZÁMOLÓ * * *

```

ÍRD BE : TÉGLALAP, HÁROMSZÖG VAGY KÖR

? TÉGLALAP

ÍRD BE A KÉT OLDAL HOSSZÁT

? 14.6, 10

TÉGLALAP TERÜLETE 146

ÍRD BE : TÉGLALAP, HÁROMSZÖG VAGY KÖR

? NEM

NEM AZT JELENTI, HOGY STOP

ON-GO TO

EZ AZ UTASÍTÁS AZ UTASÍTÁSOK VÉGREHAJTÁSÁNAK

SZOKÁSOS SORRENDJÉT TÖRI MEG α LÉNYEGÉBEN EGY TÖBBÁLLÁSÚ KAPCSOLÓ.

80 ON D+1 GO TO 100, 110, 120, 130

VÁLTOZÓ VAGY KIFEJEZÉS

EGYBE VAGY KÜLÖN ÍRVA

SORSZÁMOK LISTÁJA (NEM KIFEJEZÉSEKÉ) MINDEN MEGADOTT SORNAK LÉTEZNI KELL

A RENDSZER KISZÁMÍTJA A KIFEJEZÉS ÉRTÉKÉT ÉS:

- NÉMELY BASIC RENDSZER A LEGKÖZELEBBI EGÉSZRE KEREKÍTI AZ EREDMÉNYT.
- MÁSONK AZ EGÉSZ RÉSZÉT VESZIK.

HA AZ EREDMÉNY 1, AKKOR A RENDSZER A LISTA ELSŐ ÉRTÉKÉT HASZNÁLJA FEL (GO TO 100). HA AZ EREDMÉNY 2, AKKOR A RENDSZER A MÁSODIK SORSZÁMOT HASZNÁLJA FEL (GO TO 110) ÉS ÍGY TOVÁBB.

```
10 PRINT "ÍRJ BE EGY SZÁMOT 0 ÉS 3 KÖZÖTT"
15 PRINT "LEÁLLÁS: BREAK BILLENTYŰVEL"
30 INPUT D
40 IF D-INT(D)<>0 THEN 140
50 IF D<0 THEN 150
60 IF D>3 THEN 150
65 REM MOST D ÉRTÉKE 0, 1, 2 VAGY 3
70 PRINT "AZT ÍRTAD, HOGY";
75 REM
80 ON D+1 GO TO 100, 110, 120, 130
85 REM IDE NEM JUTHAT
100 PRINT "NULLA"
105 GO TO 10
110 PRINT "EGY"
115 GO TO 10
120 PRINT "KETTŐ"
125 GO TO 10
130 PRINT "HÁROM"
135 GO TO 10
140 PRINT "NEM EGÉSZ"
145 GO TO 10
150 PRINT "TARTOMÁNYON KÍVÜL"
160 GO TO 10
170 END
```

FOLYTATÁS A TÚLOLDALON

RUN

ÍRJ BE, EGY SZÁMJEGYET 0 ÉS 3 KÖZÖTT
LEÁLLÁS: BREAK BILLENTYŰVEL

? 0

AZT ÍRTAD HOGY NULLA

ÍRJ BE EGY SZÁMJEGYET 0 ÉS 3 KÖZÖTT
LEÁLLÁS: BREAK BILLENTYŰVEL

? 6

TARTOMÁNYON KÍVÜL

ÍRJ BE

EZ EGY HASZNOS UTASÍTÁS, DE LÉGY ÓVATOS A BASIC KÜLÖNBÖZŐ VÁLTOZATAINAK EGYEDI INTERPRETÁCIÓIVAL,

★ MERT NÉMELY BASIC RENDSZER A KIFEJEZÉS ÉRTÉKÉHEZ LEGKÖZELEBB ELSŐ EGÉSZ SZÁMOT VESZI, MÍG MÁSOK AZ ÉRTÉK EGÉSZ RÉSZÉT HASZNALJÁK, BIZTOSÍTSD, HOGY SAJÁT KIFEJEZÉSEIDNEK CSAK EGÉSZ EREDMÉNYEI LÉGYENEK (L. AZ ELŐZŐ OLDALI 40. SORT);

★ NÉMELY BASIC RENDSZER HIBÁT JELEZ KI ÉS LEÁLLÍTTJA A FUTTATÁST, HA A KIFEJEZÉS EREDMÉNYE TARTOMÁNYON KÍVÜL ESIK. MÁS RENDSZEREK A LISTA ELSŐ ELEMÉT HASZNALJÁK, HA AZ EREDMÉNY KISEBB, MINT 1; HA PEDIG AZ EREDMÉNY A LISTÁBAN SZEREPLŐ ELEMÉK SZÁMÁNÁL NAGYOBB, AKKOR AZ UTOLSÓ ELEMET VESZIK FIGYELEMBE. UGYANAKKOR MÁS BASIC RENDSZER AZ "ON"-T KÖVETŐ SORBA UGRIK (L. TÚLOLDALON A 85. SOR), HA AZ EREDMÉNY TARTOMÁNYON KÍVÜL VAN. ÍGY A "HORDOZHATÓSÁG" KEDVÉÉRT A TARTOMÁNYT ELLENŐRIZD MAGAD (L. A TÚLOLDALI 50. ÉS 60. SORT)!

AZ ALÁBBIKBAN LÁTHATÓK EZEN UTASÍTÁS MÁS (KEVÉSBÉ ISMERT) FORMÁI, AHOGY EZEKET A BASIC KÜLÖNBÖZŐ VÁLTOZATAIBAN HASZNALJÁK.

```
80 ON D+1 THEN 100, 110, 120, 130
```

```
80 GOTO 100, 110, 120, 130 ON D+1
```

```
80 GOTO D+1 OF 100, 110, 120, 130
```

AZ ALÁBBI UTASÍTÁS UNIVERZÁLISSÁ VÁLHAT A FENT LEÍRT "ON GOTO" MELLETT (AZ "ON GO SUB" JELENTÉSE VILÁGOS LESZ, AHOGY AZ 55. OLDALRA ÉRÜNK):

```
80 ON D+1 GO SUB 100, 110, 120, 130
```


FOR-NEXT

EZ AZ UTASÍTÁSPÁR AZT EREDMÉNYEZI, HOGY UTASÍTÁSOK EGY ADOTT

SOROZATA ÚJRA MEG ÚJRA VÉGREHAJTÁSRA KERÜL AZ UTASÍTÁSOK ILYEN SOROZATÁT CIKLUSNAK NEVEZZÜK.

```
10 LET M=3
20 READ X $
30 PRINT M ; X $
40 LET M=M-1
50 IF M <> 0 THEN 20
60 PRINT "KÖRTEFA"
70 DATA "TYÚK", "GALAMB", "FOGOLY"
80 END
```

RUN

```
3 TYÚK
2 GALAMB
1 FOGOLY
KÖRTEFA
```

CIKLUST ÚGY HOZHATSZ LÉTRE, HOGY FELÁLLÍTASZ EGY "SZÁMLALÓT"

(LET M=3)

MAJD FOKOZATOSAN 1-ET LEVONSZ BELOLE,

(LET M=M-1)

ELLENŐRZÖD A FENNMARADO ÉRTÉKET

(IF M <> 0),

ÉS "VISSZAFORDULSZ" EGÉSZEN ADDIG, MÍG A SZÁMLALÓ TARTALMA EL NEM FOGV.

UGYANEZT AZ EREDMÉNYT KÖNNYEBBEN IS KI LEHET HOZNI AZ ITT LÁTHATÓ MÓDON



```
10 FOR M = 3 TO 1 STEP -1
20 READ X $
30 PRINT M ; X $
40 NEXT M
50 PRINT "KÖRTEFA"
60 DATA "TYÚK", "GALAMB", "FOGOLY"
70 END
```

AZ UTASÍTÁS FORMÁJA A KÖVETKEZŐ:

BÁRMELY NUMERIKUS VÁLTOZÓ

IDE BÁRMILYEN BONYOLULT KIFEJEZÉS KERÜLHET

10 FOR M = 3 TO 1 STEP -1

EGYMÁS UTÁN VÉGREHAJTHATÓ UTASÍTÁSOK

NEXT M

UGYANAZ A VÁLTOZÓ, MINT AZ ELŐBB

HA A LÉPÉS (STEP) = +1, AKKOR A "STEP 1"-ET EGYSZERŰEN ELHAGYHATOD:

100 FOR N = P TO Q

EGYIK CIKLUST A MÁSIKBA ÁGYAZHATOD, AZ EGYMÁSBA

ÁGYAZÁS MÉLYSÉGÉNEK MINDIG VAN FELSŐ HATÁRA. JELLEGZETES HATÁR A 10, AMELY ELEGENDŐ MÉLYSÉG MÉG A LEGÖSSZETETTEBB PROGRAMOK ESETÉN IS.

```

10 PRINT " RÉGI DICCSŐSÉG "
20 FOR R = 1 TO 7
30 LET P$ = " "
40 IF R = 1 THEN GO TO 130
50 LET P$ = "*"
60 FOR C = 1 TO 8
70 PRINT P$;
80 NEXT C
90 IF R-2 * INT(R/2) = 0 THEN GO TO 130
100 FOR C = 1 TO 12
110 PRINT " = " ;
120 NEXT C
130 PRINT
140 NEXT R
150 FOR R = 1 TO 3
160 PRINT
170 FOR C = 1 TO 20
180 PRINT " = " ;
190 NEXT C
200 PRINT
210 NEXT R
220 END
    
```



NEM SZABAD, HOGY A CIKLUSOK EGYMÁSBA FONÓDJANAK.

```

10 FOR I =
20 FOR K =
60 NEXT I
70 NEXT K
    
```



KI IS UGORHATSZ A HUROKBÓL:

```

10 REM EGY ASSZONYNAK KILENC LÁNYA
20 DATA "ERZSI", "ZSUZSI", "SÁRI"
30 DATA "MARISKA", "ROZÁLI", "ELLA"
40 DATA "BELLA", "JUCI", "KAROLINA"
50 PRINT " (RD BE A NEVED "
60 INPUT N$
70 FOR L = 1 TO 9
80 READ M$
90 IF M$ = N$ THEN GO TO 130
100 NEXT L
110 PRINT N$ ; "TE NEM VAGY A LÁNYA"
120 GO TO 140
130 PRINT N$ ; L ; "-IK LÁNYA VAGY"
140 END
    
```

EZT A LISTÁBAN VALÓ KERESÉSNEK HÍVJÁK A MEGFELELŐ PÁR MEGTALÁLÁSA CÉLJÁBÓL.

A CIKLUSVÁLTOZÓ (FENTIEKBEN L) MEGTARTJA ÉRTÉKÉT, HA MÉG A CIKLUS LEFUTÁSA ELŐTT KIUGRASZ BELŐLE. DE HA AZ ALJÁN ESEL KI (EBBEN AZ ESETBEN A 110. SORNÁL), AKKOR ~~NE~~ TÉTELEZZ FEL SEMMIT A CIKLUSVÁLTOZÓ ÉRTÉKÉRŐL: A FENTI ESETBEN EZ LEHET 8, DE LEHET HOGY NEM ANNYI (ENNEK MAGYARÁZATA A TULOLDALON KÖVETKEZIK). EBBEN A BASIC RENDSZEREK ELTÉRNEK EGYMÁSTÓL.

CIKLUSOK (FOLYTATÁS)

SOHASE UGORJ A CIKLUS BELSEJÉBE :

```
120 IF A > 2 THEN 210
```

```
200 FOR I = 1 TO 6
210 PRINT "*" ;
220 NEXT I
230 LET A = A - 1
```

A BASIC RENDSZER NEM HAJTANÁ VÉGRE A "NEXT" UTASÍTÁST, HA MÁR ELŐTTE NEM TALÁLKOZOTT VOLNA A HOZZÁ TARTOZÓ "FOR"-RAL.

A LEGTÖBB BASIC RENDSZERBEN AZONBAN KIUGORHATSZ A HUROKBÓL MAJD ÚJRA VISSZA (EZ TERMÉSZETESEN NEM NEVEZHETŐ SZÉP MEGOLDÁSNAK)

```
150 FOR I = 1 TO 6
160 READ M$
170 IF M$ = N$ THEN 300
180 NEXT I
```

```
300 PRINT M$
310 GO TO 180
```

EZT A SZAKMAI ZSARGON KITERJESZTETT TARTOMÁNYÚ CIKLUSNAK NEVEZI.

ÁLTALÁBAN A CIKLUSBA A FOR UTASÍTÁSON KERESZTÜL LÉPHETSZ BE ÉS:

- VAGY A NEXT-EN KERESZTÜL LÉPSZ KI,
- VAGY KIUGRASZ ÉS KINT IS MARADSZ.

HA KITERJESZTETT TARTOMÁNYÚ CIKLUSRA VAN SZÜKSÉGED, EZT A "GO SUB" UTASÍTÁSSAL IS ELÉRHETED, AMELYET AZ 52. OLDALON ÍRUNK LE.

HA A PROGRAMNAK BIZONYOS KÖRÜLMÉNYEK KÖZÖTT EGYALTALÁN NEM KELL A CIKLUST VÉGREHAJTANIA, AKKOR BIZTOSABB, HA ELLENŐRIZZÜK EZEKET A FELTÉTELEKET, ÉS HA FENNÁLLNAK, JOBB A CIKLUST TELJESEN ELKERÜLNI.

```
10 FOR I = 1 TO 5
20 READ C
30 PRINT C ;
40 IF C <= 0 THEN 80
50 FOR R = 1 TO C
60 PRINT "x" ;
70 NEXT R
80 PRINT
90 NEXT I
100 DATA 3, 0, -1, 4, 1
110 END
```

KERÜLDEL A CIKLUST, AMELY 1-TŐL C-IG FUT, AMIKOR C <= 0

```
RUN
3 ***
0
-1
4 ****
1 *
```

KÉSŐBB MEGLÁTJUK, MIÉRT VOLTUNK ÓVATOSAK.

A

KONCEPCIÓ EGYSZERŰNEK TŰNIK, MÉGIS VAN NÉHÁNY REJTETT VESZÉLY A CIKLUSOKKAL KAPCSOLATBAN: KÜLÖNBÖZŐ BASIC RENDSZEREK KÜLÖNFÉLEKÉPPEN KEZELIK ÖKET. AZ ALÁBBIKBAN A SZABVÁNYOSNAK NEVEZHETŐ BASIC RENDSZEREKBE SZOKÁSOS INTERPRETÁCIÓT MUTATJUK BE.

TEKINTSÜK A KÖVETKEZŐ CIKLUST:


10 FOR V = A TO Z STEP S

TÖRZS

40 NEXT V

AHOL A, Z ÉS S BONYOLULT KIFEJEZÉSEK IS LEHETNEK ∞

AZ INTERPRETÁCIÓ 3 VÁLTOZÓT ÉRINT, AMELYEKET A BASIC RENDSZER HASZNÁLHAT, DE AMELYEKRE TE, A FELHASZNÁLÓ NEM HIVATKOZHATSZ ÉS MEG SEM VÁLTOZTATHATOD. EZEK LEGYENEK:



EZ A LOGIKA ÉRVÉNYES MÉG AKKOR IS, HA MEGVÁLTOZTATJUK A CIKLUS TÖRZSÉBEN A, Z ÉS S ÉRTÉKÉT; EZ NEM FOGJA A CIKLUS FUTÁSAINAK SZÁMÁT BEFOLYÁSOLNI. DE SOHASE VÁLTOZTASD MEG V ÉRTÉKÉT A CIKLUS TÖRZSÉN BELÜL! TUDVÁN, HOGY SOK BASIC RENDSZER NEM EZT A LOGIKÁT ALKALMAZZA, NE VÁLTOZTASS MEG SEMMI OLYANT A TÖRZSÖN BELÜL, AMI A V, A, Z VAGY S ÉRTÉKÉT MÓDOSÍTANA! HASZNÁLJ MINÉL EGYSZERŰBB VEZÉRLÉSI STRUKTÚRÁT!

GO SUB-RETURN

SOK PROGRAMBAN AZ UTASÍTÁSOK EGY BIZONYOS SZOROZATA TÖBB ALKALOMMAL FORDUL ELŐ. ILYEN ESETEKBE NEM KELL AZ UTASÍTÁSSOROZATOT TÖBBSZOR MEGISMÉTELNE, HANEM EGY SZUBRUTINNA FORMÁLOD AZT, AMIT A PROGRAM BÄRMELYIK SORÁBÓL ELÉRHETSZ (GO SUB), MAJD A VÉGREHAJTÁS UTÁN VISSZATÉRHETSZ ODA (RETURN), AHONNAN JÖTTÉL.

A 49. OLDALON LÉVŐ "RÉGI DICSŐSÉG" NEVŰ PROGRAMNAK VAN OLVAN RÉSZLETE, AMELY HÄROMSZOR FORDUL ELŐ:

FOR C = 1 TO valamedig
PRINT valamit
NEXT C

AMELYET SZUBRUTINNA FOGLALHATNÄNK ÖSSZE:

```
300 REM SZUBRUTIN B ÉS P$ SEGÍTSÉGEVEL  
310 FOR I = 1 TO B  
320 PRINT P$;  
330 NEXT I  
340 RETURN
```

ÉS AZ "RÉGI DISCŐSÉG" PROGRAMOT ÍGY ÖNTHETNÉNK ÚJ FORMÁBA:

```
10 PRINT "RÉGI DISCŐSÉG GO SUB ÉS RETURN UTASÍTÁSOKKAL"  
20 FOR R = 1 TO 7  
30 LET P$ = " "  
40 IF R = 1 THEN 70  
50 LET P$ = "*"   
70 LET B = 8  
80 GO SUB 300  
90 IF R - 2 * INT (R / 2) = 0 THEN 130  
100 LET B = 12  
110 LET P$ = "=="  
120 GO SUB 300  
130 PRINT  
140 NEXT R  
150 FOR R = 1 TO 3  
160 PRINT  
170 LET B = 20  
180 GO SUB 300  
200 PRINT  
210 NEXT R  
220 GO TO 900  
900 END
```

2 szóköz

szóköz

300. SORBA MEGY - 90. SORBA TÉR VISSZA.


130. SORBA TÉR VISSZA

200. SORBA TÉR VISSZA

KIHAGYJA A SZUBRUTINT

VIT A SZUBRUTIN HELYE

A Z UTASÍTÁS FORMÁJA A KÖVETKEZŐ:



A "GO SUB" BÁRHOL SZEREPELHET A PROGRAMBAN; A HÍVOTT SZUBRUTIN IS BÁRHOL LEHET (NEM FELTÉTLENÜL CSAK MAGASABB SORSZÁMÚ SOROKBAN). ÓVATOSAN KELL SZERVEZNI A DOLGOKAT, HOGY A TELJES ÖSSZEGABALYODÁST MEGELŐZZÜK.

NÉHÁNY BASIC RENDSZER MEGENGED FORMÁLIS PARAMÉTEREKET A SZUBRUTINBAN, ÉS EZEK A FÜGGVÉNYEK FORMÁLIS PARAMÉTEREIHEZ HASONLÓAN VISELKEDNEK (L. 26. OLD.). A RÉSZLETEK AZONBAN BASIC-RŐL BASIC-RE HALADVA IGEN ELTÉRŐEK LEHETNEK, EZÉRT NE HASZNÁLJ FORMÁLIS PARAMÉTEREKET, HA "HORDOZHATÓ" PROGRAMOT AKARSZ ÍRNI!

VIGYÁZZ, HOGY A SZUBRUTINBA "VÉLETLENÜL" (AZAZ "GO SUB" VÉGREHAJTÁSA NÉLKÜL) NE KERÜLJ BELE! FIGYELJÜK MEG A TÚLOLDALON SZEREPLŐ 220. SORT; HA EZT KIHAGYNÁNK, AKKOR KÖVETKEZNE BE EGY ILYEN HÍVÁS NÉLKÜLI VÉGREHAJTÁS.

GO SUB (FOLYTATÁSA)

A PROGRAMOZÁSBAN KEZDŐK EZT A 2 OLDALT A KÖNYV EL-SŐ ELOLVASÁSÁKOR KIHAGYHATJAK.

A "GO SUB" BÁRHOL MEGJELENHET A PROGRAMBAN; UGYANÍGY A HÍVOTT SZUBRUTIN IS BÁRHOL LEHET. A BASIC-NEK NINCSEN BIZTOS MÓDSZERE ARRRA, HOGY A MEGFELELŐ "RETURN" UTASÍTÁST AZZAL A "GO SUB"-BAL TÁRSÍTSA, AMELYIKHEZ TE KIVÁNOD RENDELNI. (NEM OLYAN, MINT A "FOR V=" , AMELYET A "NEXT V" KÖVET, AHOL V VILÁGOSSÁ TESZI AZ ÖSSZEFÜGGÉST.) ÍGY AZ ALÁBBIKBAN LEÍRJUK, HOGY KAPCSOLJA ÖSSZE A RENDSZER A "GO SUB" "RETURN" PÁROKAT.

NÉMELYIK BASIC RENDSZER UN. VEREMTECHNIKÁVAL KEZELI A "GO SUB" UTASÍTÁST. A VEREM A KÖVETKEZŐKÉPPEN MŰKÖDIK :



AMIKOR A RENDSZER TALÁLKOZIK EGY "GO SUB" UTASÍTÁSSAL, AZONNAL MEGJEGYZI A "GO SUB"-OT KÖZVETLENÜL KÖVETŐ UTASÍTÁS SORSZÁMÁT, ÉS EZT A SZÁMOT A VEREM LEGTETEJÉRE TESZI, A VEZÉRLÉS PEDIG A "GO SUB" UTÁN MEGNEVEZETT SORNAK ADÓDIK ÁT.




AMIKOR A RENDSZER A "RETURN" UTASÍTÁSSAL TALÁLKOZIK, EGYSZERŰEN MEGNÉZI AZT A SZÁMOT, AMELY AKKOR A VEREM LEGTETEJÉN VAN, EGYENEST ABBA A SORBA MEGY, AMELYNEK EZ A SZÁMA, MAJD A VEREM TETEJÉRŐL TÖRLI EZT A SZÁMOT.

LOGIKA AZT JELENTI, HOGY HA A BASIC A LEGELSŐ "GO SUB" ELŐTT EGY "RETURN"-NAL TALÁLKOZIK, AKKOR MÉG ÜRES A VEREM, ÍGY NINCSEN HOVA MENNI. EZ A HIBA GYAKRAN ANNAK A KÖVETKEZMÉNYE, HOGY EGY SZUBRUTINT "GOSUB" HÍVÁS NÉLKÜL HAJTOTTUNK VÉGRE. MÁSFELŐL A RENDSZER SORRA TALÁLKOZHAT A "GO SUB" UTASÍTÁSOKKAL, ÉS HA NEM TALÁLKOZIK E-LÉG "RETURN" UTASÍTÁSSAL, AKKOR A VEREM TELJESEN MEGTELEHET (A VE-REM KAPACITÁSA RENDSZERENKÉNT VÁLTOZHAT, A 10 TÍPIKUS ÉRTÉKNEK VEHEŐ).

JÓLLEHET NEM MINDEN BASIC RENDSZER HASZNÁL PONTOSAN ILYEN MECHANIZ-MUST A "GO SUB" KEZELÉSEKOR, MÉGIS SEGÍT, HA ÍGY GONDOLOD ÁT, AMIKOR A MEGÍRANDÓ PROGRAMOD LOGIKÁJÁT ELLENŐRZÖD, VAGY AMIKOR A PROGRAMBAN HIBÁT KERESSEL, ILL. MÁS PROGRAMJÁNAK SZERKEZETÉT PRÓBALOD KÖVETNI.

AZ, HOGY A TE BASIC-ED A "GO SUB" OT VEREM HASZNÁLATÁVAL KEZELI-E, CSAK AKKOR VÁLIK FONTOSSÁ, HA OLYAN SZUBRUTINT ÍRSZ, AMELY SAJÁT MA-GÁT HÍVJA. EZT REKURZIÓNAK NEVEZIK, ÉS AZ ELŐBBIEKBEN KÖRVONALAZOTT VEREM-TECHNIKÁVAL OLDHATÓ MEG.



EZ A VEREMTÁR AZT MUTATJA, HOGY A LEGUTOLSÓ VÉGREHAJTANDÓ "GO SUB" UGYANAZ, MINT AZ ELŐZŐ. EZ AZT JELENTI, HOGY A SZUBRUTIN SAJÁT MAGÁT KÖZVETLENÜL HÍVJA. ELŐZŐLEG, A SZUBRUTIN EGY MÁSIKAT HÍVOTT, AMELY AZTÁN AZ ÁBRÁN A ★ - GAL MEGJELÖLT ELSŐT HÍVTA. IGY A SZUBRUTIN INDIREKT MÓDON HÍVTA SAJÁT MAGÁT.

KIPRÓBÁLHATOD TE IS, HOGY VAJON BASIC VÁLTOZATOD MEGENGEDI-E, HOGY A SZUBRUTINOK SAJÁT MAGUKAT HÍVJÁK. PRÓBÁLD KI A KÖVETKEZŐ KIS PROGRAMOT, AMELY EUKLEIDÉSZ MÓDSZERÉT ALKALMAZVA KÉT SZÁM LEGNAGYOBB KÖZÖS OSZTÓJÁT SZÁMOLJA KI!

```

10 PRINT "ÍRJ BE 2 POZITÍV EGÉSZET"
20 INPUT M, N
30 GO SUB 70
40 PRINT "L. K. O. -JUK"; P
50 GO TO 140
60 REM A PROGRAM VEGE
70 REM A SZUBRUTIN KEZDETE
80 LET P=N
90 LET N=M-N*INT(M/N)
100 LET M=P
110 IF N=0 THEN 130
120 GO SUB 70
130 RETURN
140 END

```

SZUBRUTINT HÍV!



MEGÁLL (NEM ESİK ÁT A SZUBRUTINON)

KILÉP, HA N=0

SAJÁT MAGÁT HÍVJA HA, N ≠ 0


HA A BASIC RENDSZERREL A 120. SOR VÁLTOZTATÁSA NÉLKÜL ELFOGADTAD A PROGRAMOT, AKKOR MEGPRÓBÁLHATOD MEGTALÁJNI A VEREMMÉLYSÉGÉNEK KORLÁTJÁT. PL. HA AZT ÍROD, HOGY 85, 204, A PROGRAM KIÍRJA AZ EREDMÉNYT, AMI 17. AZONBAN, HA 85, 289-ET GÉPELSZ, AKKOR A PROGRAM A MEGENGEDHETŐ VEREMMÉLYSÉGET FELTEHETŐLEG TÚLLÉPI, BÁR A VÁLASZ TOVÁBBRA IS 17.

PRÓBÁLD "SZÁMÍTÓGÉPESDIT" JÁTSZANI CERUZÁT, PAPIRT ÉS EGY ZSEBSZÁMOLÓGÉPET HASZNÁLVA. EZ FÉNYT DERÍT EUKLEIDÉSZ MÓDSZERÉRE, ÉS AZT IS MEGMUTATJA, HOGY A VEREM HOGYAN TELIK MEG ÉS HOGYAN ÜRÜL KI.



TALÁLD KI KI

∞ JÁTÉKSZABÁLYOK ∞



AZ ITT BEMUTATOTT PROGRAM A "MOO" JÁTÉK-
RA ÉS A "GO SUB" ILLUSZTRÁLÁSÁRA KÉ-
SZÜLT.

KEZDÉSKÉNT DOBJUNK 2 DOBÓKOCKÁVAL! A
DOBÁSOK EREDMÉNYÉT VIGYÜK BE A GÉPBE A
BILLENTYŰZETEN! (EZ BIZTOSÍTJA, HOGY MIN-
DEN ALKALOMMAL MÁS LEGYEN A JÁTÉK.)

A SZÁMÍTÓGÉP KIVÁLASZT EGY 4 JEGYŰ SZÁMOT,
AMELYBEN ∞ NINCIS KÉT EGYFORMA SZÁMJEGY
∞ (MEGJEGYZÉS: AZ ELSŐ SZÁMJEGY LEHET 0 IS)

TALÁLD KI, HOGY MILYEN SZÁMOT VÁLASZTOTT A
SZÁMÍTÓGÉP, ÉS ÍRD BE A TIPPED, AMIKOR A GÉP ER-
RE KÉR.

A SZÁMÍTÓGÉP, KIJELZI, HOGY SZÁMJEGYEIDBŐL
HANY VOLT JÓ ∞ DE HIBÁS HELYEN ∞, EZEKET
FEHÉRNEK NEVEZI.

A SZÁMÍTÓGÉP KIJELZI AZT IS, HOGY SZÁMJEGYE-
IDBŐL HANY VOLT NEMCSAK HOGY JÓ, DE ∞
JÓ HELYEN IS ∞ EZEKET FEKETÉNEK NEVEZI.

MIUTÁN MEGMONDTA, HOGY HANY FEKETÉT ÉS HANY
FEHERET SZEREZTÉL, A SZÁMÍTÓGÉP ÚJBÓL TIPPELÉS-
RE SZOLÍT FEL, EGÉSZEN ADDIG MÍG 4 FEKETED
LESZ ÉS EGYETLEN EGY FEHÉRED SEM.



A PROGRAM A KÖVETKEZŐ 8

```
5 REM TALÁLD KI
10 PRINT "DOBJ KÉT KOCKÁVAL"
15 INPUT I, J
20 FOR K=0 TO 6*I+J-7
25 LET X=RND (csettes' RND(0))
30 NEXT K
35 REM ÉS MOST MAGA A PROGRAM
```

† A JÁTÉK MAGYARORSZÁGON MASTERMIND NÉVEN ISMERT.

```

110 LET X = INT ( 10000 * RND )
120 LET Y = X
130 GO SUB 1000
140 IF C > 0 THEN 110
150 REM AZ ÖNMAGÁVAL VETT ÖSSZEHASONLÍTÁSNÁL FEHÉR NEM VOLT
160 PRINT "VÁLASZTOTTAM EGY NÉGYJEGYŰ SZÁMOT"
170 LET T = 0
180 LET T = T + 1
190 PRINT "MI A TIPPED"
200 INPUT X
210 GO SUB 1000
220 PRINT B; "FEKETE ÉS"; C; "FEHÉR"
230 IF B < 4 THEN 180
240 PRINT "EHHEZ"; T; "PRÓBA KELLETT"
250 PRINT
260 GO TO 110
270 REM A BREAK BILLENTYŰVEL ÁLLÍTHATÓ LE A JÁTÉK
1000 REM SZUBROUTIN
1010 LET B = 0
1020 LET C = 0
1030 FOR I = 1 TO 4
1040 LET K = 10 ↑ I
1050 LET L = INT ( 10 * ( X - K * INT ( X / K ) + . 1 ) / K )
1060 FOR J = 1 TO 4
1070 LET M = 10 ↑ J
1075 LET P = INT ( 10 * ( Y - M * INT ( Y / M ) + . 1 ) / M )
1080 IF L <> P THEN 1130
1090 LET C = C + 1
1100 IF I <> J THEN 1130
1110 LET C = C - 1
1120 LET B = B + 1
1130 NEXT J
1140 NEXT I
1150 RETURN
1160 REM
1200 END

```

ELLENŐRZI HOGY
NINCS-E KÉT EGY-
FORMA SZÁMJEGY

SZÁMOLJA A
PRÓBÁKAT

ÖSSZEHASONLÍTTJA
A TIPPET A SZÁMMAL

új játék kezdődik

FEKETE

FEHÉR

VEDD A TIPP EGYES
JEGYEIT EGYMÁS UTÁN

HASONLÍTSD ÖSSZE A SZÁM MINDEN
SZÁMJEGYÉT A TIPP I-EDIK SZÁMJEGYÉVEL

A 0.1 A KÉREKÍTÉSI
HIBÁKAT ELLENSÚLYOZZA
 $\text{INT}(3.99999) = 3$, AZ
 $\text{INT}(3.99999 + .1)$ VISZONT
4

```

RUN
DOBJ KÉT KOCKÁVAL
? 6, 3
VÁLASZTOTTAM EGY NÉGYJEGYŰ SZÁMOT
MI A TIPPED
? 1234
0 FEKETE ÉS 4 FEHÉR
MI A TIPPED
? 5678
EHHEZ 6 PRÓBA KELLETT

```

ÉS ÍGY TOVÁBB

VÉGÜL



TÖMBÖK

TÖMBÖK

EGYSZERŰ NUMERIKUS ÉS SZÖVEGES VÁLTOZÓKON KÍVÜL TÖBB EZER, TÖMBÖKBE RENDEZETT VÁLTOZÓT IS HASZNÁLHATSZ.

AZ ALÁBBIKBAN EGY $A()$ -NAK NEVEZETT EGYDIMENZIÓS NUMERIKUS TÖMBRE LÁTHATUNK PÉLDÁT.

A(1)	3.56
A(2)	7.12
A(3)	10.68
A(4)	

$A()$ -NAK ELEMEI VANNAK, MINDEGYIKÜK UGYANÚGY TUD EGY SZÁMOT TÁROLNI, MINT BÁRMELY EGYSZERŰ VÁLTOZÓ.

100	LET	$A(1) = 3.56$
110	LET	$A(2) = 7.12$
120	LET	$A(3) = A(1) + A(2)$

AZ ELEMET EGYINDEXŰ VÁLTOZÓNAK IS HÍVHATJUK (VAGY EGYSZERŰEN INDEXELT VÁLTOZÓNAK, HA A SZÖVEGKÖRNYEZET EZT MEGENGEDI).

AZ EGYDIMENZIÓS TÖMBÖT VEKTORNAK, OSZLOPVEKTOR-NAK VAGY OSZLOP-MÁTRIXNAK IS NEVEZHETJÜK. $A()$ -T ÚGY IS FELFOGHATOD, MINT EGY SORT.

A(1)	A(2)	A(3)	A(4)
3.56	7.12	10.68	

ÉS SORVEKTORNAK VAGY SORMÁTRIXNAK IS NEVEZHETED.

AZ ALÁBBIKBAN EGY $B(,)$ -NEK NEVEZETT KÉTDIMENZIÓS NUMERIKUS TÖMBRE LÁTHATUNK PÉLDÁT, AMELYNEK 4 SORA ÉS 3 OSZLOPA VAN.

	1)	2)	3)
B(1,	8.92		
B(2,		17.84	-8.92
B(3,			
B(4,			

UGYANÚGY, MINT AZ $A()$ -NÁL $B(,)$ -NEK OLYAN ELEMEI VANNAK, AMELYEKNEK MIND EGYIKE UGYANÚGY TUD EGY SZÁMOT TÁROLNI, MINT EGY EGYSZERŰ NUMERIKUS VÁLTOZÓ.

130	LET	$B(1,1) = 8.92$
140	LET	$B(2,2) = 2 * B(1,1)$
150	LET	$B(2,3) = -B(1,1)$

AZ ILYEN KÉTDIMENZIÓS TÖMB ELEMEIT KÉTINDEXŰ VÁLTOZÓNAK NEVEZHETJÜK (VAGY EGYSZERŰEN INDEXELT VÁLTOZÓNAK, HA A SZÖVEGKÖRNYEZET EZT MEGENGEDI).

A KÉTDIMENZIÓS TÖMBÖT TÉGLALAPTÖMBNEK ÉS TÉGLALAPMÁTRIXNAK IS NEVEZHETJÜK. (HA A SOROK ÉS OSZLOPOK SZÁMA MEGEGYEZIK, AKKOR TÉGLALAP HELYETT NÉGYZETESNEK IS HÍVHATJUK.)

NÉHÁNY BASIC RENDSZER MEGENGED HÁROMDIMENZIÓS TÖMBÖKET IS.

AZ ALÁBBIKBAN EGY $T\$()$ -NAK NEVEZETT SZÖVEGES TÖMBRE LÁTHATUNK PÉLDÁT. EZ A TÖMB EGYDIMENZIÓS. (SZÁMOS BASIC RENDSZER NEM ENGED MEG KÉTDIMENZIÓS SZÖVEGES TÖMBÖT, ÍGY A HORDOZHATÓSÁG KEDVÉÉRT JOBB EZEKET MELLOZNI.)

$T\$(1)$	HO
$T\$(2)$	
$T\$(3)$	KEGRAPHA
$T\$(4)$	
$T\$(5)$	KEGRAPHA

A $T\$()$ -NEK OLYAN ELEMEI VANNAK, AMELYEK KÖZÜL MINDEGYIK UGYANÚGY (ÉS ÁLTALÁBAN UGYANOLYAN HOSSZÚSÁGBAN) TUD SZÖVEGET TÁROLNI, MINT BARMELY EGYSZERŰ SZÖVEGES VÁLTOZÓ.

160	LET	$T\$(1) = "HO"$
170	LET	$T\$(3) = "KEGRAPHA"$
180	LET	$T\$(5) = T\(3)

EGY PROGRAMBAN LEGFELJEBB 26 NUMERIKUS TÖMB SZEREPELHET $A(), B(), C(), D(), \dots, Z()$

AMELYEK KÖZÜL LEHETNEK EGYDIMENZIÓSOK ÉS KÉTDIMENZIÓSOK. EGY BETŰ EGYSZERRE NEM JELÖLHET KÉTFÉLE TÖMBÖT.

EGY PROGRAMBAN LEGFELJEBB 26 SZÖVEGES TÖMB SZEREPELHET $A\$(), B\$(), C\$(), D\$(), \dots, Z\$()$

ÉS EMELLETT MIND A 286 EGYSZERŰ NUMERIKUS VÁLTOZÓT, ILL. MIND A 26 EGYSZERŰ SZÖVEGES VÁLTOZÓT IS FELHASZNÁLHATJA, AMELYEK KÜLÖNBÖZNI FOGNAK A TÖMBÖK ELEMEITŐL. $B(2,2)$ -NEK SEMMI KÖZE A B-HEZ, A $T\$(5)$ -NEK SEMMI KÖZE A $T\$()$ -HOZ.


HATÁROZD MEG A TÚLOLDALON ELMAGYARÁZOTT MÓDON AZOKNAK A TÖMBÖKNEK A MÉRETÉT ÉS ALAKJÁT "DIM" (A DIMENZIO ROVIDÍTÉSE) UTASÍTÁSSAL, AMELYEKET HASZNÁLNI AKARSZ. A FENT SZEREPLŐ TÖMBÖKET A KÖVETKEZŐKÉPPEN HATÁROZHATNÁNK MEG:

10	REM	PROGRAM TÖMBÖKKEL
20	DIM	$A(4), B(4,3), T\$(5)$

DIM

A DIM A DIMENZIÓ RÖVIDÍTÉSE
EZ AZ UTASÍTÁS A PROGRAM TÖMBJEINEK
MÉRETEIT ÍRJA LE.

A TÖMBÖK ALÁBBI CSOPORTJÁT :



EGYETLEN "DIM" UTASÍTÁSBAN IS DEKLARÁLHATJUK :

```

10 REM PROGRAM EGY DIM UTASÍTÁSSAL
20 DIM A$(4), A(1), C(2), B(3,3)

```

DE ÍRHATUNK TÖBB "DIM" UTASÍTÁST IS :

```

10 REM PROGRAM TÖBB DIM UTASÍTÁSSAL
20 DIM C(2), A(1)
30 DIM B(3,3)
40 DIM A$(4)

```

CSAK EGÉSZ
KONSTANS: KIFEJEZÉS
NEM LEHET

SOK BASIC RENDSZER MEGENGEDI, HOGY KIHAGYD A "DIM" UTASÍTÁST, AMI-
KOR A MÉRET 10 VAGY ENNÉL KEVESEBB LESZ, DE EZT NE HASZNÁLD KI!
HA EGY RÉGEBBEN MEGÍRT PROGRAMHOZ TÉRSZ VISSZA, VAGY HA MÁS PROG-
RAMJÁT KELL MEGÉRTENED, JÓ TUDNI MINDJÁRT AZ ELEJÉN, HOGY AZ MI-
LYEN TÖMBÖKET HASZNÁL. MINDENESETRE VANNAK OLYAN BASIC REND-
SZEREK, AMELYEK MEGKÍVÁNJÁK VALAMENNYI TÖMB DEKLARÁCIÓJÁT,
BÁRMILYEN KICSIK IS LEGYENEK A TÖMBÖK.

MINDEN "DIM" UTASÍTÁSNAK A PROGRAM KEZDETÉHEZ KÖZEL KELL LEN-
NIE MÉG AZ ÉRINTETT INDEXELT VÁLTOZÓ ELSŐ HASZNÁLATA ELŐTT ∞ ÉS
BIZTOSÍTANOD KELL, HOGY A "DIM" UTASÍTÁSOK TÉNYLEGESEN VÉGREHAJTÁS-
RA KERÜLJENEK. (ÚGY, AHOGY A 26. OLDAL "DEF" UTASÍTÁSAI ESETÉBEN.)

```

50 LET P(1) = 4.5
60 DIM P(16)

```

```

10 GO TO 30
20 DIM A(40,40), B(1000)
30

```

SOK BASIC RENDSZERBEN EZEK NEM SZÁMÍTANAK, DE NÉMELYBEN IGEN ∞
ÍGY A "HORDOZHATÓSÁG" KEDVÉÉRT FOGADD EL EZEKET A JELENTÉKTELEN MEG-
KÖTÉSEKET!

MINDIG HIBA EGY TÖMBÖT EGYNÉL
TOBBSZÖR DEKLARÁLNI.

```

10 DIM A(100,20), B(60)
20 DIM B(60), C(500)

```

A BASIC-BEN A TÖMBÖK SORAIT ÉS OSZLOPAIT HAGYOMÁNYOSAN NULLÁTÓL SZÁMOZZAK ÉS NEM EGYTŐL.

	0)	1)	2)	3)
P(0,				
P(1,				
P(2,				

Q(0)	
Q(1)	
Q(2)	

A "DIM" UTASÍTÁS PEDIG, A LEGNAGYOBB INDEXEKRE VONATKOZIK ÉS NEM SOROK ÉS OSZLOPOK SZÁMÁRA :

```
10 REM HAGYOMÁNYOS BASIC
20 DIM P(2,3), Q(2)
```

UGYANAKKOR MANAPSÁG LÉTEZNEK A BASIC-NEK OLYAN VÁLTOZATAI, AMELYEK EGYTŐL SZÁMOZZAK (MINT A TÚLOLDALI ILLUSZTRÁCIÓ ESETÉBEN), ÍGY BIZTONSÁGOSABB, HA FELTÉTELEZED, HOGY VÁLTOZATOD EGYTŐL SZÁMOZ. HA VALÓBAN NULLÁTÓL SZÁMOZ, EZ AZT JELENTI, HOGY EGY KIS HELYET ELPAZAROLSZ A SZÁMÍTÓGÉP TÁRÁBAN, DE A PROGRAMOD "HORDOZHATÓ" LESZ.

NÉMELYIK BASIC MINDKÉT MÓDOT ISMERI, ÉS AZ ALÁBBI UTASÍTÁSOKKAL

VAGY

```
10 BASE 0
10 BASE 1
```

LEHET DEKLARÁLNI, HOGY NULLÁTÓL VAGY EGYTŐL KÍVÁNSZ SZÁMOZNI.

A "DIM" UTASÍTÁS HATÁSÁRA A RENDSZER A SZÁMÍTÓGÉP TÁRÁBAN MINDEN TÖMBÖD SZÁMÁRA HELYET FOGLAL EL, DE EZ NEM JELENTI AZT, HOGY TÖRLŐDIK AZ ELŐZŐ FUTTATÁSBÓL HÁTRAMARADT INFORMÁCIÓ. NÉMELYIK BASIC RENDSZER AZ INDEXES VÁLTOZÓKNAK, NULLA KEZDŐ ÉRTÉKET AD (SZÓKÖZT SZÖVEGES VÁLTOZÓ ESETÉN), MÁSOK "HATÁROZATLANKÉNT" JELÖLIK MEG (MINT AHOGY MÁR EGYSZERŰ VÁLTOZÓK ESETÉBEN A 11. OLDALON MEGTÁRGYALTUK), ISMÉT MÁSOK VISZONT TELIRAKJÁK "SZEMÉTTTEL". ÍGY, HA A PROGRAMOD EGY TÖMBBŐL OLVAS :

```
100 LET A=B(2,2)
```

ARRA SZÁMÍTVÁN, HOGY NULLÁT TALÁL OTT, AMIKOR MÉG NEM KERÜLT ODA SEMMI, BIZTOSÍTSD, HOGY A PROGRAMOD ELŐSZÖR TÖRÖLJE A TÖMBÖT! ENNEK LEGEGYSZERŰBB MÓDJÁ A 86. OLDALON LÁTHATÓ, AHONNAN EGY PÉLDÁT SZERETNÉK ELŐREBOCSÁTANI :

```
70 MAT B=ZER
```

A B(,) TÖMB ÖSSZES ELEMÉNEK 0 ÉRTÉKET AD

INDEXEK

A KÖVETKEZŐ TÖMB :

```
10 DIM B(4,3)
```

	1)	2)	3)
B(1,			
B(2,			
B(3,			
B(4,			

12 INDEXES VÁLTOZÓ TÖMBJE, AMELYEK MINDEGYIKÉT ÚGY LEHET KEZELNI, MINTHA EGYSZERŰ NUMERIKUS VÁLTOZÓ LENNE :

```
100 LET B(1,2) = B(2,2)↑2 + 6.8
110 LET B(3,3) = FNC(B(1,1) / B(2,2))
120 PRINT B(1,2); B(3,3)
130 FOR I = B(1,1) TO B(1,2) STEP B(1,3)
140 IF B(3,3) >= B(2,3) THEN 600
```

SZINTAKTIKAI PÉLDÁK; NEM ALKOTNAK ÉRTELMES PROGRAMOT

UGYANEZ ÉRVÉNYES A SZÖVEGES VÁLTOZÓKRA IS :

```
150 LET T$(6) = "PONZIO"
160 IF T$(7) <> "PILATO" THEN 200
170 PRINT T$(6); T$(7)
```

BÁRMELY OLYAN KIFEJEZÉSBEN, AHOL MEGENGEDETT AZ EGYSZERŰ VÁLTOZÓ, AZ INDEXES VÁLTOZÓ HASZNÁLATA IS MEGENGEDETT.

AZONBAN A KIFEJEZÉSEKEN KÍVÜL VANNAK OLYAN HELYEK, AHOL NEM ÁLLHAT INDEXES VÁLTOZÓ.

NEM HASZNÁLHATÓ, MINT CIKLUSVÁLTOZÓ :

```
180 FOR B(1,1) = 1 TO 3
```

```
190 DEF FNC(B(1,1)) = SQR(A↑2 + B↑2)
```

VAGY FÜGGVÉNY FORMÁLIS PARAMÉTERE.

TERMÉSZETESEN, AZ INDEXES VÁLTOZÓK IGAZI ELŐNYE NEM ABBAN ÁLL HOGY EGYSZERŰ VÁLTOZÓKAT HELYETTESÍTENEK, HANEM ABBAN HOGY

INDEXŰR LEHET VÁLTOZÓ VAGY KIFEJEZÉS

SŐT MÉG BONYOLULT, KIFEJEZÉS IS, AMELYEK ÖNMAGUK IS TARTALMAZHATNAK INDEXES VÁLTOZÓKAT.

GY KÖNNYŰ BEOLVASNI EGY
SOR- VAGY OSZLOPVEKTORT :

```

240 FOR R= 1 TO 4
250 FOR C= 1 TO 3
260 LET B(R,C)=0
270 NEXT C
280 NEXT R

```

```

200 DATA "A", "B", "C", "D"
210 FOR I = 1 TO 4
220 READ T$(I)
230 NEXT I

```

VAGY KITÖRÖLNI A TÉGLALAP ALAKÚ TÖMBÖT :

VAGY KINYOMTATNI EGY VEKTORT
(EBBEN AZ ESETBEN SORONKÉNT) :

```

290 FOR I = 1 TO 4
300 PRINT T$(I);
310 NEXT I
320 PRINT

```

```

330 FOR R= 1 TO 4
340 FOR C= 1 TO 4
350 PRINT B(R,C);
360 NEXT C
370 PRINT
380 NEXT R

```

VAGY KINYOMTATNI EGY TÉGLALAP TÖMBÖT (EBBEN AZ ESETBEN SORONKÉNT) :

VAGY EGY ADATLISTÁT ÁTFÉSÜLNI,
HOGY EGY BIZONYOS ADAT HELYÉT
MEGTALALJUK :

```

390 FOR I= 1 TO 4
400 IF T$(I)= "C" THEN 430
410 NEXT I
420 LET I = 0
430 PRINT "HELYE" ; I

```

DE VIGYÁZZ!
HA
BONYOLULT KIFEJEZÉSEK-
KET HASZNALSZ INDEX-
KÉNT :

```

440 LET X = B(A↑2/2, 3)

```

MI LENNE, HA $A \uparrow 2/2$ -BŐL 3.99999 LENNE?

- * NÉMELYIK BASIC RENDSZER AZ EREDMÉNY EGÉSZ RÉSZÉT VENNÉ, AMI $LET X = B(3,3)$ -AT JELENT
- * MÁSONKÉNT AZ EREDMÉNYHEZ LEGKÖZELEBB ESŐ EGÉSZ SZÁMOT VENNÉK, AMI $LET X = B(4,3)$ -AT EREDMÉNYEZ

AZ ALÁBBI KIS PROGRAM LEFUTTATÁSÁVAL KÖNNYEDÉN ELLENŐRIZHETED, HOGY MIT CSINÁL BASIC VÁLTOZATOD. NE ÍRJ OLYAN PROGRAMOT, AMELY A NYELV EGY SPECIÁLIS INTERPRETÁCIÓJÁRA HAGYATKOZIK! AZ INDEXEK LEGYENEK EGYSZERŰEK, ÉS HASZNÁLD AZ "INT()" UTASÍTÁST, HA ÉRTÉKÜK TÖRT RÉSZT IS TARTALMAZHAT.

```

10 DIM A(2)
20 LET A(1) = 100
30 LET A(2) = 200
40 LET I = 1.99
50 PRINT A(I)
60 END

```

RENDEZÉS SZOMSZÉDOS ELEMÉK CSERÉJÉVEL

PÉLDA AZ INDEXES VÁLTOZÓK BEMUTATÁSÁRA

SZÁMOK NÖVEKVŐ SORRENDBE VALÓ RENDEZÉSE ELMÉLETILEG IGEN EGYSZERŰ, AZONBAN MEGLEPŐEN NEHÉZ MEGSZERVEZNI ADATOK NAGY TÖMEGE ESETÉN. AZ ALÁBBI PÉLDA A LEHETŐ LEGEGYSZERŰBB MÓDSZERT HASZNÁLJA A SZOMSZÉDCSERÉS RENDEZÉST, AMELY KIS MENNYISÉGŰ (MINTEGY 100 ÉRTÉK) BASIC - TÖMBÖKBEN TÁROLT ADAT ESETÉN MEGFELELŐ.

A (1)	6.5
A (2)	13.9
A (3)	10.2
A (4)	4.6
A (5)	3.5


AZ A () TÖMBÖT, EGY OSZLOPVEKTORT NÖVEKVŐ SORRENDBE KELL RAKNI A NAGYOBB SZÁMOK AZ ALJÁRA SÜLLYEDNEK. A SORRENDET AZ "IF" UTASÍTÁSBAN SZEREPLŐ FELTÉTEL MEGFORDÍTÁSÁVAL VISSZAFORDÍTHATOD.

AZ 1-ES SORBA MUTATÓ I "INDEX"-SEL KEZDÜNK; MAJD I-T SORONKÉNT LÉPTETJÜK. MINDEN LÉPÉSNEEL MEGNÉZZÜK, HOGY AZ I ÁLTAL MUTATOTT ÉS AZ EZ ELŐTT ÁLLÓ SOR KÖZÜL MELYIK NAGYOBB, HA AZ UTÓBBI, AKKORA KÉT SZÁMOT FELCSERÉLJÜK.

BEFEJEZVÉN A SZEKTOR "ÁTFÉSÜLÉSÉT" ÚJ ÁTFÉSÜLÉST INDÍTUNK, DE AZ ELŐZŐNÉL EGY SORRAL HAMARABB ÁLLUNK MEG HISZ A LEGNAGYOBB SZÁM MÁR BIZONYÁRA A VEKTOR VÉGÉRE KERÜLT.

FOLYTATJUK AZ ÁTFÉSÜLÉST AZ ELŐZŐ ÁTFÉSÜLÉSNEEL MINDÍG EGY SORRAL KEVESEBBET FIGYELVE EGÉSZEN ADDÍG, MÍG AZ EGÉSZ VEKTORBAN NINCS TÖBB FELCSERÉLNIVALÓ, VAGY MÍG AZ ÁTFÉSÜLÉS HOSSZA A SEMMIVÉ RÖVIDÜL.

ME, ITT AZ EGÉSZ FOLYAMAT; JEL AZT MUTATJA, HOGY HOL TÖRTÉNT CSERE;



AZ ALÁBBI PROGRAM AZ N, SORÚ A() OSZLOPVEKTORT TESZI SORRENDBE. A VEKTORT A "DATA" UTASÍTÁS TÖLTI MEG. AZ ELSŐ "DATA" UTASÍTÁS ELSŐ ADATA MONDJA MEG, HOGY HÁNY SZÁMOT KELL SORRENDBE RENDEZNI. AZ ITT BEMUTATOTT GONDOLATMENETET HASZNOSABB MÓDON ÚJRA FELHASZNÁLJUK A 71. OLDALON, AHOL ELKERÜLJÜK A SORRENDBE RENDEZENDŐ SZÁMOK TÉNYLEGES MOZGATÁSÁT.

```

10 REM SZOMSZÉDCSERÉS RENDEZÉS ; MAX 100 SZÁM
20 DIM A(100)
30 DATA 5
40 DATA 6.5, 13.9, 10.2, 4.6, 3.5
50 REM SZÜKSÉG SZERINT FOLYTASD DATA UTASÍTÁSOKKAL
60 REM
1000 READ N
1010 FOR I = 1 TO N
1020 READ A(I)
1030 NEXT I
1040 REM
1050 REM
1060 FOR S = 1 TO N-1
1070 LET M = 0
1080 FOR I = 1 TO N-S
1090 IF A(I) <= A(I+1) THEN 1140
1100 LET X = A(I)
1110 LET A(I) = A(I+1)
1120 LET A(I+1) = X
1130 LET M = 1
1140 NEXT I
1150 IF M = 0 THEN 2000
1160 NEXT S
2000 REM
2010 REM
2020 FOR I = 1 TO N
2030 PRINT A(I)
2040 NEXT I
2050 END

```

SZÁMOK SZÁMA

AZ A() VEKTOR TÖLTÉSE

ÁTFÉSÜL

LÉPTESS MINDEN ÁTFÉSÜLÉSKOR

NE CSERÉLDKI!


KORAI EXIT

AZ M VÁLTOZÓ EGY "JELZŐ", AMELYET NULLÁVÁ TESZÜNK MINDEN ÁTFÉSÜLÉST MEGELŐZŐEN, CSERE ESETÉN AZONBAN ÉRTÉKE 0-TÓL KÜLÖNBÖZŐ. AZ M-ET MINDEN ÁTFÉSÜLÉS VÉGÉN ELLENŐRIZZÜK, ÉS HA AZT MUTATJA, HOGY NEM VOLT CSERE, AKKOR A VEZÉRLÉS KORAI EXITRE KERÜL. ERRE ÉS A KÖNYV HOSSZABB PÉLDAIBAN HASZNÁLT MÓDSZEREKRE VONATKOZÓAN VILÁGOS MAGYARÁZAT TALÁLHATÓ A COLIN "FORTRAN TECHNIQUES" GÍMŰ, CAMBRIDGE UNIVERSITY PRESS (1972) KIADÁSÚ KÖNYVÉBEN.

LÁNCOS AL(LAT)

A LÁNCOK JÓL MUTATJAK AZ INDEXEK HASZNOSSÁGÁT. A LÁNCOK MANIPULÁCIÓJÁT LISTAFELDOLGOZÁSNAK NEVEZIK.

A LEGEGYSZERŰBB LÁNCFAJTÁT AZ ALÁBBIKBAN MUTATJUK BE. A LISTA FEJE EGY EGYSZERŰ VÁLTOZÓ: H. EZ TÁROLJA A MUTATÓT, AMELY VALAHOL AZ A(,) TÖMB TERÜLETÉN TALÁLHATÓ INFORMÁCIÓ ELSŐ SORÁRA VÁBBA' EGY OLYAN MÁSÍK MUTATÓT TARTALMAZÓ LÁNCRÁ MUTAT, AMELY AZ A(,)-BAN TALÁLHATÓ INFORMÁCIÓ KÖVETKEZŐ SORÁRA MUTAT. A LÁNC UTOLSÓ ELEME EGY NULLÁT TARTALMAZ, AMELY AZT JELENTI, HOGY "VÉGE A LÁNCRNAK".



A KÖVETKEZŐ PROGRAMRÉSZLET AZ A(,)-BEN A H FEJŰ LÁNC ÁLTAL SZERVEZETT INFORMÁCIÓT NYOMTATJA KI, BÁRMILYEN HOSSZÚ IS LEGYEN A LÁNC.


```

155 LET R=H
160 IF R=0 THEN 180
165 PRINT A(R,1); A(R,2); A(R,3)
170 LET R=L(R)
175 GO TO 160
180 END

RUN
4.7 0 246
20 0 468
16.2 0 821
33 0 900
    
```

← VEDD FEL A KÖVETKEZŐ LÁNCRÓT!


Egy lánCR LERAJZOLÁSÁNAK LEGVILÁGOSABB MÓDJA, HOGY EGYIK KEZED-BEN A LÁNC FEJÉT, A MÁSÍKBAN AZ UTOLSÓ LÁNCRSZEMET TARTOD ÉS SZÉTHÚZOD. A FENT BEMUTATOTT LÁNC MOST ÍGY NÉZ KI:




EZT A MODELIT ARRÁ HASZNÁLJUK, HOGY A LÁNCOKON VÉGEZHETŐ NÉHÁNY MŰVELETET MAGVA-RAZZUK MEG VELE.

A Z A(,) ÚJ SORÁT A KÖVETKEZŐKÉPPEN KAPCSOLHATJUK BE A LÁNCBA :


MÁSOLDÁT A FEJ TARTALMÁT AZ ÚJ LÁNCZEMBE!
MÁSOLD LE AZ ÚJ LÁNCZEM INDEXÉT A FEJBE (ÚGY, HOGY A FEJ AZ ÚJ LÁNCZEMRE MUTASSON)!



PÉLDA :
ÚJ 5. SOR



EREDMÉNY :



A KÖVETKEZŐ SZUBRUTIN ÖSSZEKAPCSOLJA AZ N SORT AZ L() VEKTORBAN A H VÁLTOZÓJÚ FEJEL TÁROLT LÁNCVAL.

```

100 REM ÖSSZEKAPCSOLJA AZ N SORT A H FEJŰ L() LÁNCVAL
110 LET L(N)=H
120 LET H=N
130 RETURN

```

ÍGY AZ 5. SOR LÁNCVAL VALÓ ÖSSZEKAPCSOLÁSÁT (L. A FENTI RAJZON) A KÖVETKEZŐKÉPPEN KÖNNYEN EL LEHET VÉGEZNI :

```

1055 LET N=5
1060 GO SUB 100
1065

```

EGY ÚJ LÁNCOT ALAKÍTHATUNK KI, MONDJUK AZ ELSŐ 10 SOR ÖSSZEKAPCSOLÁSÁHOZ, AZ ALÁBBI MÓDON :

```

2125 LET H=0
2130 FOR N = 1 TO 10
2135 GO SUB 100
2140 NEXT N

```

KEZDJ ÜRES LÁNCVAL!

MEGFIGYELHETŐ, HOGY A LÁNCHOZ KAPCSOLÓDÓ LEGUTOLSÓ SORBÓL ÍGY AZ ELŐZŐ OLDALON SZEREPLŐ RUTIN ÁLTAL ELSŐKENT KINYOMTATOTT SOR LESZ. EZT A MECHANIZMUST AZ 54. OLDALON LEÍRT FAJTÁJÚ VEREMTÁR SZERVEZÉSÉRE LEHET FELHASZNÁLNI. A VEREMTÁR TETEJÉRŐL A KÖVETKEZŐKÉPPEN VEHETÜNK LE ELEMET :

```

3260 LET R=H
3265 LET H=L(H)


```

TÁROLD AZ ELEM SORINDEXÉT!
A FELSŐ ELEMET DOBD KI!

LÁNCOK (FOLYTATÁS)

ÉRDEMES A LÁNCOK FEJÉT UGYANABBAN A VEKTORBAN TÁROLNI, MINT A LÁNCZEMEKET, MERT ÍGY RENDEZETTEBB PROGRAMOKAT KAPUNK. EZ EGY KICSIT TÖBB HELYET IGÉNYEL (AZ ALÁBBIKBAN BEMUTATOTT MÓDON), AMI BONYOLULTABB INDEXEK HASZNALATÁVAL VISSZANYERHETŐ, HA A LÁNCOT ELVÁLASZTJUK INFORMÁCIÓTARTALMÁTÓL.

UGYANABBAN A VEKTORBAN LEHET TÖBB LÁNC IS, FELTÉVE, HA AZ ÖSSZE-KAPCSOLT SOROK KÖLCSÖNÖSEN KIZÁRJÁK EGYMÁST (PL. EGYETLEN EGY LÁNC-SZEM SEM TARTOZHAT EGYNÉL TÖBB LÁNC-HOZ). AZ ALÁBBI SZUBRUTIN A NŐKET AZ EGYIK LÁNC-AL, A FÉRFIAKAT A MÁSIK LÁNC-AL KAPCSOLJA ÖSSZE. MINDKÉT LÁNC FEJE AZ ŐKET TÁROLÓ VEKTOR ELEJÉN TALÁLHATÓ.




```


1000 REM KEVERT LÁNCOK; N EMBER
1010 LET L(1)=0
1020 LET L(2)=0
1030 FOR I=3 TO N+2
1040 LET Q=1
1050 IF S$(I)="NŐ" THEN 1070
1060 LET Q=2
1070 LET L(I)=L(Q)
1080 LET L(Q)=I
1090 NEXT I
1100 RETURN
    
```

NEM SZIGORÚ-AN ELLENŐRIZVE: BÁRKIT AKI NINGS NŐNEK DEKLARÁLVA, FÉRFIKÉNT KEZEL

A 66. OLDALON A RENDEZŐPROGRAMBAN AZ ELEMI ADATOKAT FELCSERÉLTÜK. LÁNCOLT LISTÁK ESETÉN A RENDEZÉSNEEL CSAK A LÁNCZEMEKET KELL KICSERÉLNI. TEKINTSÜK E RÖVID LÁNCOT:



AHHOZ, HOGY VALÓJÁBAN E LÁNC-AL KAPCSOLATOS TÖMB 7. ÉS 3. SORÁT FEL TUDJUK CSERÉLNI, CSUPÁN 3 MUTATÓT KELL A LÁNCBAN ELMOZDÍTANI, AHOGY AZT A TÚLOLDALI ÁBRA MUTATJA.



AZ ALÁBBI SZUBRUTIN UGYANAZT A LOGIKÁT HASZNÁLJA FEL, MINT A 67. OLDALON LEVŐ RENDEZŐPROGRAM, AZONBAN AMIKOR A SOROKAT KELL FELCSERÉLNI, A MUTATÓKAT A FENT BEMUTATOTT MÓDON IRÁNYITJA. A MUTATÓMOZGATÁS A 2100 - TÓL A 2130 SORIG TALÁLHATÓ, AHOL :

- P AZ ELŐZŐ LÁNCSZEM INDEXE,
- C A JELENLEGI LÁNCSZEM INDEXE,
- F A KÖVETKEZŐ LÁNCSZEM INDEXE,

MINDEGYIKÜK JÓL LÁTHATÓ A FENTI ÁBRÁN. EGY CSERE UTÁN A C MÁR NEM A JELENLEGI INDEX : EZ A 2140. SORBAN VÁLTOZOTT MEG.

```

2000 REM SZOMSZÉDCSERÉS RENDEZÉS L( )LÁNC; L(Z) FEJ;
2005 REM SOROK SZÁMA N(Z); RENDEZÉS A(,) K. OSZLOPA SZERINT
2010 FOR S=1 TO N(Z)-1
2020 LET M=0
2030 LET C=Z
2040 FOR I=1 TO N(Z)-5
2050 LET P=C
2060 LET C=L(C)
2070 LET F=L(C)
2080 IF A(C,K) <= A(F,K) THEN 2150
2090 LET M=1
2100 LET X=L(P)
2110 LET L(P)=L(C)
2120 LET L(C)=L(F)
2130 LET L(F)=X
2140 LET C=F
2150 NEXT I
2160 IF M=0 THEN 2180
2170 NEXT S
2180 RETURN

```

Handwritten annotations:

- 2030: A LÁNC FEJE L(Z)
- 2050: ÚJ "ELŐZŐ" = RÉGI "JELENLEGI"
- 2060: MENJ TOVÁBB A KÖVETKEZŐ LÁNC SZEMRE!
- 2070: AZ ÚJ KÖVETKEZŐ LÁNC SZEM
- 2080: AZ A(,) K OSZLOPÁT RAKJA SORRENDBE
- 2110: LÁSD A FENTI ÁBRÁT
- 2140: NE CSERÉLJÜK FEL!
- 2160: KORAI EXIT



A 67. OLDAL PROGRAMJÁTÓL VALÓ EGYÉB ELTÉRÉSEKNEK SEMMI KÖZÜK A RENDEZÉS LOGIKÁJÁHOZ. A FENTI RUTIN BÁRMELY L()-BEN TÁROLT LÁNCOT FEL TUD HASZNÁLNI, HA MEGADJUK Z ÉRTÉKÉT : A KIVÁNT LÁNC FEJE EKKOR L(Z) LESZ, AZ ELEMÉK SZÁMA PEDIG N(Z)-BEN TALÁLHATÓ MEG. EZEN TÚLMENŐEN A FENTI RUTIN AZ A(,) TÖMB BÁRMELY OSZLOPÁT SORRENDBE TUDJA RAKNI A K-BAN LEVŐ OSZLOPINDEX ALAPJÁN.

SZÉTVÁLOGATÁS

PÉLDA AZ IMÉNT LEÍRT EGYSZERŰ LIS-TAFELDOLGOZÓ TECHNIKA FELHASZNÁLÁSÁRA.

A TÍPIKUS "VEZETŐI JELENTÉS" PROGRAMJA KORRA, ISKOLAI VÉGZETTSÉGRE ÉS NEMEKRE, CSOPORTOSÍTOTT NÉPESSÉGGEL FOGLALKOZIK. A KÖVETKEZŐ PROGRAM A VALÓDI FELADAT GROTESZK PARODIAJA, AZONBAN JOL BEMUTATJA AZ ÜZLETI ADATFELDOLGOZÁS NÉHÁNY JELLEMZŐ MODSZERÉT. A PROGRAM A SZÁMOK KÜLÖNBÖZŐ OSZLOPAIT VALOGATJA KI. AZ OLVASÓNAK GYAKORLÁSKÉNT AJÁNLOM, HOGY A PROGRAMOT ALFABETIKUS RENDEZÉSRE BŐVÍTSE KI (L. ERRÓL A 41. OLDALT).

PROGRAM AZ ELŐZŐ KÉT OLDALON MEGADOTT KÉT SZUBRUTINT HÍVJA. A PROGRAM HASZNÁLATÁKOR AZ ÁRUHÁZI SZEMÉLYZET MINDEN TAGJÁNAK NEVÉT, NEMÉT, ÁRUOSZTÁLYANAK SZÁMÁT, KORÁT ÉS FIZETÉSÉT KELL MEGADNI. EZ A PROGRAM 3 VEZETŐI JELENTÉST KÉSZÍT, MINDEGYIK KÜLÖN FOGLALKOZIK A NEMEKKEL. AZ ELSŐ JELENTÉS AZ ÁRUOSZTÁLY SZÁMA SZERINT RENDEZI EL A KIINDULÁSI ADATOKAT, A MÁSODIK AZ ALKALMAZOTTAK KORA SZERINT, A HARMADIK A FIZETÉSEK SZERINT. MIND A HÁROM BESZÁMOLÓ "RENDEZŐ" OSZLOPA A NEVEK OSZLOPA MELLETT ÁLL.

ÉRDEMES EGY FONTOS DOLGOT MEGJEGYEZNI E PROGRAMMAL KAPCSOLATBAN: AZ EREDETI   BEMENŐ ADATOT SOHASEM MOZGATJA. A RENDEZÉS ÚGY TÖRTÉNIK, HOGY A MUTATÓKAT EGYETLEN MUTATÓLISTÁN BELÜL TOLOGATJUK.

```
10 PRINT "ÁRUHÁZ : SZEMÉLYZETI ANALÍZIS"
20 DIM N(2), L(100), N$(100), M$(100), S$(100), A(100,3), T$(2), R$(3)
30 DATA "NŐ", "FÉRFI", "OSZTÁLY", "KOR", "FIZETÉS"
40 READ T$(1), T$(2), R$(1), R$(2), R$(3)
50 DEF FNC(Y) = Y - 3 * INT(Y/4)
60 REM FNC CIKLUSA 1, 2, 3, 1, 2 HA Y = 1.... 5
70 PRINT "DOLGOZÓK SZÁMA"
80 INPUT N
90 PRINT "CSALÁDNÉV", "KERESZTNÉV", "NEM", "OSZTÁLY", "KOR", "FIZETÉS"
100 LET N(2) = 0
110 FOR I = 3 TO N + 2
120 INPUT N$(I), M$(I), S$(I), A(I, 1), A(I, 2), A(I, 3)
130 IF S$(I) = T$(1) THEN 150
140 LET N(2) = N(2) + 1
150 NEXT I
160 LET N(1) = N - N(2)
170 REM INPUT KÉSZ
```

SZEMÉLYZET LÉTSZÁMA


ÖSSZESZÁMOLJA A FÉRFIAKAT

NŐK SZÁMA


```

180 REM ANALÍZIS KEZDETE
190 GO SUB 1000
200 FOR K= 1 TO 3
210 PRINT
220 PRINT K;"BESZÁMOLÓ; R$(K); "SZERINT RENDEZVE"
230 PRINT
240 FOR Z= 1 TO 2
250 GO SUB 2000
260 PRINT
270 PRINT T$(Z); "DOLGOZOK"
280 PRINT "NÉV", , R$(K), R$(FNC(K+1)), R$(FNC(K+2))
290 LET R=L(Z)
300 IF R=0 THEN 340
310 PRINT N$(R), M$(R), A(R,K), A(R, FNC(K+1)), A(R, FNC(K+2))
320 LET R=L(R)
330 GO TO 300
340 NEXT Z
350 NEXT K
360 GO TO 9999

```



ITT MÁSOLD BE AZ 1000-ÉS 2000-ES SZUBRUTINOKAT

9999 END

RUN

ÁRUHÁZ : SZEMÉLYZETI ANALÍZIS
DOLGOZÓK SZÁMA

? 6

CSALÁDNÉV	KERESZTNÉV	NEM	OSZTÁLY	KOR	FIZETÉS
? KOVÁCS	RÓZSI	NŐ	6	21	2800
? TÓTH	PÉTER	FÉRFI	6	23	3750
? KIS	PÁL	FÉRFI	7	22	3750
? NAGY	VIOLA	NŐ	6	66	2000
? FAZEKAS	LÁSZLÓ	FÉRFI	7	18	4000
? HORVÁT	ANDRÁS	FÉRFI	5	70	6000

1. BESZÁMOLÓ ÁRUOSZTÁLY SZERINT RENDEZVE


NŐ DOLGOZÓK

NÉV	OSZTÁLY	KOR	FIZETÉS
NAGY VIOLA	6	66	2000
KOVÁCS RÓZSI	6	21	2800

FÉRFI DOLGOZÓK

NÉV	OSZTÁLY	KOR	FIZETÉS
HORVÁT ANDRÁS	5	70	6000
TÓTH PÉTER	6	23	3750
FAZEKAS LÁSZLÓ	7	18	4000
KIS PÁL	7	22	3750

2. BESZÁMOLÓ KOR SZERINT RENDEZVE (stb. stb.)





MÁTRIXOK

MÁTRIXOK

NE IJEDJ MEG
A "MAT" UTASÍTÁSOK A
MÁTRIXALGEBRA ISMERE-
TE NÉLKÜL IS HASZNOSAK.

A BASIC-BEN A MÁTRIX EGYSZERŰEN INDEXES VÁLTOZÓK TÉGLALAP ALAKÚ TÖMBJE :

10 DIM A(4,4), B(4,1), C(1,4), D(2,3), E(1,1)

	1)	2)	3)	4)
A(1,				
A(2,				
A(3,				
A(4,				

NÉGYZET MÁTRIX

	1)
B(1,	
B(2,	
B(3,	
B(4,	

OSZLOP-
MÁTRIX

	1)	2)	3)	4)
C(1,				

SORMÁTRIX

	1)	2)	3)		1)
D(1,					
D(2,					

TÉGLALAP MÁTRIX

EZ EGY
MÁTRIX

A LEGTÖBB BASIC RENDSZERBEN KB. 12 "MAT" UTASÍTÁS VAN, AMELYEK-
KEL ILYEN TÖMBÖKET LEHET KEZELNI.

DE VIGYÁZZ A KÖVETKEZŐKRE :

NÉMELYIK BASIC RENDSZER A "MAT" UTASÍTÁSOK HASZNÁ-
LATÁT EGYDIMENZIÓS TÖMBÖK ESETÉN IS MEGENGEDI, MASOK VI-
SZONT NEM. ÍGY MINDÍG KÉTDIMENZIÓS TÖMBÖT KÉSZÍTS, AMI-
KOR "MAT" UTASÍTÁST AKARSZ HASZNÁLNI MEG HA A DI-
MENZIÓK EGYIKÉT EGYNEK IS KELL VÁLASZTANOD (MINT A FEN-
TIEKBEN B(,), C(,) ÉS E(,) ESETÉBEN)

NE HAGYD EL A "DIM" UTASÍTÁST A TÖMBÖKNÉL MÉG AKKOR
SEM, HA A LEGTÖBB RENDSZER MEGENGEDI EZT, HA A MÉRET
10-NÉL NEM NAGYOBB (L. A 62. OLDALT)

NÉMELY BASIC RENDSZER MEGENGED NUL-
LÁS INDEXEKET, DE ENNEK ELLENÉRE "MAT"
UTASÍTÁSAIK NEM VESZIK FIGYELEMBE A NULLA-
DIK SORT ÉS A NULLADIK OSZLOPOT (EZZEL IS
MEGERŐSÍTVE A 63. OLDALON LEVŐ FIGYELMEZTE-
TÉST, AMELY SZERINT SOSE HASZNÁLJ NULLÁS IN-
DEXET). HA ESETLEG BASIC VÁLTOZATOD MEGEN-
GED NULLÁS INDEXEKET, ÉS NEM TUDOD ELDÖNTE-
NI A HASZNÁLATI UTASÍTÁS ALAPJÁN, HOGY A "MAT" UTASÍTÁSOK IS
ENNEK MEGFELELŐEN VISELKEDNEK-E, AKKOR FUTTASD LE E KIS,
RÖVID PROGRAMOT. HA 1-ET ÍR KI (ÉS NEM 0-T), AKKOR FELTEHETŐLEG
AZ ÖSSZES "MAT" UTASÍTÁSOD ÉSZREVETTE A 0-DIK SORT ÉS A 0-DIK
OSZLOPOT; EKKOR MEGLEHETŐSEN SZOKATLAN RENDSZERREL VAN DOLGUNK.

```
10 DIM A(1,1)
20 LET A(0,0)=0
30 MAT A=CON
40 PRINT A(0,0)
50 END
RUN
0
```

mátrix (matematikai fogalom): mennyiségek v. szimbólumok téglalap a-
lakú tömbje.

A MÁTRIX-UTASÍTÁSOK A "MAT" SZÓVAL KEZDŐDNEK. EZEK KÖZÜL TIZENKETTŐT A KÖVETKEZŐKBEN SOROLUNK FEL, A MAGYARÁZAT PEDIG KÉSŐBB KÖVETKEZIK. OLYAN KIFEJEZÉSEKET ISMERÜNK MEG, MINT "TRANZPONÁLÁS", "MÁTRIXSZORZÁS" ÉS "INVERZIÓ". AKIK MÉG SOHASEM TALÁLKOZTAK A MÁTRIXALGEBRÁVAL, AZOK IS MEGÉRTHETIK ÉS HASZNÁLHATJÁK A VONATKOZÓ UTASÍTÁSOKAT.

		OLDALSZÁM	
MAT	A = B	A(,) TÖMB ELEMÉIT B(,) TÖMB ELEMÉIVEL TESZI EGYENLŐVÉ.	78
MAT	A = B ± C	A(,) TÖMB ELEMÉI A B(,) ÉS A C(,) TÖMB ELEMÉINEK ÖSSZEGE VAGY KÜLONBSÉGE LESZ.	80
<div style="border: 1px solid black; border-radius: 50%; padding: 5px; display: inline-block; margin-bottom: 10px;"> EZEKBEŒN A "MAT" UTASÍTÁSOKBAN BÄRMELY BETŐ ALLHAT NEMCSAK A, B, C </div>			
MAT	A = (N) * B	A(,) TÖMB ELEMÉI A B(,) TÖMB MEGFELELŐ ELEMÉINEK N-SZERESEI LESZNEK, AHOL N EGY SZÄM, NEM MÁTRIX.	82
MAT	A = TRN(B)	A(,) TÖMB SORAI UGYANAZOK LESZNEK, MINT A B(,) TÖMB OSZLOPAI.	84
MAT	A = ZER	A(,) TÖMB ÖSSZES ELEMÉT NULLÁVÁ TESZI.	86
MAT	A = CON	A(,) TÖMB ÖSSZES ELEMÉ EGYSÉGNYI LESZ.	87
MAT	A = IDN	A(,) NÉGYZETES MÁTRIXOT EGYSÉGMÁTRIXSZÄ ALAKÍTJA. A FŐÄTLÖBÄN MINDEN ELEM 1, AZ ÄTLÖN KIVÜLI ELEMÉK NULLAK.	87
MAT	R = A * B	R(,) TÖMB AZ A(,) ÉS B(,) MÁTRIXSZORZATA LESZ.	88
MAT	A = INV(B)	A(,) TÖMB A B(,) NÉGYZETES MÁTRIX INVERZE LESZ.	92
MAT	READ A, B, C	TELJES TÖMBÖK OLVASÄSA "DATA" UTASÍTÄSOKBÖL.	94
MAT	INPUT A, B, C	TELJES TÖMBÖK BEVITELE PROGRAMON KIVÜLI FORRÄSBÖL.	96
MAT	PRINT A, B, C	TELJES TÖMBÖK KINYOMTATÄSA.	98

KÖNYVÜNK MINDEN "MAT" UTASÍTÄSNÄK MEGÄDJA, AZ EGYSZERÜ BASIC-BEN MEGÄRT MEGFELELŐJET ÜGY, HOGY EGYMÄSBA ÄGYÄZOTT CIKLUSOKAT HASZNÄL. A CIKLUSOK MINDEN EGYMÄSBA ÄGYÄZÖDÄSÄNÄL AZ A VÄLTÖZÖ VÄLTÖZIK A LEGGVORSÄBBÄN, AMELY A LEGBELSÖ CIKLUST VEZÄRLI. HA A RENDSZER A TÖMBÖKET OSZLOPOK SZERINT TÄROLJA, AKKOR JOBB, HA AZ OSZLOPINDEXEK VÄLTÖZNEK GYORSÄBBÄN. A HELYZET AZ, HOGY NÄHÄNY BASIC RENDSZER A TÖMBÖKET OSZLOPONKÄNT, MÄSOK SORONKÄNT TÄROLJÄK; ÄGY A KÖNYVBEN SINGCS RENDSZER A SOR- ÉS OSZLOPINDEXEK KIVÄLÄSZTÄSÄBÄN.

MAT A = B

EZ AZ UTASÍTÁS A TELJES TÖMB MÁ-SOLÁSÁT VÉGZI EL.

10 DIM A(3,4), B(3,4), C(2,3)

	1)	2)	3)	4)
A(1)				
A(2)				
A(3)				

	1)	2)	3)	4)
B(1)	5	-9.6	2.5	7.51
B(2)	6.3	0	1	2.34
B(3)	7.2	3.7	-1	8.86

TEGYÜK FEL, HOGY MOST AZT AKAROD, HOGY A B(,) TÖMBBEN JELENLEG TÁROLT ÉRTÉKEK AZ A(,) TÖMBBEN LEGYENEK. EZT EGYSZERŰEN ÚGY ÉRED EL, HA ELEM RŐL ELEMRE ÁTMÁSOLD AZ ADATOKAT B(,)-BŐL A(,)-BA A KÖVETKEZŐKÉPPEN :

	1)	2)	3)
C(1)	235	198	137
C(2)	642	856	705

```

200 FOR C = 1 TO 4
210 FOR R = 1 TO 3
220 LET A(R,C) = B(R,C)
230 NEXT R
240 NEXT C
    
```

OSZLOPOK CIKLUSA
SOROK CIKLUSA

DE AZ EGÉSZET ELINTÉZHETED EGYETLEN "MAT" UTASÍTÁSSAL IS A KÖVETKEZŐKÉPPEN :

100 MAT A=B

JEGYEZZÜK MEG, HOGY A-NAK ÉS B-NEK A 100. SORBAN SEMMI KÖZÜK SINCS AZ A ÉS B EGYSZERŰ VÁLTOZÓKHOZ. A "MAT" SZÓBÓL TUDJA A RENDSZER, HOGY A ÉS B AZ A(,) ÉS B(,) TÖMBÖK.

A FENTIEKBEN ÁBRÁZOLT TÖMBÖK ESETÉN HIBÁS LENNE AZ ALÁBBI 110. SOR ; B(,) TÚL NAGY AHHOZ, HOGY ELFÉRJEN A C(,)-BEN. AZONBAN A 120. SOR MEGENGEDETT. B(,) DIMENZIÓIT 2 SORRA ÉS 3 OSZLOPRA VÁLTOZTATJA MEG :

2 ÉS 3 MOST A **JELENLEGI DIMENZIÓJA**, AMELYET MAS "MAT" UTASÍTÁSSAL LEHET BEÁLLÍ-

110 MAT C=B
120 MAT B=C

	1)	2)	3)
B(1)	235	198	137
B(2)	642	856	705

ZSUGORODIK
B(2,4) TÖBBÉ NEM LÉTEZIK


TANI, FELTÉVE, HA ÉRTÉKEIK SOHASEM HALADJAK MEG AZ EREDETI "DIM" UTASÍTÁSBAN SZEREPLŐKÉT.

ÚJRADIMENZIONÁLÁS

EZ MAJD-
NEM

MINDEGYIK "MAT" UTASÍTÁSRA ÉRVÉNYES.

A TÚLOLDALI 120. SOR UTÁN A $B(,)$ TÖMBÖT ÚJRADIMENZIONÁLTUK, HOGY AKKORA LEGYEN, MINT A $C(,)$ TÖMB. MOST HIBÁS LENNE AZ ALÁBBI UTASÍTÁS:

130 LET $B(2,4) = 13$ 

MERT $B(2,4)$ MEGSZÜNT. ENNEK ELLENÉRE NÉMELYIK BASIC RENDSZER NEM JELZI A HIBÁT, ÉS ELŐRE NEM LÁTHATÓ EREDMÉNYEKET AD ∞ A 13-AS SZÁM MÁS HELYEN FOG KISÉRTETÉSZERŰEN MEGJELENNI.

TOVÁBBRA IS MEGENGEDETT A KÖVETKEZŐ:

140 MAT $B = A$

BÁR $A(,)$ TOVÁBBRA IS 3 SORRAL ÉS 4 OSZLOPPAL RENDELKEZIK, MIALATT $B(,)$ -NEK JELENLEG 2 SORA ÉS 3 OSZLOPA VAN. A "MAT" UTASÍTÁS (A SZOKÁSOS "LET" UTASÍTÁSTÓL ELTÉRŐEN) ÚJRADIMENZIONÁLJA A $B(,)$ -T AZ AKTUÁLIS 3 SORRA ÉS 4 OSZLOPRA.

ÁLTALÁBAN A RENDSZER ÚJRADIMENZIONÁLJA AZ EGYENLŐSÉGJEL BAL OLDALÁN LEVŐ TÖMBÖT, A JOBB OLDALON LEVŐ TÖMBÖK AKTUÁLIS DIMENZIÓINAK MEGFELELŐEN, FELTÉVE HOGY AZ ÚJ DIMENZIÓK A "DIM" UTASÍTÁSBAN MEGADOTT ÉRTÉKEKET NEM HALADJÁK MEG. (EZÉRT ROSSZ A TÚLOLDALI 110. SOR, AZONBAN A FENTI 140. SOR HELYES).

VAN AZONBAN EGY KIS PROBLÉMA: ÚGY TŰNIK, HOGY A LEGTÖBB BASIC RENDSZER AZ "EREDETI MÉRETEEN" AZ EREDETI TÖMB ELEMEINEK ÖSSZES SZÁMÁT ÉRTI: EZÉRT, HA $P(1,16)$ SZEREPELT A "DIM" UTASÍTÁSBAN, AKKOR $P(,)$ AZ ÚJRADIMENZIONÁLÁSSAL 4 SORBÓL ÉS 4 OSZLOPBÓL ÁLLÓ NÉGYZETES TÖMBBÉ VÁLHATNA. EZ NEM VALAMI SZÉP. JOBB ÉS BIZTOSABB, HA MINDEN DIMENZIÓT AZ ÚJRADIMENZIONÁLÁS KÜLÖN-KÜLÖN KORLÁTJÁNAK TEKINTÜNK ∞ , EZÉRT AZ EREDETI "DIM" UTASÍTÁSBAN SZEREPLŐ $P(1,16)$ -NAK AKÁRHÁNY OSZLOPA LEHET EGÉSZEN 16-IG, DE SOHASEM LEHET EGYNÉL TÖBB SORA. HA AZ A SZÁNDÉKOD, HOGY A $P(,)$ 16 ELEMŰ EGY SORÚ TÖMBÖT 4 ELEMŰ 4 SOROSRA ALAKÍTSD, AKKOR A "DIM" UTASÍTÁSBAN $P(4,16)$ -KÉNT KELL DEKLARÁLNOD.

MAT A = B ± C

10 DIM A(3,2), B(3,2), C(3,2), D(2,3), E(3,3)

	1)	2)
A(1,	4.3	0
A(2,	8.5	4
A(3,	9.4	4.3

	1)	2)
B(1,	1.5	-2.7
B(2,	3.6	7.8
B(3,	8.4	4.3

	1)	2)
C(1,	2.8	2.7
C(2,	4.9	-3.8
C(3,	1	0

	1)	2)	3)
D(1,			
D(2,			

	1)	2)	3)
E(1,			
E(2,			
E(3,			

TEGYÜK FEL, HOGY AZT AKARJUK: AZ A(,) TÖMB ELEMET A B(,) ÉS A C(,) TÖMBÖK MEGFELELŐ ELEMINEK ÖSSZEJEI (VAGY KÜLÖNBSEGEI) LE-
GYENEK. EZT A KÖVETKEZŐKÉPPEN ÉRHHETED EL?

```

200 FOR C = 1 TO 2
210 FOR R = 1 TO 3
220 LET A(R,C) = B(R,C) + C(R,C)
230 NEXT R
240 NEXT C
    
```

MINUSZJEL, HA A KÜLÖNBSEGET AKAROD

A FELADAT MEGOLDHATÓ AZONBAN EGYETLEN "MAT" UTASÍTÁSSAL IS IGY,

```

100 MAT A = B + C
    
```

VAGY MINUSZJEL

A B(,) ÉS A C(,) TÖMBÖK VÁLTOZATLANOK MARADNAK & AZ A(,) TÖMB A FENTI ÁBRÁN LÁTHATÓ.

FIGYELJÜK MEG, HOGY A 100. SORBAN LEVŐ A, B, ÉS C-NEK SEMMI KÖ-
ZÜK SINCS AZ A, B ÉS C EGYSZERŰ VÁLTOZÓKHOZ. A "MAT" SZÓ TU-
DATJA A RENDSZERREL, HOGY AZ A(,), B(,) ÉS C(,) TÖMBÖKRŐL VAN
SZÓ.

HA AZ EGYENLŐSÉGJEL JOBB OLDALÁN MEGNEVEZETT KÉT TÖMB JELENLE-
GI DIMENZIÓI NEM AZONOSAK, AKKOR A BASIC RENDSZER MEGTAGADJA
AZ UTASÍTÁS VÉGREHAJTÁSÁT.

```

110 MAT A = B + D
    
```

DIM B(3,2) KONTRA D(2,3)

AZ EGYENLŐSÉGJEL BAL OLDALÁN MEGNEVEZETT TÖMB DIMENZIÓI LEGYENEK LEGALÁBB, AKKORAK, MINT A JOBB OLDALÁN MEGNEVEZETT TÖMBÖK DIMENZIÓI! EZÉRT AZ ALÁBBI 130. SOR HIBÁS; A 140. SOR MEGENGEDETT, DE A RENDSZER ÚGY DIMENZIONÁLNA ÚJRA AZ E(,) TÖMBÖT, HOGY ANNAK 3 SORA ÉS 2 OSZLOPA LEGYEN; IGY HIBA LENNE E(3,3)-RA HIVATKOZNI. AZ ÚJRADIMENZIONÁLÁS KÖVETKEZMÉNYEIRŐL A 79. OLDALON VOLT SZÓ.

130	MAT	$D = B + C$
140	MAT	$E = B + C$
150	LET	$E(3,3) = 13$

NÉMELYIK BASIC RENDSZER (NEM MINDEGYIK) MEGENGEDEI, HOGY AZ EGYENLŐSÉGJEL MINDKÉT OLDALÁN UGYANAZ A TÖMB LEGYEN; A "HORDOZHATÓSÁG" KEDVÉÉRT KERÜLD EZT!

160	MAT	$A = A + A$
170	MAT	$A = A - C$
180	MAT	$A = C - A$

NE KOCKÁZTASD MEG!

AZ UTASÍTÁS FORMÁJÁT NEM LEHET MEGVÁLTOZTATNI. NE PRÓBÁLD MEG ELŐJELET ÉS BETŰKET HOZZÁTENNI.

190	MAT	$A = B + C$	\leftarrow MAT $A = C - B$
200	MAT	$A = B + C + F$	\leftarrow MAT $X = B + C$ MAT $A = X + F$
210	MAT	$A = -B$	\leftarrow MAT $A = (-1) \times B$ l. a. túloldalon

MAT A = (SKALÁR KIFEJEZÉS) * B

10 DIM A(3,2), B(3,2), C(4,3), D(2,2)

	1)	2)
A(1,	15.6	4.5
A(2,	-9	0
A(3,	6	-6

	1)	2)
B(1,	5.2	1.5
B(2,	-3	0
B(3,	2	-2

	1)	2)	3)
C(1,			
C(2,			
C(3,			
C(4,			

	1)	2)
D(1,		
D(2,		

JEGYÜK FEL, HOGY AZT AKARJUK: AZ A(,) TÖMB MINDEN ELEME, A B(,) TÖMB MEGFELELŐ ELEMEINEK HÁROMSZOROSA LEGVEN. EZT IGEN KÖNYVEN A KÖVETKEZŐ MÓDON OLDHATOD MEG:

```

200 FOR C=1 TO 2
210 FOR R=1 TO 3
220 LET A(R,C) = 3.0 * B(R,C)
230 NEXT R
240 NEXT C
    
```

A FELADAT AZONBAN EGYETLEN "MAT" UTASÍTÁSSAL IS MEGOLDHATÓ A KÖVETKEZŐKÉPPEN:

```

100 MAT A = (3.0) * B
    
```

A B(,) TÖMB NEM VÁLTOZIK MEG, A(,) TARTALMA A FENTI ÁBRÁN LÁTHATÓ. MINDEN ELEMET A ZÁRÓJELBEN LEVŐ SKALÁR (AZAZ NEMVEKTOR VAGY NEMMÁTRIX) KIFEJEZÉSSEL SZOROZTUNK MEG.

JEGYEZZÜK MEG, HOGY A 100. SORBAN SZEREPLŐ A -NAK ÉS B -NEK SEMMI KÖZE AZ A ÉS B EGYSZERŰ VÁLTOZÓKHOZ; A "MAT" SZÓ TUDATJA A RENDSZERREL, HOGY AZ A(,) ÉS B(,) TÖMBÖKRŐL VAN SZÓ. A ZÁRÓJELBEN BELÜL AZONBAN A ÉS B EGYSZERŰ VÁLTOZÓK LENNÉNEK: AZ ALÁBBI 130. SORNAK PONTOSAN UGYANAZ LESZ A HATÁSA, MINT A FENTI 100. SORNAK; (A/B) A SKALÁR MENNYISÉGET (3.0) JELKÉPEZL.

```

110 LET A = 6
120 LET B = 2
130 MAT A = (A/B) * B
    
```

AZ EGYENLŐSÉGJEL, BAL OLDALÁN MEGNEVEZETT TÖMBNEK A "DIM" UTASÍTÁSBAN LEGALÁBB AKKORA DIMENZIÓKKAL KELL RENDELKEZNIÉ, MINT A JOBB OLDALON MEGNEVEZETT TÖMB JELENLEGI DIMENZIÓI. EZÉRT AZ ALÁBBI 140. SOR HIBÁS, A 150. SOR MEGENGEDETT, DE A BASIC RENDSZER ÚGY DIMENZIONÁLNA ŰRA A C(,) TÖMBÖT, HOGY ENNEK 3 SORA ÉS 2 OSZLOPA LEGYEN. EZÉRT HIBÁS LENNE A C(4,2)-RE HIVATKOZNI. AZ ŰRADIMENZIONÁLÁS KÖVETKEZMÉNYEIRŐL A 79. OLDALON VAN SZÓ.

140	MAT	$D = (0.5) * B$
150	MAT	$C = (-1) * B$
160	LET	$C(4,2) = 13$

NÉMEZYIK BASIC RENDSZER (NEM MINDEGYIK) MEGENGEDEI, HOGY AZ EGYENLŐSÉGJEL MINDKÉT OLDALÁN UGYANAZT A TÖMBÖT NEVEZZÜK MEG; A "HORDOZHATÓSÁG" KEDVEÉRT NE TEDD EZT.

170	MAT	$A = (-X) * A$
-----	-----	----------------

NE TEDD EZT!

A ZÁRÓJELBEN LEVŐ KIFEJEZÉS OLYAN BONYOLULT LEHET, AMILYENT CSAK AKARSZ, FELTÉVE, HA EGYETLEN SKALÁR ÉRTÉKET KÉPVISEL.

180	MAT	$A = (-X + \text{SQR}(3 * B(2 * I, J))) * B$
-----	-----	--

EZEN UTASÍTÁS FORMÁJA PONTOSAN OLYAN, MINT AZ ELŐZŐ OLDALON SZEREPLŐ FEJLÉCÉ. AZ ALÁBBIÁKBAN NÉHÁNY FORMAI HIBÁT MUTATTUNK BE.

185	LET	$X = 3$	<p>EZ A "MÁTRIXSZORZÁS" FORMÁJA, L. A 88. OLDALT</p> <p>MAT $A = (-X) * B$</p> <p>MAT $A = (X * Y) * B$</p> <p>MAT $A = (X) * B$</p>
190	MAT	$A = X * B$	
192	MAT	$A = -(X) * B$	
194	MAT	$A = (X) * (Y) * B$	
196	MAT	$A = B * (X)$	

MAT A = TRN(B)

10 DIM A(3,2), B(2,3), C(3,3), D(2,2)

	1)	2)
A(1,	4.2	-3.8
A(2,	15.6	8.7
A(3,	0	-4

	1)	2)	3)
B(1,	4.2	15.6	0
B(2,	-3.8	8.7	-4

	1)	2)	3)
C(1,			
C(2,			
C(3,			

	1)	2)
D(1,		
D(2,		

TEGYÜK FEL: AZ A(,) TÖMB SORAIT EGYENLŐVÉ AKAROD TENNI A B(,) TÖMB OSZLOPAIVAL \Leftrightarrow MÁ S ZÓ VAL A(,) - T, B(,) TRANSZPONÁLTJÁVÁ AKAROD ÁTALAKÍTANI. A KÖVETKEZŐKÉPPEN MÁSOLHATNÁD ÁT EGYESÉVEL AZ ELEMÉKET:

```

200 FOR I = 1 TO 2
210 FOR J = 1 TO 3
220 LET A(J,I) = B(I,J)
230 NEXT J
240 NEXT I
    
```

FIGYELD MEG AZ INDEXEKET: (J, I) ÉS (I, J)

A FELADAT MEGOLDHATÓ EGYETLEN "MAT" UTASÍTÁSSAL IS A KÖVETKEZŐKÉPPEN:

100 MAT A = TRN(B)

A B(,) TÖMB VÁLTOZATLAN MARAD; AZ A(,) TÖMB OLYAN LESZ, MINT AZ OLDAL ELEJÉN MÁR BEMUTATOTT.

FIGYELJÜK MEG, HOGY A 100. SORBAN LEVŐ A - NAK ÉS B - NEK SEMMI KÖZÜK SINCS AZ A ÉS B EGYSZERŰ VÁLTOZÓKHOZ; A "MAT" SZÓ TUDATJA A BASIC RENDSZERREL, HOGY EZ ALATT AZ A(,) ÉS B(,) TÖMBÖKRŐL VAN SZÓ.

AZ EGYENLŐSÉGJEL BAL OLDALÁN MEGNEVEZETT TÖMBNEK LEGALÁBB AKKORA MÉRETŰNEK KELL LENNIE A "DIM" UTASÍTÁSBAN, MINT A JOBB OLDALON MEGNEVEZETT TÖMB JELENLEGI DIMENZIÓI. EZÉRT AZ ALÁBBI 100. SOR HIBÁS; A 120. SOR MEGENGEDETT, DE A BASIC ÚGY DIMENZIONÁLJA ÚJRA A C(,) TÖMBÖT, HOGY ANNAK 3 SORA ÉS 2 OSZLOPA LESZ (N.B. NEM 2 SORA ÉS 3 OSZLOPA!), ÉS HIBÁS LENNE A C(3,3)-RA HIVATKOZNI.

AZ ÚJRADIMENZIONÁLÁS KÖVETKEZMÉNYEIRŐL A 79. OLDALON VAN SZÓ.


110	MAT	$D = \text{TRN}(B)$
120	MAT	$C = \text{TRN}(B)$
130	LET	$C(3,3) = 13$

NEM SZEREPELHET UGYANAZ A TÖMBNÉV AZ EGYENLŐSÉGJEL MINDKÉT OLDALÁN, MÁ S ZÓVAL NEM TRANSZPONÁLHATOD SAJÁT MAGÁRA A MÁTRIXOT:

140	MAT	$B = \text{TRN}(B)$
-----	-----	---------------------

VAN AZONBAN OLYAN BASIC RENDSZER IS, AMELY MEGENGEDI EZT. GYAKORLÁSKÉNT TALÁN SZÓRAKOZTATÓ LESZ MEGÍRNI EGY RUTINT, AMELY FELCSERELI A TÖMBÖT SAJÁT TRANSZPONÁLTJÁVAL ANÉLKÜL, HOGY ELŐTTE MÁ S TÖMBBE MÁSOLNÁNK ÁT (MÁ S, SZÓVAL A TÖMBÖT "HELYBEN" KELL TRANSZPONÁLNO D). TRÜKKÖ S MEGOLDÁ S, DE NEM LEHETETLEN.

TRANSZPONÁLÁS IGEN HASZNOS MŰVELET A MÁTRIXALGEBRÁBAN: KÜLÖNÖSKÉPPEN, HA A KOORDINÁTÁKAT EGYIK KOORDINÁTA RENDSZERBŐ L A MÁ SIKBA AKARJUK TRANSZFORMÁLNI. ENNEK LELEGYSZERŐ BB PÉLDÁJÁT AZ ALÁBBIÁKBAN MUTATJUK BE.




A P PONT KOORDINÁTÁJA U-HO Z É S V-HEZ VISZONYÍTVÁ u É S v, MIK LESZNEK P KOORDINÁTÁI AZ X É S Y TENGELYHEZ VISZONYÍTVÁ?

A TRIGONOMETRIÁT FELHASZNÁLVA:

$$\begin{aligned} x &= u \cos \theta - v \sin \theta \\ y &= u \sin \theta + v \cos \theta \end{aligned}$$

AMELY MÁTRIXFORMÁBAN AZ ALÁBBIÁK SZERINT ÍRHATÓ LE:

$$\begin{bmatrix} x \\ y \end{bmatrix} = \begin{bmatrix} \cos \theta & -\sin \theta \\ \sin \theta & \cos \theta \end{bmatrix} \begin{bmatrix} u \\ v \end{bmatrix} \dots (1)$$



UGYANCSAK A TRIGONOMETRIÁT FELHASZNÁLVA, UGYANILYEN KÖNNYŰ EGY OLYAN EGYENLETPÁRT TALÁ LNI, AMELY MEGADJA u-t É S v-t x É S y FÜGGVÉNYÉBEN.

$$\begin{bmatrix} u \\ v \end{bmatrix} = \begin{bmatrix} \cos \theta & \sin \theta \\ -\sin \theta & \cos \theta \end{bmatrix} \begin{bmatrix} x \\ y \end{bmatrix} \dots (2)$$

JEGYEZZÜ K MEG, HOGY E NÉGYZETES-MÁTRIXOK MINDEGYIKE A MÁ SIK TRANSZPONÁLTJA.

ÁLTALÁBAN IGAZ, ORTOGONÁLIS TENGELYTRANSZFORMÁCIÓKRA, HOGY A VISZTA TRANSZFORMÁCIÓHO Z CSUPÁN A TRANSZFORMÁLÓ MÁTRIXOT KELL TRANSZPONÁLNI. TANKÖNYVEKBEN A MÁTRIXOKAT GYAKRAN VASTAG BETŰKKEL, A TRANSZPONÁLTÁKAT PEDIG FELSŐ VESSZŐVEL JELÖLIK. A FENTI KÉT EGYENLET ÁLLHAT PL. ILYEN ALAKBAN: $X = TU$ É S $U = T'X$.

MAT A = ZER

10 DIM A(3,4), B(2,3)

	1)	2)	3)	4)
A(1,	0	0	0	0
A(2,	0	0	0	0
A(3,	0	0	0	0

	1)	2)	3)
B(1,	1	1	1
B(2,	1	1	1

A TÖMB ÖSSZES ELEMÉT NULLÁZHATJUK A KÖVETKEZŐ MÓDON:

```
200 FOR R = 1 TO 3
210 FOR C = 1 TO 4
220 LET A(R,C) = 0
230 NEXT C
240 NEXT R
```

A FELADATOT EGYETLEN "MAT" UTASÍTÁSSAL IS MEGOLDHATOD A KÖVETKEZŐKÉPPEN:

```
100 MAT A = ZER
```

AHOL "ZER" A NULLA SZÓ RÖVIDÍTÉSE, ÉS AZ A BETŰNEK SEMMI KÖZE AZ A EGYSZERŰ VÁLTOZÓHOZ. A "MAT" SZÓ TUDATJA A BASIC RENDSZERREL, HOGY AZ A(,) TÖMBRŐL VAN SZÓ.

LÉGY ÓVATOS EZZEL AZ UTASÍTÁSSAL, VALAMINT AZ ALÁBBIK BÀN KIFEJTETT "CON" ÉS "DN" UTASÍTÁSOKKAL! NÉMELYIK BASIC RENDSZER FELTÉTELEZI, HOGY AZ EGYENLŐSÉGGEL BAL OLDALÁN MEGNEVEZETT TÖMB MEGTARTJA JELENLEGI DIMENZIÓIT, AMELYEK KISEBBEK LEHETNEK, MINT A "DIM" UTASÍTÁSBAN SZEREPLŐK. ÍGY, HACSAK NEM EZ AZ ELSŐ VÉGREHAJTANDÓ, "MAT" UTASÍTÁS, BIZTOSABB AZ ALÁBBI ALTERNATÍV FORMÁT HASZNÁLNI:

```
110 MAT A = ZER (2,3)
```

ITT A KIFEJEZÉSEK IS MEGENGEDETEK

AMELY AZ EGYENLŐSÉGGEL BAL OLDALÁN MEGNEVEZETT TÖMBÖT ÚRADIMENZIONÁLJA, ELEMEINEK PEDIG NULLA ÉRTÉKET AD. AZ ÚRADIMENZIONÁLÁS KÖVETKEZMÉNYEIRŐL A 79. OLDALON VOLT SZÓ. HA A MÉRTEKKNÉL KIFEJEZÉSEKET HASZNÁLSZ, BIZTOSÍTSD (PL. "INT"ET HASZNÁLVA), HOGY KIFEJEZÉSEID EGÉSZ EREDMÉNYEKET ADJANAK. NÉHÁNY BASIC RENDSZER AZ EREDMÉNYHEZ LEGKÖZELEBB ESŐ EGÉSZET HASZNÁLJA, MÁS RENDSZEREK AZONBÁN AZ EREDMÉNY EGÉSZ RÉSZÉT VESZIK.

MAT A = CON

AZ EGYETLEN KÜLÖNBSÉG EZEN UTASÍTÁS ÉS A "MAT A = ZER" UTASÍTÁS KÖZÖTT AZ, HOGY AZ EREDMÉNYŰL KAPOTT TÖMB EGYESEKSEL LESZ TELE ÉS NEM NULLÁKKAL.

```
120 MAT B = CON(2,3)
```

A "CON" A CONSTANS SZÓ RÖVIDÍTÉSE. A KÉT "MAT" UTASÍTÁSSAL MINDEN ELEMHEZ BÁRMELY KONSTANS ÉRTÉKET RENDELHETSZ.

```
130 MAT A = CON(2,3)
140 MAT B = (-5) * A
```

B(,) ÖSSZES ELEMÉNEK ÉRTÉKE -5

MAT A = IDN

AZ "IDN" AZ IDENTITÁS (AZONOSSÁG) SZÓ RÖVIDÍTÉSE. (A MÁTRIXALGEBRA "EGYSÉG MATRIXA" MEGFELEL AZ ÁLTALÁNOS ALGEBRA EGYSÉGÉNEK; EZT A 91. OLDAL MUTATJA BE.) AZ EGYSÉG MÁTRIX ÁTLÓJÁBAN EGYESEK VANNAK (AHOL A SOR- ÉS AZ OSZLOPINDEXEK EGYENLŐEK), ÉS NULLÁK VANNAK AZ ÁTLÓN KÍVÜL (AHOL AZ INDEKES KÜLÖNBÖZŐEK). ÍME ENNEK EGYFAJTA PROGRAMOZÁSI MÓDJA :

	1)	2)	3)
A(1,	1	0	0
A(2,	0	1	0
A(3,	0	0	1

```
200 FOR R = 1 TO 3
210 FOR C = 1 TO 3
220 LET A(R,C) = 1 - ABS(SGN(R-C))
230 NEXT C
240 NEXT R
```

A FELADATOT EGYETLEN "MAT" UTASÍTÁSSAL IS MEGOLDHATOD :

```
150 MAT A = IDN(3,3)
```

AZ EGYSÉG MÁTRIX MINDIG NÉGYZETES

A MÉRETEKET ELHAGYHATJUK, CSUPÁN "MAT A = IDN" IS ELÉG, HA A(,) JELENLEG NÉGYZETES MÁTRIX.

A MÉRETEKRE KIFEJEZÉSEKET IS LEHET HASZNÁLNI, A FENT LEÍRT EGÉSZEK HELYETT. PL. "MAT A = IDN(X, 2 * Y + 3)". KIÉRTEKELESKOR AZONBAN A KÉT KIFEJEZÉS ÉRTÉKÉNEK UGYANANNAK KELL LENNI, MERT CSAK NÉGYZETES EGYSÉG MÁTRIX VAN. E-ZEN FELÜL BIZTOSÍTANI KELL, HOGY KIFEJEZÉSEID EGESZ ÉRTÉKŰEK LEGYENEK A TÚLOLDALON LEÍRT OKOK MIATT.

MAT R = A * B

ÍME A "MAT" UTASÍTÁS NÉLKÜL VÉGREHAJTOTT "MÁTRIXSZORZÁS" EGYSZERŰ ILLUSZTRÁCIÓJA.

3 ELADÓ 4 TERMÉKET AD EL 3 HETI ELADÁSAIKAT ÍGY LEHET TÁBLÁZATBA FOGLALNI 3

ELADÓK	TERMÉKEK			
	A	B	C	D
KOVÁCS GÉZA	5	2	0	10
BÁN EDÉNE	3	5	2	5
HORVÁT ILONA	20	0	0	0

HETI ELADÁS : A(3,4) TÖMB

ÁRJEGYZÉK: B(4,2)

TERMÉK	ÁR	JUTALÉK
A	1.50	0.20
B	2.80	0.40
C	5.00	1.00
D	2.00	0.50

AZ ÁRJEGYZÉK ÉS AZ ELADÁSI JUTALÉK (PÉNZBEN ÉS NEM SZÁZALÉKOSAN) A KÖVETKEZŐ TÁBLÁZATOKAT ALKOTJÁK 3

ÍGY A BEVÉTEL 3

KOVÁCS GÉZA $5 \times 1.50 + 2 \times 2.80 + 0 \times 5.00 + 10 \times 2.00 = 33.10$

BÁN EDÉNE $3 \times 1.50 + 5 \times 2.80 + 2 \times 5.00 + 5 \times 2.00 = 38.50$

HORVÁT ILONA $20 \times 1.50 + 0 \times 2.80 + 0 \times 5.00 + 0 \times 2.00 = 30.00$

BELSO SZORZATOK

ÉS A MEGKERESETT JUTALÉKAIK 3

KOVÁCS GÉZA $5 \times 0.20 + 2 \times 0.40 + 0 \times 1.00 + 10 \times 0.50 = 6.80$

BÁN EDÉNE $3 \times 0.20 + 5 \times 0.40 + 2 \times 1.00 + 5 \times 0.50 = 7.10$

HORVÁT ILONA $20 \times 0.20 + 0 \times 0.40 + 0 \times 1.00 + 0 \times 0.50 = 4.00$

HA EZT GÉPESÍTENÉNK, A HETI ELADÁSOKAT AZ A(,) TÖMB TÁROLNÁ, AZ ÁRJEGYZÉK A B(,) TÖMB LENNE, ÉS AZ EREDMÉNYEK AZ R(,) TÖMBBÉ KERÜLNÉNEK A KINYOMTATÁST MEGELŐZŐEN. A PROGRAM A KÖVETKEZŐ LEHETNE 3

```

10 DIM A(3,4), B(4,2), R(3,2)
20 MAT R = ZER(3,2)
200 FOR I = 1 TO 2
210 FOR J = 1 TO 3
220 FOR K = 1 TO 4
230 LET R(J,I) = R(J,I) + A(J,K) * B(K,I)
240 NEXT K
250 NEXT J
260 NEXT I
    
```

UGYANAZOK AZ EREDMÉNYEK

1) 2)

R(1,	33.1	6.8
R(2,	38.5	7.1
R(3,	30	4

AZ R(,) OSZLOPAI LÁTHATÓAN A FENTI HOSSZADALMAS, KÉZI


MÓDSZERREL KAPOTT EREDMÉNYI TARTALMAZZÁK.

A 200. - TÓL A 260. SORIG TERJEDŐ PROGRAM-RÉSZLET EGYETLEN "MAT" UTASÍTÁSSAL HELYETTESÍTHETŐ :

100 MAT R = A * B

AZ A(,) SORAI A SKALÁR SZORZÁS SZABÁLYAI SZERINT SZORZÓDNAK ÖSSZE A B(,) OSZLOPAIVAL. MI KERÜL AZ R(r,c)-BE? AZ A(,) r-EDIK SORÁNAK ÉS B(,) c-EDIK OSZLOPÁNAK SKALÁR SZORZATA.

EZEN "MAT" UTASÍTÁS MŰKÖDÉSÉNEK SZÜKSÉGES FELTÉTELE A TÖMBÖK AKTUALIS MÉRETEINEK ÖSSZEFÉRHETŐSÉGE.



$$R(r,c) = A(r,s) * B(s,c)$$

A SOROK SZÁMA AZ EREDMÉNYBEN

UGYANAZ A BELSŐ MÉRET

AZ OSZLOPOK SZÁMA AZ EREDMÉNYBEN

A TÚLOLDALI PÉLDÁBAN A(,) DIMENZIÓJA (3,4) ÉS B(,) -NEK PEDIG (4,2) IGY A "BELSŐ" "4" A(,) -T ÉS B(,) -T ÖSSZEFÉRHETŐVÉ TESZI. TELJESEN HIBÁS LENNE AZONBAN A KÖVETKEZŐ :

110 MAT R = B * A 

A KÖZÖNSÉGES ALGEBRÁBAN AZ $a \times b = b \times a$ SZORZÁS "KOMMUTATÍV" A MÁTRIXALGEBRÁBAN VISZONT AZ IS LEHET, HOGY $b \times a$ NEM IS LÉTEZIK. A FENTI 110. SOR TELJESEN ÉRTELMETLEN.

FELTÉTELEZVE AZT, HOGY A(,) ÉS B(,) ÖSSZEFÉRHETŐ, AZ R(,) TÖMBÖT ÚGY DIMENZIONÁLJUK ÚJRA, HOGY ANNYI SORA LEGYEN, MINT A(,) PILLANATNYI ELSŐ MÉRETE, ÉS ANNYI OSZLOPA, MINT B(,) PILLANATNYI MÁSODIK MÉRETE. HA EZEK AZ ÉRTÉKEK R(,) -RE NÉZVE MEGHALADJÁK A "DIM" UTASÍTÁSBAN SZEREPLŐKET, AKKOR A RENDSZER HIBÁT JELEZ ÉS LEALL A VÉGREHAJTÁSSAL. AZ ÚJADIMENZIONÁLÁS MÁS KÖVETKEZMÉNYEIRŐL A 79. OLDALON OLVASHATUNK.

AZ UTASÍTÁS EGYSZERŰ FORMÁJÁN NEM VÁLTOZTATHATUNK, ÉS NEM ÁLLHAT UGYANAZ A TÖMB AZ EGYENLŐSÉGJEL MINDKÉT OLDALÁN.

120 MAT P = A * P 

INVERZIO

ELŐSZÖR IS EGY RÖVID
BEVEZETÉS AZ
"INVERZ" SZÓ JELEN-
TÉSÉBE.

KÖZÖNSÉGES ALGEBRAI EGYENLETET ÍGY ÍRHATUNK LE:

$$4.5 \cdot x = 37$$

A MEGOLDÁS A KÖVETKEZŐ:

$$\begin{aligned}x &= (4.5)^{-1} \cdot 37 \\x &= 0.2222 \cdot 37 \\x &= 8.222\end{aligned}$$

AZ ELJÁRÁSBAN AZ x EGYÜTTHATÓJÁNAK "RECIPROKÁT" KELL VENNI ÉS MEG KELL SZOROZNI A JOBB OLDALLAL. ÉRDEMES MEGFIGYELNI, HOGY AZ EREDETI EGYÜTTHATÓT, 4.5-T "INVERZÉVEL" MEGSZOROZVA 0.9999...-ET KAPUNK (IDEÁLIS ESETBEN PONTOSAN 1-ET).

A MÁTRIXALGEBRÁBAN HASONLÓAN KÖZELÍTJÜK MEG AZ EGYENLETREND-
SZEREKET. TEKINTSÜK AZ ALÁBBI HÁROM EGYENLETET:

$$\begin{aligned}15x + 10y + 5z &= 3.1 \\12x + 24y + 8z &= 4.5 \\6x + 0y + 36z &= 6.3\end{aligned}$$

MÁTRIXALAKBAN A KÖVETKEZŐKÉPPEN ÍRHATÓK FEL.

$$\begin{bmatrix} 15 & 10 & 5 \\ 12 & 24 & 8 \\ 6 & 0 & 36 \end{bmatrix} * \begin{bmatrix} x \\ y \\ z \end{bmatrix} = \begin{bmatrix} 3.1 \\ 4.5 \\ 6.3 \end{bmatrix}$$

(HA, AZ ELSŐ MÁTRIX 1. SORÁNAK ÉS A MÁSODIK 1. OSZLOPÁNAK SKALÁRSZORZATÁT KÉPZED, $15 \cdot x + 10 \cdot y + 5 \cdot z$ -T KAPSZ EREDMÉ-
NYÜL, AMINT AZT A SKALÁRSZORZATOKKAL KAPCSOLATBAN A 88.
OLDALON ELMONDTUK.)

AZ EGYENLETEK MEGOLDÁSA A KÖVETKEZŐ:

$$\begin{bmatrix} x \\ y \\ z \end{bmatrix} = \begin{bmatrix} 15 & 10 & 5 \\ 12 & 24 & 8 \\ 6 & 0 & 36 \end{bmatrix}^{-1} * \begin{bmatrix} 3.1 \\ 4.5 \\ 6.3 \end{bmatrix}$$

AHOL A " -1 " KITEVŐ A MÁTRIX FELSŐ INDEXEKÉNT A MÁTRIXNAK AZ
INVERZÉT JELÖLI. EZ (AHOGYAN MAJD MEGMUTATJUK) A KÖVETKEZŐ LESZ:

$$\begin{bmatrix} x \\ y \\ z \end{bmatrix} = \begin{bmatrix} .1029 & -.0429 & -.00476 \\ -.0457 & .0607 & -.00714 \\ -.0171 & .00714 & .0286 \end{bmatrix} * \begin{bmatrix} 3.1 \\ 4.5 \\ 6.3 \end{bmatrix}$$

ENNEK ALAPJÁN BÁRMELY JOBB OLDAL ESETÉN MEGKAPHATOD A MEGOL-
DÁSOKAT MÁTRIX SZORZÁST ALKALMAZVA. PL. y (2. SOR 1. OSZLOP) KIFE-
JEZÉSE: $y = -0.0457 \cdot 3.1 + 0.0607 \cdot 4.5 - 0.00714 \cdot 6.3 = 0.4598$

AZÉRT HOGY TELJESSÉ TEGYÜK A KÖZÖNSÉGES ALGEBRÁVAL FENNÁLLÓ ANALÓGIÁT, SZOROZZUK MEG AZ EGYÜTTHATÓK EREDETI MÁTRIXÁT SAJÁT INVERZÉVEL:

$$\begin{bmatrix} 15 & 10 & 5 \\ 12 & 24 & 8 \\ 6 & 0 & 36 \end{bmatrix} * \begin{bmatrix} .1029 & -.0429 & -.00476 \\ -.0457 & .0607 & -.00714 \\ -.0171 & .00714 & .0286 \end{bmatrix} = \begin{bmatrix} 1.001 & -.0008 & -.0008 \\ .0012 & .9991 & .00032 \\ .0018 & -.00036 & 1.001 \end{bmatrix}$$

EZ LÁTHATÓAN KÖZEL ESİK AZ EGYSÉGMÁTRIXHOZ (IDEÁLIS ESETBEN AZ EREDMÉNY PONTOSAN AZ EGYSÉGMÁTRIX LENNE, AZONBAN MI CSAK 4 ÉRTÉKŰ JEGYRE SZÁMOLTUNK).

TEHÁT A FENT BEMUTATOTT INVERZET ÚGY KAPJUK MEG, HOGY 3 EGYENLETRENDSZERT OLDUNK MEG, AMELYNEK JOBB OLDALAI AZ EGYSÉGMÁTRIX OSZLOPAI. AZ INVERTÁLÁS TULAJDONKÉPPEN AZ EGYSÉGMÁTRIX ÉS AZ EGYÜTTHATÓMÁTRIX "HÁNYADOSÁNAK" MEGHATÁROZÁSA.

$$\begin{aligned} 15 \times X + 10 \times Y + 5 \times Z &= 1 \text{ ILL. } 0, \text{ ILL. } 0 \\ 12 \times X + 24 \times Y + 8 \times Z &= 0 \quad 1 \quad 0 \\ 6 \times X + 0 \times Y + 36 \times Z &= 0 \quad 0 \quad 1 \end{aligned}$$

A SZÁMÍTÁST ÚGY VÉGEZZÜK EL, AHOGY AZ ISKOLÁBAN TANULTUK, ILL. TALÁN SZIGORÚBB SORRENDEN. ELŐSZÖR X EGYÜTTHATÓT KÜSZÖBÖLJÜK KI A 2. ÉS 3. EGYENLETBŐL. SZOROZZUK MEG A 2. EGYENLETET (EQ.2.-T) 15-TEL, ÖSSZUK EL 12-VEL, ÉS VONJUK LE EZT AZ ÁTALAKÍTOTT EGYENLETET AZ ELSŐ EGYENLETBŐL (EQ.1.-BŐL). ÍGY EGY ÚJ EQ.2.-T KAPUNK. HASONLÓ MÓDON ALAKÍTHATUNK KI EGY ÚJ EQ.3.-T ÚGY, HOGY MEGSZOROZZUK EQ.3.-T 15-TEL ELÖSZTJÜK 6-TAL, ÉS EZT KIVONJUK EQ.1.-BŐL.

$$\begin{aligned} EQ2 = EQ1 - (15/12) \times EQ2 \dots & \begin{aligned} 15 \times X + 10 \times Y + 5 \times Z &= 1 & 0 & 0 \\ 0 \times X - 20 \times Y - 5 \times Z &= 1 & -1.25 & 0 \end{aligned} \\ EQ3 = EQ1 - (15/6) \times EQ3 \dots & \begin{aligned} 0 \times X + 10 \times Y - 85 \times Z &= 1 & 0 & -2.5 \end{aligned} \end{aligned}$$

MOST PEDIG KÜSZÖBÖLJÜK KI AZ ÚJ EGYENLET 3. Y-JÁNAK EGYÜTTHATÓJÁT. SZOROZZ MEG A 3. EGYENLETET -20-SZAL, OSZD EL 10-ZEL, ÉS VOND KI A 2. EGYENLETBŐL.

$$\begin{aligned} EQ3 = EQ2 - (-20/10) \times EQ3 \dots & \begin{aligned} 15 \times X + 10 \times Y + 5 \times Z &= 1 & 0 & 0 \\ 0 \times X - 20 \times Y - 5 \times Z &= 1 & -1.25 & 0 \\ 0 \times X + 0 \times Y - 175 \times Z &= 3 & -4.25 & -5 \end{aligned} \end{aligned}$$

ÍGY AZ EGYÜTTHATÓK EREDETI MÁTRIXÁNAK ÁTLÓ ALATTI ÖSSZES ELEME NULLA. MOST "HELVETTESÍTSD VISSZA" A 3. EGYENLETNÉL KEZDVE! Z: A 3. EGYENLET ISMERETLENJE, ÍGY OSZD EL A JOBB OLDALAKAT -175-TEL!

$$Z = -.0171 \quad .00714 \quad .0286$$

MOST A 2. EGYENLET JOBB OLDALÁNAK MINDEN TAGJÁBÓL VOND LE Z MEGFELELŐ ÉRTÉKÉNEK (-5)-SZÖRÖSÉT, ÉS OSZD EL Y EGYÜTTHATÓJÁVAL, AMI (-20)! (PL. AZ ELSŐ $[1 - (-5) \times (-.0171)] \div (-20) = -.0457$)

$$Y = -.0457 \quad .0607 \quad -.00714$$

VÉGÜL AZ 1. EGYENLET JOBB OLDALÁNAK MINDEN TAGJÁBÓL VOND LE A Z MEGFELELŐ ÉRTÉKÉNEK 5-SZÖRÖSÉT, Y MEGFELELŐ ÉRTÉKÉNEK 10-SZERESÉT, ÉS OSZD EL 15-TEL (PL. AZ ELSŐ $[1 - 5 \times (-.0171) - 10 \times (-.0457)] \div 15 = .1029$)

$$X = .1029 \quad -.0429 \quad -.00476$$

ÍGY KAPJUK MEG A FORDÍTOTT SORRENDEN KISZÁMOLT INVERZ HÁROM SORÁT.

MAT A = INV(B)

10 DIM A(3,3), B(3,3), I(3,3)

	1)	2)	3)
A(1,			
A(2,			
A(3,			

	1)	2)	3)
B(1,	15	10	5
B(2,	12	24	8
B(3,	6	0	36

ME EGY RUTIN, AMELYIK EGY B(,)-BEN TÁROLT N*N DIMENZIÓJÚ MÁTRIXOT INVERTÁL: AZ INVERZ A(,)-BAN ÉPÜL FEL. A HASZNALT MÓDSZERT AZ ELŐZŐ OLDAL PÉLDAJÁBAN ISMERTÜK MEG.

```

200 REM A(,) = INVERZ B(,); N x N-ES
210 FOR I = 1 TO N
220 FOR J = 1 TO N
230 LET A(I,J) = 1 - ABS(SGN(I-J))
240 NEXT J
250 NEXT I

260 FOR I = 1 TO N-1
270 IF ABS(B(I,I)) > 0.00001 THEN 300
280 PRINT "NINCS MEGOLDÁS"; B(I,I)
290 STOP
300 FOR J = I+1 TO N
310 LET X = B(J,I) / B(I,I)
320 FOR K = 1 TO N
330 LET B(J,K) = B(J,K) - X * B(I,K)
340 LET A(J,K) = A(J,K) - X * A(I,K)
350 NEXT K
360 NEXT J
370 NEXT I

380 FOR I = N TO 1 STEP -1
390 FOR K = 1 TO N
395 IF I = N THEN 430
400 FOR J = I+1 TO N
410 LET A(I,K) = A(I,K) - B(I,J) * A(J,K)
420 NEXT J
430 LET A(I,K) = A(I,K) / B(I,I)
440 NEXT K
450 NEXT I
    
```

KÉPEZZ A(,)-BAN EGYSÉG MÁTRIXOT!

1	0	0	0
0	1	0	0
0	0	1	0

HÁROMSZÖGESÍTSD B(,)-T!

m	n	o	p	q
o	m	n	o	p
o	o	m	n	o
o	o	o	m	n
o	o	o	o	m

"HELYETTESÍTS VIGSZA"!

DE EZ EGYETLEN "MAT" UTASÍTÁSSAL EGYSZERŰBEN MEGOLDHATÓ:

100 MAT A = INV(B)

HA A MÁTRIXNAK NINCSEN INVERZE (HA "SZINGULÁRIS"), NEMELY BASIC RENDSZER HIBÁT JELEZ ÉS LEÁLLÍTJA A VÉGREHAJTÁST: MÁSOEK VISZONT NEM JELZIK A HIBÁT, DE OLYAN LEHETŐSÉGET NYÚJTANAK, AMELYNEK SEGÍTSÉGÉVEL A MÁTRIX DETERMINÁNSÁT MEG LEHET VIZSGÁLNI. A NULLA DETERMINÁNS SZINGULÁRIS MÁTRIXOT JELENT. (EGY DETERMINÁNST A 43. OLDALON LÁTHATUNK)

```

110 LET      D = DET (B)
120 IF      D > .001 THEN 150
130 PRINT   "SZINGULARIS MÁTRIX"
140 STOP

```

FOLYTASD?

A "DET" ÁLTAL SZOLGÁLTATOTT ÉRTÉK A PROGRAM ÁLTAL UTOLSÓKÉNT INVERTÁLT (VAGY UTOLSÓKÉNT MEGKISÉRELT) MÁTRIX DETERMINÁNSÁNAK AZ ÉRTÉKE. NEM MINDEN BASIC RENDSZER RENDELKEZIK AZONBAN "DET" UTASÍTÁSSAL.

A NEM NÉGYZETES MÁTRIXNAK NINCSEN INVERZE ∞ ENNEK KISZÁMÍTÁSÁRA TETT KISÉRLETEK HIBÁKNAK MINŐSÜLNEK. A LEGTÖBB BASIC RENDSZER NEM HAJTJA VÉGRE A MÁTRIX SAJÁT INVERZÉVEL TÖRTÉNŐ HELYETTESÍTÉSÉT.

```

150 MAT     A = CON (2,3)
160 MAT     B = INV (A)
170 MAT     B = INV (B)

```

NEM NÉGYZETES

MINDKÉT OLDALON UGYANAZ A NÉV

AZ INVERZÍÓ PONTOSSÁGÁRÓL ÚGY GYŐZŐDHETSZ MEG, HOGY KINYOMTATOD AZ EREDETI MÁTRIX ÉS INVERZÉNEK SZORZATÁT. DEFINÍCIÓ SZERINT ENNEK PONTOSAN AZ EGYSÉG MÁTRIXNAK KELL LENNIE, AZONBAN ÁLTALÁBAN ELTÉR ETTŐL, AZ ELŐZŐKÉBEN BEMUTATOTT "KEREKÍTÉSI HIBÁK" MIATT. (EKKOR 4 SZIGNIFIKÁNS SZÁMJEGYET HASZNÁLTUNK.)

```

180 MAT     A = INV (B)
190 MAT     I = B * A
195 MAT     PRINT I

```

A "MAT PRINT" UTASÍTÁS LEÍRÁSA A 98. OLDALON TALÁLHATÓ

A "DIM" UTASÍTÁSBAN AZ A(,) MÉRETEINEK LEGALÁBB OLYAN NAGYNAK KELL LENNIÜK, MINT A FENTI 180. SORBAN SZEREPLŐ B(,) PILLANATNYI DIMENZIÓI. AZ ÚJRADIMENZIONÁLÁS KÖVETKEZMÉNYEIRŐL A 79. OLDALON SZÓLTUNK.

AZ ELŐZŐ OLDAL 200. SORÁTÓL KEZDŐDŐ RUTIN HATÁSÁBAN ELTÉR A "MAT A = INV (B)"-TÓL. ELŐSZÖR IS A B(,) TÖMBÖT TELJESEN ÖSSZEZAGYVÁLJA & A "MAT" UTASÍTÁSNAK NEM SZABAD EZT TENNIE. (HA A FENTI 180. SORNÁL KEZDŐDŐ TESZT-PROGRAMRÉSzt LEFUTTATOD, HAMAR KIDERÜL, HOGY A BASIC VÁLTOZATOD B(,)-T MÓDOSÍTTA-E VAGY SEM.) MÁSOADSZOR, AZ ELŐZŐ OLDALI RUTIN CSAK AKKOR MŰKÖDIK JÓL, HA A B(,) FŐÁTLÓJÁBAN LÉVŐ SZÁMOK MINDEGYIKE KB. UGYANAKKORA, AZ ÁTLÓN KIVÜLIEK PEDIG ENNÉL KISEBBEK. (EGYÁLTALÁN NEM FOG MŰKÖDNI JÓL, HA A B(,) NULLA.)

A "MAT" UTASÍTÁSNAK A LEGJOBB OSZTÓKAT KELL KIVÁLASZTANIA, NEMCSAK A FŐÁTLÓBAN LEVŐT HASZNÁLNI, AHOGY EZT A PROGRAM 310. SORÁBAN AZ EGYSZERŰSÉG KEDVÉÉRT TETTÜK.

MAT READ

EZ AZ UTASÍTÁS UGYAN-
ABBÓL A "DATA" UTASÍTÁSBÓL OLVAS BE,
MINT A KÖZÖNSÉGES "READ"

10	DIM	A(2,3), B(3,1), C(1,3)
20	DATA	1.5, 2.3, 3.2, 4.6, 5.7, 6.5, 7.1
30	DATA	8.9, 9.0, 11.7, 12.6, 13, 14, 15

	1)	2)	3)
A(1,	1.5	2.3	3.2
A(2,	4.6	5.7	6.5

	1)
B(1,	7.1
B(2,	8.9
B(3,	9.0

	1)	2)	3)
C(1,	11.7	12.6	13

TEGYÜK FEL, HOGY AZ A(,) TÖMBÖT SORONKÉNT AKARJUK FELTÖLTENI A 20. SORSZÁMÚ UTASÍTÁS ADATSOROZATÁBÓL. ÍGY IS LEHETNE CSINÁLNI :

```

200 FOR R = 1 TO 2
210 FOR C = 1 TO 3
220 READ A(R,C)
230 NEXT C
240 NEXT R
    
```

AZ EGÉSZET EGYETLEN "MAT" UTASÍTÁSSAL A KÖVETKEZŐKÉPP OLDHATOD MEG:

100 MAT READ A

A(,)SORONKÉNT
KERÜL BEOLVAGÁSRA

VALÓJÁBAN AKÁRHÁNY TÖMBÖT MEGTÖLTHETSZ (SORONKÉNT) EGYETLEN "MAT" UTASÍTÁST HASZNÁLVA :

100 MAT READ A, B, C

MINDIG VAN VEGSZŐ
A LISTÁBAN

AZ A(,), B(,) ÉS C(,) TÖMBÖK TARTALMÁT A LAP JETEJÉN MÁR BEMUTAT-
TUK. FIGYELJÜK MEG, HOGY AZ A, A B ÉS A C BETŰKNEK SEMMI KÖZÜK AZ
A, B ÉS C EGYSZERŰ VÁLTOZÓKHOZ!

TÖMBÖKET RÉSZLEGESEN ÚGY TÖLTHETSZ MEG, HOGY A "MAT" UTASÍTÁSBAN
ÚJ MÉRETEKET HATÁROZOL MEG, FELTÉVE HOGY EZEK NEM NAGYOBBAK, MINT
A "DIM" UTASÍTÁSBAN SZEREPLŐK.

100 MAT READ A(2,2), B, C(1,1)

	1)	2)	
A(1,	1.5	2.3	
A(2,	3.2	4.6	

	1)
B(1,	5.7
B(2,	6.5
B(3,	7.1

	1)	
C(1,	8.9	

HA, A "MAT READ"-BÓL KIHAGYOD A MÉRETEKET, A RENDSZER A TÖMB AK-
TUÁLIS DIMENZIÓIT HASZNÁLJA ☞ EZEK LEHET, HOGY KISEBBEK, MINT A
"DIM" UTASÍTÁSBELIEK.

100	MAT	A = ZER (1, 1)
110	MAT	READ A

A FENTI KÉT UTASÍTÁS TEHÁT CSUPÁN 1 SZÁMOT, OLVAS BE AZ EGYSZER EGYES A(,) TÖMBBE. AZ ILYEN ÚJRADIMENZIONÁLÁS KÖVETKEZMÉNYEIT A 79. OLDAL TÁRGYALJA.

A "MAT" UTASÍTÁSBAN DIMENZIÓKÉNT VÁLTOZÓKAT ÉS KIFEJEZÉSEKET IS FELHASZNÁLHATSZ ?

10	DIM	E (10, 10)	DIMENZIÓK DATAKÉNT
20	DATA	2, 3	
30	DATA	1.1, 2.2, 3.3, 4.4, 5.5, 6.6	KIFEJEZÉSEK ITT MEGENGEDETEK
40	READ	R, C	
50	MAT READ	E (R, C)	

DE MINDIG BIZTOSÍTSUK (PL. "INT" UTASÍTÁST HASZNÁLVA), HOGY A KIFEJEZÉSEK EGÉSZ ÉRTÉKEKET ADJANAK, MERT NÉMELYIK BASIC RENDSZER AZ EREDMÉNYHEZ LEGKÖZELEBB ESŐ EGÉSZET, MÁSONK ANNAK EGÉSZ RÉSZÉT HASZNÁLJÁK FEL.

A "DATA" UTASÍTÁSOKBAN CSUPÁN EGY ADATSOROZAT VAN, ÉS MINDEN "READ" VAGY "MAT READ" A VÉGREHAJTÁSRA EBBŐL ANNYIT HASZNÁL FEL, AMENNYIRE SZÜKSÉGE VAN. A "RESTORE" UTASÍTÁS AZ OLVASÁST AZ ELSŐ SOR ELEJÉRE IRÁNYÍTTJA, AHOGY EZT A 17. OLDALON ELMAGYARÁZTUK.

A "MAT READ" UTASÍTÁS IGEN HASZNOS "ÁLLAPOTTÁBLÁZATOK" FELÁLLÍTÁSÁHOZ OLYAN PROGRAMOKBAN, AMELYEK ILYEN ESZKÖZÖKET HASZNÁLNAK FEL. ERRE EGY TELJES PÉLDA A 102. OLDALON TALÁLHATÓ.

MAT INPUT

EZ AZ UTASÍTÁS
ADAT BEOLVASÁSA-
KOR TELJES TÖM-
BÖKET KÉR BE.

```
10 DIM A(3,3)
```

	1)	2)
A(1,	1.5	2.6
A(2,	3.7	4.8

MOST EGY MÓDSZERT MUTATUNK ARRRA, HOGYAN LEHET MEGTÖLTENI EGY TÖMBNEK EGY BIZONYOS RÉSZÉT A BILLENTYŰZETRŐL BEKÉRT ADATOKKAL.

```
200 PRINT "ÍRD BE A MÉRETEKET"  
210 INPUT R, C  
220 FOR I = 1 TO R  
230 PRINT "ÍRD BE A"; I; " SORT, SORONKÉNT 1 SZÁMOT"  
240 FOR J = 1 TO C  
250 INPUT A(I, J)  
260 NEXT J  
270 NEXT I  
280 END
```

RUN

ÍRD BE A MÉRETEKET

? 2,2

ÍRD BE A 1. SORT, SORONKÉNT 1 SZÁMOT

? 1.5

? 2.6

ÍRD BE A 2 SORT, SORONKÉNT 1 SZÁMOT

? 3.7

? 4.8

ÜGYANEZT MÁS MÓDON A "MAT INPUT" UTASÍTÁS FELHASZNÁLÁSÁVAL IS MEGOLDHATTAD VOLNA:

```
100 PRINT "ÍRD BE A MÉRETEKET"  
110 INPUT R, C  
120 PRINT "ÍRD BE A"; R; "SOR"; C; "DARAB ELEMÉT"  
130 MAT INPUT A(R, C)  
140 END
```

RUN

ÍRD BE A MÉRETEKET


? 2,2

ÍRD BE A 2 SOR 2 DARAB ELEMÉT

? 1.5 ; 2.6

? 3.7 ; 4.8

A A BILLENTYŰZETRŐL AKARSZ ADATOKAT BEKÉRNI, JOBB, HA EGY SAJÁT INPUT RUTINT ÍRSZ (MINT AMILYEN, A 200. SORNÁL KEZDŐDIK), MINTHOGY A "MAT INPUT" UTASÍTÁST HASZNÁLD. ENNEK HÁROM OKA VAN :

A 230. SORHÓZ HASONLÓ HASZNOS KÖZBENSŐ ÜZENETEKET KÜLDHETSZ KI  EZ NEM LEHETSÉGES A "MAT INPUT" HASZNÁLATÁVAL.

A BEVITELNÉL MINDEN SZÁM ÉRTÉKÉT ELLENŐRIZHETED ÉS MÉG A TELJES TÖMB BEVITELE ELŐTT KÖZBELEPHETSZ. PL. HA TUDOD, HOGY MINDEN ELEMNEK KISEBBNEK KELL LENNIE ± 10 -NÉL :

```
252 IF ABS(A(1, J)) < 10 THEN 260
254 PRINT "TARTOMÁNYON KÍVÜL ; ÍRD ÚJRA A " ; J ; "OSZLOPOT"
256 GO TO 250
```

ÉS EZEKET BESZÚRHATOD A TÚLOLDALI RUTINBA.

A SORONKÉNT CSUPÁN EGY SZÁM GÉPELÉSE MEGOLDJA AZT A PROBLÉMÁT, AMI AKKOR ADÓDNA, HA AZ EGÉSZ SZÁMSOROZAT NEM FÉR KI EGY SORBA. NÉMELYIK BASIC RENDSZER MÁS MEGOLDÁSOKAT NYÚJT ; MÁS BASIC RENDSZER MEGENGEDI A & JELET A SOR VÉGÉN, AMIAZT JELENTI ; "NEM FEJEZTEM BE A SORT".

ÍGY A HORDOZHATÓSÁG KEDVÉÉRT, HA EGY MÓD VAN RÁ, NE HASZNÁLD A "MAT INPUT" UTASÍTÁST AZ ADATOK BILLENTYŰZETRŐL TÖRTÉNŐ BEKÉRÉSÉNÉL ! AZ UTASÍTÁS MÉG A KÉSŐBBIEKBEN BEMUTATÁSRA KERÜLŐ MÁS CÉLOKRA HASZNOS LESZ.

A "MAT INPUT" UTASÍTÁST KÖVETŐ LISTA AKÁRHÁNY TÖMBNEVET TARTALMAZHAT ; E NEVEK BÁRMELYIKÉT DIMENZIÓK KÖVETHETIK. MINT A "MAT READ" ESETÉBEN A DIMENZIÓK LEHETNEK EGÉSZ SZÁMOK, VÁLTOZÓK VAGY KIFEJEZÉSEK, FELTÉVE, HOGY ÉRTÉKEIK NEM HALADJÁK MEG A "DIM" UTASÍTÁSBELIEKET. HA KIFEJEZÉSEKET HASZNÁLSZ MÉRETKÉNT, BIZTOSÍTSD, HOGY EZEK MINDIG EGÉSZ ÉRTÉKET EREDMÉNYEZZENEK, MERT NÉMELYIK BASIC AZ EREDMÉNYHEZ LEGKÖZELEBB ESŐ EGÉSZET, MÁSOK PEDIG AZ ÉRTÉK EGÉSZ RÉSZÉT HASZNÁLJÁK FEL. AZ ÚJRADIMENZIONÁLÁS MÁS KÖVETKEZMÉNYEIRŐL A 79. OLDALON OLVASHATUNK.

```
150 MAT INPUT A, B(2, 1), C(2 * X, Y)
```

MINDIG VESSZŐ

MAT PRINT

EZ AZ UTASÍTÁS
A TELJES TÖM-
BÖK KINYOMTATÁSÁT
EREDMÉNYEZI.

```

10 DIM A(3,3), B(2,3), C(2,3)
20 MAT A = IDN(2,2)
30 MAT B = CON
40 MAT C = (-1.5) * B
    
```

	1)	2)	
A(1,	1	0	
A(2,	0	1	

	1)	2)	3)
B(1,	1	1	1
B(2,	1	1	1

	1)	2)	3)
C(1,	-1.5	-1.5	-1.5
C(2,	-1.5	-1.5	-1.5

A KÖVETKEZŐ RUTIN SORONKÉNT NYOMTATJA KI AZ A(,) TÖMBBEN TÁROLT 2x2-ES MÁTRIXOT.

```

200 PRINT
210 FOR R = 1 TO 2
220 FOR C = 1 TO 2
230 PRINT A(R,C);
240 NEXT C
250 PRINT
260 PRINT
270 NEXT R
    
```

ÚJ SORT KEZD

; BIZTOSÍTTJA, HOGY AZ EGY SORBAN LEVŐ ELEMÉK EGYMÁS MELLE ÍRÓDJANAK

ÚJ SORT KEZD

A SOROK KÖZÖTT EGY ÜRES SOR

A FELADAT EGYETLEN "MAT" UTASÍTÁSSAL IS MEGOLDHATÓ:

```

100 MAT PRINT A;
    
```

AZ EREDMÉNYEKET SZOROSAN EGYMÁS MELLE NYOMTATJA

UGYANCSAK EGY "MAT" UTASÍTÁSSAL TÖBB TÖMBÖT IS KI LEHET IRATNI. A KIÍRÁSI FORMA RÉSZLETES MAGYARÁZATÁT A KÖVETKEZŐ OLDALON TALÁLHATJUK.

```

110 MAT PRINT A; C,
    
```

C(,) -T ZÓNÁKBAN NYOMTATJA

A TÖMBÖK SORONKÉNT KERÜLNEK KINYOMTATÁSRA. (HA SOROKKÉNT AKAROD KIIRATNI EGY TÖMB OSZLOPAIT, HASZNÁLD ELŐSZÖR A "MAT B=TRN(A)" UTASÍTÁST!) A TÖMBÖK AKTUÁLIS MÉRETEIKNEK MEGFELELŐEN KERÜLNEK KINYOMTATÁSRA. EZÉRT AZ A(,) TÖMBNEK KIÍRÁSKOR 2 SORA LESZ, AMELYNEK MINDEGYIKE 2 ELEM BŐL ÁLL: EZT A MÉRETET A FENTI 20. SORBAN ADTUK MEG. A C(,) TÖMBNEK 2 SORA LENNE, MINDEGYIKBEN 3 ELEMEL.

ME ÍRJUNK MÉRETEKET AZ UTASÍTÁSBA? A RENDSZER ISMERI MINDEN TÖMB AKTUÁLIS MÉRETÉT, ÉS NEM SZERETI, HA ERRE ÚJRA EMLÉKEZTETIK.

```

120 MAT PRINT A(2,2); C
    
```

ALAPFELTÉTELEZÉSKÉNT IDE VESSZŐT GONDOLUNK!

A KINYOMTATOTT OLDAL FORMÁJA AZ ALÁBBI SZABÁLYOKAT KÖVETI. BÁR NÉMELYIK BASIC RENDSZER KISEBB ELTÉRÉSEKET MUTATHAT.

A TÖMB MINDEN SORA A NYOMTATÁSBAN ÚJ SORBAN KEZDŐDIK.

A TÖMB MINDEN KINYOMTATOTT SORÁT EGY ÜRES SOR KÖVETI.


HA A "MAT PRINT" UTÁN A LISTÁBAN A TÖMB NEVÉT PONTOSVESSZŐ KÖVETI, AKKOR A SZÁMOK KÖZVETLENÜL EGYMÁS UTÁN KERÜLNEK KINYOMTATÁSRA; HA A NEVET VESSZŐ KÖVETI, AKKOR A SZÁMOK ZÓNÁBAN (ÁLTALÁBAN 15 KARAKTER SZÉLESSÉGBEN) ÍRÓDNAK KI. A PONTOSVESSZŐK ÉS VESSZŐK PONTOS HATÁSÁT EBBEN AZ ÖSSZEFÜGGÉSBEN A 29. OLDALON TÁRGYALJUK.

HA A TÖMB EGY SORA NEM FÉR KI EGY KINYOMTATOTT SORBA (ÁLTALÁBAN 72 KARAKTER), AKKOR A KÖVETKEZŐ SORBAN FOLYTATÓDIK.

HA A "MAT PRINT" UTASÍTÁSBAN FELSOROLT UTOLSÓ NÉV UTÁN NINCS ÍRÁSJEL, AKKOR A BASIC RENDSZER EZT VESSZŐVEL PÓTOLJA. (A LISTA ELEMÉIT ELVÁLASZTÓ ÍRÁSJELEK SZEREPE LÉNYEGBEVÁGÓ.)

LEHET, HOGY ÚGY TALÁLOD, HOGY AZ UTASÍTÁSSAL BEVEZETETT MEREV FORMA KEVÉS LEHETŐSÉGET AD TETSZETŐS OLDAL-ELRENDEZÉSEK MEGTERVEZÉSÉRE. AZONBAN AZ UTASÍTÁS NAGYON HASZNOS AZ ÚJ PROGRAMOK FEJLESZTÉSÉNÉL.

A 110. SOR ÁLTAL GENERÁLT KIMENET, A TÚLOLDAL ELEJÉN SZEREP-LŐ TÖMBÖKKEL A KÖVETKEZŐ LENNE:





TELJES PÉLDAPROGRAMOK

MCDXCII



PROGRAM A RÓMAI SZÁMOK ISMERT ARAB SZÁMOKKÁ VALÓ ÁTALAKÍTÁSÁRA.

EZ A PÉLDA A PROGRAMOZÁS SZOKÁSOS ESZKÖZÉT, AZ ÁLLAPOTTÁBLÁZAT HASZNÁLATÁT MUTATJA BE.

TEGYÜK FEL, HOGY MINDEN ÉRVÉNYES RÓMAI SZÁM A KÖVETKEZŐ ELEMEBŐL ÁLL SOHASEM HASZNÁL EGYNÉL TÖBBET AZ ALÁBBI DOBOZOK EGYIKÉBŐL SEM :

<p>EZRESEK</p> <p>M = 1000 MM = 2000 MMM = 3000</p> <p>TELSZÓLÉGES FELSO HATAR</p>	<p>SZÁZASOK</p> <p>D = 500 C = 100 DC = 600 CC = 200 DCC = 700 CCC = 300 DCCC = 800 CD = 400 CM = 900</p>	<p>TIZESEK</p> <p>L = 50 X = 10 LX = 60 XX = 20 LXX = 70 XXX = 30 LXXX = 80 XL = 40 XC = 90</p>	<p>EGYESEK</p> <p>V = 5 I = 1 VI = 6 II = 2 VII = 7 III = 3 VIII = 8 IV = 4 IX = 9</p>
---	--	--	---

ÚGY TŰNIK, HOGY A KLASSZIKUS RÓMA RITKÁN HASZNÁLTA A IV-BEN REJLŐ KIVONÁSI ÉLVEZT INKÁBB A IIII-ET KEDVELTÉK, PROGRAMUNK AZONBAN NÉGY AZONOS, EGYMÁST KÖVETŐ BETŰ KEZELÉSÉT MEGTAGADJA

A PROGRAM LOGIKÁJÁT A KÖVETKEZŐ ÁLLAPOTTÁBLÁZAT FOGLALJA ÖSSZE :

		"SZIMBÓLUM"							
		M	D	C	L	X	V	I	
"ÁLLAPOT"	KEZDETI ÁLLAPOT →	01	1000 & 02	500 & 03	100 & 09	50 & 05	10 & 10	5 & 07	1 & 11
		02	1000 & 02	500 & 03	100 & 09	50 & 05	10 & 10	5 & 07	1 & 11
		03	HIBA	HIBA	100 & 09	50 & 05	10 & 10	5 & 07	1 & 11
		04	HIBA	HIBA	100 & 04	50 & 05	10 & 10	5 & 07	1 & 11
		05	HIBA	HIBA	HIBA	50 & 06	10 & 10	5 & 07	1 & 11
		06	HIBA	HIBA	HIBA	HIBA	10 & 06	5 & 07	1 & 11
		07	HIBA	HIBA	HIBA	HIBA	HIBA	5 & 08	1 & 11
		08	HIBA	HIBA	HIBA	HIBA	HIBA	HIBA	1 & 08
		09	800 & 05	300 & 05	100 & 04	50 & 06	10 & 10	5 & 08	1 & 11
		10	HIBA	HIBA	80 & 07	30 & 07	10 & 06	5 & 08	1 & 11
		11	HIBA	HIBA	HIBA	HIBA	8 & 00	3 & 00	1 & 08

VEGYÜK PÉLDAKÉNT A CIX RÓMAI SZÁMOT: KEZDJÜNK A NULLÁS ÉRTÉKKEL! A 01-ES ÁLLAPOTBAN VAGY (AHOVA A NYIL MUTAT) A C SZIMBÓLUM ALATT TALÁLJUK A 100 & 09-ET, AMELY AZT MONDJA "ADJ 100-AT AZ ÉRTÉKHEZ ÉS MENJ 09-ES ÁLLAPOTBA". TEHÁT ADJ 100-AT A 0-HOZ ÉS VIDD A KIS NYILAT A 09-HEZ! MOST AZ I SZIMBÓLUM ALATT 1 & 11-ET TALÁLUNK; TEHÁT ADJ AZ ÉRTÉKHEZ EGYET (100+1=101) ÉS VIDD A KIS NYILAT A 11-ES ÁLLAPOTHOZ! VÉGÜL AZ X SZIMBÓLUM ALATT 8 & 00 ÁLL, ÍGY AZ ÉRTÉKHEZ 8-AT ADJ HOZZÁ! (101+8=109) A 00 AZT JELENTI HOGY VÉGEZTÉL.

A TÁBLÁZAT RÉSZÉ A PROGRAMNAK, FELÉPÍTÉSÉT AZ ALÁBBIKBAN MUTATJUK BE. HELYMEGTAKARÍTÁS CÉLJÁBÓL A(,) MINDEN ELEMÉ MAGÁBAN FOG-LALJA MIND A HOZZÁADANDÓ SZÁMOT, MIND PEDIG AZ ÚJ ÁLLAPOT KÓDJÁT. EZÉRT 5 & 07-BŐL $100 * 5 + 07 = 507$, LESZ. AZ "ERROR VAGY HIBA" BEJEGYZÉSEK -1-KÉNT KERÜLNEK A TÁBLÁZATBA.

10	REM	RÓMAI SZÁMOK DEKÓDOLÁSA						
20	DIM	A (11,7), C\$(7)						
30	REM	MINDEN ELEM = 100 * ÖSSZEADÁS + ÚJ ÁLLAPOT						
100	REM	M,	D,	C,	L,	X,	V,	I
110	DATA	100002,	50003,	10009,	5005,	1010,	507,	111
120	DATA	100002,	50003,	10009,	5005,	1010,	507,	111
130	DATA	-1,	-1,	10009,	5005,	1010,	507,	111
140	DATA	-1,	-1,	10004,	5005,	1010,	507,	111
150	DATA	-1,	-1,	-1,	5006,	1010,	507,	111
160	DATA	-1,	-1,	-1,	-1,	1006,	507,	111
170	DATA	-1,	-1,	-1,	-1,	-1,	508,	111
180	DATA	-1,	-1,	-1,	-1,	-1,	-1,	108
190	DATA	80005,	30005,	10004,	5006,	1010,	508,	111
200	DATA	-1,	-1,	8007,	3007,	1006,	508,	111
210	DATA	-1,	-1,	-1,	-1,	800,	300,	108
220	REM							
230	MAT READ	A(11,7)						
240	DATA	"M",	"D",	"C",	"L",	"X",	"V",	"I"
250	READ	C\$(1), C\$(2), C\$(3), C\$(4), C\$(5), C\$(6), C\$(7)						
260	REM							

A C\$() SZÖVEGES TÖMB MOST A KÖVETKEZŐ:

C\$(1)	C\$(2)	C\$(3)	C\$(4)	C\$(5)	C\$(6)	C\$(7)
M	D	C	L	X	V	I


IDEÁLIS ESETBEN MOST MÁR ELÉG LENNE EGYETLEN "MCDXCII"-SZERŰ SZÖ-VEGET BEVINNI, ÉS ENNEK BETŪIT EGYENKÉNT AZONOSÍTHATNÁNK A C\$() TÖMB

ELEMEIVEL. SAJNOS A BASIC VÁLTOZATOK EBBEN NEM EGYSÉGESEK. TEGYÜNK FEL, HOGY A KÖVETKEZŐ SZÖ-VEGÜNK VAN: LET P\$ = "ABCDEFGHI"

ÉS AZ FGH-T AZ A\$-BA AKAROD TENNI.

LET A\$ = SUBSTR(P\$,6,3)
 LET A\$ = P\$(6,8)
 LET A\$ = STR\$(P\$(6,3))
 LET A\$ = MID\$(P\$,6%,3%)
 LET A\$ = EXT\$(P\$,6,8)
 LET A\$ = P\$(:6,3)
 LET A\$ = SUB\$(P\$,6,3)

NÉHÁNY MEGOLDÁS A SOK KÜLÖN-BŐZŐ KÖ-ZÜL.



ÍGY MEGELEGESZÜNK A BETŪK EGYENKÉNTI BEVITELÉVEL.

▲ TT A RÓMAI SZÁMOK MEGFEJTÉSÉRE KÉSZÜLT PROGRAM TÖRZSE 3

```

300 REM A, FŐPROGRAM ELKEZDŐDIK
310 PRINT "ÍRD BE BETŰNKÉNT A RÓMAI SZÁMOKAT"
320 PRINT "SZÁM VÉGE: * PROGRAM VÉGE: * *"
330 PRINT "IV - GYEL ÉS IX - CEL VÉGZŐDŐ SZÁMOKNÁL NEM KELL *"
340 PRINT
350 PRINT "START"
360 LET R = 1
370 LET M = 0
380 LET C = 0
390 LET P = 0
400 INPUT T$
410 IF T$ = "*" THEN GOTO 610
420 IF T$ = " " THEN GOTO 580
430 FOR K = 1 TO 7
440 IF T$ = C$(K) THEN GOTO 480
450 NEXT K
460 PRINT "HIBÁS RÓMAI SZÁM"
470 GOTO 340
480 LET X = A(R, K)
490 IF X < 0 THEN GOTO 460
500 REM EGY BETŰ CSAK HÁROMSZOR FORDULHAT ELŐ
510 LET C = (1 - ABS(SGN(K - P))) * (1 + C)
520 IF C > 2 THEN GOTO 460
530 LET P = K
540 REM AZ EREDMÉNYT AZ M-BEN GYŰJTI ÖSSZE; R-BEN TÁROLJA
550 LET M = M + INT(X/100)
560 LET R = X - 100 * INT(X/100)
570 IF R <> 0 THEN GOTO 400
580 REM KINYOMTATJA AZ EREDMÉNYT
590 PRINT M
600 GOTO 340
610 END

```

KEZDETI ÁLLAPOT A TÁBLÁZATBAN

AZ EREDMÉNYT M-BEN GYŰJTI ÖSSZE

EGYFORMÁN EGYMÁSÁT KÖVETŐ BETŰK ELŐFORDULÁSÁT SZÁMOLJA

"ELŐZŐ" BETŰ

RUN VÉGE

AZ M-BEN ÖSSZEGYŰJTÖTT EREDMÉNYT KIÍRJA

ÚJ SZÁMOT VÁR

KIEMELI A TÁBLÁZAT ELEMÉIT

-1 A TÁBLÁZATBAN HIBÁNAK FELEL MEG

HA AZ "ELŐZŐ" = "JELENLEGI" AKKOR C = C + 1 EGYÉBKÉNT 0

AZ "ELŐZŐ" ÉRTÉKÉT A "JELENLEGI"RE ÁLLÍTJA.

A TAG ELSŐ RÉSE

MÁSODIK RÉSZ = ÚJ ÁLLAPOT

BEÍRNI A KÖVETKEZŐ BETŰT

ÚJ SZÁMOT NÉZ

RUN

ÍRD BE BETŰNKÉNT A RÓMAI SZÁMOKAT
 SZÁM VÉGE: * PROGRAM VÉGE: * *
 A IV - GYEL ÉS IX - CEL VÉGZŐDŐ SZÁMOKNÁL NEM KELL *

START
 ? M
 ? M
 ? I
 ? *
 2001

START

? M

? C

? M

? L

? X

? X

? X

? I

? V

1984

NEM KELL X, MERT
A SZÁM IV-RE VÉGZŐDIK

START

? V

? *

5

START

? **

A PROGRAM VÉGE

Ebben a példában az állapottáblázat minden elemének első része egyszerűen olyan szám, amelyet az "M" változóhoz kell adni. Bonyolultabb esetben ez egy szubrutin sorszám lenne. Egy elem kiválasztását követően (úgy, ahogy ez a 480. sorban látható) egy "ON" utasítás arra a szubrutinra adná a vezérlést, amelyet ezen elem előfordulásakor végre kell hajtani. A szubrutinból visszatérve, egy utasítás az 560. sorhoz hasonlóan beállítaná az új állapotot.


LEGROVIDEBB NAZAŰT

AZ IPARBAN ÉS A

KERESKEDELEMBEN ISMERT PROBLÉMA.

ÍME PÉLDA EGY "ALGORITMUSRA" (SZÁMÍTÓGÉPES TERMINOLÓGIA A "MÓDSZER" SZÓRA), AMELLYEL KIDERÍTHETED A LEGROVIDEBB UTAT A "CSOMÓPONTOK" ÉS AZ "ELÁGAZÁSOK" HÁLÓZATÁN KERESZTÜL. (A BEMUTATOTT MÓDSZER MÁS ALKALMAZÁSA A "KRITIKUS ÚT-ANALÍZIS".) EZ AZ ALGORITMUS AKKOR MŰKÖDIK, HA BARMELY 2 CSOMÓPONT KÖZÖTT, EGY IRÁNYBAN CSUPÁN EGY ÚT LÉTEZIK.



START



**HOME
OPTION**

ENNEK AZ ÚTNAK A MEGTÉTELÉHEZ SZÜKSÉGES IDŐ

AZ ALGORITMUS A KÖVETKEZŐ: AZ ADATOK NYILVÁNTARTÁSÁT A 2 CSOMÓPONT ESETÉN AZ ALABBI RAJZ MUTATJA:



A 3. CSOMÓPONTNÁL KEZDVE MENJ VÉGIG MINDEN CSOMÓPONTON 3, 4, 5, 6, FOLYTASD TOVÁBB ÉRINTÉSÜKET 1, 2, 3, 4, 5, 6, 1, 2 STB. EGÉSZEN ADDIG, AMIG MINDEN KAPCSOLÓ KIKAPCSOLT ÁLLAPOTBA NEM KERÜL. MINDEN CSOMÓPONTBAN VÉGEZD EL AZT, AMIT A 2. CSOMÓPONTRA ÍRUNK LE!

- ✱ NÉZD MEG A KAPCSOLÓT A (2,1)-NÉL! HA KI VAN KAPCSOLVA, AKKOR MENJ TOVÁBB A KÖVETKEZŐ CSOMÓPONTHOZ! HA EZ BE VAN KAPCSOLVA, AKKOR:
- ✱ VÁLASZD KI AZ EDDIGI LEGJOBB IDŐT: EZ AZ A(2,2)-BEN VAN.
- ✱ AZ A(2,4)-BEN HASZNÁLD FEL A LÁNC FEJÉT, HOGY A 2. CSOMÓPONTHOZ KAPCSOLÓDÓ MINDEN CSOMÓPONTOT ÁTNÉZHESSE! A LÁNC MINDEN SZEMÉNÉL ÚGY JÁRJ EL, AHOGY EZT MOST AZ 5. CSOMÓPONTRA LEÍRJUK:

- ★ VEDD A CSOMÓPONT SZÁMÁT (MOST EZ 5)!
- ★ VÁLÁSZD KI AZ EHEZ A CSOMÓPONTHOZ VEZETŐ ÚT UTAZÁSI IDEJÉT (KEZDETBEN EZ 8.0), ÉS ADD EZT A (2,2)-BŐL MÁR KIVÁLASZTOTT IDŐHÖZ! EZ AZ EREDMÉNY LEGYEN T (AZ 5. CSOMÓPONTNAK A 2. CSOMÓPONTON KERESZTÜL VEZETŐ ELÉRÉSI IDEJE)!
- ★ GYŐZDJI MEG ARRÓL, HOGY A(5,2)-NÉL MOST JOBB UTAT TALÁLTÁL-E! HA IGEN:
 - KAPCSOLD "BE" A KAPCSOLÓT A(5,1)-NÉL!
 - A RÉGI LEGJOBB IDŐT (A(5,2)) HELYETTESÍTSD T-VEL!
 - A 2 SZÁMU CSOMÓPONTOT RAKD BE, AZ A(5,3)-BA, ÍGY ÉPÍTVE FEL A LEGJOBB ÚT CSOMÓPONTJAIN KERESZTÜL VEZETŐ LÁNCOT!

MIUTÁN EZT, MINDEN CSOMÓPONTNÁL ELVÉGEZTED, "KAPCSOLD KI" A CSOMÓPONT KAPCSOLÓJÁT!

AZ A(,) ÉS B(,) TÖMBÖKET ELŐ KELL KÉSZÍTENI, MIELŐTT A CSOMÓPONTOKAT A FENTIEKBEN LEÍRT MÓDON SORRAVESSZÜK. MINDEN KAPCSOLÓT BE KELL KAPCSOLNI, A RENDKÍVÜL NAGY UTAZÁSI IDŐKET AZ A(,) MÁSODIK OSZLOPÁBA KELL TENNI ÉS MINDEN CSOMÓPONTOT A BELŐLE ELÉRHETŐ CSOMÓPONTOKKAL KELL ÖSSZEKÖTNI. A KEZDETI CSOMÓPONT UTAZÁSI IDEJÉNEK A NULLA ÉRTÉKET KELL ADNI.

```

10 REM  LEGRÖVIDEBB HAZAÚT : CSAK AZ ALGORITMUST MUTATJA
20 REM  (NEM SZÜRJÜK KI A LEHETETLEN ADATOKAT)
30 REM  50 CSOMÓPONTRA ÉS 120 ÚTVONALRA VAN HELY
40 DIM  A(50,4), B(120,3)
50 PRINT "CSOMÓPONTOK SZÁMA, ÚTVONALAK SZÁMA, KEZDETI CSOMÓPONT, CÉLCSOMÓPONT"
60 INPUT          N,          E,          S,          H
70 FOR    I = 1 TO N
80 LET    A(I,1) = 1
90 LET    A(I,2) = 1000000
100 LET   A(I,3) = 0
110 LET   A(I,4) = 0
120 NEXT  I
130 LET   A(S,2) = 0


```

Annotations:

- BEKAPCSOL (points to line 80)
- NAGYON MAGAS IDŐ (points to line 90)
- TÖRLI A FEJEKET ÉS AZ UTAK LÁNCÁT (points to lines 100 and 110)
- TÖRLI A KEZDETI CSOMÓPONTOKNÁL AZ IDŐT (points to line 130)

A TÚLOLDALON BEMUTATOTT PÉLDÁBAN AZ ADATOK ELSŐ SORA A KÖVETKEZŐ LENNE: ? 6, 9, 3, 6; AZ A(,) TÖMB TARTALMA PEDIG:

	1)	2)	3)	4)
A(1,	1	1000000	0	0
A(2,	1	1000000	0	0
A(3,	1	0	0	0
A(4,	1	1000000	0	0
A(5,	1	1000000	0	0
A(6,	1	1000000	0	0



OST OLVASD BE AZ ÚTVONALAK ADATAIT (BÁRMILYEN SORRENDENBEN), ÉS A CSOMÓPONTOK MINDEGYIKÉT KAPCSOLD ELŐDJÜK LÁNCÁHOZ!

AZ INDULÓ ADATOKAT AZ ITT LÁTHATÓ MODON LEHET ELŐKÉSZÍTENI.

CSOMÓK SZÁMA	ÚTVONALAK SZÁMA	KEZDETI CSOMÓPONT	CÉL- CSOMÓPONT
6	9	3	6

ÚTVONALAK		ÚTVONALRA VONATKOZÓ IDŐ
A	B	
3	1	10.0
3	2	16.0
1	2	5.0
1	5	14.0
2	4	6.0
2	5	8.0
5	6	9.0
4	6	10.0
4	5	1.0

```

140 REM INPUT ÉS LÁNCOLÁS (AZ ADATOKAT NEM ELLENŐRZI)
150 PRINT "A---->--B, ÚTVONAL-IDŐ "
160 FOR P = 1 TO E
170 INPUT A, B, R
180 LET B(P,2) = B
190 LET B(P,3) = R
200 LET B(P,1) = A(A,4)
210 LET A(A,4) = P
220 NEXT P

```


A KÉT TÖMB EKKOR A KÖVETKEZŐ:

	1)	2)	3)	4)
A(1,	1	1000000	0	0 4
A(2,	1	1000000	0	0 6
A(3,	1	0	0	0 2
A(4,	1	1000000	0	0 9
A(5,	1	1000000	0	0 7
A(6,	1	1000000	0	0

	1)	2)	3)
B(1,	0	1	10.0
B(2,	1	2	16.0
B(3,	0	2	5.0
B(4,	3	5	14.0
B(5,	0	4	6.0
B(6,	5	5	8.0
B(7,	0	6	9.0
B(8,	0	6	10.0
B(9,	8	5	1.0

A(,) 4. OSZLOPÁBAN TALÁLHATÓ ÁTHÚZOTT SZÁMOK A LÁNC SZEMEK EGYMÁS UTÁNI KAPCSOLÓDASAIT MUTATJÁK.

A PROGRAM FŐ RÉSE KÖVETKEZIK. A K VÁLTOZÓ AZ 1. CSOMÓPONTON VALÓ ÁTHALADÁSOKAT SZÁMOLJA, MIKÖZBEN CIKLIKUSAN BEJÁRJUK A CSOMÓPONTOKAT. K NULLÁVÁ VÁLIK, HA AZ ÚT MEGVÁLTOZIK. ÍGY AMIKOR K ELÉRI A 2-T, MINDEN KAPCSOLÓ KI VAN KAPCSOLVA, ÉS AZ EREDMÉNYT KI LEHET NYOMTATNI.



A PROGRAM A TÚLOLDALI ADATOKKAL TÖRTÉNŐ LEFUTTATÁSA A KÖVETKEZŐ EREDMÉNYT ADJA:

3-TÓL 6-BA VEZETŐ LEGRÖVIDEBB ÚT 31
6, 5, 4, 2, 1, 3 -T ÉRINTVE

A FUTÁS VÉGÉN AZ A(,) MÁSODIK OSZLOPA TARTALMAZZA A HÁLÓZAT S CSOMÓPONTJÁTÓL AZ ÖSSZES TÖBBI CSOMÓPONTIG VEZETŐ LEGRÖVIDEBB IDEJŰ UTAKAT.




PARANCSONK ÉS BEJELENTKEZÉS

BEJELENTKEZÉS

NÉZZ UTANA A KÉZIKÖNYVBEN ∞ MINDEN RENDSZERNEK MEGVAN A MAGA SZABÁLYA.

A BASIC EGYFORMÁN FUTTATHATÓ KIS- ÉS NAGYSZÁMÍTÓGÉPEKEN.



HASZNÁLHATSZ MODERN "ASZTALI" SZÁMÍTÓGÉPET, AMELYET KIZÁRÓLAG A BASIC PROGRAMOK KIFEJLESZTÉSÉRE ÉS FUTTATÁSÁRA TERVEZTEK.

(EZEK KÖZÜL NÉMELYEK BILLENTYŰÍN A NYELV SZAVAI SZEREPELNEK A PROGRAMÍRÁS MEGKÖNNYÍTÉSÉRE.)

VAGY HASZNÁLHATSZ EGY "TERMINÁLT", AMELY AZ ÉPÜLETBEN LEVŐ VAGY EGY KÜLSŐ SZÁMÍTÓGÉPPAL TELEFONON KERESZTÜL VAN ÖSSZEKAPCSOLVA.

NÉMELYIK TERMINÁLNAK BILLENTYŰZETE ÉS A TELEVÍZIÓHOZ HASONLÓ KÉPERNYŐJE VAN: MÁSOKNAK PEDIG BILLENTYŰZETE ÉS ÍRÓMŰVE VAN, AMELY PAPIRRRA NYOMTAT.



A NAGYSZÁMÍTÓGÉPEK A NYELVEK GAZDAG VÁLASZTÉKÁT NYÚJTJÁK ∞ A BASIC EZEK KÖZÜL AZ EGYIK. VALAHOGY MEG KELL MONDANOD A SZÁMÍTÓGÉPNEK A BEJELENTKEZÉSKOR, HOGY A BASIC NYELVET AKAROD HASZNÁLNI.

AMIKOR BEKAPCSOLSZ EGY TERMINÁLT (ÉS HA EZ TELEFONVONALBA VAN KAPCSOLVA, A SZÁMÍTÓGÉP-KÖZPONT SZÁMÁT TÁRCSÁZOD), AZ ELSŐ ÉRTHETŐ VÁLASZT A SZÁMÍTÓGÉP "OPERÁCIÓS RENDSZERÉTŐL" KAPOD. EKKOR OLYAN NYELVEN KELL UTASÍTÁSOKAT ADNOD, AMELYET AZ OPERÁCIÓS RENDSZER MEGÉRT. SAJNOS EZEK A NYELVEK OPERÁCIÓS RENDSZERENKÉNT IGEN KÜLÖNBÖZŐEK. VAN, AMELYIK KÉRDÉSEK FELTEVÉSÉVEL ÉS GYORS VÁLASZADÁSSAL VAN SEGÍTSÉGEDRE. A KÖVETKEZŐ PÉLDA EGY ILYEN KÉPZELETBELI RENDSZERRE VONATKOZIK, DE SZERENCSESNEK KELL AHHOZ LENNED, HOGY EGY OLYAN RENDSZERT TALÁJ, AMELYIK ENNYIRE ÉRTHETŐ.

BESZÉLGETÉS EGY KÉPZELETBELI OPERÁCIÓS RENDSZERREL :

ÖN A NECROPOLITAN BIZTOSÍTÓ TÁRSASÁG MORSVILLE -1
"THANATOS" RENDSZERÉVEL VAN ÖSSZEKAPCSOLVA
KÉREM ÍRJA BE A SZÁMLASZÁMÁT
? 123021/6
PASSWORD (JELSZÓ)
? DAWYJONES

NÉMELYIK RENDSZER KARAKTEREK FEKETE ÖSSZEVISSZASÁGÁT ÍRJA KI ÉS ÍRJA,
ÁT MÉG EGYSZER, AMIKOR A JELSZAVADAT KÉRI, ÍGY AMIT AZ ÖSSZEVISSZASÁG-
RA RÁGÉPELSZ, KIVÁNCSI SZEMLÉLŐDŐ NEM BOGOZHATJA KI:

PASSWORD (JELSZÓ)

? ■ ■ ■ ■ ■ ■ ■ ■

123021 SZ. 6. ALSZÁMLASZÁMU FELHASZNÁLÓ 17.03-KOR JELENTKEZETT BE
MILYEN PROCESSZORVAGY NYELV

? BASIC

AZ OPERÁCIÓS RENDSZERREL FOLYTATOTT BESZÉLGETÉSNEK MÉG NINCSEK VÉGE
AZONBAN EZ AZ A PONT, AMELYNÉL AZ EGYSZERŰBB OPERÁCIÓS RENDSZEREK
(AZON SZÁMÍTÓGÉPEK ESETÉN, AMELYEK BASIC-RE ÉPÜLNEK) A BEKAPCSOLÁS-
KOR KEZDENEK.

ÚJ VAGY RÉGI PROGRAM

? NEW

ÚJ PROGRAM NEVE

? EXAMPL

A NÉV HOSSZÚSÁGÁT A LEGTÖBB OPERÁCIÓS RENDSZER 6 (BETŰ ÉS SZÁM-
JEGY) KARAKTERRE LIMITÁLJA, AZ ELSŐ KARAKTER MINDIG BETŰ.

A BASIC MUNKÁRA KÉSZ (BASIC IS READY)

EZ AZ A PONT, AHOLO E KÖNYV PÉLDÁI ÁLTALÁBAN KEZDŐDNEK, AZAZ A GÉP
KÉSZEN ÁLL OLYAN ÚJ PROGRAM BEFOGADÁSÁRA, AMELYET A "RUN" KIADÁSA
ELŐTT SORONKÉNT VISZNEK BE.

A BEJELENTKEZÉST, A PROGRAM BEVITELÉT ÉS LEFUTTATÁSÁT KÖVETŐEN KI A-
KÁRSZ MAJD LÉPNI A RENDSZERBŐL. SOK RENDSZERNÉL EZT A "BYE" VAGY A
"GOODBYE" SZAVAKKAL ÉRED EL.

HA NEM I-
GAZODSZ KI
A RENDSZER-

BYE

123021 SZ. 6. ALSZÁMLASZÁMU FELHASZNÁLÓ 17.15-KOR LÉPETT KI
FELHASZNÁLT IDŐ = 12 PERC, PROCESSZOR IDŐ = 1.383 SEC.


REL FOLYTATOTT BESZÉLGETÉSBEN, PRÓBÁLD MEG A "HELP" (SEGÍTSÉG) SZÓT BEGÉPELNI.
NÉMELYEK SEGÍTŐKÉSZEN VÁLASZOLNAK: MÁSOK AZT MONDJÁK: WHAT? (MICSODA?) - azért kiprobálni mégis érdemes.

PARANCSONK

MINDEN PARANCS BEVITELÉNEK PILLANATÁBAN KERÜL VÉGREHAJTÁSRA.

A BASIC LEGFONTOSABB PARANCSA A "RUN", VANNAK AZONBAN MÁSOK IS, AMELYEK KÖZISMERTÉK A KÜLÖNBÖZŐ BASIC VÁLTOZATOKBAN, ÉS EZÉRT ITT RÖVIDEN BEMUTATJUK ŐKET. A LEGTÖBB BASIC RENDSZERBEN VANNAK OLYAN HASZNOS PARANCSONK, AMELYEK AUTOMATIKUSAN BESZÁMOZZÁK A SOROKAT, ÚJRA SZÁMOZZÁK A PROGRAMSOROKAT ÉS ÖSSZEILLESZTIK A PROGRAMOKAT. EZEK AZONBAN RÉSZLETEIKBEN TÚLSÁGOSAN ELTÉRŐEK AHHOZ, HOGY E KÖNYVBE BEKERÜLJENÉK. A BASIC RENDSZER PARANCSAIT ILLETOÉN A KÉZIKÖNYVET KELL FELÜTNI.

```
BASIC IS READY
10 REM EGY PÉLDA CSUPÁN
20 FOR A = 1 TO 8
30 PRINT A;
40 EMD
40 END
32 NEXT A
10
RUN
1 2 3 4 5 6 7 8
```




A SORSZÁM NÉLKÜLI "RUN" NEM A PROGRAM RÉSZÉ, HATÁSAIRA A PROGRAM AZONNÁL VÉGREHAJTÓDIK

A ÚJRA KIADNÁNK A "RUN"-T, UGYANAZT AZ EREDMÉNYT KAPNÁNK: A BASIC NEM FELEJTI EL A PROGRAMOT A VÉGREHAJTÁS UTÁN.

```
RUN
1 2 3 4 5 6 7 8
```

A "LIST" PARANCS HATÁSÁRA, A RENDSZER KIÍRJA A PROGRAMOT. EZ A LISTA A PROGRAM LEGUTOLSÓ VÁLTOZATA, AMELYBEN A BASIC RENDSZER MÁR ÁTVEZETTE A JAVÍTÁSOKAT, A HIBÁS SOROKAT ELTÁVOLÍTOTTA, ÉS AZ UTASÍTÁSOKAT SORSZÁM SZERINT RENDEZTE. HASONLÍTSD ÖSSZE A FENTI EREDETI PROGRAMOT AZ ALÁBBIVAL (L. TOVÁBBÁ AZ 7. OLDALT)!

```
LIST
20 FOR A = 1 TO 8
30 PRINT A;
32 NEXT A
40 END
```



MÉLYIK BASIC RENDSZERBEN A "LIST 30" HATÁSÁRA CSUPÁN A 30. SOR KERÜL KINYOMTATÁSRA: MÁ S RENDSZEREK BÉN A 30. SOR ÉS AZ EZT KÖVETŐ SOROK NYOMTATÓDNÁNK KI. EGYÉ S RENDSZEREK BÉN A "LIST 20, 32" (MÁSOKNÁL "LIST 20 - 32") HATÁSÁRA CSUPÁN A PROGRAM MEGADOTT RÉSZÉ KERÜL KINYOMTATÁSRA.

A "SAVE" PARANCS KIADÁSKOR A BASIC RENDSZER AZ AKTUÁLIS PROGRAM EGY PÉLDA NYÁT ADATÁLLOMÁNYBAN TÁROLJA. A SZÁMÍTÓGÉPEK-NEK MUNKA TERÜLETE (ÁLTALÁBAN A KÖZPONTI TÁRBAN) ÉS ADATÁLLOMÁNY-TERÜLETE (ÁLTALÁBAN MÁGNESSZALAGRA ÉS MÁGNESLEMEZRE ÉPÜLŐ HÁTTÉRTÁRON) VAN. A "SAVE" PARANCS HATÁSÁRA A MUNKA TERÜLETEN LEVŐ PROGRAM MÁSOLATA ÁTKERÜL AZ ADATÁLLOMÁNY-TERÜLETRE. INNEN A PROGRAM NEVE ALAPJÁN A KÉSŐBBIEKBEN TÁRGYALT MÓDON VISSZANYERHETŐ.

```

LIST
20 FOR A = 1 TO 8
30 PRINT A
32 NEXT A
40 END
SAVE

```

A PROGRAM MÁSOLATÁT A 113. OLDALON MEGADOTT "EXAMPL" NÉV ALATT KÜLDI ÁT AZ ADATÁLLOMÁNY-TERÜLETRE

AL. A 117. OLDALT IS

MA A PROGRAMOD MÁR AZ ADATÁLLOMÁNY-TERÜLETEN VAN, NEM FOG ELVESZNI, AMIKOR KIJELENTKEZEL: A TÁROLÁSÉRT AZONBAN ADDIG, AMEDDIG E TERÜLETET HASZNÁLOD, FIZETNEK KELL (ÁLTALÁBAN NAPONKÉNT).

A "CATALOG" PARANCCSAL MEGNÉZHETED, HOGY MIT TÁROLTÁL EDDIG AZ ADATÁLLOMÁNY-TERÜLETEN. MINDEN RENDSZER KÜLÖNBÖZŐKÉPPEN MUTATJA BE AZ INFORMÁCIÓT, JELLEMZŐK AZONBAN A KÖVETKEZŐK:

CATALOG						
FELHASZNÁLÓ		123021/6		TÁRGY ADATÁLLOMÁNYOK		
NÉV	TÍPUS	MÉRET	A LÉTREHOZÁS DÁTUMA			
NAME	TYPE	SIZE	DATE	CREATED		
EXAMPL	FORRÁS KÓD	1	01	MAR	77	
PROG 27	FORRÁS KÓD	2	28	FEB	77	
DATA 65	ADAT	27	13	JUN	75	
DAT4	ADAT	15	24	AUG	72	

A "FORRÁSKÓD" KIFEJEZÉS AZ EREDETI NYELVŰ PROGRAMRA EBBEN AZ ESETBEN A BASIC-RA UTAL.

AZ ADATÁLLOMÁNYOK LÉTREHOZÁSÁNAK ÉS TÁROLÁSÁNAK ESZKÖZEIT A 120. OLDALON TÁRGYALJUK.



A TÁROLÁS NAPI DÍJÁT A MÉRETBŐL HATÁROZZUK MEG: A MÉRET EGYSÉGE LEHET BLOKKOKBAN, "PÁLYÁKBAN" VAGY MÁS MÉRTÉKEGYSÉGBEN, A HASZNÁLATBAN LEVŐ ESZKÖZ FAJTÁJÁTÓL FÜGGŐEN.

PARANCSONK (FOLYTATÁS)

AZ ÁLLOMÁNYT TÖRÖLHETED AZ ADATÁLLOMÁNY-TERÜLETRŐL, MEGSZÜNTETHETED AZ "UNSAVE" PARANCCSAL, AMELYET AZ ADATÁLLOMÁNY NEVE VAGY VESSZŐVEL ELVÁLASZTOTT NEVEK LISTÁJA KÖVET. NÉMELYIK RENDSZER-BEN EZ A SZÓ "PURGE" (TISZTOGAT); MÁSOKBAN "DESTROY" (MEGSEMMISÍT).

UNSAVE DATA65, DATA4

HA MOST ÚJRA BEÍRJUK, HOGY "CATALOG", A DATA65 ÉS A DATA4 MÁR NEM SZEREPEL A LISTÁN, ÉS EZEK TÁROLÁSI KÖLTSÉGE IS MEGSZÚNIK. SOK RENDSZER-BEN AZ ADATÁLLOMÁNYRÓL LYUKSZALAGRA VAGY LYUKKÁRTVÁRA KÉSZÍTHETSZ MÁ-SOLATOT: EZEKEN AZ ADATHORDOZÓKON TÁROLÁSI KÖLTSÉG NÉLKÜL TÁ-ROLHATOD AZ ADATÁLLOMÁNYOKAT ÉS, HA SZÜKSÉG VAN RAJUK, BE LEHET OL-VASNI ÖKET AZ ADATÁLLOMÁNY-TERÜLETRE. AZ ITT LEÍRT MÓDOK SZÁMÍTÓ-GÉPRŐL SZÁMÍTÓGÉPRE ELTÉRNEK, EZÉRT E KÖNYVBEN NEM MUTATJUK BE.

HA AZ ADATÁLLOMÁNYT AZ ADATMEZŐBŐL A MUNKA TERÜLETRE AKARJUK ÁT-VINNI, AKKOR AZ "OLD" UTASÍTÁST ADJUK KI, AMELYET A KÍVÁNT A-DATÁLLOMÁNY NEVE KÖVET:

OLD PROG27

AMI EDDIG A MUNKA TERÜLETEN VOLT, AZ MOST ELTŰNIK, EZÉRT HA A MUNKA TERÜLETRŐL EGY MÁ-SOLATOT MEG A-KARSZ TARTANI, AKKOR NE FELEJTSD EL AZ "OLD" ELŐTT A "SAVE" PARAN-CSOT HASZNÁLNI!

MOST KIADHATOD A "LIST" PARANCSOT, HOGY A PROG27-RŐL KÉSZÜLT MÁ-SOLATOT MEGNÉZHESD, VAGY ÍRHATSZ "RUN"-T, HOGY VÉGRE HAJTSD A PROG27-ET. LEHETŐSÉG VAN A BASIC NYELVŰ PROGRAM SORAINAK MEG-VÁLTOZTATÁSÁRA, ILL. A PROGRAM ÚJ SOROKKAL VALÓ BŐVÍTÉSÉRE IS. A-MIKOR KIÍRTAD AZ "OLD" SZÓT, ENNEK PONTOSAN UGYANAZ A HATA'SA, MINTHA BEVINNÉD A MEGNEVEZETT PROGRAMOT. A PROGRAM MEGMARAD AZ ADATÁLLOMÁNY-TERÜLETEN IS: AZ "OLD" UTASÍTÁS A PROGRAM EGY MÁ-SOLATÁT KÜLDI ÁT A MUNKA TERÜLETRE.

MI TÖRTÉNIK AKKOR, HA MEGVÁLTOZTATOD A PROG27-T A MUNKA TERÜLE-TEN ÚGY, HOGY NÉHÁNY ÚJ BASIC SORT ÍRSZ BE, MAJD EZUTÁN ÚJRA A "SAVE" SZÓT HASZNÁLOD? VAN MÁR EGY PROG27 AZ ADATMEZŐBEN ÉS EGY MÁSIK A MUNKA TERÜLETEN. ILYENKOR A BASIC RENDSZER HIBAÜZE-NETET ÍR KI, AMELY SZERINT A PROG27-T MÁR TÁROLTA.

A LEGTÖBB "SAVE" UTASÍTÁS AZONBAN MEGENGEDI A KÖVETKEZŐT:

```
SAVE   PROG27A
```

EZ AZT JELENTI, HOGY "TÁROLD A MUNKATERÜLETEN LEVŐ PROGRAMOT PROG27A NÉVEN". AZ ADATÁLLOMÁNY-TERÜLET MOST MINDKETTŐT, A PROG27-ET ÉS A PROG27A-T IS TARTALMAZZA, A MUNKATERÜLETEN SZEREP-LŐ PROGRAM NEVE TOVABBRA IS PROG27 MARAD.

A KETTŐS NEVEK PROBLÉMÁJÁRA MÁSIK MEGOLDÁS IS LÉTEZIK. AZ ÁBRA AZT A KISÉRLETET MUTATJA BE, AMELY MEGPRÓBÁL KÉT PROGRAMOT UGYANA-ZON NÉV ALATT TÁROLNI.

```
SAVE  
*** ERROR *** PROG27 ALREADY SAVED  
UNSAVE PROG27  
SAVE
```

PROG27 MÁR
VAN AZ ADATÁLLOMÁNY-
TERÜLETEN

LEGTÖBB RENDSZERBEN AZ "UNSAVE" UTASÍTÁS CSAK AZ ADATÁLLOMÁNY-TERÜLETEN MŰKÖDIK ÉS NEM ÉRINTI A MUNKATERÜLETEN LEVŐ ADATÁLLOMÁNYOKAT. HA MOST KIADOD A "CATALOG" UTASÍTÁST, AKKOR A PROG27-ET ÚJRA AZ ADATÁLLOMÁNYTERÜLETEN TALÁLOD, AZONBAN LÉTRE-HOZÁSÁNAK IDEJE A MAI DÁTUM LENNE.

A MUNKATERÜLET TELJES TÖRLÉSÉRE A "NEW" UTASÍTÁS SZOLGÁL.

```
NEW
```

HA, EZT NEM HASZNÁLJUK, AKKOR A BEÍRT BASIC SOROK A MUNKATERÜ-LETEN JELENLEG OTT LEVŐ PROGRAMOT MÓDOSÍTJÁK.



FIGYELJÜK MEG, HOGY A 113. OLDALON BEMUTATOTT BEJELENTKEZÉS NEM ENGED MEG MÁS ELSŐ PARANCSOT, CSAK AZ "OLD"-OT VAGY A "NEW"-T!




ADATÁLLOMÁNYOK
« FILE - OK »


ADATÁLLOMÁNYOK

☞ A BILLENTYŰZETEN KIVÜL MÁS LEHETŐSÉG IS VAN ADATOK BEVITELÉRE, MINT AHOGY A TERMINÁL SEM AZ EGYETLEN ADATKIVITELI LEHETŐSÉG.

EDDIG A KÖVETKEZŐKET LÁTHATTUK :



LEHETŐSÉG VAN ARRRA IS, HOGY AZ ADATÁLLOMÁNY-TERÜLETEN LEVŐ EGYIK ÁLLOMÁNNYAL HOZZUNK LÉTRE ADATCSERÉT, AMELY RENDSZERINT MÁGNESES LEMEZEKEN HELYEZKEDIK EL :



AZ ÁLLOMÁNYOKAT FŐKÉNT A KÖVETKEZŐKRE HASZNÁLJÁK :

★ KOMMUNIKÁCIÓ: EGY PROGRAM EREDMÉNYEI MÁS PROGRAMOKHOZ KÉSŐBBIEKBEN FELHASZNÁLÁSRA KERÜLŐ BEMENŐ ADATKÉNT TÁROLÓDNAK AZ ÁLLOMÁNYBAN.

★ HÁTTÉRTÁROLÁS: EGY PROGRAM TÖBB KÖZBENSŐ INFORMÁCIÓT GENERÁLHAT, MINT AMENNYIT A BASIC TÖMBÖK FORMÁJÁBAN TÁROLNI KÉPES (MINDEN RENDSZERNEK MEGVAN A MAGA TÖMBMÉRETRE VONATKOZÓ KORLÁTJA).

AHHOZ, HOGY A PROGRAM TÖMBJEI ÉS AZ ADATMEZŐK KÖZÖTT ADATOKAT VIGYÜNK ÁT, AZ ALÁBBI UTASÍTÁSOKAT HASZNÁLJUK :

INPUT	(18. OLDAL)
MAT INPUT	(96. OLDAL)
PRINT	(28. OLDAL)
PRINT USING	(34. OLDAL)
MAT PRINT	(98. OLDAL)

KIEGÉSZÍTVE AZ "INPUT" VAGY A "PRINT" UTÁN ÁLLÓ CSATORNASZÁMMAL (AMELYET KETTŐSPONT KÖVET) .

AZ ALÁBBI SZUBRUTIN AZ A(,) TÖMB ELSŐ N SORÁT VISZI ÁT A 7 CSATORNÁN LÉVŐ ÁLLOMÁNYBA :

```

1000 REM SZUBRUTIN A(,) N SORÁNAK ÁTVITELEA
1010 REM 7-ES CSATORNÁN LÉVŐ FILE-BA
1020 FOR I = 1 TO N
1030 PRINT 7: A(I,1), A(I,2), A(I,3)
1040 NEXT I
1050 RETURN
    
```

	1)	2)	3)
A(1,			
A(2,			
A(3,			
...			
A(N,			

HA A SZUBRUTINT MÁSODSZORRA IS MEGHÍVJUK, AKKOR AZ ÚJ HÍVÁS TOVÁBBI SZÁMSOROZATOKAT FŰZ AZ ELŐZŐ HÍVÁSKOR ÁTVITT ÁLLOMÁNYHOZ. ÍGY SOKKAL HÖSSZABB FORMÁBAN TUDOD A TÖMBÖT AZ ADATÁLLOMÁNYBAN TÁROLNI, MINT AMIT A TÖMBRE VONATKOZÓ "DIM" UTASÍTÁS TARTALMAZ. A KÖVETKEZŐ SZUBRUTIN AZ R. REKORDOT OLVASSA BE A 6. CSATORNÁN EGY ADATÁLLOMÁNYBÓL. FELTÉTELEZZÜK, HOGY ENNEK AZ ÁLLOMÁNYNAK UGYANAZ A FELÉPÍTÉSE, MINT A 7. CSATORNA ÁLLOMÁNYÁNAK A FENTI ÁBRÁN. AZ ADATÁLLOMÁNY EGYSZERŰEN FOGALMAZVA EGYEDI TÉTELEK SOROZATA; FELÉPÍTÉSÉT A FELHASZNALÓNAK KELL MEGSZERVEZNIÉ (MONDJUK, HOGY MINDEN REKORD EGY HÁROMELEMŰ SORBÓL ÁLL, MINT EBBEN A PÉLDÁBAN).

```

2000 REM SZUBRUTIN AZ R. REKORD BEVITELÉRE B(,) BE
2010 REM A 6. CSATORNARÓL
2020 REM ELŐSZOR ÁLLJ RÁ AZ ÁLLOMÁNY 1. REKORDJÁRA.
2030 RESET 6
2040 REM PÖRGESD VÉGIG AZ ELSŐ (R-1) REKORDOT
2050 FOR I = 1 TO R-1
2060 INPUT 6: A, B, C
2070 NEXT I
2080 REM MOST VIDD BE AZ R. REKORDOT B(,) BE
2090 INPUT 6: B(1), B(2), B(3)
2100 RETURN
    
```

R-1
REKORD
KIHAGYÁSA



B(1)	B(2)	B(3)

A PÉLDA TARTALMAZZA A "RESET" UTASÍTÁST, AMELY SOK BASIC RENDSZERBEN HASZNÁLHATÓ, BÁR VAN OLYAN VÁLTOZAT, AMELY A "RESTORE" SZÓT HASZNÁLJA. ÁLTALÁBAN MEGTEHETED, HOGY EGYETLEN UTASÍTÁSSAL ÁLLÍTSZ VISSZA TÖBB CSATORNÁT:

```
100 RESET 1, 3, 6
```

AZ UTASÍTÁS EGY KÉPZELETBELI MUTATÓT A KIVÁLASZTOTT CSATORNÁN LÉVŐ ADATÁLLOMÁNY ELEJÉRE ÁLLÍT. ÚGY, HOGY A KÖVETKEZŐ "INPUT" UTASÍTÁS AZ ADATÁLLOMÁNYHOZ AZ ELSŐ ADATOT OLVASSA. NE HASZNÁLD A "RESET" UTASÍTÁST EGY ÁLLOMÁNY LISTÁZÁSA KÖZBEN!

FILE-NEVEK * CSATORNÁK

HA A BASIC PROGRAMBAN ADATÁLLOMÁNYOK SZEREPELNEK, KAPCSOLATOT KELL TEREMTENI A "CATALOG" ODBAN MEGJELENŐ ÁLLOMÁNYNEVEK ÉS AZ "INPUT", VALAMINT A "PRINT" UTASÍTÁSOKBAN HASZNÁLT CSATORNA SZÁMOK KÖZÖTT. (A "CATALOG" UTASÍTÁS MAGYARÁZATÁT L. A 115. OLDALON.) ENNEK LEGALÁBB ANNYIFÉLE MÓDJA VAN, MINT AHÁNYFÉLE BASIC RENDSZER LÉTEZIK. TEGYÜK FEL, HOGY A "MYDATA"-NAK NEVEZETT ÁLLOMÁNYBÓL A 6. CSATORNÁN AKARSZ ADATOKAT BEVINNI, MAJD EGY "RESULT"-NAK NEVEZETT ÁLLOMÁNYT A 7. CSATORNÁN AKARSZ KINYOMTATNI. TEKINTSÜK A FELADAT ELVÉGZÉSÉNEK NÉHÁNY ELTÉRŐ VÁLTOZATÁT, AMELYEK KÜLÖNBÖZŐ BASIC-RENDSZEREKBE ÉRVÉNYESEK? (AZ ÁLTALAD KITALÁLHATÓ NEVEK ÁLTALÁBAN 6 BETŰRE ÉS SZÁMJEGYRE KORLÁTOZÓDNAK, AMELYEK KÖZÜL AZ ELSŐNEK BETŰNEK KELL LENNIE.)

```
10 FILE # 6 : "MYDATA"
20 FILE # 7 : "RESULT"
```

N.C.G.
"STANDARD
BASIC"

```
10 FILES # 6 = "MYDATA", # 7 = "RESULT"
```

```
10 FILES 1, 2, 3, 4, 5, "MYDATA"/10, "RESULT"/10
```

SZÁMOZOTT
KAZETTA'S
ÁLLOMÁNYOK

MEGNEVEZETT
LEMEZES
ÁLLOMÁNYOK

AMELYEK MÉRETE
1024 KARAKTERES
BLOKKOKBAN VAN MEGADVA

A CSATORNASZÁ-
MOT A LISTÁN
ELFOGLALT
POZÍCIÓ
HATÁROZZA MEG

```
10 FILES A; B; C; D; E; MYDATA; RESULT
```

```
10 OPEN 6 = "MYDATA", INPUT
20 OPEN 7 = "RESULT", OUTPUT
```

A KIMENŐ
ÁLLOMÁNYOKTÓL
MEGKÜLÖNBÖZTETETT
BEMENŐ ÁLLOMÁNYOK

```
10 OPEN "MYDATA" FOR INPUT AS FILE 6
20 OPEN "RESULT" FOR OUTPUT AS FILE 7
```

```
10 OPEN "MYDATA" TO : 6, INPUT
20 OPEN "RESULT" TO : 7, PRINT
```

TOVÁBBI BONYODALMAT JELENT AZ, HOGY TÖBB BASIC RENDSZERNÉL ÉRTENI KELL A MUNKAVEZÉRLŐ NYELVET (AZAZ AZT A NYELVET, AMELYET A SZÁMÍTÓGÉP OPERÁCIÓS RENDSZERE MEGÉRT) AHHOZ, HOGY DEKLARÁLJ ÉS RÉSZLETEKET MEGADJ MINDEN OLYAN ADATÁLLOMÁNYRÓL, AMELYRE A BASIC PROGRAMOD HIVATKOZIK: TEHÁT KÉTSZER KELL A RÉSZLETEKET MEGADNI.

ADATÁLLOMÁNYOK (FOLYTATÁS)

SZÖVEGEKET ÉS SZÁMOKAT EGYARÁNT TÁROLHATSZ FIZIKAI ADATÁLLOMÁNYOKRA:

```
100 REM ÁLLOMÁNY NEVEK ÉS ADATOK
110 PRINT 7: "ÜGYFÉL NEVE"; N$; S$
120 MAT PRINT 7: A
```

NÉHÁNY BASIC RENDSZER AZONBAN CSAK AKKOR FOGADJA EL A SZÖVEGEK ÉS SZÁMOK ILYEN KEVERÉKÉT, HA AZOK A KÉSŐBB DEFINIÁLT KARAKTER FORMAJÚ ÁLLOMÁNYOKBA KERÜLNEK.

AZ ELŐZŐ PÉLDÁK ALAPJÁN NYILVÁNVÁLÓ, HOGY MILYEN KÖNNYŰ EGY ÁLLOMÁNYT FELÍRNI, MAJD ROSSZ FÁZISBAN ((ELTOLVA)) OLVASNI VISSZA: NÉ FELEDD, HOGY NEM LÁTHATOD AZ ÁLLOMÁNY TARTALMÁT. BEMUTATUNK EGY RUTINT, AMELYIK A 6. CSATORNÁN OLVASSA BE AZ ADATÁLLOMÁNYT, AMELYIKET AZ ELŐZŐ, 100-TÓL 120-IG TARTÓ UTASÍTÁSOK ÍRTAK A 7. CSATORNÁRA.

```
200 REM OLVASD VISSZA AZ ÜGYFÉL NEVÉT ÉS ADATAIT
210 INPUT 6: M$, Q$
220 MAT INPUT 6: A
```

ITT AZONBAN VAN EGY NAGY HIBA. AZ "ÜGYFÉL NEVE" SZÖVEG, AZ N\$-BAN ÉS AZ S\$-BAN TÁROLT ÜGYFÉL KÉT NEVE ELÉ KERÜLT AZ ADATÁLLOMÁNYON: A 210. SOR ERRÓL NEM VESZ TUDOMÁST, ÍGY LESZ AZ S\$-BÓL A BEMENETI MÁTRIX ELSŐ TAGJA, AMI NYILVÁN ROSSZ.

AZ ADATÁLLOMÁNYOKAT HASZNÁLÓ PROGRAM FEJLESZTÉSÉNél, A FOLYAMATOS ELLENŐRZÉS CÉLJÁBÓL ÉRDEMES MINDEN ÁTVITEL KÖR MÖND-JUK AZ ELSŐ VAGY UTOLSÓ ELEMET KIKÜLDENI A TERMINÁLRA. A FENTI RUTIN ((KIJAVÍTOTT)) KÖZBENSŐ VÁLTOZATA A KÖVETKEZŐ LESZ:

```
200 REM OLVASD VISSZA AZ ÜGYFÉL NEVÉT ÉS ADATAIT
210 INPUT 6: T$, M$, Q$
211 PRINT 210; Q$
220 MAT INPUT 6: A
221 PRINT 220; A(1,1)
```

211. ÉS 221. SOROK
CSUPÁN A BELÖVÉS
CÉLJÁBÓL

ADATÁLLOMÁNY FAJTÁK

ALAPVETŐEN 4 FAJTA LÉTEZIK: NÉMELYIK BASIC RENDSZER CSUPÁN EGYET ISMER, MÁSONK TÖBBET. EZEK VÁZLATOS ÁTTEKINTÉSE A KÖVETKEZŐ:




AZ ADATÁLLOMÁNYOK TERÜLETÉN "VERTIKÁLIS" OSZTÁLYOZÁS VAN, AMELY ELVÁLASZTJA A SZEKVENCIALIS ADATÁLLOMÁNYOKAT A KÖZVETLEN HOZZÁFÉRÉSŰTŐL.

SZEKVENCIALIS ADATÁLLOMÁNYOK



MINDEN ADATÁLLOMÁNYNAK VAN EGY KÉPZELETBELI MUTATÓJA, AMELY AZ ÁLLOMÁNY ELEJÉRŐL INDUL, ÉS AMELYET A "RESET" UTASÍTÁSSAL BÁRMIKOR AZ ÁLLOMÁNY ELEJÉRE LEHET ÁLLÍTANI.

AMIKOR AZ ÁLLOMÁNYBA A "PRINT" UTASÍTÁSSAL INFORMÁCIÓT KÜLDESZ, AZ ÚJ INFORMÁCIÓ AZ ÁLLOMÁNY VÉGÉRE KERÜL, ÉS A MUTATÓ KÖZVETLENŰL AZ ÍGY LÉTREHOZOTT ÁLLOMÁNY VÉGE UTANI HELYRE MUTAT.




AMIKOR AZ ADATÁLLOMÁNYBÓL INFORMÁCIÓT OLVASOL, AKKOR AZT AZ INFORMÁCIÓT KAPOD, AMIRE A MUTATÓ MUTAT. A MUTATÓ PEDIG TOVÁBBMEGY A KÖVETKEZŐ INFORMÁCIÓEGYSÉGHEZ, ÉS KÉSZEN ÁLL A KÖVETKEZŐ BEMENETI UTASÍTÁS FOGADÁSÁRA.

EGYÉRTELMEŰ, HOGY ADDIG NEM OLVASHATSZ ADATOT A FELÍRÁS ALATT LEVŐ ÁLLOMÁNYRÓL, MÍG BE NEM FEJEZTED A FELÍRÁST, ÉS NEM ÁLLÍTOTTAD A KÉPZELETBELI MUTATÓT "RESET"-TEL AZ ADATÁLLOMÁNY ELEJÉRE. NÉMELYIK BASIC-BEN EZ AZ "OUTPUT" ÁLLOMÁNY LEZÁRÁSÁT ÉS ENNEK "INPUT" ÁLLOMÁNYKÉNT VALÓ ÚJRA MEGNYITÁSÁT JELENTI UGYANAZON VAGY MÁS CSATORNÁN. AZ ELŐZŐ OLDALON SZEREPLŐ PÉLDÁK A SZEKVENCIALIS ADATÁLLOMÁNYOK HASZNÁLATÁT MUTATTÁK BE.

KÖZVETLEN HOZZÁFÉRÉSŰ ADATÁLLOMÁNYOK (EZEKET NÉHA RANDOM ACCESS (VÉLETLEN ELÉRÉSŰ) ADATÁLLOMÁNYOKNAK NEVEZIK: TÉVES ELNEVEZÉS, HISZEN SENKI SEM HASZNÁL VÉLETLENSZERŰEN KIVÁLASZTOTT REKORDOKAT.) A KÖZVETLEN HOZZÁFÉRÉSŰ ADATÁLLOMÁNYOKBAN VEZÉRELHETED A MUTATÓ HELYZETÉT. AMIKOR A MUTATÓT BEÁLLÍTOTTAD, ÚGY KEZELHETED AZ ADATÁLLOMÁNYT, MINTHA AZ SZEKVENCIALIS LENNE. TEHÁT A KÖZVETLEN HOZZÁFÉRÉSŰ ÁLLOMÁNYOKAT IS KEZELŐ BASIC RENDSZEREKNEK OLYAN KÜLÖNLEGES UTASÍTÁSAINAK IS KELL LENNI, AMELYEK "N REKORD"-RA ÁLLÍTJÁK A MUTATÓT, ÉS OLYAN SZOLGÁLTATÁSAINAK, AMELYEK MEGHATÁROZZÁK, HOGY HOVA KERÜLT A MUTATÓ. A BASIC-BEN A KÖZVETLENEK KEVÉSBÉ GYAKORIAK, MINT A SZEKVENCIALISOK, EZEKRŐL A TOVÁBBIAKBAN NEM LESZ SZÓ.

LÉTEZIK AZ ADATÁLLOMÁNY TERÜLETÉNEK "HORIZONTÁLIS" FELOSZTÁSA IS, AMELY A KÓDOLT ÁLLOMÁNYOKAT VALASZTJA EL A BINÁRIS ÁLLOMÁNYOKTÓL.



KÓDOLT ADATÁLLOMÁNYOK EZEKET A TERMINÁLON VAGY A SORNYOMTATÓN KINYOMTATHATOD MINDEN BETŰ, SZÁMJEGY ÉS SZIMBÓLUM AZ ÁLLOMÁNYBAN EGYENKENT KERÜL TÁROLÁSRA; ÁLTALÁBAN A.S.C.I.I. KÓDBAN. A KÓDOLT ÁLLOMÁNYOK PROBLÉMÁJA AZ, HOGY AZOK A SZÁMÍTÓGÉPEK, AMELYEK BINÁRIS ARITMETIKÁT HASZNÁLNAK, KÓDOLT DECIMÁLISBÓL BINÁRISSÁ KELL ALAKÍTSÁK A SZÁMOKAT A BEVITEL FOLYAMÁN, ÉS BINÁRISBÓL KÓDOLT DECIMÁLISSÁ A KIVITEL SORÁN. EZ FELESLEGES MŰNKA, HA NINCS SZÜKSÉG OLVAHATÓ EREDMÉNYEK NYOMTATÁSÁRA; TOVÁBBÁ ELŐFORDULHAT, HOGY A KÉT KONVERTÁLÁSNÁL PONTOSSÁG-VEZTÉSÉG IS FELLÉP.

BINÁRIS ADATÁLLOMÁNYOK EZEK SOKKAL TÖMÖREBBEN TÁROLJÁK AZ ADATOKAT, MINT AHOGY EZ KÓDOLT ÁLLOMÁNYOKKAL LEHETSÉGES. BE-ÉS KIVITEL ESETÉBEN NINCS KONVERZIÓRA SZÜKSÉG. MÁSFELŐL ZAGYVASÁGOT KAPNÁNK, HA A TERMINÁLON KI TUDNÁNK ŐKET NYOMTATNI. A BINÁRIS ÁLLOMÁNYOK SZIGORÚAN VÉVE, SZÁMÍTÁSOK KÖZBENSŐ EREDMÉNYEINEK TÁROLÁSÁRA ÉS BEOLVASÁSÁRA VALÓK, TOVÁBBI SZÁMÍTÁSOK ELVÉGZÉSE CÉLJÁBÓL. SOK BASIC-RENDSZERNEK, MELY A KÓDOLT ÁLLOMÁNYOK KIEGÉSZÍTÉSEKÉNT BINÁRIS ÁLLOMÁNYOKKAL IS DOLGOZIK, MEGKÜLÖNBÖZTETETT UTASÍTÁSAI VANNAK BINÁRIS BEVITELHEZ, ÉS KIÍRÁSHOZ. TÍPIKUS PÉLDÁK:

A "GET" SZÓ AZ "INPUT" HELYETT
A "PUT" SZÓ AZ "PRINT" HELYETT

VANNAK RENDSZEREK, AMELYEK A "READ" ÉS A "WRITE" SZAVAKAT IS HASZNÁLJÁK. NÉMEZYIK BASIC VÁLTOZATBAN MEGENGEDNEK SZÖVEGBŐL ÁLLÓ BINÁRIS ÁLLOMÁNYOKAT IS. ISMÉT MÁS RENDSZERBEN SZÁMOK ÉS SZÖVEGEK KEVERÉKÉBŐL ÁLLÓ BINÁRIS ÁLLOMÁNYOK IS LÉTEZNEK.

NEM VALÓSZÍNŰ, HOGY SIKERÜL OLYAN TELJESEN HORDOZHATÓ BASIC PROGRAMOKAT ÍRNI, AMELYEK ÁLLOMÁNYOKAT HASZNÁLNAK, AZONBAN HA RAGASZKODSZ A KÓDOLT SZEKVENCIÁLIS ADATÁLLOMÁNYOK HASZNÁLATÁHOZ, PROGRAMOD NEM IGÉNYEL SOK VÁLTOZTATÁST AHHOZ, HOGY MÁS RENDSZEREN IS FUTHASSON.



SZINTAXIS

SZINTAXIS

EZ A RÉSZ A SZINTAXIS, A KÖNYVBEN LEÍRT BASIC ÖSSZEFOGLALÓJA. AZ OLVASÓ BASIC VÁLTOZATA FELTEHETŐLEG ELTÉR ETTŐL, DE HA VAN AZ ALÁBBIHOZ HASONLÓ SZINTAXISDEFINÍCIÓJA, AKKOR A KÜLÖNBBSÉGEK LEGTÖBBJÉT KÖNNYEN ÉSZRE LEHET VENNI AZ ÖSSZEHASONLÍTÁSNÁL.

EGY MÓDOSÍTOTT "BACKUS-NAUR" JELÖLÉSRENDSZERT HASZNÁLTUNK AZ ÖSSZEFOGLALÁSRA. SOK JELÖLÉSRENDSZER JÖTT LÉTRE A BASIC SZINTAXISÁNAK DEFINIÁLÁSÁRA, ÉS NÉMELYIKÜK ELÉG NEHEZEN OLVASHATÓ. MEGPRÓBÁLTAM EZT A VÁLTOZATOT A LEHETŐ LEGOLVASHATÓBBÁ TENNI ANÉLKÜL, HOGY SZIGORÚSÁGOMBÓL VESZÍTETTEM VOLNA, DE AZ OLVASÓ MÉG ÍGY IS NEHÉZKESNEK TALAÍHATJA.

SZIMBÓLUMOK

⇒

|

[]

{ }

AMIKET A DEFINÍCIÓBAN HASZNÁLUNK

AZT JELENTI, "DEFINÍCIÓ SZERINT"

AZT JELENTI, HOGY "VAGY"

SZÖGLETES ZÁRÓJELBE TETTÜK AZOKAT A DOLGOKAT, AMELYEK EGYSZER SZEREPELNEK, VAGY ELHAGYHATÓK A DEFINIÁLT FOGALOMBÓL.

KAPCSOS ZÁRÓJELBE TETTÜK AZOKAT A DOLGOKAT, MELYEK EGYSZER VAGY TÖBBSZÖR SZEREPELNEK, VAGY ELHAGYHATÓK A DEFINIÁLT FOGALOMBÓL.

BETŰTÍPUSOK

Kisbetűk

NAGYBETŰK
(+ - / * ↑)
< = : ; , . \$ ' " >
0 1 2

AMIKET A DEFINÍCIÓBAN HASZNÁLUNK

OLYAN MAGYARÁZATOKBAN SZEREPELNEK, AHOL A HELYZET EGYÉRTELMŰ, VAGY AHOL A SPECIÁLIS JELÖLÉSMÓD NEM FELELNE MEG.

BETŰK, SZÁMJEGYEK ÉS SZIMBÓLUMOK JELELÉSÉRE HASZNÁLTUK, AMELYEK VÁLTOZATLAN FORMÁBAN SZOLGÁLNAK PÉLDAKÉNT A DEFINIÁLT FOGALOMRA.

MEGJEGYZÉSEK * PÉLDÁK.

☾ ☽

AZ ÁRNYÉKOLT ZÁRÓJELEK OLYAN MEGJEGYZÉSEKET ÉS PÉLDAKAT FOGNAK KÖZRE, AMELYEK NEM TARTOZNAK DEFINÍCIÓKHOZ.

számjegy	⇒ egy a 0-tól 9-ig terjedő számjegyek közül	
betű	⇒ egy az A-tól Z-ig terjedő betűk közül	
előjel	⇒ + -	
operátor	⇒ + - * / ↑	20
választó	⇒ , ;	28
relációs operátor	⇒ = < > <= >= <>	41
szöveg	⇒ "bármely karakter, kivéve idézőjelet"	12
szám	⇒ egészszám 1 - 9999 -ig	7
függvény	⇒ SGN SIN COS TAN ATN EXP ABS LOG SQR INT RND FN betű	22, 24
konstans	⇒ RND FN betű	25, 26
(A DEFINÍCIÓ MEGENGEDI RND-T ÉS RND(X)-ET VALAMINT FNA-T ÉS FNA(X)-ET IS)		

MÁSODSZOR AZ ÖSSZETETT FOGALMAK (AZ ELEMECTŐL SZÁNDÉKOSAN MEGKÜLÖNBÖZTETVE) :

egész	⇒ számjegy { számjegy }	
(pl.: 0, 012, 87654 : A HOSSZ A NYELV EGYES VÁLTOZATAIBAN KÜLÖN-KÜLÖN KORLÁTOZOTT)		
exponens	⇒ E [előjel] egész	
szám	⇒ egész [.] [egész] [exponens] . egész [exponens]	9
(pl.: 12, 12.2, 12.2 E+6, .12 E-6, 12., 12.E6)		
adat	⇒ [előjel] szám szöveg	16
(pl.: 2, -2.5, "ABC" : MINT A "DATA" UTASÍTÁSOKBAN)		
változó	⇒ numerikus szöveges	
numerikus	⇒ betű [számjegy] betű (kifejezés [, kifejezés])	10, 60
(pl.: A, A5, A(4+I), A(I, 2*J))		
szöveges	⇒ betű \$ [kifejezés]	13, 60
(pl.: A\$, A\$(2+I))		
lexikális	⇒ szöveg szöveges változó	41
(pl.: "ABC", A\$, A\$(2+I) : MINT AZ "IF" UTASÍTÁSOKBAN)		
tag	⇒ szám numerikus függvény (kifejezés) konstans (kifejezés)	
(pl.: 6.5, A(I, J), RND, INT(2+B), (-3*I+J))		
kifejezés	⇒ [előjel] tag { operátor tag }	20
(pl.: A, + A(I, J), + A(L, J) * INT(3*A+B))		
deklaráció	⇒ betű (egész [, egész]) betű \$ (egész)	62
(pl.: A(4), A(2, 30), A\$(26) : MINT A "DIM" UTASÍTÁSOKBAN)		
kinyomtatható	⇒ kifejezés lexikális TAB (kifejezés)	28
(pl.: A(I, J) * INT(ABS(1+P)), "ABC", A\$(Q), TAB(X))		
beállítás	⇒ (kifejezés, kifejezés)	79
(pl.: (2 * A, B(I, J) / 6) : MINT BIZONYOS "MAT" UTASÍTÁSOKBAN)		

SZINTAXIS

(FOLYTATÁSA)

MOST PEDIG A BASIC UTASÍTÁSOK KÖVETKEZNEK.
OLDALSZÁM: 62

TÖMBÖK DIMENZIÓI :

sorszám DIM deklaráció { , deklaráció }

ÉRTÉKADÁS :

sorszám LET numerikus = kifejezés 11
sorszám LET szöveges = szöveges változó | szöveg 13
sorszám DEF FN betű [(betű [számjegy])] = kifejezés 26

BEVITEL :

sorszám DATA adat { , adat } 16
sorszám READ változó { , változó } 16
sorszám RESTORE ((" DATA " UTASÍTÁSOKHOZ)) 17
sorszám INPUT [egész :] változó { , változó } 18, 120
sorszám RESET egész { , egész } ((FILE - OKNÁL)) 121

KIVITEL :

sorszám PRINT [egész :] USING sorszám { , változó } 34, 120
sorszám : ((A SOR-KÉP FELÉPÍTÉSE TÚL VÁLTOZATOS AHHOZ, HOGY ITT DEFINIÁLJUK)) 34-37
sorszám PRINT [egész :][kinyomtatható { elválasztó kinyomtatható } [elválasztó]] 28-32
((NE HASZNÁLJUNK VEGSZÓT A TAB() UTÁN)) 120

MÁTRIXOK :

sorszám MAT betű = betű 76
sorszám MAT betű = betű + betű 78
sorszám MAT betű = betű - betű 80
sorszám MAT betű = (kifejezés) * betű 80
sorszám MAT betű = TRN (betű) 82
sorszám MAT betű = ZER [beállítás] 84
sorszám MAT betű = IDN [beállítás] 86
sorszám MAT betű = CON [beállítás] 87
sorszám MAT betű = betű * betű 87
sorszám MAT betű = INV (betű) 88
sorszám MAT READ betű [beállítás]{ , betű [beállítás]} 94
sorszám MAT INPUT [egész :] betű [beállítás]{ , betű [beállítás]} 96
sorszám MAT PRINT [egész :] betű {elválasztó betű}{elválasztó} 98

V EGYES :

sorszám REM { bármilyen karakter }
sorszám END

OLDALSZÁM:

8
7

V E Z É R L É S :

sorszám GO TO sorszám (("GOTO" ÁLTALÁBAN MEGENGEDETT)) 39
sorszám GO SUB sorszám (("GOSUB" ÁLTALÁBAN MEGENGEDETT)) 40
sorszám RETURN 52
sorszám ON kifejezés GO TO sorszám { , sorszám } 52
((VIGYÁZZ A KÜLÖNBÖZŐ BASIC VÁLTOZATOK HASZNÁLATÁVAL)) 46
sorszám IF kifejezés relációs operátor kifejezés THEN sorszám 41
sorszám IF lexikális relációs operátor lexikális THEN sorszám 41
(("GO TO" GYAKORI A "THEN" HELYETT))
sorszám FOR betű [szám] = kifejezés TO kifejezés [STEP kifejezés] 48
sorszám NEXT betű [szám] 48
sorszám STOP 42

A D A T Á L L O M Á N Y K E Z E L É S :

((TÚL SZERTEÁGAZÓ AHHOZ, HOGY ITT DEFINIÁLNI LEHESSEN))

119

P A R A N C S O K :

RUN 111
LIST 114
CATALOG 115
SAVE [file] 115, 117
UNSAVE file 116
OLD file 116
NEW 117


file ⇒ rendszertől függő szintaxisú adatállománynév

((A PARANCSONK SZÁMBAN ÉS SZINTAXISBAN NAGYMÉRTÉKBEN
ELTÉRNEK A KÜLÖNBÖZŐ, BASIC -ET SZOLGÁLTATÓ RENDSZEREK
ESETÉN))

FÜGGELÉK

A KÖNYV ELŐZŐ FEJEZETEI A BASIC NYELV ÁLTALÁNOS LEÍRÁSÁT TARTALMAZZÁK. SOK HELYEN UTAL ARRRA A SZERZŐ, HOGY EGY-EGY UTASÍTÁS VAGY PARANCS VÉGREHAJTÁSA A BASIC NYELVEKBEN ELTÉRŐ.

A FÜGGELEK CÉLJA A SOK LEHETSÉGESBŐL EGY KONKRÉT MEGVALÓSÍTÁS, AZ



SZEMÉLYI SZÁMÍTÓGÉPEN[†] HASZNÁLT BASIC VÁLTOZAT BEMUTATÁSA.

HOGYAN HASZNÁLD A FÜGGELEKET ?

OLVASD A KÖNYV FEJEZETEIT, ÉS AZ OLYAN HELYEKEN, AHOL A BASIC -EK KÜLÖNBÖZŐSÉGÉRE UTALÓ MEGJEGYZÉST TALÁLSZ, LAPOZD FEL A FÜGGELEKBEN A HOZZÁ TARTOZÓ MEGJEGYZÉST, KIEGÉSZÍTÉST VAGY MÓDOSÍTÁST !

A FÜGGELEK MEGISMÉTLI A HIVATKOZOTT LAPSZÁMOT ÉS CÍMET, PL. **22 FÜGGVÉNYEK** AZT JELENTI, HOGY A

22. LAPON KEZDŐDŐ FÜGGVÉNYEK FEJEZETTEL KAPCSOLATOSAK A FÜGGELEK MEGJEGYZÉSEI.

A KÖNYV A FÜGGELEKKEL KIEGÉSZÍTVE SEM ISMERTETI AZ **MO8X** BASIC ÖSSZES LEHETŐSÉGÉT.


A NYELV RÉSZLETEKBE MENŐ MEGISMERÉSÉHEZ HASZNÁLD AZ **MO8X BASIC** KÉZIKÖNYVET !

(SZÁMÍTÁSTECHNIKAI KOORDINÁCIÓS INTÉZET, 1983)

[†] AZ **MO8X** ÉS A **proper 8** A SZÁMÍTÁSTECHNIKAI KOORDINÁCIÓS INTÉZET ÁLTAL FEJLESZTETT PROFESSZIONÁLIS SZEMÉLYI SZÁMÍTÓGÉP CSALÁD ELSŐ KÉT, EGYMÁSSAL PROGRAMKOMPATIBILIS TAGJA. EZÉRT ÉRTELEMSZERŰEN A TOVÁBBIKBAN AZ **MO8X** ELNEVEZÉS EGYÚTTAL A CSALÁDRA IS UTAL.

5 BILLENTYŰZET

∞ AZ MOBX BILLENTYŰZETE ∞



AZ MOBX BILLENTYŰZETÉN ÖNMAGUKBAN LEÜTÖTT BETŰK KÉPE, KISBETŰ. HA A SHIFT BILLENTYŰVEL EGYÜTT ÜTÖD LE, NAGYBETŰ JELENIK MEG, A KÉPERNYŐN. PROGRAMSOROK BEGÉPELÉSÉNÉL BARMELVIKET HASZNÁLHATOD, A BASIC EGYFORMÁN ÉRTI.

AZ EL (RUB OUT) BILLENTYŰVEL A KÉPERNYŐN A CURSORT BALRA MOZGATHATJUK ÉS KARAKTEREKET TÖRÖLHETÜNK.

A DEL BILLENTYŰ LEÜTÉSÉVEL AZ ÉPPEN BEÍRT KARAKTERT TÖRÖLHETED.

HA ÚJ SORT AKARSZ KEZDENI, ÜSD LE A CR BILLENTYŰT!

HA EGY PROGRAMSORBA TÖBB KARAKTERT AKARSZ ÍRNI, MINT AMENNYI A KÉPERNYŐ SORÁBA ELFÉR, A → BILLENTYŰT HASZNÁLD, ÉS FOLYTASD A GÉPELÉST A KÖVETKEZŐ SORBAN!


HA ARITMETIKAI MŰVELETEKET AKARSZ VÉGREHAJTANI, MŰVELETI JELEKNEK A KÖVETKEZŐKET HASZNÁLD:

+	=	add össze
-	=	vond ki
*	=	szorozd össze
/	=	oszd el
^	=	emeld hatványra

6 CÉPELÉS

AZ **MOBX** BASIC LEGFELEBB 255 KARAKTERBŐL ÁLLÓ SORT ENGED MEG.

NAGY A SZABADSÁG A SZÓKÖZÖK HASZNÁLATÁBAN. CSAK AZ UTASÍTÁSSZAVAKBAN (LET, FOR, INPUT stb.) NEM SZABAD A BETŰKET SZÓKÖZZEL ELVÁLÁSZTANI.




7 SORSZÁMOK

Egy programban a legnagyobb megadható sorszám 65529.

9 SZÁMOK

AZ **MOBX** BASIC-BEN A SZÁMOK SZIMPLA ÉS DUPLA PONTOSSÁGÚAK LEHETNEK.



★ AZ **MOBX** BASIC A TIZES RENDSZERŰ SZÁMOKON KÍVÜL A TIZENHATOS RENDSZERŰ ÉS NYOLCAS RENDSZERŰ SZÁMOK HASZNÁLATÁT IS MEGENGEDLI.

16-OS $\&H20 = 32$ 8-AS $\&34 = 28$

10 VÁLTOZÓK

AZ **MOBX** BASIC-BEN A VÁLTOZÓK NEVE 40 ÉRVÉNYES KARAKTERIG ADHATÓ MEG. AZ ELSŐ KARAKTER CSAK BETŰ LEHET. A VÁLTOZÓK SZÁMOT VAGY SZÖVEGET JELÖLNEK.

	név	típus
10	LET	HATÁR % = 50
20	LET	ÁT MÉRŐ = 6.5
30	LET	HOSSZ! = 27.25
40	LET	PI# = 3.14
50	LET	A\$ = "EREDMÉNY"

a! elhagyható

TÍPUSOK :

%	EGÉSZ SZÁM
!	VALÓS SZIMPLA PONTOSSÁGÚ
#	VALÓS DUPLA PONTOSSÁGÚ
\$	SZÖVEG (STRING)

13 SZÖVEGES VÁLTOZÓK

AZ **MOBX** BASIC-BEN A SZÖVEGES VÁLTOZÓK SZÁMA SINCSE KORLATOZVA BÁRMILYEN \$ JELRE VÉGZŐDŐ NÉV HASZNÁLHATÓ ERRE A CÉLRA.

AZ EGY VÁLTOZÓBAN TÁROLT SZÖVEG HOSSZA MAXIMUM 255 KARAKTER LEHET.

10 LET Z\$ = "255 KARAKTER TÁROLHATÓ"

17 RESTORE

A "RESTORE" UTASÍTÁS ARGUMENTUMAKÉNT MEGADHATÓ EGY SORSZÁM.

A KÖVETKEZŐ "READ" UTASÍTÁS AZ EBBEN A SORBAN LEVŐ "DATA" UTASÍTÁS ADATAIVAL FOLYTATJA AZ OLVASÁST.

```

10 DATA 1, 2, 3
20 DATA 4, 5, 6
30 DATA 7, 8, 9
40 READ A, B, C, D, E, F
50 PRINT A; B; C; D; E; F
60 RESTORE 20
70 READ G, H, I
80 PRINT G; H; I
90 END

```

RUN

```

1 2 3 4 5 6
4 5 6

```

18 INPUT

AZ ADAT BEÍRÁSA UTÁN A  BILLENTYŰT NYOMD LE! HA NEM VÁLASZOLSZ, A RENDSZER A VÉGTELENSÉGIG VÁR.

19 ?

MI TÖRTÉNIK, HA HIBÁS A VÁLASZ?

```
10 INPUT A, B$, C
RUN
? 2, ÉV, 26, 3
" ? REDO FROM START "
? 2, ÉV
" ? REDO FROM START "
? ÉV, HÓ, 26
" ? REDO FROM START "
```

TÚL SOK ADAT

TÚL KEVÉB ADAT

ROSSZ ADATFAJTA

A BEOLVASOTT ADATOKAT AZ MOBX BASIC NEM RENDELI HOZZÁ A VÁLTOZÓKHOZ, MÍG NEM HIBATLAN A TELJES VÁLASZ.

21 KIFEJEZÉSEK


FIGYELEMRE MÉLTÓ PONTOK :

HA EGY "LET A = B/C" UTASÍTÁSBAN A KIFEJEZÉS KIÉRTÉKELÉSE SORÁN C=0, AZ MOBX BASIC KÉPERNYŐJÉN MEGJELENIK A "DIVISION BY ZERO" ÜZENET. EZEK UTÁN "A" ÉRTÉKE A SZÁMLÁLÓ ELŐJELEVEL AZONOS "GÉP VÉGTELEN" ÉRTÉKET VÁP, VAGYIS $\pm 10^{38}$. A FELDOLGOZÁS FOLYTATÓDIK.

HASONLÓAN, HA HATVÁNYOZÁS SORÁN NULLÁT AKARUNK NEGATÍV HATVÁNYRA EMELNI, AZ EREDMÉNY $+ 10^{38}$, ÉS A FELDOLGOZÁS FOLYTATÓDIK A "DIVISION BY ZERO" ÜZENET KIÍRÁSA UTÁN.


22 FÜGGVÉNYEK

AZ **MO8X** BASIC A BEÉPÍTETT FÜGGVÉNYEK NAGY VÁLASZTÉKÁT NYÚJTJA.



AZ **ELEMI** FÜGGVÉNYEK \otimes SGN, ABS, SQR, INT, LOG, EXP \otimes LEÍRÁSÁT LÁSD A 23. OLDALON.

A **TRIGONOMETRIKUS** FÜGGVÉNYEK \otimes SIN, COS, TAN, ATN \otimes LEÍRÁSÁT A 24. OLDAL TARTALMAZZA.



★ RÉSZLETES LEÍRÁSUK AZ **MO8X** BASIC KÉZIKÖNYVÉBEN.

SZÖVEG KEZELŐ FÜGGVÉNYEK

10 LET A \$ = "FÜGGVÉNYEK"

SZÖVEGBŐL ADOTT RÉSZ KI-
HASÍTÁSÁT, EGY TERÜLET
SZÓKÖZZEL VAGY ADOTT
KARAKTERREL VALÓ FEL-
TÖLTÉSÉT STB. VÉGIK.

LEFT\$ (X\$, I)

20 PRINT LEFT\$ (A\$, 4)



FÜGG

X\$ BAL OLDALI I KARAK-
TERÉT ADJA.

MID\$ (X\$, I, J)

30 PRINT MID\$ (A\$, 5, 4)



VÉNY

X\$ I-EDIK KARAKTERÉRŐL
J KARAKTERT AD.

RIGHT\$ (X\$, I)

40 PRINT RIGHT\$ (A\$, 6)



VÉNYEK

X\$ JOBB OLDALI I KARAK-
TERÉT ADJA.

SPACE\$ (X)

50 PRINT SPACE\$ (3); "25"



25

X DARAB "SZÓKÖZ"
KARAKTERT AD.

STRING\$ (I, J)

60 PRINT STRING\$ (4, "A")



AAAA

I DARAB
J KARAKTERT AD.

CHR\$ (I)

70 PRINT CHR\$ (65)



A

AZ I ÉRTÉKÉNEK MEGFELE-
LŐ KARAKTERT ADJA.
(ASC II KÓD)

KONVERTÁLÓ

FÜGGVÉNYEK

GYAKRAN ELŐFORDUL, HOGY NUMERIKUS ADATOT SZÖVEGES VÁLTOZÓBAN TÁROLUNK. MIVEL ARITMETIKAI MŰVELETEKET, CSAK "SZÁM" TÍPUSÚ VÁLTOZÓKKAL VÉGEZHETÜNK, GYAKRAN VAN SZÜKSÉG ARRRA, HOGY A VÁLTOZÓK TÍPUSÁT MEGVÁLTOZTASSUK.

6.28

Számból szám

6

AB

6566

Szövegből szám

33.34

Számból szöveg

VAL (X\$)

SZÖVEGES TÍPUSÚ VÁLTOZÓ SZÁMÉRTÉKÉT ADJA.

```
10 INPUT A$
20 A1 = VAL(A$) * 3
30 PRINT A$;A1
40 END
```

aritmetikai művelet csak így végezhető A\$-al

RUN

2.8

8 24

STR\$(X)

EGÉSZ SZÁM SZÖVEG TÍPUSÚ FORMÁJÁT ADJA.

```
10 A = 2643
20 B$ = LEFT$(STR$(A),2)
30 PRINT B$
```

"szöveg"-művelet csak így végezhető számmal

RUN

26

HA ÉRDEKEL AZ [M08X] BASIC TÖBBI KONVERTÁLÓ FÜGGVÉNYE CDBL, CINT, CSNG, FIX, ASC, CVI, CVS, CVD, HEX\$, OKT\$, MKI\$, MKS\$, MKD\$, NÉZZ UTANA AZ [M08X] BASIC KEZIKÖNYVBEN!

TÁRRAL

KAPCSOLATOS

FÜGGVÉNYEK

FRE (0)

A BASIC ÁLTAL NEM HASZNÁLT TÁR NAGYSÁGÁT ADJA VÁLASZUL.

PEEK (1)

A TÁR I-EDIK BYTE-JÁNAK TARTALMAT ADJA VISSZA.

25 RAD

MEGJEGYZÉS : AZ **MOBX** BASICBEN AZ RND-NEK ADHATÓ ARGUMENTUM IS. RND (X)

- $X > 0$ A VÉLETLENSZÁM - CIKLUS KÖVETKEZŐ SZÁMÁT GENERÁLJA. EZ AZONOS AZ ARGUMENTUM NÉLKÜLI RND-VEL.
- $X = 0$ A CIKLUS UTOLJÁRA HASZNÁLT SZÁMÁT GENERÁLJA ÚJRA.
- $X < 0$ MINDÍG AZONOS CIKLUS INDUL ÚJRA MINDEN $X < 0$ ESETÉN.

RANDOMIZE

A "RANDOMIZE" UTASÍTÁSNAK IS VAN ARGUMENTUMA.

10 RANDOMIZE (2) A VÉLETLENSZÁM - GENERÁTOR CIKLUSÁT AZ ARGUMENTUMKÉNT SZEREPLŐ KIFEJEZÉS ÉRTÉKE SZERINT INDÍTJA. ILYENKOR AZ RND ARGUMENTUMÁT HA VAN ∞ FIGYELMEN KÍVÜL HAGYJA.

HA "RANDOMIZE" UTASÍTÁST ARGUMENTUM NÉLKÜL HASZNÁLUNK, AKKOR ARGUMENTUMOT A PROGRAM FUTÁSA KÖZBEN KELL MEGADNI.

```

10 RANDOMIZE
20 FOR I = 1 TO 3
30 PRINT RND
40 NEXT
50 END

RUN
RANDOM NUMBER SEED? 4
.88598 .484668 .586328
OK
RUN
RANDOM NUMBER SEED? 4
OK

```

BASIC KÉRDEZI

BEGÉPELT VÁLASZ

26 DEF

AZ **MOBX** BASIC-BEN A DEFINIÁLT FÜGGVÉNYEK SZÁMA NINCS KORLÁTOZVA.

a függvény neve max. 40 karakter, az első csak betű lehet (é. változó)

```
10 DEF FN PYTH(A,B)=SQR(A^2+B^2)
```

egynél több formális argumentum is megengedett

★ HA LEFUTTATOD A 26. OLDAL ALJÁN LEVŐ KIS PROGRAMOT, AZ **MOBX** BASIC NULLÁT ÍR KI.

34 PRINT USING

AZ **MO8X** BASIC "PRINT USING" UTASÍTÁSA NAGYMÉRTÉKBEN ELTÉR A KÖNYV ÁLTAL LEÍRTAKTÓL. EZERT AZ ALÁBBIK NEMCSAK A KÜLÖNBSEGEKET, HANEM A TELJES UTASÍTÁSLEÍRÁST TARTALMAZZÁK.

A KIÍRNI KÍVÁNT SZÁMOK VAGY SZÖVEG FORMÁTUMÁT, A **MINTA**-T **NEM KÜLÖN UTASÍTÁSSOR** TARTALMAZZA, HANEM MAGA A PRINT USING UTASÍTÁS KÉT " " JEL KÖZÉ ZÁRVA.

```
10 LET A=2
20 PRINT USING "###" ; 634 , A , 13
30 END
RUN
634 2 13
```

"minta" a "minta"t a listától ; választja el.

kiírandók listája

EGY "PRINT USING" UTASÍTÁS EGYETLEN MINTÁT TARTALMAZHAT.

SZÁM KIÍRÁSÁHOZ A **MINTA**-BAN TÖBB SPECIÁLIS KARAKTER IS LEHET :

JELÖLI A SZÁMJEGYEKET.

```
10 PRINT USING "####" ; 4612 , 24
20 END
RUN
4612 24
```

2 szóköz

kiírás jobbra igazítva

○ **T**IZEDESPONT A **MINTA**-BAN BÁRHOL LEHET.

```
10 PRINT USING "##.##" ; 10.2 , .78 , 66.789
20 END
RUN
10.20 0.78 66.79
```

tizedespont előtt legalább egy számjegy kiíródik

a tizedes törtet a MINTA pontosságáig kerekíti

tizedespont után minden számjegyet kiír

+ P LUSZ JEL LEHET A **MINTA** ELEJÉN VAGY VÉGÉN.

```
10 PRINT USING "+##.##"; -68.95, 2.4, -.9  
20 END
```

RUN

-68.95 + 2.40 - 0.90

az előjelet mindig kiírja

- MÍNUSZ JEL IS LEHET A **MINTA** ELEJÉN VAGY VÉGÉN.

```
10 PRINT USING "##.##-"; -68.95, 22.449, -7.01  
20 END
```

RUN

68.95- 22.45 7.01-

pozitív előjel helyére szóköz kerül.

csak a negatív előjel íródik ki.

*** * N** A A **MINTA** ELEJÉRE KÉT CSILLAG KARAKTERT ÍRUNK, A SZÁMMEZŐNEK AZOKAT A HELYEIT, AHOVA NEM KERÜL SZÁMJEGY, "X" KARAKTERREL TÖLTI FÖL.

```
10 PRINT USING "**##.##"; 12.39, 0.9, -765.1  
20 END
```

RUN

** 12.4 *** 0.9 - 765.1

a csillagot is helyettesítheti számjegy

\$ \$ K ARAKTEREK A **MINTA** ELEJÉN AZT EREDMÉNYEZIK, HOGY A KIÍRT SZÁMOT MEGELŐZI EGY \$ JEL.

```
10 PRINT USING "$$###.##"; 456.78, 5112  
20 END
```

RUN

\$ 456.8 \$ 5112.0

a \$ helyére is kerülhet számjegy



MA A **MINTA**-BA A SZÁMJEGY KARAKTEREK (#) UTÁN ^^^^ KARAKTEREKET ÍRUNK, A KIÍRT SZÁM E-FORMAJÚ LESZ. A NÉGY ^ JEL AZ E+XX SZÁMÁRA BIZTOSÍT HELYET.

```
10 PRINT USING "#.## ^^^^"; 234.56
20 PRINT USING "#### ^^^^"; 888888
30 PRINT USING "+.## ^^^^"; 123
40 END
```

RUN


```
2.35 E +02
.8889 E +06
+.12 E +03
```

A ALÁHÚZÁS KARAKTER A **MINTA**-BAN BÁRHOL LEHET. A KÖVETKEZŐ KARAKTER MEGJELENIK A KIÍRT SZÖVEGBEN.

```
10 PRINT USING "_ !##.##_!"; 12.34
20 END
```

RUN

```
!12.34!
```



MA NAGYOBB SZÁMOT AKARUNK KIIRATNI, MINT AMEKKORÁT A **MINTA** LEHETŐVÉ TESZ, A SZÁM KIÍRÁSA ELŐTT % JEL JELENIK MEG.

```
10 PRINT USING "##.##"; 111.22
20 END
```

RUN

```
%111.22
```

túl nagy szám!
módosítsd a "minta"-t!

SZÖVEG

KIÍRÁSÁHOZ IS HASZNÁLHATJUK A "PRINT USING" UTASÍTÁST, HA A SZÖVEGNEK CSAK AZ ELSŐ KARAKTERÉT AKARJUK KIÍRNI.

EZT A ~~MINTA~~ -BAN MEGADOTT  KARAKTERREL JELEZZÜK.

```
10 A$="IDE" : B$="NÉZZ"  
20 PRINT USING "!" ; A$ ; B$ ; "!"  
RUN  
IN!
```

*vigyázz!
ez nem minta!*

40 GO TO

A PROGRAMOT MEGÁLLÍTANI EGYÜTTES LEÜTÉSÉVEL LEHET.

A

CTRL

+

C

BILLENTYŰK

41 IF - THEN

★ AZ **MO8X** BASIC MEGENGEDI AZ **IF-GOTO** UTASÍTÁS HASZNÁLATÁT IS.

```
20 IF A = B * C GOTO 50
```

★ HASZNÁLHATSZ **IF-THEN-ELSE** UTASÍTÁST IS.

```
30 IF A = B THEN 40 ELSE 100
```

itt folytasd, ha a feltétel igaz

különben, ha a feltétel nem igaz, itt folytasd

★ MIND A **THEN**, MIND AZ **ELSE** UTÁN KÖVETKEZHETNEK UTASÍTÁSOK IS.

```
40 IF X < Y THEN PRINT "KISEBB" ELSE PRINT "NAGYOBB"  
50 INPUT C
```

ha a feltétel igaz, hajtsd végre ezt az utasítást, utána folytasd a következő sorban (itt 50)

máskülönben hajtsd végre ezt az utasítást, majd folytasd a következő sorban (50)

★ A **THEN** ÉS AZ **ELSE** UTÁN KÖVETKEZŐ UTASÍTÁSOK SZÁMÁT CSAK A SOR HOSSZA (MAX. 255 KARAKTER) KORLÁTOZZA.

42 STOP

"STOP" UTASÍTÁS VÉGREHAJTÁSA UTÁN MEGJELENIK A KÉPERNYÓN A KÖVETKEZŐ ÜZENET:

```
BREAK IN LINE NNNNN
```

annak a sornak a száma, amelyben a végrehajtást megállító STOP utasítás van.

46

ON - GOTO

UTASÍTÁSBAN AZ "ON" UTÁN ÁLLÓ KIFEJEZÉS KIÉRTÉKELÉSEKOR AZ **MO8X** BASIC A SZÁM TÖRT-RÉSZÉT (HA VAN ILYEN) KEREKÍTI.

```
10 LET A = 1.6
20 ON A GOTO 50, 80
```

A értéke 2 lesz

a következő utasítás a listában másodiknak megadott 80 sor-számú utasítás lesz.

★ HA A KIFEJEZÉS ÉRTÉKE NULLA VAGY NAGYOBB, MINT AZ ELÁGAZÁSI CÍMEK SZÁMA, A VÉGREHAJTÁS A KÖVETKEZŐ SORBAN FOLYTATÓDIK.

★ HA A KIFEJEZÉS ÉRTÉKE NEGATÍV VAGY NAGYOBB, MINT 255, A PROGRAM NEM FOLYTATÓDIK.
A KÉPERNYŐN MEGJELENŐ HIBAÜZENET: "ILLEGAL FUNCTION CALL"

60

TÖMBÖK

AZ, **MO8X** BASIC NEMCSAK 1 VAGY 2 DIMENZIÓS, HANEM MAXIMUM 255 DIMENZIÓS TÖMBÖK HASZNÁLATÁT IS MEGENGEDI. EGY DIMENZION BELÜL AZ ELEMÉK SZÁMA (AZ INDEX) LEGFELJEBB 767 LEHET.

```
10 DIM A(15, 767, 100, ..... , 5)
```

1 2 3 255

max. indexérték

maximális dimenziószám

62

DIM

AZ `MO8X` BASIC MEGENGEDI, HOGY EGYDIMENZIÓS, 10-NEL NEM NAGYOBB INDEXŰ TÖMBÖKET "DIM" UTASÍTÁS NÉLKÜL IS HASZNÁLHASSUK.

63

DIM

AZ `MO8X` BASIC AZ INDEXEK SZÁMOZÁSÁT NULLÁVAL KEZDI.

HA EGYTŐL SZERETNÉL SZÁMOZNI, HASZNÁLD AZ **OPTION BASE 1** UTASÍTÁST!

HIBA, HA A VÁLTOZÓBAN HASZNÁLT DIMENZIÓ VAGY A DIMENZIÓN BELÜLI INDEX NAGYOBB A "DIM" UTASÍTÁSBAN DEFINIÁLTNÁL.

10	DIM	A(2, 1)
20	LET	A(1, 1, 1) = 36
30	LET	A(1, 3) = 83

MINDKÉT ESETBEN HIBAÜZENET JELENIK MEG A KÉPERNYŐN:

"SUBSCRIPT OUT OF RANGE"

★ **A** "DIM" UTASÍTÁSBAN HASZNÁLT VÁLTOZÓK KEZDŐ-ÉRTÉKÉT AZ `MO8X` BASIC NULLÁRA ÁLLÍTTJA.

76

MATRIXOK

AZ MOBXI BASIC - BEN NINCSENEK "MAT" UTASÍTÁSOK. TÖBBMŰVELETEKHEZ HASZNÁLJUK AZ 5. FEJEZETBEN AZ EGYES "MAT" MŰVELETEKNÉL LEÍRT CIKLUSOKBÓL LÉTREHOZOTT SZUBRUTINOKAT!



MAT A=B HELYETTESÍTÉSÉRE SZOLGÁLÓ SZUBRUTIN.

```
200 FOR O=1 TO K
210 FOR S=1 TO L
220 LET A(O,S)=B(O,S)
230 NEXT S
240 NEXT O
250 RETURN
```

OSZLOPOK CIKLUSA

SOROK CIKLUSA


A SZUBRUTIN HÍVÁSA ELŐTT K-NAK ÉS L-NEK ÉRTÉKET ADUNK.

A SZUBRUTINT A 78. OLDALI PÉLDA MEGOLDÁSÁRA ÍGY HASZNÁLHATJUK:

```
10 DIM A(3,4), B(3,4)
20 LET L=3 : K=4
30 GOSUB 200
40 END
```


112 BEJELENTKEZÉS

AZ **MO8X1** EGY PROFESSZIONÁLIS SZEMÉLYI SZÁMÍTÓGÉP. (PSZSZ.)




A BILLENTYŰZET ÉS A KÉPERNYŐ A SZÁMÍTÓGÉPPEL FOLYÓ PÁRBESZÉD KÉT ALAPVETŐ ESZKÖZE. A SZÁMÍTÓGÉP AZONBAN CSAK AKKOR VÁLIK KÉPESSÉ PÁRBESZÉDRE, HA BETÖLTÖTTÜK OPERÁCIÓS RENDSZERÉT.

AZ **MO8X1** OPERÁCIÓS RENDSZERÉT  MÁS PROGRAMOKHOZ HASONLÓAN  HAJLÉKONY (FLOPPY) LEMEZEN TÁROLJUK.



 TEGYÜK TEHÁT A KÖVETKEZŐKET :



 KAPCSOLJUK BE A SZÁMÍTÓGÉPET!




 TEGYÜK AZ "A" LEMEZEGYSÉGBE AZ OPERÁCIÓS RENDSZERT ÉS A BASIC-ET TARTALMAZÓ LEMEZT!




★ NYOMJUK MEG A  ÉS 
BILLENTYŰKET EGYSZERRE!

★ ÜSSÜK LE AZ , MAJD A 
(NULLA) BILLENTYŰT!



az operációs
rendszer
munkára
kész!

★ BØ8 BEGÉPELÉssel KÉRJÜK A BASIC
INTERPRETER BETÖLTÉSÉT.




a BASIC várja,
hogy kezdjük
a munkát

a rendelkezésünkre
álló szabad hely

HA A PROGRAMUNKAT BEGÉPELTÜK, LEFUTTATTUK,
ÉS A MUNKÁT ABBA AKARJUK HAGYNI, HAJLÉKONY
LEMEZÜNKET KIVESSZÜK A LEMEZEGYSÉGBŐL, ÉS
A GÉPET KIKAPCSOLJUK.

WISZONTLÁTÁSRA!

DE AZÉRT TEKINTSÜK MÉG ÁT, AZ  BASIC
PARANCSAIT ÉS AZ ADATÁLLOMÁNYOK KEZELÉSÉHEZ
NYÚJTOTT ESZKÖZEIT...! »

114 PARANCSONK

AZ MO8X BASIC-NEK 19 PARANCSA VAN.

RUN

PARANCSNAK ADHATÓ ARGUMENTUM IS AZ MO8X BASIC-BEN.

RUN 210

a program végrehajtását a 210. sorban indítsd!

RUN "PROG 1"

lemezről hívd be a PROG 1 nevű programot! és indítsd!

AUTO

AUTOMATIKUS SORSZÁMOZÁS

AUTO

10 REM SORSZÁM

20 FOR A = 1 TO 8

30

argumentum nélkül

sorszámzás tízesével

AUTO 50

50 REM SOR50

100 FOR A = 1 TO 2

150

argumentummal

sorszámzás az argumentum szerint

★ AZ AUTOMATIKUS SORSZÁMOZÁST A BILLENTYŰK LEÜTÉSÉVEL SZÜNTETHETJÜK MEG.



DELETE

TÖRLI AZ ADOTT SORSZÁMÚ SORT VAGY SOROKAT A PROGRAMBÓL.

DELETE 10

törli a tízes sort

DELETE 20-50

minden sort töröl a 20-50-ig.

LIST

KILISTAZZA AZ EGÉSZ PROGRAMOT
VAGY A PROGRAM ADOTT RÉSZÉT.

```
10 REM LIST - DELETE
20 FOR I = 1 TO 10
30 PRINT A
40 NEXT I
35 PRINT I
50 END
```

```
DELETE 30
LIST 20-40
```

```
20 FOR I = 1 TO 10
35 PRINT I
40 NEXT I
```

a listázás helyes
sorrendbe rendezi
a programsorokat

RENUM

ÁTSZÁMOZZA A PROGRAMSOROKAT.

```
RENUM
10 REM LIST - DELETE
20 FOR I = 1 TO 10
30 PRINT I
40 NEXT I
50 END
```

argumentum nélkül
tizesével

első programsor új száma

első program
régi száma

```
RENUM 20, 10, 20
20 REM LIST - DELETE
40 FOR I = 1 TO 10
60 PRINT I
80 NEXT I
100 END
```

növekedés

SAVE

MÁSOLATOT KÉSZÍT A PROGRAMRÓL A LEMEZEN MEGADOTT NÉVVEL.

```
10 INPUT A, B, C
20 PRINT (A + B + C) / 3
30 END
```

```
SAVE "ABC.BAS"
```

név megadása kötelező!

FILES

A TÁROLT ADATÁLLOMÁNYOKRÓL AD INFORMÁCIÓT. A 115. O. CATALOG PARANCSÁNAK FELEL MEG.

```
FILES
```

az összes adatállomány információit kiírja

```
FILES "PROG1"
```

a PROG1 állomány adatait közli

KILL

ADATÁLLOMÁNYT TÖRÖL A LEMEZRŐL. (A 116. OLDALON ISMERTETETT "UNSAVE" PARANCSNAK FELEL MEG.)

```
KILL "PROG3"
```

LOAD

ADOTT NEVŰ ADATÁLLOMÁNYT VISZ A LEMEZRŐL A TÁRBA. (A 116. OLDALON LEÍRT "OLD" PARANCS MEGFELELŐJE.)

```
LOAD "PROG2"
```

NAME

ADATÁLLOMÁNYOK ÁTNEVEZÉSÉRE SZOLGÁL.

```
NAME "PÉLDA" AS "PROG5"
```

régi név

új név

a lemezen meglévő adatállomány neve

eddig még nem használt állománynév

HA A "NAME" PARANCS UTÁN "FILES" PARANCSOT ADUNK, A FELSOROLT ADATÁLLOMÁNYNEVEK KÖZÖTT SZEREPELNI FOG A "PROG5", DE HIÁNYZIK A "PÉLDA".

NEW

TÖRLI A MUNKATERÜLETET.
(AZONOS A 117. OLDALON ISMERTETETT
"NEW" PARANCCSAL.)

MERGE

LEMEZEN LEVŐ ADATAÁLLOMÁNYT EGYESÍT A
MUNKATERÜLETEN LEVŐ PROGRAMMAL.
AZ EGYESÍTÉS A PROGRAMOK UTASÍTÁSAINAK SORSZÁMA ALAP-
JÁN TÖRTÉNIK.

A MUNKATERÜLETEN LEVŐ
PROGRAM

```
10 REM TÖMB-ÖSSZEADÁS
20 DIM A(3,2), B(3,2)
30 DIM C(3,2)
40 LET K=3
50 LET L=2
60 GOSUB 100
200 PRINT "KÉSZ"
210 END
```

A LEMEZEN LEVŐ
PROGRAM

```
100 REM SZUBRUTIN
110 FOR C=1 TO L
120 FOR R=1 TO K
130 LET A(R,C)=B(R,C)+C(R,C)
140 NEXT R
150 NEXT C
160 RETURN
```

```
MERGE "MATPLUS"
```

A "MERGE" PARANCCSAL BEMÁSZOLJUK A "MATPLUS" MÁTRIX-
ÖSSZEADÓ RUTINT A PROGRAMUNKBA.

LIST

PARANCCSAL A PROGRAMOT SORNYOMTA-
TÓN LISTÁZHATJUK KI.

AZ ITT ISMERTETETT PARANCSOKRÓL, ILLETVE A NEM
ISMERTETETT PARANCSOKRÓL (CLEAR, CONT, EDIT,
RESET, SYSTEM, TRON-TROFF) RÉSZLETES INFORMÁCIÓT
AD AZ MO8X BASIC KÉZIKÖNYV.

120 ADATÁLLOMÁNYOK


AZ **MO8X** BASIC SZEMÉLYI SZÁMÍTÓGÉPNÉL ADATOK BEVITELÉRE A

- BILLENTYŰZET ÉS A
- HAJLÉKONY LEMEZ

SZOLGÁL.

ADATKIVITELI LEHETŐSÉGET A

- KÉPERNYŐ A
- HAJLÉKONY LEMEZ ÉS A
- NYOMTATÓ



SZOLGÁLTAT.

EDDIG BILLENTYŰZETRŐL BEÍRT ÉS KÉPERNYŐRE KIÍRT ADATOKKAL FOGLALKOZTUNK.

ADATÁLLOMÁNYOKAT AZONBAN EGY HÁTTÉRTÁROLÓN **MO8X** ESETÉN HAJLÉKONY LEMEZEN IS ELHELYEZHETÜNK, AZOKBA ADATOT ÍRHATUNK, ILLETVE OLVASHATUNK.

AZ EDDIG MEGISMERT UTASÍTÁSOK KÖZÜL A KÖVETKEZŐKET HASZNÁLHATJUK ERRE A CÉLRA IS:

INPUT
PRINT
PRINT USING

MINDÖSSZE EGY # JELLEL KELL KIEGÉSZÍTENÜNK AZ UTASÍTÁS NEVÉT ANNAK JELZÉSÉRE, HOGY LEMEZEGYSÉGRE VONATKOZIK, VALAMINT EGY SZÁMMAL KELL AZONOSÍTANUNK AZT AZ ÁLLOMÁNYT, AMELYHEZ HOZZÁ AKARUNK FÉRNI.

1030 PRINT # 1, A(I,1), A(I,2), A(I,3)

a többi az 1 számú állományban akarjuk tárolni

2060 INPUT # 6, A,B,C


a 6 számú állományból olvasunk 3 numerikus elemet

122 FILE-NEVEK+SZÁMOK

HA PROGRAMUNK ADATAÁLLOMÁNNYAL KOMMUNIKÁL, KAPCSOLATOT KELL TEREMTENÜNK A "FILES" PARANCCSAL KIÍRT ÁLLOMÁNY-NEV ÉS AZ "INPUT#" VAGY A "PRINT#" UTASÍTÁSBAN MEGADOTT ÁLLOMÁNYSZÁM KÖZÖTT. EZT AZ ÖSSZERENDELEST AZ "OPEN" UTASÍTÁS VÉGZI EL.

OPEN

UTASÍTÁSNAK KELL MEGELŐZNI AZ ADAT-ÁLLOMÁNYBA ÍRÁST VAGY OLVASÁST.



AZ EGYES PARAMÉTEREK VÁLTOZÓBAN IS MEGADHATÓK:

10 LET	A\$ = "ADAT 3"
20 LET	M\$ = "I"
30 LET	FN = 3
40 OPEN	M\$, FN, A\$

AZ "OPEN" UTASÍTÁSSAL MEGNYITOTT ÁLLOMÁNYHOZ CSAK A MEGADOTT MÓDON FÉRHETÜNK HOZZÁ. MÓD: "O" ESETÉN CSAK ÍRHA-TUNK AZ ÁLLOMÁNYBA. HA OLVASNI SZERETNÉNK, ELŐBB LE KELL ZÁRNI AZ ÁLLOMÁNYT, MAJD "I" MÓDBAN ÚJRA MEGNYITNI.

CLOSE

AZ ÁLLOMÁNY LEZÁRÁSÁRA SZOLGÁLÓ UTASÍTÁS.

1
100 CLOSE 1,3

ÍRNI SEM OLVASNI NEM LEHET.

AZ UTASÍTÁS VÉGREHAJTÁSA UTÁN AZ 1 ÉS 3 SZÁMÚ ÁLLOMÁNYBA SEM

1 10 CLOSE

az összes állományt lezárja

A "CLOSE" UTASÍTÁSSAL AZONOS HATA-SÚ A "RESET" PARANCS.

124 ADATÁLLOMÁNY

FAJTÁK

AZ **MO8X** BASIC SZEKVENCIAÁLIS ÉS KÖZVETLEN HOZZÁFÉRÉSŰ ÁLLOMÁNYOKAT KEZEL.

A SZEKVENCIAÁLIS ADATÁLLOMÁNY KÉPZELETBELI MUTATÓJÁT AZ "OPEN" UTASÍTÁS ÁLLÍTTJA AZ ÁLLOMÁNY ELEJÉRE.

A KÖZVETLEN HOZZÁFÉRÉSŰ ADATÁLLOMÁNY ADOTT REKORDJÁT

GET

UTASÍTÁSSAL OLVASHATJUK BE.

200 GET # 3, 10

a 3. számú állomány 10. rekordját hozza be a rendszer által definiált területre.

PUT

UTASÍTÁSSAL ÍRHAUNK KÖZVETLEN HOZZÁFÉRÉSŰ ADATÁLLOMÁNYBA.

250 PUT 3, 15

a 3. számú adatállomány 15. rekordjába írja a rendszer által definiált területen lévő adatokat

HOGYAN VIHETJÜK BE, ILL. OLVASHATJUK KI A RENDSZER ÁLTAL DEFINIÁLT BE- ÉS KIVITELI PUFFERTERÜLETRŐL AZ ADATOKAT?

ERRE SZOLGÁL A

FIELD

UTASÍTÁS.

A "FIELD" UTASÍTÁS MEZŐKET DEFINIÁL A PUFFERTERÜLETEN SZÖVEGES VÁLTOZÓKNAK.

60 FIELD 2, 20 AS NÉV \$, 15 AS FOGLALKOZÁS \$

AZ UTASÍTÁS A 2 SZÁMÚ ADATÁLLOMÁNY REKORDTARTALMÁT OSZTJA FEL LOGIKAILAG EGY 20 KARAKTERES "NÉV", ÉS EGY 15 KARAKTERES "FOGLALKOZÁS" MEZŐRE.



AZ ADATÁLLOMÁNYOKKAL KAPCSOLATOS MŰVELETEKRŐL RÉSZLETESEBB INFORMÁCIÓT AZ **MO8X** BASIC KÉZIKÖNYVBEN TALÁLSZ!

TÁRGYMŰTATÓ

A

- ABS (ABSZOLÚT ÉRTÉK) FÜGGVÉNY 23
- ADATÁLLOMÁNY 120... 125, 158... 160
 - FAJTAI 124, 125, 160
 - LÉTREHOZÁSA 120
 - TERÜLETE 124
 - TÖRLÉSE 116
- ADATÁLLOMÁNYOK HÁTTÉRTÁROLÁSA 120
 - INPUT UTASÍTÁSAI 120... 123
 - PRINT UTASÍTÁSAI 120
 - PRINT USING UTASÍTÁSAI 120-123
 - TÁROLÁSA 115, 117
- ADATÁLLOMÁNYOKBAN LEVŐ SZÖVEG 123
- ADATOK A DATA UTASÍTÁSBAN 17
- ALFAMETRIKUS SZÖVEG 12
 - - VÁLTOZÓ 13
- ALGORITMUS 106
- ÁLLAPOTTÁBLAZATOK 95, 102... 105
- Á. N. S. I. x3J2, *vi*
- ÁPOSZTRÓFOK 12
- ÁSZTALI SZÁMÍTÓGÉPEK
- ATN (TANGENS) FÜGGVÉNY 24

B

- "BACKUS - NAUR" JEJÖLÉSRENDSZER 128
- BASIC ELEMÉK 129
 - EREDETE *vi*
 - SZINTAXISA 128... 131
- BEJELENTKEZÉS 4, 112, 113, 152
- BELSŐ FÜGGVÉNY 22
 - SZORZATOK 89, 90
- BETŰTÍPUSOK 128
- BILLENTYŰZET 5, 136
- BINÁRIS ADATÁLLOMÁNYOK 125
 - SZÁMJEGYEK 9

C, CS

- CATALOG PARANCs 115, 122
- CIKLUSOK 48... 51
 - , EGYMÁSBA FONÓDOTT 49

- , EGYMÁSBA ÁGYAZOTT 49, 77
- CON (KONSTANS) MAT UTASÍTÁS UTÁN 87
- COS (COSINUS) FÜGGVÉNY 24
- CSATORNÁSZÁM 120, 122

D

- DATA UTASÍTÁS 16, 17, 94
- DEF (DEFINIÁLT) UTASÍTÁS 26, 143
- DETERMINÁNSOK 43, 93
- DIM (DIMENZIÓ) UTASÍTÁS 62, 121, 150

E, É

- EGÉSZ RÉSZ 23, 35, 46, 47, 65, 95, 97
 - FORMAJÚ MASZK TÖLTÉSE 35
- EGÉSZÉK 9
 - KINYOMTATÁSA 30, 34, 35
- EGYDIMENZIÓS NUMERIKUS TÖMB 60
- EGYINDEXŰ VÁLTOZÓ 60
- EGYMÁSBA ÁGYAZOTT CIKLUSOK 49, 77
 - - FONÓDOTT CIKLUSOK 49
- END UTASÍTÁS 7, 42
- ÉRTÉKTELEN NULLÁK 30
- EXP (EXPONENCIÁLIS) FÜGGVÉNY 23

F

- FELTÉTELEK 41
- FOR UTASÍTÁS 48
- FORMÁLIS ARGUMENTUM 26
 - PARAMÉTEREK 53
- FORMÁTUM 34
- FORRÁSKÓD 115
- FÜGGVÉNY 22... 27, 140... 142
 - , ABS 23
 - , ATN 24
 - , COS 24
 - , EXP 23
 - , INT 23
 - , LOG 23

(FOLYTATÁS)

- , RND 25
- , SGN 23
- , SIN 24
- , SQR 23
- , TAB 32
- , TAN 24

- , TRIGONOMETRIKUS 24
- FÜGGVÉNYEK DEFINIÁLÁSA 128
- SAJÁT DEFINIÁLÁSA 26, 27

G, QY

- GÉPELÉS 6, 18, 19, 137
- GO SUB UTASÍTÁS 52... 55
- GO TO UTASÍTÁS 40, 46, 47, 147
- GRAFIKONOK KÉSZÍTÉSE 33
- GYORSAN VÁLTOZÓ INDEXEK 77

H

- HATÁROZATLAN VÁLTOZÓK ÉS TÖMBÖK 1, 13, 63
- HÁTTÉRTÁROLÁS ADATAÁLLOMÁNYOKKAL 120
- HATVÁNYOZÁS PRIORITÁSA HELYETTESÍTÉS 11
- , MÁTRIXOKÉ 78
- , VÁLTOZÓKÉ 11, 13
- HIBÁK ÁLLAPOTTÁBLÁZATOKBAN 103
- TÖRLÉSE 7
- "HORDOZHATÓ" PROGRAM 27
- "HORDOZHATÓSÁG" vii

I

- IDÉZETT KARAKTERLÁNC 12
- IDÉZŐJELEK HASZNÁLATA 12
- IDN (AZONOSSÁGI MÁTRIX) UTASÍTÁS MAT UTÁN 87
- IF UTASÍTÁS 41
- INDEXEK 18, 64, 65
- INDEXELT VÁLTOZÓK 60, 64
- INPUT UTASÍTÁS 18, 19, 139
- - , ADATAÁLLOMÁNYOKÉ 120... 123
- - MAT UTÁN 96, 97
- INVERZ 90
- INVERZIO 90

INT (EGÉSZ RÉSZ) FÜGGVÉNY 23

J

JELSZÓ 113

K

- KAPCSOLÓK 106, 107
- KARAKTER HELYZETEI 28
- KARAKTERLÁNC 12
- KARAKTERLÁNC - VÁLTOZÓ 13
- KÉP 34
- KÉPZELETBELI MUTATÓ 121, 124
- KEREKÍTÉSI HIBÁK 57, 93
- KÉTDIMENZIÓS NUMERIKUS TÖMB 60
- KÉTINDEXŰ VÁLTOZÓ 60
- KEZDETI ÉRTEK 11, 13, 63
- KIFEJEZÉSEK 20, 21, 64, 139
- KIJELENTKEZÉS 113
- KIS SZÁMOK 9, 30
- KITERJESZTETT TARTOMÁNY 50
- KIVONÁS, MÁTRIXOKÉ 80, 81
- - PRIORITÁSA 20
- KÓDOLT ADATAÁLLOMÁNYOK 125
- KOMMUNIKÁCIÓ ADATAÁLLOMÁNYOKKAL 120
- KOMMUTATÍV SZORZÁS 89
- KOORDINÁTÁK TRANSZFORMÁCIÓJA 85
- KÖZVETLEN HOZZÁFÉRÉSŰ ADATAÁLLOMÁNYOK 124
- KRITIKUS ÚT - ANALÍZIS 106
- KULCSSZÓ 6

L

- LÁNC 12, 68... 73
- LIST PARANCS 114
- LISTAFELDOLGOZÁS 68... 73
- LISTÁBAN VALÓ KERESÉS (ÁTFÉSÜLÉS) 49
- LOG (LOGARITMUS) FÜGGVÉNY 23

M

- MÁGNESLEMEZES ADATAÁLLOMÁNY 120
 - MASZK 34
 - MAT UTASÍTÁSOK LISTÁJA 77
 - MÁTRIXINVERZIO 90... 93
- (FOLYTATÁS)

MÁTRIXOK 60, 76, 151
 — JELENLEGI DIMENZIÓI 78
 — SZORZÁSA 89
 — ÚJRADIMENZIONALÁSA 79
 MEGENGEDETT SZÓKÖZ 6, 9, 19, 41
 MEGFELELÉS 49
 MINIMÁLIS BASIC v.ii
 MINTASOR 34
 MÓDSZER 106
 MUNKATERÜLET 115... 117
 MUNKAVEZÉRLŐ (JOB CONTROL) NYELV 122
 MUTATÓ 68, 72
 —, KÉPZELETBELI 121, 124
 MUTATÓLISTA 72

N, NY

NAGY SZÁMOK 9
 — — KINYOMTATÁSA 30, 35
 NÉGYZETES MÁTRIX 60, 87, 93
 NÉGYZETTÖMB (VAGY MÁTRIX) 60, 76, 77
 NEW PARANCs 117
 NEXT UTASÍTÁS FOR UTÁN 48... 51
 NYELV v.ii
 NYOMTATÓFEJ HELYZETE 33, 34

O, Ö

OLD PARANCs 116
 ON UTASÍTÁS 47, 149
 — — TARTOMÁNYA 47
 OPERÁCIÓS RENDSZER 112, 113
 OSZLOPVEKTOR 60
 OSZTÁS PRIORITÁSA 20
 ÖSSZEADÁS PRIORITÁSA 20
 ÖSSZEFÉRHETŐ TÖMBÖK (MÁTRIXOK) 89

P

PARANCs 114... 117, 154... 157
 —, CATALOG 115, 112
 —, LIST 114, 155
 —, NEW 117, 157
 —, OLD 116
 —, RUN 4, 114, 154
 —, SAVE 115... 117, 156
 —, UNSAVE 116

PÉNZJEGYJEL 36
 PONTOSVESSZŐK MAT PRINT-BEN 98, 99
 — PRINT UTASÍTÁSBAN 12, 28, 29
 PRINT UTASÍTÁS 28... 33
 — — MAT UTÁN 98
 PRINT USING UTASÍTÁS 34... 37, 144..
 PROGRAM v.ii, 8 .. 146
 — LEÁLLÍTÁSA 40, 42
 PROGRAMSORSZÁM 34
 PROMPTLEKÉRDEZÉS (INPUT UTASÍTÁS
 ESETÉN) 18
 PSZUDOVÉLETLEN SZÁMOK 25

R

RADIÁNOK 24
 RANDOM ACCESS 124
 RANDOMIZE UTASÍTÁS 25, 143
 READ UTASÍTÁS 16, 17
 — — MAT UTÁN 94, 95
 REAL FORMA 9
 REKORDOK ADATAÁLLOMÁNYBAN 121
 REKURZIÓ 54, 55
 REM (MEGJEGYZÉS) UTASÍTÁS 8
 RENDEZÉS 7, 66, 67, 71... 73
 RESET UTASÍTÁS 121, 124
 RESTORE UTASÍTÁS 17, 121, 138
 RETURN UTASÍTÁS 52, 53
 RND (VÉLETLENSZÁMOK) FÜGGVÉNY 25, 143
 RUN PARANCs 4, 114

S

SAVE PARANCs 115... 117
 SGN (SIGNUM) FÜGGVÉNY 23
 SIN (SINUS) FÜGGVÉNY 24
 SKALÁRIS SZÁM 82, 83
 SKALÁRSZORZÁS 82, 83
 SOR 16
 SORMÁTRIX 60
 SOROK HOSSZÁNAK KORLÁTJA 6, 99
 SORSZÁMOK 7, 137
 SORVEKTOR 60
 SQR (NÉGYZETGYÖK) FÜGGVÉNY 23
 STANDARD BASIC SPECIFIKÁCIÓJA v.ii
 STOP UTASÍTÁS 42, 148

(FOLYTATÁS)

SZ

- SZÁMÍTÓGÉP TERMINÁLJAI 112
- SZÁMÍTÓGÉPEK ADATÁLLOMÁNYTERÜLETE 115
 - MUNKA TERÜLETE 115
- SZÁMÍTÓGÉP KÖZPONT 112
- SZÁMOK 9, 137
 - FORMÁI 9
 - E FORMÁJA 9
 - HATVÁNYKITEVŐS FORMÁJA 9, 36
 - PONTOSSÁGA 9
 - VALÓS FORMÁJA 9
- SZÁMVEGI BETŰKÖZÖK
- SZEKVENCIAÁLIS ADATÁLLOMÁNYOK 124, 125
- SZÉTVÁLOGATÁS 72
- SZIGNIFIKÁNS JEGY 9
- SZIMBÓLUM ÁLLAPOTTÁBLÁZATOK 95, 102...105
- SZIMBÓLUMOK 128
- SZIMULTÁN EGYENLETEK 43, 90, 91
- SZINGULÁRIS MATRIX 93
- SZINTAXIS DEFINIÁLÁSA 26
- SZÓHOSSZÚSÁG 9
- SZÓKÖZÖK, MEGENGEDETT 6, 9, 19, 41
- SZORZÁS PRIORITÁSA 20
- SZÖVEGEK 12
- SZÖVEGES TÁROLÓ 13
 - TÖMB 61, 138
 - VÁLTOZÓK 13
- SZUBRUTINHÍVÁS 53, 121
 - ÁLLAPOTTÁBLÁZATOKKAL 105

T

- TAB FÜGGVÉNY 32, 33
- TAN ((TANGENS)) FÜGGVÉNY 24
- TARTOMÁNY TESZTJE 97
- TÉGLALAPMATRIX 60, 76
- TÉGLALAPTÖMB 60, 76
- TERMINÁL 112
- TIZEDES FORMÁJÚ MASZK TÖLTÉSE 35
- TIZEDESPONT KÖTÖTT FORMÁBAN 34
 - SZÁMOKBAN 9
- TÖBBÁLLÁSÚ KAPCSOLÓK 46
- TÖMB MASOLÁSA 78

- TÖMBÖK 60, 149
 - ÖSSZEADÁSA 80, 81
- TRIGONOMETRIKUS FÜGGVÉNYEK 24, 140
- TRN ((TRANZPONÁLÁS)) UTASÍTÁS MAT UTÁN 84, 85

U, Ő

- UGRÁSOK A CIKLUSOKBAN 49, 50
- ÚJRADIMENZIONÁLÁS 79
- UNSAVE PARANCSS 116, 117
- UTASÍTÁSOK 8, 159
 - , DATA 16
 - , DEF 26
 - , DIM 62, 150
 - , FOR 48
 - , IF 41, 148
 - , INPUT 18, 139
 - , GO SUB 52
 - , GO TO 40, 46, 47, 147
 - , MAT 77
 - , NEXT 48
 - , ON 46, 149
 - , PRINT 28... 33
 - , RANDOMIZE 25
 - , READ 16, 17
 - , REM 8
 - , RESET 121, 124
 - , RESTORE 17, 121, 138
 - , RETURN 52, 53
 - , RND 25
 - , STOP 42, 148
 - , TRN 84, 85
 - , ZER 86
- UTASÍTÁSOK SORA 8
 - SORÁNAK MEGVÁLTOZTATÁSA 40, 41
 - SZINTAXISA 130... 131
- ÜRES PRINT LISTÁK 29
 - SOROK 8, 29

V

- VALTOZOK 10, 138
- , EGYSZERU NUMERIKUS 10, 11
- HELYETTESITESE 11, 13
- VEGREHAJTHATO UTASITASOK 8
- VEKTOR 60
- VELETLEN EGESZ 27
- ELRENDEZESU ADATALLOMANY 124
- SZAMOK 25
- - CIKLUSA 25
- VEREMTAR 54, 55
- VEREMTECHNIKA 54, 69
- VESSZO DATA UTASITAS ESETEN 17
- , INPUT UTASITAS ESETEN 19
- , MAT PRINT UTASITAS ESETEN 98
- , PRINT UTASITAS ESETEN 12, 28
- , PRINT USING UTASITAS ESETEN 34

Z

- ZAROJELEK HASZNALATA 20
- ZER (ZERO MATRIX) UTASITAS MAT
UTAN 86
- ZONAK A PRINT UTASITASBAN 28
- # JEL 34
- ? JEL 19, 139

VISZONTLÁTÁSRA

3 17 38 6 10 43

12

KIADJA A MŰSZAKI KÖNYVKIADÓ

FELELŐS KIADÓ: FISHER HERBERT IGAZGATÓ

RÉVAI NYOMDA BUDAPEST

FELELŐS VEZETŐ: MORVÁTH JÓZSEFNÉ IGAZGATÓ (6538)

MŰSZAKI VEZETŐ: KÖRIZS KÁROLY

MŰSZAKI SZERKESZTŐ: MÓZER ÁSTVÁN

A KÖNYVET ÉS A BORÍTÓT KÉSZÍTETTE: BELECSÉNYI ZSÓLT

A KÖNYV FORMÁTUMA: B 5 — ÍVTERJEDELME: 16,125 (A 5)

PAPÍR MINŐSÉGE: 120 g AURÓRA

AZONOSSÁGI SZÁM: 61318

MŰ: 3518-h-8486

A KÉZIRAT LEZÁRVA: 1983. OKTÓBER

KÉSZÜLT AZ MSZ 5601 ÉS 5602 SZERINT

Ára: 60,— Ft

EMHEZ A BASIC
TANKÖNYVHÖZ...
ÖNNEK
...MÁR CSAK EGY

MO8X
8
16
32

PROFESSZIONÁLIS SZEMÉLYI
SZÁMÍTÓGÉPRE VAN SZÜKSÉGE!

Szki

SZÁMÍTÁSTECHNIKAI
KOORDINÁCIÓS
INTÉZET
1054. BUDAPEST V.
AKADÉMIA U. 17.
T: 129-600

Sci-L

SZÁMÍTÁSTECHNIKAI
INFORMATIKAI FEJLESZTŐ
LEÁNYVÁLLALAT
1011 BUDAPEST I.
ISKOLA U. 10.
T: 260-000